
Введение

Изучение	программирования	на	ассемблере	может	оказаться	обескураживаю
щим,	но	совсем	не потому,	что	это	язык,	не прощающий	ошибок,	ведь	компью
тер	будет	«одобрять»	ваши	действия	при	каждом	удобном	случае.	А если	это	
не так,	то,	возможно,	гдето	в программе	скрывается	необнаруженная	ошибка,	
которая	«укусит»	вас	во	время	выполнения	программы.	Сверх	всего	прочего	
кривая	сложности	обучения	весьма	крута,	язык	загадочный	и не сразу	понят
ный,	официальная	документация	Intel	ошеломляюще	велика,	а доступные	ин
струменты	разработки	обладают	весьма	специфическими	особенностями.

Эта	 книга	 научит	 вас	 программировать	 на	 ассемблере,	 начиная	 с  самых	
прос	тых	программ	и  постепенно	 осваивая	 путь	 к  овладению	программиро
ванием	 с  использованием	 расширенной	 системы	 команд	 Advanced	 Vector	
Extensions	(AVX).	Прочитав	эту	книгу	полностью,	вы	сможете	писать	и читать	
код	 на	 ассемблере,	 ассемблерный	 код,	 объединенный	 с  языками	 высокого	
уровня,	поймете,	что	такое	AVX	и многое	другое.	Цель	этой	книги –	показать,	
как	используются	инструкции	языка	ассемблера.	Это	не руководство	по	сти
лю	программирования	или	по	оптимизации	производительности	кода.	После	
того	как	вы	освоите	базовые	знания	об	ассемблере,	можно	будет	продолжить	
обучение	по	теме	оптимизации	кода.	Эта	книга	не должна	быть	вашей	первой	
книгой	по	программированию:	если	вы	никогда	раньше	не программировали,	
то	отложите	эту	книгу	на	время	и изучите	основы	программирования	на	ка
комлибо	языке	высокого	уровня,	например	на	C.

Весь	 исходный	 код,	 используемый	 здесь,	 доступен	 по	 ссылке	 Download	
Source	 Code	 на	 сайте	 www.apress.com/9781484250754.	 Исходный	 код	 в  книге	
представлен	в настолько	простом	виде,	насколько	это	вообще	возможно,	т. е.	
без	 какихлибо	 графических	 пользовательских	 интерфейсов,	 без	 излишеств	
(«бантиков	 и  рюшечек»),	 без	 средств	 проверки	на	 ошибки.	Добавление	 всех	
этих	замечательных	функциональных	возможностей	увело	бы	нас	от	истинной	
цели:	изучение	языка	ассемблера.

Теоретическая	часть	сведена	к необходимому	минимуму:	немного	инфор
мации	о двоичных	(бинарных)	числах,	краткое	описание	логических	операто
ров	и коечто	об	основах	линейной	алгебры.	Здесь	не рассматриваются	опера
ции	преобразования	чисел	с плавающей	точкой.	Если	требуется	преобразовать	
двоичные	или	шестнадцатеричные	числа,	то	найдите	вебсайт,	который	сдела
ет	это	за	вас.	Не теряйте	вермени	на	вычисления	вручную.	Помните	о главной	
цели:	изучение	ассемблера.

Исходный	ассемблерный	код	представлен	в виде	 завершенных	программ,	
так	что	вы	можете	протестировать	их	на	своем	компьютере,	поэксперименти
ровать,	изменять	их	и «ломать»…

Кроме	того,	будут	рассматриваться	инструментальные	средства,	которыми	
можно	воспользоваться,	и потенциальные	проблемы	при	использовании	этих	

http://www.apress.com/9781484250754


инструментов.	Выбор	правильных	инструментов	чрезвычайно	важен	для	пре
одоления	крутой	кривой	сложности	обучения.	Иногда	будут	встречаться	ссыл
ки	на	книги,	статьи,	прочие	документы	и вебсайты,	которые	могут	оказаться	
полезными	при	обучении	или	содержать	более	подробные	описания.

Автор	не намеревался	предоставить	исчерпывающий	курс	по	всем	инструк
циям	ассемблера.	Это	невозможно	сделать	в одной	книге	 (взгляните	на	раз
меры	справочных	руководств	компании	Intel).	Читателю	предоставлена	воз
можность	 попробовать	 на	 практике	 основные	 компоненты,	 чтобы	получить	
представление	 о  том,	 куда	 двигаться	 дальше.	 При	 работе	 с  этой	 книгой	 вы	
получите	знания,	необходимые	для	самостоятельного	дальнейшего	глубокого	
изучения	определенных	предметных	областей	ИТ.	После	завершения	чтения	
этой	книги	вы	сможете	изучать	справочные	руководства	компании	Intel	и по
нимать	(во	всяком	случае,	пытаться	понять)	смысл	их	содержимого.

Основная	 часть	 книги	 содержит	 информацию	 о  применении	 ассемблера	
в Linux,	потому	что	это	самая	простая	и удобная	платформа	для	изучения	язы
ка	 ассемблера.	В  заключительной	части	книги	представлено	несколько	 глав,	
описывающих	методику	использования	ассемблера	в Windows.	Вы сами	убе
дитесь	в том,	что,	вооружившись	ассемблером	Linux,	гораздо	проще	осваивать	
ассемблер	в Windows.

Существует	несколько	трансляторов	ассемблера,	доступных	для	использова
ния	с процессорами	Intel,	например	FASM,	MASM,	GAS,	NASM	и YASM	(список	
далеко	не полный).	В этой	книге	используется	транслятор	NASM,	поскольку	он	
многоплатформенный:	доступен	для	Linux,	Windows	и macOS.	Кроме	того,	он	
обладает	относительно	большой	пользовательской	базой.	Но не стоит	беспо
коиться;	если	вы	знаете	один	ассемблер,	то	с легкостью	освоите	любой	другой	
«диалект».

Исходный	код	в книге	тщательно	проверен	и протестирован.	Но если	обна
ружатся	какиелибо	опечатки	в тексте	или	ошибки	в программах,	мы	не несем	
за	это	никакой	ответственности.	Автор	обвиняет	в этом	двух	своих	котов,	ко
торые	любят	прыгать	на	клавиатуру	во	время	его	работы.

Идеи	и мнения,	представленные	в этой	книге,	принадлежат	лично	автору	
и не всегда	представляют	позицию,	стратегии	или	мнения	компании	IBM.

Прежде чем начать

Для	чтения	этой	книги	вы	должны	знать	некоторые	основные	темы.
	� Вы	должны	уметь	устанавливать	и настраивать	программное	обес-
печение	 виртуализации	 (VMware,	 VirtualBox	 или	 аналогичное	 ПО). 
Если	эти	программы	вам	неизвестны,	то	загрузите	бесплатное	ПО	Oracle	
VirtualBox	 (https://www.virtualbox.org),	 установите	 его	 и  научитесь	 его	 ис
пользовать,	установив,	например,	Ubuntu	Desktop	Linux	как	гостевую	опе
рационную	систему	(ОС).	ПО	виртуализации	позволяет	устанавливать	раз
личные	гостевые	ОС	на	основном	компьютере,	и если	вы	чтото	напутали		
в такой	гостевой	ОС,	то	можете	ее	удалить	и установить	заново.	Или	если	
имеются	 мгновенные	 снимки	 состояния	 системы,	 то	 можно	 вернуться	
к предыдущей	версии	гостевой	ОС.	Другими	словами,	при	экспериментах	
с гостевой	ОС	не будет	нанесено	никакого	вреда	основной	операционной	
системе.	В интернете	можно	найти	огромное	количество	ресурсов,	описы
вающих	VirtualBox	и другие	решения	с использованием	ПО	виртуализации.

	� Вы	должны	обладать	базовыми	знаниями	об	использовании	интер-
фейса	 командной	 строки	 Linux.	 В  книге	 используется	Ubuntu	Desktop	
Linux	и командная	строка	этой	ОС,	начиная	с главы	1.	Если	хотите,	можете	
работать	в другом	дистрибутиве	Linux,	но	при	этом	необходимо	убедиться	
в том,	что	имеется	возможность	установить	все	инструментальные	сред
ства,	применяемые	в данной	книге	(NASM,	GCC,	GDB,	SASM	и т.	д.).	Требу
ются	следующие	базовые	знания:	как	установить	ОС,	как	устанавливается	
дополнительное	ПО,	как	запустить	терминал	с приглашением	(промптом)	
командной	строки,	а также	как	создавать,	перемещать,	копировать	и уда
лять	 каталоги	 и  файлы	 в  командной	 строке.	 Также	 необходимо	 уметь	
использовать	утилиты	tar,	grep,	find,	 ls,	 time	и т.	п.	Вы должны	знать,	как	
запустить	и использовать	текстовый	редактор.	Не требуется	никаких	про
двинутых	знаний	о Linux,	нужны	только	самые	простые,	базовые	навыки	
выполнения	 задач,	 для	 того	 чтобы	 следовать	 описаниям,	 приведенным	
в этой	книге.	Если	вы	незнакомы	с ОС	Linux,	то	немного	поработайте	в ней,	
чтобы	освоить	ее	использование.	В интернете	существует	множество	не
больших	по	объему,	но	качественных	руководств	для	начинающих	(напри
мер,	https://www.guru99.com/unix-linux-tutorial.html).	Вы сами	убедитесь	в том,	
что	после	изучения	ассемблера	на	компьютере	с ОС	Linux	освоение	ассемб
лера	в других	ОС	станет	не таким	уж	трудным	делом.

	� Вы	должны	обладать	некоторыми	базовыми	знаниями	в области	
программирования	на	языке	C.	В книге	в некоторых	случаях	применя
ются	функции	на	языке	C для	упрощения	примеров	ассемблерного	кода.	
Кроме	того,	будет	показано,	как	организовать	интерфейс	с языком	вы
сокого	уровня,	таким	как	C.	Если	вы	не знаете	C,	но	намерены	в полной	
мере	 освоить	 содержимое	 этой	 книги,	то	 рекомендуется	изучить	пару	



бесплатных	курсов	по	C,	например	tutorialspoint.com.	Нет	необходимости	
проходить	 полный	 курс,	 просто	 внимательно	 изучите	 несколько	 про
грамм	на	этом	языке.	В дальнейшем	всегда	можно	вернуться	к курсу	по	
C,	чтобы	узнать	больше	подробностей.

зачем нужнО изучать аССемблер
Знание	ассемблера	дает	некоторые	преимущества:

	� вы	узнаете,	как	работает	(центральный)	процессор	(ЦПУ)	и оперативная	
память;

	� вы	узнаете,	как	совместно	работают	компьютер	и операционная	система;
	� вы	поймете,	как	компиляторы	языков	высокого	уровня	генерируют	код	
на	машинном	языке,	а эти	знания	могут	помочь	писать	более	эфектив
ный	код;

	� вы	получите	 более	 эффективные	 средства	для	 анализа	 ошибок	 в  про
граммах;

	� вы	получите	 огромное	 удовольствие,	 когда,	 наконец,	 ваша	 программа	
заработает	правильно;

	� и	причина,	по	которой	я написал	эту	книгу:	если	необходимо	исследо
вать	вредоносное	ПО,	то	в вашем	распоряжении	есть	только	машинный	
код	без	исходного	кода.	Если	вы	хорошо	понимаете	код	ассемблера,	то	
сможете	проанализировать	вредоносную	программу,	 выполнить	необ
ходимые	действия	и принять	превентивные	меры.

рукОвОдСтва кОмпании Intel
Справочные	руководства	компании	Intel	содержат	все,	что	может	потребовать
ся	 при	 программировании	микропроцессоров	 Intel.	 Но  информация	 весьма	
сложна	для	освоения	начинающими	программистами.	По мере	чтения	данной	
книги	вы	обнаружите,	что	описания	в этих	руководствах	Intel	постепенно	ста
новятся	все	более	понятными.	В книге	часто	встречаются	ссылки	на	эти	солид
ные	тома	информации.

Справочные	руководства	Intel	можно	найти	здесь:

https://software.intel.com/en-us/articles/intel-sdm.

Только	не пытайтесь	распечатать	их	на	бумаге –	пожалейте	деревья,	кото
рые	вы	можете	уничтожить.	Бегло	просмотрите	руководства,	чтобы	убедиться	
в том,	насколько	исчерпывающими,	подробными	и формализованными	доку
ментами	они	являются.	Попытка	изучения	ассемблера	по	этим	руководствам	
может	оказаться	обескураживающе	неудачной.	Особый	интерес	для	нас	пред
ставляет	Volume	2	 (том	2),	 в  котором	вы	найдете	подробные	описания	про
граммных	инструкций	ассемблера.

Полезный	источник	информации	можно	найти	здесь:	https://www.felixcloutier.
com/x86/index.html.	На этом	сайте	представлен	список	всех	инструкций	с крат
кими	описаниями	их	использования.	Если	предоставленная	здесь	информация	
окажется	недостаточной,	то	вы	всегда	можете	вернуться	к справочным	руко
водствам	Intel	или	обратиться	к вашему	надежному	другу	Google.

Глава 1
Самая первая программа

Многие	поколения	разработчиков	начали	свою	карьеру	программиста	с изуче
ния	способа	вывода	на	экран	компьютерного	дисплея	фразы	hello, world.	Эта	
традиция	была	заложена	в 1970е	гг.	Брайаном	Керниганом	(Brian	W.	Kernighan)	
в  книге,	 которую	 он	 написал	 вместе	 с  Деннисом	 Ритчи	 (Dennis	 Ritchie)	
«The C Programming	Language»	(«Язык	программирования	C»).	Керниган	при
нимал	участие	в разработке	языка	программирования	С в	компании	Bell	Labs.	
С тех	пор	язык	C значительно	изменился,	но	он	остается	языком,	с которым	
должен	быть	знаком	каждый	уважающий	себя	программист.	Большинство	«но
вейших»,	«модных»	языков	программирования	в той	или	иной	степени	явля
ются	наследниками	языка	C.	Язык	C иногда	называют	переносимым	языком	
ассемблера,	и  как	программист,	 стремящийся	к овладению	ассемблером,	 вы	
должны	знать	C.	Уважая	традицию,	начнем	с программы	на	ассемблере,	выво
дящей	фразу	hello, world	на	экран.	В листинге	1.1	показан	исходный	код	версии	
на	языке	ассемблера	этой	программы,	которую	мы	будем	анализировать	далее	
в этой	главе.

Листинг 1.1.	hello.asm

;hello.asm
section .data
    msg    db      "hello, world",0
section .bss
section .text
    global main
main:
    mov    rax, 1       ; 1 = запись.
    mov    rdi, 1       ; 1 = в поток стандартного вывода stdout.
    mov    rsi, msg     ; Выводимая строка в регистре rsi.
    mov    rdx, 12      ; Длина строки без конечного 0.
    syscall             ; Вывод строки.
    mov    rax, 60      ; 60 = код выхода из программы.
    mov    rdi, 0       ; 0 = код успешного завершения программы.
    syscall             ; Выход из программы.



редактирОвание, аССемблирОвание,  
Связывание и запуСк (или Отладка)
Существует	 множество	 хороших	 текстовых	 редакторов,	 как	 бесплатных,	 так	
и коммерческих.	Следует	искать	редактор,	поддерживающий	подсветку	син
таксиса	для	версии	ассемблера	NASM	64bit.	В большинстве	случаев	придется	
скачать	и установить	некоторый	подключаемый	модуль	(plugin)	или	дополни
тельный	пакет,	обеспечивающий	подсветку	синтаксиса.

Примечание. В  этой книге мы будем писать код для версии Netwide Assembler 
(NASM). Существуют и другие версии ассемблера, например YASM, FASM, GAS или 
MASM компании Microsoft. И как обычно в мире ИТ, иногда возникают оживленные 
дискуссии о том, какая версия ассемблера является самой лучшей. В этой книге ис-
пользуется версия NASM, потому что она доступна для ОС Linux, Windows и macOS, 
а кроме того, из-за наличия большого сообщества пользователей NASM. Справочное 
руководство по NASM можно найти здесь: www.nasm.us.

В	этой	книге	будет	использоваться	текстовый	редактор	gedit	с установленным	
дополнительным	файлом	 (модулем)	 подсветки	 синтаксиса	 ассемблера.	 Gedit –	
стандартный	текстовый	редактор,	доступный	в  системе	Linux1, –	 здесь	исполь
зуется	Ubuntu	Desktop	18.04.2	LTS.	Файл	(модуль)	поддержки	подсветки	синтак
сиса	можно	найти	здесь:	https://wiki.gnome.org/action/show/Projects/GtkSourceView/
LanguageDefinitions.	Загрузите	файл	asm-intel.lang,	скопируйте	его	в каталог	/usr/
share/gtksourceview*.0/language-specs/,	заменив	символ	звездочки	(*)	на	номер	вер
сии,	установленной	в вашей	системе.	При	первом	запуске	gedit	можно	выбрать	
поддерживаемый	 язык	 программирования,	 в  нашем	 случае	 Assembler	 (Intel),	
в нижней	части	окна	gedit.

На	экране файл	hello.asm,	приведенный	в листинге	1.1,	будет	выглядеть	так,	
как	показано	на	рис. 1.1.	

Рис. 1.1.	Содержимое файла hello.asm в текстовом редакторе gedit

1	 Автор	не совсем	точен –	в Linux	gedit	доступен	только	при	наличии	установленной	
рабочей	 среды	 GNOME.	 Для	Windows	 и macOS	 все	 необходимое	 для	 работы	 gedit	
включено	в дистрибутив. –	Прим. перев.

Согласитесь,	с подсветкой	синтаксиса	исходный	код	на	ассемблере	немного	
проще	читать.

При	 написании	 ассемблерной	 программы	 на	 экране  открыто	 два	 окна  –	
окно	gedit,	содержащее	исходный	код	на	ассемблере,	и окно	с приглашением	
(промптом)	командной	строки	в каталоге	проекта,	так	что	можно	с легкостью	
переключаться	между	редактированием	и управлением	файлами	проекта	(вы
полнять	ассемблирование	и запускать	программу,	заниматься	отладкой	и т.	п.).	
Разумеется,	что	в более	крупных	и  сложных	проектах	такой	подход	вряд	ли	
можно	применять –	потребуется	интегрированная	среда	разработки	(integrated	
development	environment –	IDE).	Но прямо	сейчас	работы	с простым	текстовым	
редактором	и интерфейсом	командной	строки	(CLI –	command	line	interface)	
вполне достаточно.	Преимуществом	этого	процесса	становится	тот	факт,	что	
мы	 можем	 сосредоточиться	 на	 изучении	 ассемблера,	 а  не  многочисленных	
функциональных	 возможностей	 и  особенностей	 IDE.	 В  последующих	 главах	
будут	рассматриваться	полезные	инструментальные	средства	и утилиты,	не
которые	из	которых	обладают	графическим	пользовательским	интерфейсом,	
прочие	же	связаны	с интерфейсом	командной	строки.	Как	бы	то	ни	было,	опи
сание	и использование	IDE	не относится	к тематике	этой	книги.

Для	любого	упражнения	из	данной	книги	используется	отдельный	каталог	
project,	содержащий	все	необходимые	для	проекта	и генерируемые	файлы.

Разумеется,	 в дополнение	 к текстовому	 редактору	 необходимо	проверить	
наличие	некоторых	других	установленных	инструментальных	средств,	таких	
как	GCC,	GDB,	make	и NASM.	Сначала	потребуется	GCC –	используемый	по	умолчанию	
в Linux	компилятор	и редактор	связей	(линкер).

GCC –	это	сокращение	от	GNU	Compiler	Collection –	стандартного	инструмен
тального	средства	компиляции	и редактирования	связей	в Linux.	(Аббревиатура	
GNU	является	рекурсивной:	GNU	is	Not	Unix.	Использование	рекурсивных	аббре
виатур	для	имен	стало	общепринятым	в среде	разработчиков	еще	в 1970е	 гг.,	
и все	началось	с программистов	LISP.	Да,	эти	старые	шутки	уже	не смешны.)

В	 командной	 строке	 введите	 gcc -v.	 Компилятор	GCC	 в  ответ	 выведет	 не
сколько	сообщений,	если	он	уже	установлен.	Если	комплект	GCC	отсутствует,	
то	необходимо	установить	его	с помощью	следующей	команды:

sudo apt install gcc

Далее	выполните	то	же	самое	для	gdb -v	и make -v.	Если	вы	не понимаете	смысл	
этих	команд,	то,	прежде	чем	продолжать	чтение,	вам	необходимо	пополнить	
знания	о Linux.

Также	 необходимо	 установить	 NASM	 и  пакет	 buildessential,	 содержащий	
ряд	инструментальных	средств,	которые	будут	использоваться	в дальнейшем.	
В Ubuntu	Desktop	18.04	это	делается	так:

sudo apt install build-essential nasm

После	установки	команда	nasm -v	выведет	номер	версии,	если	пакет	NASM	
был	установлен	корректно.	После	установки	всех	перечисленных	выше	про
граммных	средств	вы	полностью	готовы	к реализации	своей	первой	програм
мы	на	ассемблере.

Введите	 исходный	 код	 программы,	 показанной	 в  листинге	 1.1,	 в  тексто
вом	редакторе,	 которым	предпочитаете	пользоваться,	 сохраните	 введенный	
код	в файле	с именем	hello.asm.	Как	уже	было	отмечено	ранее,	для	сохранения	
файлов	этого	первого	проекта	используется	отдельный	каталог.	Каждая	стро
ка	кода	будет	объяснена	немного	позже	в этой	главе.	Обратите	внимание	на	
следующие	характеристики	исходного	кода	на	ассемблере	(«исходный	код» –	
это	содержимое	файла	hello.asm	с программными	инструкциями,	которые	вы	
только	что	ввели):

	� в	исходном	коде	можно	использовать	символы	табуляции,	пробела	и пе
рехода	на	новую	строку,	чтобы	сделать	код	более	удобным	для	чтения;

	� на	каждой	строке	записывается	только	одна	инструкция;
	� текст	 после	 точки	 с  запятой	 является	 комментарием.	 Другими	 слова
ми,	описанием,	предназначенным	для	чтения	человеком.	Компьютеры	
не обращают	никакого	внимания	на	комментарии.

В	текстовом	редакторе	создайте	еще	один	файл,	содержащий	строки,	при
веденные	в листинге	1.2.

Листинг 1.2.	makefile для hello.asm

#makefile для hello.asm
hello: hello.o
      gcc -o hello hello.o -no-pie
hello.o: hello.asm
      nasm -f elf64 -g -F dwarf hello.asm -l hello.lst

На	рис. 1.2	показано,	как	этот	файл	выглядит	в редакторе	gedit.

Рис. 1.2.	Содержимое файла makefile в редакторе gedit

Сохраните	этот	файл	с именем	makefile	в том	же	каталоге,	где	находится	файл	
hello.asm,	и закройте	окно	редактора.

Файл	 makefile	 будет	 использоваться	 командой	 make	 для	 автоматизации	
сборки	программы.	Сборка	(building)	означает	проверку	исходного	кода	на	
ошибки,	 добавление	 всех	 необходимых	 сервисов	 операционной	 системы	
и преобразование	исходного	 кода	 в последовательность	инструкций,	 рас
познаваемых	 компьютером.	 В  этой	 книге	 будут	 использоваться	 простые	
файлы	makefile.	 Если	вы	хотите	 узнать	больше	о файлах	makefile,	то	 спра
вочное	руководство	находится	здесь:

https://www.gnu.org/software/make/manual/make.html.

Учебное	руководство	можно	найти	здесь:

https://www.tutorialspoint.com/makefile/.

Чтобы	 понять,	 что	 именно	 делает	 makefile,	 необходимо	 читать	 его	 сни
зу	вверх.	Упрощенное	описание:	утилита	make	работает	с деревом	зависимо
стей.	Она	определяет,	что	файл	программы	hello	зависит	от	объектного	файла	
hello.o.	Затем	обнаруживается,	что	файл	hello.o	 зависит	от	файла	исходного	
кода	hello.asm	и что	файл	hello.asm	ни	от	чего	не зависит.	Далее	make	сравнивает	
даты	последнего	изменения	файлов	hello.asm	и hello.o,	и если	дата	изменения	
hello.asm	более	поздняя,	то	make	выполняет	строку	после	имени	hello.o,	т. е.	hel-
lo.asm.	Затем	make	снова	начинает	процедуру	чтения	makefile	и обнаруживает,	
что	дата	изменения	файла	hello.o	более	поздняя,	чем	дата	изменения	файла	
hello.	Поэтому	выполняется	строка	после	имени	hello,	т. е.	hello.o.

В	самой	последней	строке	файла	makefile	NASM	используется	как	ассемблер	
(программа	ассемблирования,	т. е.	конечного	этапа	сборки).	За ключом	-f	сле
дует	формат	вывода,	в данном	случае	elf64,	означающий	Executable	and	Linkable	
Format	for	64bit	(выполняемый	и связываемый	формат	для	64битовой	систе
мы).	 Ключ	 -g	 означает,	 что	 необходимо	 включить	 отладочную	информацию	
в специальном	формате	отладки,	определенном	после	ключа	-F.	Здесь	исполь
зуется	отладочный	формат	dwarf.	Похоже,	что	программисты,	разработавшие	
этот	формат,	являются	большими	поклонниками	книг	«Хоббит»	и «Властелин	
колец»	Дж.	Р.	Р.	Толкиена,	возможно,	именно	поэтому	они	решили,	что	DWARF	
(гном)	должен	стать	великолепным	дополнением	к ELF	(эльфу),	на	всякий	слу
чай,	если	вам	это	интересно.	Если	говорить	более	серьезно,	то	DWARF –	это со
кращение	от	Debug	With	Arbitrary	Record	Format2	 (отладка	с использованием	
произвольного	формата	записей).

STABS  –	 это	 еще	 один	 отладочный	 формат,	 не  имеющий	 ничего	 общего	
с кровавыми	битвами	или	ярким	эльфийским	светом	(stab –	ранение,	нанесе
ние	ран;	режущий	глаза,	ослепительный	свет)	из	романов	Толкиена,	это	назва
ние	происходит	от	Symbol	Table	Strings	(строки	таблицы	символов).	Здесь	мы	
не будем	использовать	формат	STABS,	чтобы	вы	не запутались	окончательно.

Ключ	 -l	 сообщает	NASM	 о  необходимости	 генерации	файла	 листинга	 .lst.	
Файлы	.lst	будут	использоваться	для	исследования	результатов	ассемблирова
ния.	NASM	создает	объектный	файл	с расширением	.o.	В дальнейшем	этот	объ
ектный	файл	будет	использоваться	редактором	связей	(линкером).

Примечание. Часто случается так, что NASM выдает несколько непонятных сообще-
ний и отказывается сгенерировать объектный файл. Иногда NASM начинает выда-
вать такие «жалобы» настолько часто, что может поставить программиста почти на 
грань безумия. В таких случаях чрезвычайно важно сохранять хладнокровие, выпить 
очередную чашечку кофе и еще раз внимательно просмотреть исходный код, потому 
что именно вы сделали что-то неправильно. Ассемблируя свою программу раз за 
разом, вы будете находить ошибки все быстрее и быстрее.

Когда	вы	наконец	убедите	NASM	принять	 созданный	объектный	файл,	он	
будет	сразу	же	обработан	редактором	связей	(линкером).	Редактор	связей	рас

2	 «Википедия»	 (англ.)	 дает	 другое	 толкование	 аббревиатуры	 DWARF  –	 Debug	 With	
Attributed	Record	Format –	отладка	с использованием	формата	записей	с атрибутами.	
Смысл	несколько	иной,	но	суть	дела	не меняется. –	Прим. перев.

сматривает	предложенный	объектный	код	и выполняет	поиск	в системе	дру
гих	необходимых	файлов,	обычно	системных	сервисов	или	прочих	объектных	
файлов.	 Эти	 файлы	 объединяются	 со	 сгенерированным	 объектным	 кодом,	
и  редактор	 связей	 создает	 выполняемый	файл.	 Разумеется,	 редактор	 связей	
выдаст	все	возможные	сообщения	об	отсутствующих	компонентах	и т.	п.	Если	
это	произошло,	выпейте	еще	одну	чашечку	кофе	и проверьте	свой	исходный	
код	и содержимое	makefile.

В	 рассматриваемом	 здесь	 примере	 используется	 функциональность	 GCC	
(ниже	воспроизводятся	соответствующие	строки	из	makefile):

hello: hello.o
      gcc -o hello hello.o -no-pie

Последние	версии	компилятора	и линкера	GCC	по	умолчанию	генерируют	
выполняемый	код,	не зависящий	от	положения	в памяти,	или	перемещаемый	
код	(position	independent	executable –	PIE).	Это	делается	для	защиты	от	хаке
ров,	исследующих,	как	память	используется	программой,	что	в итоге	позво
ляет	им	воздействовать	на	выполнение	программы.	В рассматриваемом	здесь	
примере	 не  создается	 перемещаемый	 выполняемый	 код,	 потому	 что	 такая	
программа	слишком	сложна	для	анализа	(усложнение	делается	преднамерен
но	из	соображений	обеспечения	безопасности).	Поэтому	в makefile	добавлен	
ключ	-no-pie.

В	makefile	можно	добавлять	комментарии,	начинающиеся	с символа	«решет
ка»	#:

#makefile for hello.asm

Здесь	используется	GCC	для	упрощения	доступа	к функциям	стандартной	
библиотеки	 C  из	 ассемблерного	 кода.	 Иногда	мы	 будем	 пользоваться	функ
циями	языка	C,	чтобы	сделать	примеры	ассемблерного	кода	более	простыми.	
Но вы	должны	знать,	что	в Linux	существует	и другой	широко	известный	GNU	
линкер	ld.

Если	 несколько	 предыдущих	 абзацев	 оказались	 для	 вас	 непонятными,	
не волнуйтесь,	выпейте	еще	чашечку	кофе	и продолжайте	чтение –	это	всего	
лишь	вспомогательная	информация,	которая	не очень	важна	на	текущем	эта
пе.	Просто	помните,	что	makefile –	ваш	друг,	который	выполняет	за	вас	огром
ную	работу,	а единственное,	о чем	следует	беспокоиться	сейчас, –	не делать	
собственных	ошибок.

В	 командной	 строке	 перейдите	 в  каталог,	 где	 сохранены	файлы	 hello.asm 
и makefile.	Выполните	команду	make	для	ассемблирования	и сборки	програм
мы,	затем	запустите	созданную	программу,	набрав	./hello	в командной	строке.	
Если	появилось	сообщение	hello, world	перед	очередным	промптом	командной	
строки,	то	все	работает	правильно.	Если	сообщение	не выведено,	то	в исходном	
коде	была	допущена	опечатка	или	какаято	другая	ошибка,	поэтому	необхо
димо	еще	раз	проверить	содержимое	файлов	hello.asm	и makefile.	Наполните	
очередную	чашечку	кофе	и приступайте	к отладке.

На	рис. 1.3	показан	пример	вывода	ожидаемого	сообщения	на	экран.

Рис. 1.3.	Вывод строки hello, world

Структура прОграммы на аССемблере
Рассматриваемая	здесь	первая	программа	демонстрирует	базовую	структуру	
любой	 ассемблерной	 программы.	 Ниже	 перечислены	 основные	 части	 про
граммы	на	ассемблере:

	� section .data
	� section .bss
	� section .txt

Раздел section .data
В	разделе	section .data	объявляются	и определяются	инициализируемые	дан
ные	в следующем	формате:

  <variable name>       <type>       <value>

Если	переменная	включена	в раздел	section .data,	для	нее	выделяется	память	
при	ассемблировании	и связывании	исходного	кода	для	создания	выполняе
мого	кода.	Переменные	имеют	символьные	имена	и ссылки	на	локации	в па
мяти,	при	этом	переменная	может	занимать	одну	или	несколько	ячеек	памяти.	
Имя	переменной	обозначает	начальный	адрес	переменной	в памяти.
Имена	переменных	должны	начинаться	 с  буквы,	 за	 которой	 следуют	 буквы,	
цифры	или	некоторые	специальные	символы.	В табл. 1.1	перечислены	возмож
ные	типы	данных.

Таблица 1.1.	Типы данных

Тип Длина Название

db 8 бит Байт

dw 16 бит Слово

dd 32 бита Двойное слово

dq 64 бита Учетверенное (двойное длинное) слово

В	рассматриваемом	здесь	примере	программы	section .data	содержит	одну	
переменную	msg,	символьное	имя	которой	указывает	на	адрес	памяти,	по	ко
торому	размещается	'h',	первый	байт	строки	"hello, world",0.	Таким	образом,	
msg	указывает	на	букву	'h',	msg+1	указывает	на	букву	'e'	и т.	д.	Эта	перемен

ная	 называется	 строкой	 (string),	 которая	 является	 непрерывным	 списком	
символов.	 Строка –	 это	 «список»	 или	 «массив»	 символов	 в  памяти.	 В  дей
ствительности	любой	непрерывный	 список	 значений	в памяти	может	 счи
таться	строкой,	при	этом	символы	могут	быть	видимыми	или	невидимыми	
для	человеческого	глаза,	а сама	строка	может	иметь	смысл	для	человека	или	
не иметь	никакого	смысла.

Для	удобства	добавляется	ноль,	обозначающий	конец	строки,	видимой	для	
человека.	Можно	не записывать	завершающий	ноль	на	свой	страх	и риск.	Этот	
завершающий	0	не соответствует	значению	ASCII	0,	это	числовой	ноль,	так	что	
в памяти	он	представлен	восемью	нулевыми	битами.	Если	аббревиатура	ASCII	
вызвала	у вас	недоумение,	обратитесь	к поиску	Google.	Знание	смысла	акро
нима	ASCII	весьма	важно	в программировании.	Краткое	описание:	символы,	
используемые	человеком,	представлены	в виде	специальных	(числовых)	кодов	
в компьютерах.	Прописная	(заглавная)	буква	A (латиница)	имеет	код	65,	B соот
ветствует	код	66	и т.	д.	Для	символа	перехода	на	новую	строку	определен	код	10,	
а невидимый	символ	NULL	получил	код	0.	Таким	образом,	мы	завершаем	строку	
символом	NULL.	Если	в командной	строке	ввести	man ascii,	то	Linux	выведет	таб
лицу	символов	и кодов	ASCII.

Раздел	section .data	также	может	содержать	константы,	т. е.	значения,	кото
рые	невозможно	изменить	в программе.	Константы	определяются	в следую
щем	формате:

  <constant name>      equ      <value>

Пример:

   pi equ 3.1416

Раздел section .bss
Аббревиатура	bss	означает	Block	Started	by	Symbol	(блок,	начинающийся	с сим
вола)	и ведет	свое	происхождение	с 1950х	гг.,	когда	этот	блок	был	компонентом	
языка	ассемблера,	разработанного	для	IBM	704.	В этот	раздел	помещаются	не
инициализированные	переменные.	Для	таких	неинициализированных	пере
менных	выделяемое	пространство	памяти	объявляется	в следующем	формате:

<variable name>      <type>      <number>

В	табл. 1.2	перечислены	возможные	типы	данных	bss.

Таблица 1.2.	Типы данных bss

Тип Длина Название

resb 8 бит Байт

resw 16 бит Слово

resd 32 бита Двойное слово

resq 64 бита Учетверенное (двойное длинное) слово

Например,	следующая	инструкция	объявляет	пространство	памяти	для	мас
сива	из	20	двойных	слов:

  dArray resd 20

Переменные	в разделе	section .bss	не содержат	какихлибо	значений,	значе
ния	будут	присваиваться	им	в дальнейшем	во	время	выполнения	программы.	
Блоки	памяти	для	этих	переменных	резервируются	не во	время	компиляции,	
а во	время	выполнения.	В последующих	примерах	будет	продемонстрировано	
практическое	использование	раздела	section .bss.	Когда	программа	начинает	
выполняться,	она	запрашивает	у операционной	системы	необходимую	память,	
выделяемую	переменным	из	раздела	section .bss	и инициализируемую	нуля
ми.	Если	во	время	выполнения	не существует	доступной	памяти,	достаточной	
для	размещения	переменных	.bss,	то	программа	завершается	аварийно.

Раздел section .txt
Все	действия	происходят	в разделе	section .txt.	Этот	раздел	содержит	код	про
граммы	и начинается	со	следующих	инструкций:

            global main
  main:

Часть	main:	называется	меткой	(label).	Если	метка	расположена	в строке,	где	
после	нее	нет	других	 символов,	то	после	 слова	должно	быть	 записано	двое
точие,	иначе	ассемблер	выведет	предупреждающее	сообщение.	А предупреж
дающие	сообщения	не следует	игнорировать.	Если	за	меткой	следуют	другие	
инструкции	(в той	же	строке),	то	двоеточие	не обязательно,	но	все	же	лучше	
выработать	 полезную	 привычку	 завершать	 все	 метки	 символом	 двоеточия.	
К тому	же	это	повышает	удобство	чтения	исходного	кода.

В	 рассматриваемом	 здесь	 примере	 исходного	 кода	 hello.asm	 после	 метки	
main:	регистры	rdi,	rsi	и rax	подготавливаются	для	вывода	сообщения	на	экран.	
Более	подробно	о регистрах	вы	узнаете	в главе	2.	Здесь	просто	выводится	стро
ка	на	экран	с использованием	системного	вызова.	То есть	мы	предлагаем	опе
рационной	системе	выполнить	эту	работу.

	� В	регистр	rax	записывается	код	системного	вызова	1,	означающий	write	
(запись).

	� Для	записи	(размещения)	некоторого	значения	в регистр	используется	
инструкция	mov.	В действительности	эта	инструкция	ничего	не переме
щает	(move –	перемещение),	она	создает	копию	источника	(source)	и за
писывает	эту	копию	в цель	(destination).	Формат	команды	mov:

       mov destination, source

	� Инструкцию	mov	можно	использовать	следующим	образом:
	� mov	регистр,	непосредственное_значение
	� mov	регистр,	адрес_памяти
	� mov	адрес_памяти,	регистр
	� недопустимое	использование:	mov	адрес_памяти,	адрес_памяти

	� В	 рассматриваемом	 здесь	 примере	 целевое	 устройство	 вывода	 для	 
записи	сообщения	сохраняется	в регистре	rdi,	а значение	1	обозначает	
устройство	стандартного	вывода	(в данном	случае	это	вывод	на	экран).	

	� Адрес	(памяти)	выводимой	строки	записывается	в регистр	rsi.
	� В	регистр	rdx	помещается	длина	сообщения.	Количество	символов	в стро
ке	hello, world,	не включая	кавычки,	в которые	заключена	строка,	и завер
шающий	0.	Если	в количество	включить	завершающий	0,	то	программа	
попытается	вывести	на	экран	NULLбайт,	что	не имеет	особого	смысла.

	� Затем	выполняется	системный	вызов	syscall,	и строка	msg	выводится	на	
устройство	стандартного	вывода.	syscall –	это	запрос	функционально
сти,	предоставляемой	операционной	системой.

	� Чтобы	 избежать	 вывода	 сообщений	 об	 ошибках	 при	 завершении	 вы
полнения	программы,	необходим	корректный	(«чистый»)	выход	из	нее.	
Для	этого	сначала	в регистр	rax	записывается	код	60,	обозначающий	exit 
(выход).	Код	«успешного»	выхода	(завершения	программы)	помещается	
в регистр	rdi,	затем	выполняется	системный	вызов.	Программа	заверша
ет	работу	без	какихлибо	сообщений.

Системные	 вызовы	 (system	 calls)	 используются	 для	 запросов	 к  операци
онной	 системе	для	 выполнения	 некоторых	 конкретных	действий.	 В  каждой	
операционной	 системе	 существует	 собственный	 набор	 параметров	 систем
ных	вызовов,	и системные	вызовы	в Linux	отличаются	от	системных	вызовов	
в Windows	или	macOS.	В этой	книге	мы	будем	использовать	системные	вызовы	
Linux	для	версии	x64,	более	подробную	информацию	о них	можно	найти	здесь:	
http://blog.rchapman.org/posts/Linux_System_Call_Table_for_x86_64/.

Следует	учитывать,	что	системные	вызовы	32битовой	версии	отличаются	
от	системных	вызовов	64битовой	версии.	При	чтении	исходного	кода	всегда	
проверяйте,	написан	ли	код	для	32битовых	или	для	64битовых	систем.

Перейдите	 в  командную	 строку	 операционной	 системы	 и  найдите	 файл	
hello.lst.	Этот	файл	сгенерирован	во	время	ассемблирования,	но	перед	связы
ванием	(линковкой),	как	определено	в makefile.	Откройте	hello.lst	в редакторе –	
и увидите	листинг	исходного	кода	на	ассемблере,	но	в самом	левом	столбце	
показаны	относительные	адреса	инструкций	кода,	а во	втором	слева	столбце	
содержится	код,	переведенный	в машинный	язык	(в шестнадцатеричном	фор
мате).	На рис. 1.4	показано	содержимое	файла	hello.lst.

Рис. 1.4.	Файл hello.lst в редакторе

28	  Самая первая программа



Здесь	можно	видеть	столбец	с номерами	строк	и следующий	столбец	шири
ной	8	знакомест.	Этот	столбец	представляет	адреса	блоков	памяти.	Когда	ас
семблер	(программа	ассемблирования	кода)	создает	объектный	файл,	он	пока	
еще	не знает,	какие	адреса	памяти	будут	использоваться.	Поэтому	ассемблер	
начинает	нумерацию	с адреса	0	для	различных	разделов.	Для	раздела	section 
.bss	память	не выделяется.

Во	 втором	 столбце	 содержится	 результат	 преобразования	 инструкций	 ас
семблера	 в шестнадцатеричный	 (машинный)	 код.	Например,	 инструкция	 mov 
rax	преобразована	в B8,	 а mov rdi	 в BF.	Это	шестнадцатеричное	представление	
машинных	инструкций.	Следует	также	обратить	внимание	на	преобразование	
строки	 msg	 в шестнадцатеричные	ASCIIсимволы.	Немного	 позже	 вы	 узнаете	
больше	 о  шестнадцатеричном	 формате	 записи.	 Первая	 инструкция	 должна	
выполняться,	начиная	с адреса	00000000,	она	занимает	пять	байт:	B8 01 00 00 00.	
Здесь	двойные	нули	нужны	для	заполнения	и выравнивания	по	адресам	памя
ти.	Выравнивание	по	адресам	памяти –	это	функциональная	особенность,	ис
пользуемая	ассемблерами	и компиляторами	для	оптимизации	кода.	Ассембле
рам	и компиляторам	можно	передавать	разнообразные	флаги,	чтобы	получить	
наименьший	возможный	размер	выполняемого	файла,	самый	быстрый	выпол
няемый	код	или	объединение	этих	характеристик.	В следующих	главах	будет	
рассматриваться	оптимизация	с целью	увеличения	скорости	выполнения	кода.

Следующая	инструкция	начинается	с адреса	00000005	и т.	д.	Адреса	памяти	
содержат	восемь	знакомест	(т. е.	8	байт),	в каждом	байте	8	бит.	Поэтому	адреса	
имеют	длину	64	бита,	 разумеется,	 если	используется	64битовый	ассемблер.	
Теперь	рассмотрим,	как	представлена	ссылка	на	строку	msg.	Поскольку	реаль
ный	адрес	памяти	msg	пока	еще	неизвестен,	ссылка	на	эту	строку	обозначена	
как	[0000000000000000].

Вероятно,	вы	согласитесь,	что	мнемонические	символьные	коды	ассембле
ра	и символьные	имена	адресов	памяти	(переменных)	немного	легче	читать	
и  запоминать,	 чем	 шестнадцатеричные	 значения,	 учитывая	 существование	
сотен	 кодов	 с  разнообразными	 операндами,	 каждый	 из	 которых	 приводит	
к генерации	еще	большего	количества	шестнадцатеричных	инструкций.	Когда	
компьютеры	только	еще	начинали	применяться,	программисты	использова
ли	только	машинный	язык,	язык	программирования	первого	поколения.	Язык	
ассемблера	с мнемоническими	кодами,	«более	простыми	для	запоминания», –	
это	язык	программирования	второго	поколения.

резюме
В	этой	главе	вы	узнали:

	� об	основной	структуре	программы	на	ассемблере,	с ее	различными	раз
делами;

	� об	использовании	памяти	с символьными	именами	для	адресов;
	� о	регистрах;
	� об	одной	из	инструкций	ассемблера	mov;
	� об	использовании	системного	вызова	syscall;
	� о	различиях	между	ассемблерным	и машинным	кодами.

Глава 2
Двоичные 

и шестнадцатеричные  
числа и регистры

В	 современных	 компьютерах	 бит	 (bit)	 является	 наименьшим	 фрагментом	
информации.	Бит	может	иметь	одно	из	значений:	1	или	0.	В этой	главе	рас
сматривается,	как	объединяются	биты	для	представления	данных,	таких	как	
целые	числа	или	значения	с плавающей	точкой.	Десятичное	представление	
чисел,	 весьма	 удобное	 для	 людей,	 не  является	 наиболее	 подходящим	 для	
компьютерной	обработки.	При	использовании	двоичной	(бинарной)	систе
мы,	в которой	возможны	только	два	значения	(1	или	0),	гораздо	более	эф
фективно	 выполняется	 работа	 со	 степенями	 2.	 Когда	мы	 обсуждаем	исто
рию	поколений	компью	теров,	то	вспоминаем	8битовые	ЦПУ	(центральные	
процессорные	устройства,	или	просто	центральные	процессоры)	(23),	16би
товые	 ЦПУ  (24),	 32битовые	 ЦПУ	 (25)	 и  ныне  преобладающие	 64битовые	
ЦПУ (26).	Но для	людей	работа	с длинными	строками	единиц	и нулей	весьма	
неудобна	и чаще	всего	даже	невозможна.	В этой	главе	будет	показано,	как	
преобразовать	биты	в десятичные	или	шестнадцатеричные	значения,	с ко
торыми	гораздо	проще	работать.	После	этого	будут	рассматриваться	реги
стры	(registers) –	области	хранения	данных,	которые	помогают	процессору	
выполнять	логические	и арифметические	инструкции.

краткий ввОдный курС пО двОичным чиСлам
Компьютеры	используют	двоичные	(бинарные)	цифры	(нули	и единицы)	для	
выполнения	требуемой	работы.	Восемь	бинарных	цифр	объединяются	и на
зываются	байтом	(byte).	Но двоичные	числа	слишком	длинны	для	того,	чтобы	
человек	работал	с ними,	не говоря	уже	о запоминании.	Шестнадцатеричные	
числа	(немного)	более	удобны,	и не в последнюю	очередь	потому,	что	любой	
8битовый	байт	можно	представить	в виде	всего	лишь	двух	шестнадцатерич
ных	цифр.

Если	 необходимо	 увидеть	 двоичное,	 десятичное	 или	 шестнадцатеричное	
значение	различных	форматов	вывода,	то	нужно	воспользоваться	инструмен



тальным	средством	преобразования.	В интернете	можно	найти	огромное	ко
личество	калькуляторов	с функциями	преобразования.	Ниже	перечислено	не
сколько	таких	калькуляторов,	которые	наиболее	просты	в использовании:

	� www.binaryconvert.com;
	� https://www.binaryhexconverter.com;
	� https://babbage.cs.qc.cuny.edu/IEEE-754.

В	табл. 2.1	приведены	преобразования	чисел	с 0	до	15,	эти	преобразования	
полезно	запомнить.

Таблица 2.1.	Числа от 0 до 15 в десятичном, шестнадцатеричном и двоичном форматах

Десятичное Шестнадцатеричное Двоичное

0 0 0000

1 1 0001

2 2 0010

3 3 0011

4 4 0100

5 5 0101

6 6 0110

7 7 0111

8 8 1000

9 9 1001

10 a 1010

11 b 1011

12 c 1100

13 d 1101

14 e 1110

15 f 1111

Целые числа
Существуют	два	типа	целых	чисел:	знаковые	(signed)	и беззнаковые	(unsigned).	
Для	знаковых	целых	чисел	крайний	левый	бит	равен	1,	если	число	отрицатель
ное,	или	0,	если	число	положительное.	Беззнаковые	целые	числа –	это	0	и все	
положительные	целые,	для	знакового	бита	место	не предусмотрено.	Для	кор
ректной	работы	 с целочисленной	 арифметикой	отрицательные	целые	числа	
используются	в так	называемом	представлении	числа	в дополнительном	дво
ичном	 коде	 (two’s	 complement).	 Получить	 двоичное	 представление	 отрица
тельного	числа	можно	следующим	образом:

1)	записать	абсолютное	значение	числа	в двоичном	виде;
2)	перевести	в дополнительный	код	(заменить	все	1	на	0,	а все	0	на	1);
3)	прибавить	1.

Ниже	приведен	пример	с использованием	16битовых	чисел	вместо	64би
товых	(чтобы	пример	выглядел	более	понятно):

decimal number =     17
binary number =      0000    0000    0001    0001
hexadecimal number =   0       0       1       1    = 11
decimal number =                      -17
binary number absolute value =        0000    0000    0001    0001
complement =                          1111    1111    1110    1110
add 1 =                               1111    1111    1110    1111
hexadecimal =                         f       f       e       f      = ffef

Verify:    -17       11111111 11101111
add:       +17       00000000 00010001
equals:    0         00000000 00000000

Обычно	 перед	 шестнадцатеричными	 числами	 записывается	 префикс	 0x,	
чтобы	отличать	их	от	десятичных	чисел,	например	-17	в шестнадцатеричном	
представлении	выглядит	как	0xffef.	Если	при	просмотре	листинга	на	машин
ном	 языке	 в файле	 .lst	 вы	 видите	 число	 0xffef,	 то	 из	 контекста	 необходимо	
определить,	является	ли	это	значение	знаковым	или	беззнаковым	целым	чис
лом.	Если	это	знаковое	целое,	то	оно	означает	-17	в десятичном	представле
нии.	Если	это	беззнаковое	целое,	то	оно	соответствует	десятичному	числу	65519.	
Разуме	ется,	если	это	адрес	памяти,	то	число	беззнаковое	(понятно,	почему?).	
Иногда	 в  ассемблерном	 коде	 будут	 встречаться	 некоторые	 другие	 форматы	
записи	 (нотации),	 такие	 как	 0800h,	 которые	также	 представляют	шестнадца
теричное	 число,	 10010111b –	двоичное	 число	 или	 420o –	 восьмеричное	 число.	
Да,	не удивляйтесь,	восьмеричные	числа	также	можно	использовать.	Мы бу
дем	применять	восьмеричные	числа,	когда	будем	писать	код	для	файла	ввода/
вывода.	При	необходимости	преобразования	целых	чисел	не трудитесь	сами,	
пользуйтесь	упомянутыми	выше	вебсайтами.

Числа с плавающей точкой
Числа	с плавающей	точкой	записываются	в двоичном	или	шестнадцатеричном	
формате	 в  соответствии	 с  международным	 стандартом	 IEEE754.	 Этот	 про
цесс	еще	более	сложен,	чем	для	целых	чисел.	Если	хотите	узнать	об	этом	более	 
подробно,	то	рекомендуется	начать	здесь:

http://mathcenter.oxford.emory.edu/site/cs170/ieee754/.

Но	и в	этом	случае	при	необходимости	преобразования	чисел	с плавающей	
точкой	рекомендуется	пользоваться	вебсайтами,	перечисленными	в преды
дущем	разделе.	Здесь	мы	не будем	углубляться	в подробности.

краткий ввОдный курС пО региСтрам
Центральное	процессорное	 устройство	 (ЦПУ),	 или	просто	процессор, –	мозг	
компьютера –	выполняет	инструкции	программы,	весьма	активно	используя	
регистры	и оперативную	память	и производя	математические	и логические	

операции	в этих	регистрах	и в	памяти.	Следовательно,	важно	обладать	основ
ными	знаниями	о регистрах	и памяти	и о	том,	как	они	используются.	Здесь	при
водится	краткий	обзор	регистров,	подробности	об	использовании	регистров	
будут	постепенно	выясняться	в  следующих	 главах.	Регистры	 (registers) –	 это	
области	(локации),	используемые	процессором	для	хранения	данных,	инструк
ций	или	адресов	памяти.	Количество	регистров	невелико,	но	процессор	может	
считывать	и записывать	их	чрезвычайно	быстро.	Можно	считать	регистры	не
которой	разновидностью	блокнота	для	процессора,	где	он	хранит	временную	
информацию.	 Рекомендуется	 запомнить	 правило:	 если	 важна	 скорость,	 то	
процессор	может	получать	доступ	к регистрам	намного	быстрее,	чем	доступ	
к оперативной	памяти.

Не беспокойтесь,	если	этот	раздел	не совсем	понятен,	все	постепенно	станет	
ясным,	когда	мы	начнем	использовать	регистры	на	практике	в следующих	главах.

Регистры общего назначения
Существует	16	регистров	общего	назначения,	и каждый	из	них	может	исполь
зоваться	 как	 64битовый,	 32битовый,	 16битовый	 или	 8битовый	 регистр.	
В табл.  2.2	перечислены	имена	 всех	 этих	регистров,	 обозначающие	 соответ
ствующие	 размеры.	 Четыре	 регистра	 rax,	 rbx,	 rcx,	 rdx	 могут	 иметь	 два	 типа	
8битовых	регистров –	нижние	8	бит	в нижней	половине 16битового	регистра	
и верхние	8	бит	в верхней	половине 16битового	регистра.

Таблица 2.2.	Имена регистров общего назначения

64-битовый 32-битовый 16-битовый Нижний 
8-битовый

Верхний 
8-битовый

Комментарий

rax eax ax al ah

rbx ebx bx bl bh

rcx ecx cx cl ch

rdx edx dx dl dh

rsi esi si sil –

rdi edi di dil –

rbp ebp bp bpl – Указатель базы 
(адреса)

rsp esp sp spl – Указатель стека

r8 r8d r8w r8b –

r9 r9d r9w r9b –

r10 r10d r10w r10b –

r11 r11d r11w r11b –

r12 r12d r12w r12b –

r13 r13d r13w r13b –

r14 r14d r14w r14b –

r15 r15d r15w r15b –

Несмотря	на	то	что	регистры	rbp	и rsp	называются	регистрами	общего	на
значения,	с ними	нужно	работать	с особой	осторожностью,	так	как	они	исполь
зуются	процессором	во	время	выполнения	программы.	Правильное	использо
вание	регистров	rbp	и rsp	будет	подробнее	рассмотрено	в следующих	главах.

Каждый	64битовый	регистр	содержит	набор	из	64	бит,	нулей	и/или	единиц,	
т. е.	8	байт.	При	записи	числа	60	в регистр	rax	в программе	hello, world	этот	ре
гистр	содержит	следующее	значение:

00000000 00000000 00000000 00000000 00000000 00000000 00000000 00111100

Это	двоичное	представление	числа	60	в 64битовом	регистре.
Каждый	32битовый	регистр –	это	набор	32	нижних	(если	считать	справа)	бит	

64битового	регистра.	Точно	так	же	16битовый	регистр	и 8битовый	регистр	
состоят	из	нижних	16	и нижних	8	бит	(соответственно)	64битового	регистра.

Примечание. Запомните: «нижние» биты – это всегда крайние правые биты.

Бит	с номером	0 –	это	крайний	правый	бит,	отсчет	начинается	справа	с ин
декса	0,	а не 1.	Таким	образом,	крайний	левый	бит	64битового	регистра	имеет	
индекс	63,	а не 64.

Если	регистр	rax	содержит	значение	60,	то	можно	также	сказать,	что	регистр	
eax	теперь	содержит	следующее	значение:

      00000000 00000000 00000000 00111100

или	что	регистр	ax	содержит:

      00000000 00111100

а	в регистре	al	хранится	значение

      00111100

Регистр счетчика команд (rip)
Процессор	постоянно	отслеживает	следующую	выполняемую	инструкцию,	со
храняя	ее	адрес	в регистре	rip.	 Значение	в регистре	rip	можно	изменить	на	
любое	на	свой	страх	и риск	(я вас	предупредил).	Более	безопасный	способ	из
менения	значения	rip –	использование	инструкций	переходов	(jump),	которые	
рассматриваются	в следующей	главе.

Регистр флагов
Существует	 также	 регистр	 флагов	 (процессора)	 rflags,	 возможные	 значения	
для	 которого	 приведены	 в  табл.  2.3.	 После	 выполнения	 очередной	 инструк
ции	программа	может	проверить,	установлен	ли	конкретный	флаг	(например,	
ZF=1),	чтобы	затем	действовать	соответственно.

Таблица 2.3.	Значения флагов процессора

Имя Символьное 
обозначение

Бит Описание

Carry CF 0 В предыдущей инструкции был выполнен перенос (раз-
рядов)

Parity PF 2 Последний байт содержит четное число единиц

Adjust AF 4 Операции BCD (в двоично-десятичном коде)

Zero ZF 6 Результат предыдущей инструкции равен нулю

Sign SF 8 В результате выполнения предыдущей инструкции самый 
значимый бит равен 1

Direction DF 10 Направление операций со строками (инкремент или 
декремент)

Overflow OF 11 В результате выполнения предыдущей инструкции воз-
никло переполнение

Более	 подробное	 описание	 флагов	 и  их	 практического	 использования	
см. немного	позже.

Существует	еще	один	регистр	флагов	с именем	MXCSR,	который	будет	ис
пользоваться	в одной	инструкции	со	многими	потоками	данных	(SIMD),	в со
ответствующей	главе	регистр	MXCSR	будет	описан	более	подробно.

Регистры xmm и ymm
Эти	регистры	используются	для	операций	с числами	с плавающей	точкой	и в	
инструкциях	со	многими	потоками	данных	(SIMD).	В дальнейшем	регистр	xmm 
и связанный	с ним	регистр	ymm	будут	активно	использоваться,	начиная	с изуче
ния	инструкций	обработки	чисел	с плавающей	точкой.

В	дополнение	ко	всем	описанным	выше	регистрам	существуют	и другие	ре
гистры,	но	мы	не будем	их	использовать	в этой	книге.

С	теорией	покончено,	наступило	время	для	практической	работы.

резюме
В	этой	главе	вы	узнали,	как:

	� представить	 значение	 в десятичном,	двоичном	и шестнадцатеричном	
форматах;

	� использовать	регистры	и флаги.

Глава 3
Анализ программ  

с помощью отладчика: GDB

В	этой	главе	представлено	введение	в процесс	отладки	программ	на	ассембле
ре.	Отладка –	это	важный	навык,	потому	что	с отладчиком	можно	исследовать	
содержимое	регистров	и оперативной	памяти	в шестнадцатеричном,	двоич
ном	или	десятичном	представлении.	Из предыдущей	главы	вам	уже	известно,	
что	процессор	интенсивно	использует	регистры	и память,	а отладчик	позво
лит	вам	выполнять	инструкции	шаг	за	шагом,	наблюдая	при	этом,	как	изме
няется	содержимое	регистров,	памяти	и флагов.	Возможно,	вы	уже	наблюдали	
аварийное	 завершение	 своей	 первой	 программы	 на	 ассемблере	 с  выводом	
непонятного	сообщения	типа	«Memory	Segmentation	Fault»	 (Нарушение	сег
ментации	памяти).	С помощью	отладчика	вы	сможете	постепенно	(пошагово)	
пройти	по	программе	во	время	ее	выполнения	и точно	определить	местона
хождение	и причину	ошибки.

началО Отладки
После	ассемблирования	и связывания	(линковки)	программы	hello, world	без	
ошибок	на	этом	этапе	вы	получаете	выполняемый	файл.	С помощью	инстру
ментаотладчика	можно	загрузить	выполняемую	программу	в память	компью
тера	и  выполнять	 ее	 строку	 за	 строкой,	 наблюдая	при	 этом	 за	 содержимым	
различных	регистров	и блоков	памяти.	В настоящее	время	доступно	несколько	
бесплатных	и коммерческих	отладчиков.	В Linux	прародителем	всех	отладчи
ков	является	GDB –	программа	с интерфейсом	командной	строки	с весьма	за
мысловатыми	командами.	Но так	даже	интереснее.	В следующих	главах	будет	
использоваться	 SASM,	 инструментальное	 средство	 с  графическим	пользова
тельским	интерфейсом,	основанное	на	GDB.	Но освоение	базовых	знаний	о са
мом	отладчике	GDB	может	оказаться	полезным,	потому	что	не вся	функцио
нальность	GDB	доступна	в SASM.

В	 своей	дальнейшей	карьере	программиста	на	 ассемблере	 вы	наверняка	
будете	 обращать	 внимание	на	 разнообразные	отладчики	 с  превосходными	
пользовательскими	 интерфейсами,	 причем	 каждый	 отладчик	 ориентиро



ван	на	конкретную	платформу,	как,	например,	Windows,	Mac	или	Linux.	Эти	
GUIотладчики	помогут	отлаживать	длинные	и сложные	программы	быстрее	
и  проще	 по	 сравнению	 с  отладчиком	 с  интерфейсом	 командной	 строки.	
Но GDB	представляет	собой	всеобъемлющий,	однако	при	этом	«самый	пря
мой»	путь	к освоению	отладки	в Linux.	GDB	установлен	в большинстве	систем	
разработки	на	платформе	Linux,	 а  если	он	отсутствует,	то	 его	можно	легко	
установить.	 GDB	 практически	 не  увеличивает	 нагрузку	 на	 систему.	 Сейчас	
мы	будем	использовать	GDB,	чтобы	познакомить	вас	с его	самыми	важными	
свойствами,	а также	для	более	плавного	перехода	к другим	инструменталь
ным	средствам	в следующих	главах.	Единственное	замечание:	GDB	выглядит	
так,	как	будто	предназначен	для	отладки	программ	на	языках	высокого	уров
ня,	некоторые	его	функциональные	возможности	не помогут	при	отладке	ас
семблерных	программ.

На	 первых	 порах	 отладка	 программы	 с  использованием	 отладчика	 в  ко
мандной	строке	может	показаться	чрезвычайно	трудной.	Но не отчаивайтесь,	
читая	эту	главу, –	продвигаясь	все	дальше	и дальше,	вы	увидите,	как	все	стано
вится	проще.

Чтобы	начать	отладку	программы	hello	в командной	строке,	перейдите	в ка
талог,	где	была	сохранена	эта	программа.	После	промпта	в командной	строке	
введите:

gdb hello

GDB	загрузит	выполняемый	файл	hello	в память	и выведет	собственное	при
глашение	 (промпт)	 (gdb),	 после	 чего	 будет	 ожидать	 ввода	 инструкций.	 Если	
ввести	следующую	команду:

list

то	GDB	выведет	некоторое	количество	строк	исходного	кода	из	загруженного	
файла.	Введите	команду	list	еще	раз,	и GDB	выведет	следующие	строки	и т. д.	
Для	 вывода	 конкретной	 строки,	 например	 самой	 первой	 строки	 исходного	
кода,	введите	команду	list 1.	На рис. 3.1	показан	пример.

Если	вывод	на	ваш	экран	отличается	от	показанного	на	рис. 3.1	и содержит	
множество	символов	%,	значит,	в вашем	случае	GDB	сконфигурирован	для	ис
пользования	синтаксиса	ассемблера	версии	AT&T.	Мы будем	использовать	син
таксис	ассемблера	версии	Intel,	который	более	интуитивно	понятен	(для нас).	
Ниже	будет	показано,	как	изменить	версию	ассемблера.

После	ввода	следующей	команды

run

GDB	запустит	программу	hello,	выведет	строку	hello, world	и вернется	к своему	
промпту	(gdb).	На рис. 3.2	показан	результат	выполнения	команды	run	и вывод	
на	экран.

Рис. 3.1.	Вывод команды list отладчика GDB

Рис. 3.2.	Вывод команды run отладчика GDB

Для	выхода	из	GDB	введите	команду	quit.
Теперь	начнем	делать	некоторые	более	интересные	вещи	с помощью	GDB.
Но	сначала	изменим	версию	дизассемблера –	это	необходимо	сделать,	только	

если	вы	видите	много	символов	%	в выводе	предыдущего	примера.	Загрузите	
выполняемый	файл	hello	в отладчик	GDB,	если	он	еще	не загружен.

Введите	команду

      set disassembly-flavor intel

Эта	команда	преобразует	дизассемблируемый	код	в формат,	который	нам	
уже	 знаком.	 Можно	 сделать	 версию	 Intel	 форматом	 ассемблера	 по	 умолча
нию	для	 отладчика	 GDB,	 используя	 соответствующие	 параметры	 настройки	
в  профиле	 командной	 оболочки	 Linux.	 См.	 документацию	 по	 конкретному	
дистрибутиву	Linux.	В Ubuntu	18.04	необходимо	в своем	домашнем	каталоге	
создать	файл	.gdbinit,	содержащий	приведенную	выше	инструкцию	set.	Чтобы	
ассемблер	Intel	использовался	по	умолчанию,	необходимо	выйти	из	отладчика	
и снова	войти	в него.

Запустите	отладчик	GDB	с файлом	hello,	чтобы	начать	анализ	программы.	
Как	вам	уже	известно,	программа	hello, world	сначала	инициализирует	неко
торые	данные	в разделах	section .data	и section .bss,	затем	переходит	к метке	
main.	Именно	здесь	начинается	настоящее	выполнение,	поэтому	начнем	иссле
дование	с этой	метки.

После	промпта	(gdb)	введите	команду

      disassemble main

Отладчик	GDB	возвращает	исходный	код	в более	или	менее	похожем	виде.	
Возвращаемый	исходный	код	не совсем	похож	на	код,	написанный	изначаль
но.	Это	слегка	удивляет.	Что	произошло?	Ситуация	требует	анализа.

На	рис. 3.3	показан	код,	который	GDB	вывел	на	экран	в ответ	на	команду	
дизассемблирования.

Рис. 3.3.	Вывод результата дизассемблирования отладчика GDB

Длинные	числа	слева,	начинающиеся	с 0x00…, –	это	адреса	памяти,	т. е.	лока
ций,	 в  которых	 сохранены	машинные	инструкции	программы.	Здесь	можно	
видеть	по	адресам	и по	<+5>	во	второй	строке,	что	для	первой	инструкции	mov 
eax, 0x1	необходимо	пять	байт	памяти.	Но в нашем	исходном	коде	была	запи
сана	инструкция	mov	rax,	1.	Почему	появился	регистр	eax?

Если	вернуться	к таблице	регистров	в главе	2	(табл. 2.2),	то	мы	увидим,	что	
eax	 является	нижней	 32битовой	частью	регистра	 rax.	Ассемблер	достаточно	
умен,	для	того	чтобы	понять,	что	64битовый	регистр –	это	излишний	расход	

ресурса	для	хранения	числа	1,	поэтому	использует	32битовый	регистр.	То же	
самое	относится	к использованию	регистров	edi	и edx	вместо	rdi	и rdx.

64битовый	ассемблер –	это	расширение	32битового	ассемблера,	поэтому	
в дальнейшем	вы	увидите,	что	во	всех	возможных	случаях	ассемблер	будет	ис
пользовать	32битовые	инструкции.

Значение	0x1 –	это	шестнадцатеричное	представление	десятичного	числа	1,	
0xd –	представление	десятичного	числа	13,	 а 0x3c	 соответствует	десятичному	
числу	60.

Инструкция	 nop	 означает	 «no	 operation»	 (отсутствие	 операции),	 ассемблер	
вставил	ее	здесь	по	причине,	связанной	с управлением	памятью.

А	что	произошло	со	строкой	msg?	Инструкция	mov rsi, msg	заменена	на	ин
струкцию	movabs rsi, 0x601030.	Пока	не обращайте	особого	внимания	на	movbs –	
эта	 команда	появилась	 здесь	изза	 64битовой	 адресации,	 она	применяется	
для	прямой	передачи	(значения)	в регистр.	Число	0x601030 –	это	адрес	памяти,	
по	которому	сохранена	строка	msg	в конкретном	компьютере.	В вашем	случае	
это	может	быть	совершенно	другой	адрес.

После	промпта	(gdb)	введите	команду

      x/s 0x601030 (или x/s 'адрес_памяти_в_вашем_случае')

GDB	отвечает,	как	показано	на	рис. 3.4.

Рис. 3.4.	Вывод отладчика GDB

Команда	x означает	examine	(обследовать),	ключ	s –	string	(строка).	GDB	отве
чает,	что	0x601030	является	начальным	адресом	строки	msg,	и пытается	вывести	
всю	строку	в целом	до	завершающего	ее	0.	Теперь	вам	известна	одна	из	причин,	
по	которой	после	строки	hello, world	записывается	завершающий	0.

Кроме	того,	можно	ввести	команду

      x/c 0x601030

Вывод	этой	команды	показан	на	рис. 3.5.

Рис. 3.5.	Вывод отладчика GDB

Ключ	c позволяет	вывести	один	символ.	Здесь	GDB	возвращает	первый	сим
вол	строки	msg,	перед	которым	выводится	его	десятичный	код	ASCII.	Найдите	
с помощью	Google	таблицу	кодов	ASCII,	чтобы	проверить	правильность	выве
денного	кода,	и сохраните	таблицу –	она	будет	полезна	в будущем,	а запоми
нать	ее	нет	никакой	необходимости.	Или	откройте	еще	одно	окно	терминала	
и в	командной	строке	введите	man ascii.

Рассмотрим	несколько	других	примеров.
Следующая	команда	выводит	13	символов,	начиная	с заданного	адреса	па

мяти	(см.	рис. 3.6):

      x/13c 0x601030

Рис. 3.6.	Вывод отладчика GDB

А	эта	команда	выводит	13	символов	в десятичном	представлении,	начиная	
с заданного	адреса	памяти	(см.	рис. 3.7):

      x/13d 0x601030

Рис. 3.7.	Вывод отладчика GDB

Еще	одна	команда	позволяет	получить	шестнадцатеричное	представление	
тех	же	13	символов	(см.	рис. 3.8):

      x/13x 0x601030

Рис. 3.8.	Вывод отладчика GDB

Строку	можно	вывести	и по	ссылке	на	ее	имя	msg	(см.	рис. 3.9):

      x/s &msg

Рис. 3.9.	Вывод отладчика GDB

Вернемся	к дизассемблированному	листингу.	Введите	команду

    x/2x 0x004004e0

Выводится	шестнадцатеричное	представление	содержимого	по	двум	адре
сам	памяти,	начиная	с 0x004004e0	(см.	рис. 3.10).

Рис. 3.10.	Вывод отладчика GDB

Это	 первая	 инструкция	 программы	 mov eax, 0x1	 на	 машинном	 языке.	 Это	
представление	инструкции	мы	видели	при	просмотре	файла	hello.lst.

двигаемСя дальше
Сделаем	еще	один	шаг	по	коду	программы	с помощью	отладчика.	Если	про
грамма	не загружена	в отладчик,	то	еще	раз	загрузите	ее.

Сначала	мы	прерываем	работу	программы,	приостанавливая	выполнение,	
что	позволит	обследовать	ее	некоторые	внутренние	элементы.	Введите	ко
манду

      break main

В	ответ	GDB	выводит	информацию,	показанную	на	рис. 3.11.

Рис. 3.11.	Вывод отладчика GDB

Введите	следующую	команду:

      run

На	рис. 3.12	показан	вывод	результата.

Рис. 3.12.	Вывод отладчика GDB

Отладчик	останавливается	в заданной	точке	останова	и показывает	следую
щую	инструкцию,	которая	должна	быть	выполнена.	Таким	образом,	инструк
ция	mov rax, 1	пока	еще	не выполнена.

Введите	команду

      info registers

GDB	выводит	информацию,	показанную	на	рис. 3.13.

Рис. 3.13.	Отладчик GDB вывел информацию о регистрах

	Сейчас	содержимое	регистров	для	нас	не важно,	за	исключением	регистра	
rip,	 счетчика	 (указателя)	 инструкций.	 В  регистре	 rip	 содержится	 значение	
0x4004e0 –	это	адрес	памяти	следующей	инструкции,	которая	должна	быть	вы
полнена.	Проверьте	листинг	дизассемблированного	кода:	адрес	0x4004e0	(в рас
сматриваемом	 здесь	 примере)	 указывает	 на	 первую	 инструкцию	 mov rax, 1.	
GDB	останавливается	прямо	перед	этой	инструкцией	и ожидает	ввода	команд	
пользователя.	Очень	важно	помнить,	что	инструкция,	на	которую	указывает	
регистр	rip,	пока	еще	не выполнена.

В	вашем	случае	GDB	может	вывести	адрес,	отличающийся	от	0x4004e0.	Это	
нормально,	 так	 как	 это	 адрес	 конкретной	 строки	 в  памяти,	 который	может	
быть	различным	в конкретной	конфигурации	компьютера.

Чтобы	продвинуться	на	один	шаг,	введите	команду

step

Для	вывода	содержимого	регистров	можно	пользоваться	сокращенной	вер
сией	команды	info registers:

i r

На	рис. 3.14	показан	вывод	этой	команды.

Рис. 3.14.	Отладчик GDB вывел информацию о регистрах

Разумеется,	 теперь	 регистр	 rax	 содержит	 значение	 0x1,	 а  в	 rip	 находится	
адрес	следующей	выполняемой	инструкции.

Самостоятельно	пройдите	дальше	по	программе	и обратите	внимание	на	
то,	как	регистр	rsi	принимает	адрес	строки	msg,	как	выводится	на	экран	стро
ка	hello, world	и происходит	выход	из	программы.	Еще	раз	следует	отметить,	
что	при	каждом	шаге	регистр	rip	указывает	на	следующую	выполняемую	ин
струкцию.

некОтОрые дОпОлнительные кОманды Отладчика GDB
Команда	break	или	b устанавливает	точку	останова	(временного	прекращения	
выполнения	программы),	как	мы	наблюдали	в предыдущем	разделе.

      disable breakpoint число
      enable breakpoint число
      delete breakpoint число

Команда	continue	или	c продолжает	выполнение	до	следующей	точки	оста
нова.

Команда	step	или	s –	шаг	в текущую	строку,	в итоге	выполняется	переход	
в вызываемую	функцию.

Команда	next	или	n –	шаг	через	текущую	строку	(из	текущей	строки)	и оста
нов	перед	следующей	строкой.

Команда	help	или	h –	вывод	справочной	информации.
Команда	tui enable –	разрешение	использования	простого	текстового	поль

зовательского	интерфейса.	Для	отключения	этого	интерфейса	введите	коман
ду	tui disable.

Команда	print	или	p –	вывод	значения	переменной,	регистра	и т.	д.
Несколько	примеров:
	� вывод	содержимого	регистра	rax:	p $rax;
	� вывод	содержимого	регистра	rax	в двоичном	формате:	p/t $rax;
	� вывод	содержимого	регистра	rax	в шестнадцатеричном	формате:	p/x $rax.

Еще	одно	важное	замечание,	касающееся	отладчика	GDB:	для	его	коррект
ного	использования	необходимо	обязательно	вставить	в исходный	код	пролог	
функции	 (function	prologue)	и эпилог	функции	 (function	epilogue).	В следую
щей	главе	будет	показано,	как	это	сделать,	а в	очередной	главе	пролог	и эпилог	
функции	будут	рассматриваться	при	обсуждении	кадров	стека.	Для	коротких	
программ,	таких	как	hello, world,	проблемы	не возникают.	Но в длинных	про
граммах	GDB	будет	демонстрировать	непредвиденное	поведение	при	отсут
ствии	пролога	и эпилога.

Поэкспериментируйте	с отладчиком	GDB,	ознакомьтесь	с онлайновым	ру
ководством	 (в  командной	 строке	 введите	 man gdb),	 в  общем,	 осваивайте	GDB	
в полной	мере,	потому	что	даже	если	вы	используете	отладчик	с графическим	
пользовательским	 интерфейсом,	 некоторые	 функциональные	 возможности	
могут	оказаться	недоступными.	Или	возможна	ситуация,	когда	в системе	во
обще	не установлен	отладчик	с GUI.

немнОгО улучшенная верСия прОграммы hello, worlD
Вероятно,	вы	заметили,	что	после	вывода	фразы	hello, world	промпт	появля
ется	в той	же	строке.	Необходимо,	чтобы	фраза	hello, world	выводилась	на	от
дельной	строке,	а командный	промпт	появлялся	в новой	строке.

В	листинге	3.1	приведен	код,	позволяющий	сделать	это.

Листинг 3.1.	Улучшенная версия программы hello, world

;hello2.asm
section .data
      msg    db    "hello, world",0
      NL     db    0xa  ; Код ASCII для символа перехода на новую строку.
section .bss
section .text
     global main
main:
    mov     rax, 1        ; 1 = запись
    mov     rdi, 1        ; 1 = в устройство стандартного вывода stdout.
    mov     rsi, msg      ; Выводимая строка.
    mov     rdx, 12       ; Длина строки без завершающего 0.
    syscall               ; Вывод строки.
    mov     rax, 1        ; 1 = запись
    mov     rdi, 1        ; 1 = в устройство стандартного вывода stdout.

    mov     rsi, NL       ; Вывод символа перехода на новую строку.
    mov     rdx, 1        ; Длина строки.
    syscall               ; Вывод строки.
    mov     rax, 60       ; 60 = выход.
    mov     rdi, 0        ; 0 = код успешного завершения программы.
    syscall               ; Выход.

Введите	этот	код	в текстовом	редакторе	и сохраните	его	в файле	hello2.asm 
в новом	каталоге.	Скопируйте	ранее	созданный	makefile	в этот	новый	каталог,	
затем	в этой	копии	makefile	 замените	все	имена	файлов	hello	на	hello2	и со
храните	файл.

В	 программу	 добавлена	 переменная	 NL,	 содержащая	 шестнадцатеричное	
значение	0xa –	код	ASCII	для	символа	перехода	на	новую	строку,	и инструкции	
вывода	этой	переменной	NL	сразу	после	вывода	строки	msg.	Ассемблируйте	этот	
код	и выполните	программу	(результат	см.	на	рис. 3.15).

Рис. 3.15.	Результат выполнения улучшенной версии программы hello, world

Другой	способ	выполнения	этой	задачи –	изменение	самой	строки	msg,	как	
показано	ниже:

      msg   db      "hello, world",10,0

Здесь	10 –	это	десятичное	представление	кода	символа	перехода	на	новую	
строку	(шестнадцатеричного	кода	0xa).	Попробуйте	и этот	вариант.	Но не за
будьте	 увеличить	 значение	для	регистра	rdx	до	13,	 чтобы	включить	в  строку	
дополнительный	символ	с кодом	10.

В	листинге	3.2	показан	код	этой	версии.	Сохраните	код	в файле	hello3.asm 
в отдельном	каталоге,	скопируйте	и измените	соответствующим	образом	файл	
makefile,	затем	выполните	сборку	и запуск	программы.

Листинг 3.2.	Еще одна версия программы hello, world

;hello3.asm
section .data
      msg    db    "hello, world",10,0
section .bss
section .text
     global main
main:
    mov     rax, 1        ; 1 = запись
    mov     rdi, 1        ; 1 = в устройство стандартного вывода stdout.
    mov     rsi, msg      ; Выводимая строка.
    mov     rdx, 13       ; Длина строки без завершающего 0.
    syscall               ; Вывод строки.

    mov     rax, 60       ; 60 = выход.
    mov     rdi, 0        ; 0 = код успешного завершения программы.
    syscall               ; Выход.

Но	 использование	 этой	 версии	 означает,	 что	 символ	 перехода	 на	 новую	
строку	является	частью	выводимой	строки,	а это	не всегда	приемлемо,	потому	
что	переход	на	новую	строку –	это	инструкция	форматирования,	которая,	ве
роятнее	всего,	предназначена	только	для	вывода	строки,	но	не для	выполнения	
других	операций	обработки	строк.	С другой	стороны,	исходный	код	становится	
короче	и проще.	Выбор	за	вами.

резюме
В	этой	главе	вы	узнали,	как:

	� использовать	 GDB  –	 отладчик	 с  текстовым	 интерфейсом	 командной	
строки;

	� вывести	символ	перехода	на	новую	строку.

Глава 4
Следующая программа:  

Alive and Kicking

После	того	 как	 вы	овладели	основными	навыками	отладки	 с помощью	GDB	
и узнали,	что	представляет	собой	программа	на	ассемблере,	попробуем	доба
вить	сложности.	В этой	главе	будет	показано,	как	получить	длину	строковой	
переменной,	а также	как	вывести	значения	целого	числа	и числа	с плавающей	
точкой,	используя	функцию	printf.	Кроме	того,	мы	немного	расширим	знания	
о командах	отладчика	GDB.

Листинг 4.1	содержит	пример	исходного	кода,	который	будет	использовать
ся	для	демонстрации	того,	как	можно	определить	длину	строки	и как	числовые	
значения	хранятся	в памяти.

Листинг 4.1.	Исходный код программы alive.asm

;alive.asm
section .data
      msg1   db    "Hello, World!",10,0       ; Строка с NL и 0.
      msg1Len      equ    $-msg1-1     ; Мера длины без 0.
      msg2   db    "Alive and Kicking!",10,0  ; Строка с NL и 0.
      msg2Len      equ    $-msg2-1     ; Мера длины без 0
      radius dq    357                 ; Это не строка, можно ли вывести?
      pi     dq    3.14                ; Это не строка, можно ли вывести?
section .bss
section .text
      global main
main:
    push           rbp            ; Пролог функции.
    mov            rbp,rsp        ; Пролог функции.
    mov            rax, 1         ; 1 = запись
    mov            rdi, 1         ; 1 = в устройство стандартного вывода stdout.
    mov            rsi, msg1      ; Выводимая строка.
    mov            rdx, msg1Len   ; Длина строки.
    syscall                       ; Вывод строки.
    mov            rax, 1         ; 1 = запись
    mov            rdi, 1         ; 1 = в устройство стандартного вывода stdout.
    mov            rsi, msg2      ; Вывод строки.
    mov            rdx, msg2Len   ; Длина строки.
    syscall                       ; Вывод строки.
    mov            rsp,rbp        ; Эпилог функции.



    pop            rbp            ; Эпилог функции.
    mov            rax, 60        ; 60 = выход
    mov            rdi, 0         ; 0 = код успешного завершения.
    syscall                       ; Выход.

Введите	этот	исходный	код	в текстовом	редакторе	и сохраните	его	в фай
ле	alive.asm.	Создайте	файл	makefile,	содержащий	строки,	показанные	в лис
тинге 4.2.

Листинг 4.2.	makefile для программы alive.asm

#makefile для alive.asm
alive: alive.o
      gcc -o alive alive.o -no-pie
alive.o: alive.asm
      nasm -f elf64 -g -F dwarf alive.asm -l alive.lst

Сохраните	этот	файл	и выходите	из	текстового	редактора.
После	промпта	командной	строки	введите	команду	make,	чтобы	ассемблиро

вать	и создать	выполняемый	файл,	затем	запустите	его	из	командной	строки:	
./alive.	Если	вы	увидите	вывод,	показанный	на	рис. 4.1,	то	все	работает,	как	
предполагалось,	в противном	случае	вы	гдето	допустили	опечатку	или	ошиб
ку.	Начинайте	отладку.

Рис. 4.1.	Вывод программы alive.asm

анализ прОграммы alIve
В	первой	программе	hello.asm	мы	поместили	длину	строки	msg,	равную	13	сим
волам,	в регистр	rdx,	чтобы	вывести	строку	msg.	В исходном	коде	alive.asm	ис
пользуется	весьма	удобная	функциональная	возможность	вычисления	длины	
переменных,	как	показано	ниже:

msg1Len equ $-msg1-1

Часть	инструкции	$-msg1-1	означает	следующее:	взять	эту	локацию	(адрес)	
памяти	($)	и вычесть	адрес	памяти,	где	находится	строка	msg1.	Результатом	яв
ляется	длина	строки	msg1.	Эта	длина	-1	(вычитается	позиция	нуля,	завершаю
щего	строку)	сохраняется	в константе	msg1Len.

Обратите	внимание	на	использование	пролога	и эпилога	функции	в этом	
коде.	 Это	 необходимо	для	 обеспечения	 корректной	 работы	 отладчика	GDB,	
о чем	было	сказано	в предыдущей	главе.	Код	пролога	и эпилога	будет	объяс
нен	в следующей	главе.

Приступим	 к  исследованию	 оперативной	 памяти	 с  использованием	 GDB.	
Введите	команду

gdb alive

Далее	после	промпта	(gdb)	введите	команду

disassemble main

На	рис. 4.2	показан	вывод	этой	команды.

Рис. 4.2.	Дизассемблированный код программы alive

Здесь	видно,	что	на	моем	компьютере	переменная	msg1	расположена	в памя
ти	по	адресу	0x601030,	это	можно	проверить	следующей	командой:

x/s 0x601030

Вывод	показан	на	рис. 4.3.

Рис. 4.3.	Адрес места расположения в памяти строки msg1

Пара	символов	\n	обозначает	«переход	на	новую	строку».	Другой	способ	про
верки	переменных	в GDB:

x/s &msg1

Вывод	показан	на	рис. 4.4.

Рис. 4.4.	Адрес места расположения в памяти строки msg1

А	как	насчет	числовых	значений?

x/dw      &radius
x/xw      &radius

Вывод	показан	на	рис. 4.5.

Рис. 4.5.	Вывод числовых значений

Таким	образом,	мы	получили	десятичное	и шестнадцатеричное	представле
ния	значения,	хранящегося	в локации	памяти	radius.	Для	переменных,	содер
жащих	числа	с плавающей	точкой,	используются	следующие	команды:

x/fg &pi
x/fx &pi

Вывод	показан	на	рис. 4.6.

Рис. 4.6.	Вывод значений с плавающей точкой

(Вы	обратили	внимание	на	ошибочное	значение	числа	с плавающей	точкой?)
Здесь	существует	одна	тонкость,	о которой	всегда	следует	помнить.	Для	ее	

наглядной	демонстрации	откроем	только	что	сгенерированный	файл	листинга	
alive.lst.	См.	рис. 4.7.

Рис. 4.7.	Содержимое файла листинга alive.lst

Взгляните	на	строки	10	и 11,	в которых	слева	можно	видеть	шестнадцате
ричное	представление	значений	radius	и pi.	Вместо	0165	записано	6501,	а вместо	
40091EB851EB851F	мы	видим	1F85EB51B81E0940.	Получается,	что	байты	(1	байт	со
держит	два	шестнадцатеричных	числа)	хранятся	в обратном	порядке.

Эта	особенность	называется	порядком	следования	байтов	(endianness).	В фор
мате	с обратным	порядком	байтов	(от	старшего	к младшему –	big	endian)	числа	
хранятся	в том	виде,	в котором	мы	привыкли	их	видеть,	т. е.	старшие	разряды	
(наиболее	значимые	цифры)	начинаются	слева.	В формате	с прямым	порядком	
байтов	(от	младшего	к старшему –	little	endian)	младшие	разряды	(наименее	зна
чимые	цифры)	начинаются	слева.	Процессоры	Intel	используют	формат	с пря
мым	порядком	байтов	(little	endian),	и это	может	приводить	к весьма	сущест
венным	затруднениям	при	чтении	кода	в шестнадцатеричном	представлении.

Но	почему	для	обозначения	порядка	байтов	используются	такие	странные	
английские	термины:	big	endian	и little	endian?

В	1726	г.	Джонатан	Свифт	написал	свой	знаменитый	роман	«Путешествия	
Гулливера».	В одной	из	частей	этого	романа	описаны	два	вымышленных	остро
ва:	Лилипутия	и Блефуску.	Обитатели	Лилипутии	воюют	с народом	Блефуску	
изза	разногласий	в способе	разбивания	вареных	яиц:	с острого	или	с тупого	
конца.	Лилипуты	предпочитают	разбивать	яйцо	с острого	конца –	они	остро
конечники	 (little	 endians).	 Жители	 Блефуску  –	 тупоконечники	 (big	 endians).	
Характерный	пример	того,	как	современные	информационные	технологии	от
дают	дань	традициям,	корни	которых	уходят	в далекое	прошлое.

Выделите	немного	времени	для	пошагового	прохода	по	 этой	программе	
(break main,	run,	next,	next,	next…).	Вы увидите,	как	GDB	пошагово	проходит	по	
прологу	 функции.	 Отредактируйте	 исходный	 код,	 удалив	 пролог	 и  эпилог	
функции,	потом	снова	соберите	программу	командой	make.	После	этого	по
пробуйте	еще	раз	пошагово	пройти	по	программе	с помощью	GDB.	В этом	

случае	GDB	отказывается	от	пошагового	прохода	и сразу	выполняет	програм
му	полностью.	При	ассемблировании	с использованием	YASM,	другой	про
граммы	ассемблирования	на	основе	NASM,	можно	безопасно	пропустить	код	
пролога	и эпилога	и выполнять	пошаговый	проход	с помощью	GDB.	Иногда	
это	необходимо	для	экспериментов,	дерзайте,	и Google	вам	в помощь.

вывОд
Рассматриваемая	здесь	программа	alive	выводит	две	строки:

Hello, World!
Alive and Kicking!

Но	в программе	существуют	еще	две	переменные,	которые	не определены	
как	строки:	radius	и pi.	Вывод	значений	этих	переменных	немного	сложнее,	
чем	вывод	строк.	Чтобы	вывести	числа,	хранящиеся	в этих	переменных,	таким	
же	способом,	как	это	сделано	для	msg1	и msg2,	необходимо	преобразовать	значе
ния	radius	и pi	в строки.	Вполне возможно	добавить	код	для	такого	преобразо
вания	в нашу	программу,	но	тогда	она	станет	слишком	сложной,	а это	весьма	
нежелательно	в настоящий	момент,	поэтому	применим	небольшую	хитрость.	
Мы позаимствуем	хорошо	известную	функцию	printf	из	языка	программиро
вания	C и	включим	ее	в свою	программу.	Если	такой	подход	разочаровывает	
вас,	потерпите	немного.	Когда	вы	станете	более	опытным	программистом	на	
ассемблере,	то	сможете	писать	собственные	функции	преобразования	и выво
да	чисел.	Или,	возможно,	придете	к выводу,	что	при	написании	собственной	
функции	printf	затрачивается	слишком	много	времени…

Для	демонстрации	использования	функции	printf	в ассемблерном	коде	нач
нем	с простой	программы.	Изменим	первую	программу	hello.asm,	как	показа
но	в листинге	4.3.

Листинг 4.3.	hello4.asm

; hello4.asm
extern      printf     ; Объявление функции как внешней.
section .data
      msg    db   "Hello, World!",0
      fmtstr db   "This is our string: %s",10,0 ; Формат вывода строки.
section .bss
section .text
      global main
main:
      push  rbp
      mov   rbp,rsp
      mov   rdi, fmtstr      ; Первый аргумент для функции printf.
      mov   rsi, msg         ; Второй аргумент для функции printf.
      mov   rax, 0           ; Регистры xmm не применяются.
      call  printf           ; Вызов (внешней) функции.
      mov   rsp,rbp
      pop   rbp
      mov   rax, 60          ; 60 = выход.
      mov   rdi, 0           ; 0 = код успешного завершения.
      syscall                ; Выход.

Здесь	мы	начали	с сообщения	ассемблеру	(и линкеру)	о том,	что	будет	ис
пользоваться	внешняя	функция	с именем	printf.	Создается	строка,	определя
ющая,	как	printf	будет	выводить	msg.	Синтаксис	строки	формата	похож	на	син
таксис	языка	C.	Если	у вас	есть	опыт	программирования	на	C,	то	вы,	разумеется,	
узнали	строку	формата:	%s –	это	шаблон	для	подстановки	выводимой	строки.

Не забывайте	о прологе	и эпилоге	функции.	Адрес	msg	помещается	в регистр	
rsi,	адрес	fmtstr –	в регистр	rdi.	Далее	обнуляется	регистр	rax,	в данном	случае	
это	означает,	что	в xmmрегистрах	registers	нет	чисел	с плавающей	точкой	для	
вывода.	Работа	с числами	с плавающей	точкой	и с	xmmрегистрами	registers 
будет	описана	в главе	11.

В	листинге	4.4	показано	содержимое	makefile.

Листинг 4.4.	makefile для программы hello4.asm

#makefile для hello4.asm
hello4: hello4.o
      gcc -o hello4 hello4.o -no-pie
hello4.o: hello4.asm
      nasm -f elf64 -g -F dwarf hello4.asm -l hello4.lst

Необходимо	убедиться,	что	флаг	-no-pie	добавлен	в makefile,	иначе	использо
вание	функции	printf	приведет	к ошибке.	В главе	1	было	отмечено,	что	совре
менная	версия	компилятора	gcc	генерирует	позиционно	независимый	выпол
няемый	(pie)	код	для	улучшения	его	защиты	от	взлома.	Одним	из	последствий	
такой	методики	компиляции	является	невозможность	простого	использования	
внешних	функций.	Для	устранения	этого	затруднения	используется	флаг	-no-pie.

Выполните	сборку	и запустите	программу.	Найдите	информацию	о функции	
printf,	чтобы	получить	представление	о возможных	форматах.	Вы узнаете,	что	
printf	предоставляет	возможности	гибкого	форматирования	при	выводе	целых	
чисел,	чисел	с плавающей	точкой,	строк,	данных	в шестнадцатеричном	виде	
и т.	д.	Функция	printf	требует,	чтобы	строка	обязательно	завершалась	0 (NULL).	
Если	0	не указан,	то	printf	будет	выводить	все	подряд,	пока	не обнаружит	0.	За
вершение	строки	нулем	не является	требованием	ассемблера,	это	необходимо	
для	printf,	GDB,	а также	для	некоторых	инструкций	SIMD	(инструкции	SIMD	
будут	рассматриваться	в главе	26).

На	рис. 4.8	показан	вывод	программы.

Рис. 4.8.	Вывод программы hello4

Но	вернемся	к программе	alive.	Теперь	с помощью	функции	printf	можно	
вывести	значения	переменных	radius	и pi.	В листинге	4.5	показан	исходный	
код.	Теперь	вы	знаете,	что	делать:	создать	файл	исходного	кода,	скопировать	
или	создать/отредактировать	makefile,	в общем,	все	как	раньше.

Листинг 4.5.	makefile для программы alive2.asm

; alive2.asm
section .data
      msg1        db    "Hello, World!",0
      msg2        db    "Alive and Kicking!",0
      radius      dd    357
      pi          dq    3.14
      fmtstr      db    "%s",10,0      ; Формат для вывода строки.
      fmtflt      db    "%lf",10,0     ; Формат для вывода числа с плавающей точкой.
      fmtint      db    "%d",10,0      ; Формат для вывода целого числа.
section .bss
section .text
extern     printf
      global main
main:
    push   rbp
    mov    rbp,rsp
; print msg1
    mov    rax, 0            ; Без числа с плавающей точкой.
    mov    rdi, fmtstr
    mov    rsi, msg1
    call   printf
; print msg2
    mov    rax, 0            ; Без числа с плавающей точкой.
    mov    rdi, fmtstr
    mov    rsi, msg2
    call   printf
; print radius
    mov    rax, 0            ; Без числа с плавающей точкой.
    mov    rdi, fmtint
    mov    rsi, [radius]
    call   printf
; print pi
    mov    rax, 1            ; Используется 1 регистр xmm.
    movq   xmm0, [pi]
    mov    rdi, fmtflt
    call   printf
    mov    rsp,rbp
    pop    rbp
ret

Здесь	добавлены	три	строки	для	форматирования	вывода.	Строка	форма
та	помещается	в регистр	rdi,	регистр	rsi	указывает	на	выводимый	элемент,	 
запись	0	в регистр	rax	означает,	что	числа	с плавающей	точкой	не использу
ются.	Затем	вызывается	внешняя	функция	printf.	Для	вывода	числа	с плаваю
щей	точкой	его	значение	помещается	в регистр	xmm0	с помощью	специальной	
инструкции	movq.	Используется	только	один	xmmрегистр,	поэтому	в регистр	
rax	записывается	1.	В следующих	главах	будет	рассматриваться	более	подроб
но	использование	регистров	XMM	для	вычислений	с плавающей	точкой	и для	
выполнения	инструкций	SIMD.

Обратите	внимание	на	квадратные	скобки	[	],	в которые	помещены	имена	
переменных	radius	и pi.

mov rsi, [radius]

Это	означает:	взять	содержимое	по	адресу	radius	и поместить	его	в регистр	
rsi.	Для	функции	printf	необходим	адрес	памяти	выводимой	строки,	но	для	
вывода	чисел	ожидается	значение,	а не адрес	памяти.	Об этом	следует	пом
нить	всегда.

Выход	из	этой	программы	изменен.	Вместо	уже	знакомого	кода

      mov  rax, 60    ; 60 = выход.
      mov  rdi, 0     ; 0 = код успешного завершения.
      syscall         ; Выход.

используется	равнозначная	инструкция:

      ret

Важное	замечание:	функция	printf	принимает	строку	формата,	которая	мо
жет	принимать	разнообразные	формы	и преобразовывать	сущность	выводи
мых	значений	(целые	числа,	числа	с двойной	точностью,	числа	с плавающей	
точкой	 и  т.	 д.).	 Иногда	 такое	 преобразование	 является	 непреднамеренным	
и может	приводить	к путанице.	Если	вам	действительно	необходимо	знать	ис
тинное	значение	регистра	или	переменной	(области	памяти)	в программе,	то	
следует	воспользоваться	отладчиком	и обследовать	требуемый	регистр	или	об
ласть	памяти.

На	рис. 4.9	показан	вывод	программы	alive2.

Рис. 4.9.	Вывод программы alive2

резюме
В	этой	главе	вы	узнали:

	� о	дополнительной	функциональности	отладчика	GDB;
	� о	прологе	и эпилоге	функции;
	� о	прямом	и обратном	порядках	следования	байтов;
	� об	использовании	функции	printf	из	стандартной	библиотеки	языка C	
для	вывода	строк,	целых	чисел	и чисел	с плавающей	точкой.

Глава 5
Ассемблер основан на логике

А	теперь	освежим	в памяти	теоретические	знания	о математической	логике.	
Не волнуйтесь –	здесь	будет	рассматриваться	только	то,	что	действительно	
необходимо:	логические	операторы	NOT,	OR,	XOR	и AND.

В	этой	главе	0	обозначает	ложь	(false),	1	обозначает	истину	(true).

лОгичеСкий ОператОр not
Таблица 5.1

A 0 1

NOT A 1 0

Преобразование	каждого	0	в 1	и каждой	1	в 0.
Пример:

A =        11001011
NOT A =    00110100

лОгичеСкий ОператОр or
Таблица 5.2

A 0 1 0 1

B 0 0 1 1

A OR B 0 1 1 1

Если	A или	B или	и A,	и B	равны	1,	то	результат	равен	1.
Пример:

A =        11001011
B =        00011000
A OR B =   11011011



лОгичеСкий ОператОр Xor
Таблица 5.3

A 0 1 0 1

B 0 0 1 1

A XOR B 0 1 1 0

Исключающее	ИЛИ:	если	или	A,	или	B равно	1,	то	результат	равен	1.	Если	и A,	
и B	одновременно	равны	1	или	0,	то	результат	равен	0.

Пример:

A =         11001011
B =         00011000
A XOR B =   11010011

Логический	оператор	XOR	можно	применять	как	ассемблерную	инструкцию	
для	очистки	(обнуления)	регистра.

A =         11001011
A =         11001011
A XOR A =   00000000

Таким	образом,	xor rax, rax –	это	то	же	самое,	что	mov rax, 0.	Но инструкция	xor 
выполняется	быстрее,	чем	mov.	Также	можно	использовать	xor	для	изменения	
знака	числа	с плавающей	точкой.

Пример	смены	знака	для	32битового	числа	с плавающей	точкой:

A            = 17.0  = 0x41880000 = 01000001 10001000 00000000 00000000
B            = -0.0  = 0x80000000 = 10000000 00000000 00000000 00000000
A XOR B      = -17.0 = 0xC1880000 = 11000001 10001000 00000000 00000000

Для	проверки	правильности	результата	этой	операции	используйте	онлай
новое	инструментальное	средство,	размещенное	здесь:	www.binaryconvert.com/
result_float.html.

Следует	отметить,	что	если	необходимо	изменить	знак	целого	числа,	то	нуж
но	просто	вычесть	его	из	нуля	или	воспользоваться	инструкцией	neg.

лОгичеСкий ОператОр anD
Таблица 5.4

A 0 1 0 1

B 0 0 1 1

A AND B 0 0 0 1

Если	A и	B одновременно	равны	1,	то	результат	равен	1,	во	всех	прочих	слу
чаях	результат	равен	0.

Пример:

A =         11001011
B =         00011000
A AND B =   00001000

Инструкция	AND	может	использоваться	как	маска	для	выбора	и проверки	от
дельных	битов.

В	приведенном	ниже	примере	B используется	как	маска	для	выбора	битов	3	
и 6	из	A (младший,	крайний	справа	бит	имеет	индекс	0):

A =         11000011
B =         01001000
A AND B =   01000000

По	результату	этой	операции	можно	сделать	вывод	о том,	что	бит	6	уста
новлен	(равен	1),	а бит	3	не установлен	(равен	0).	Более	подробно	об	этом	мы	
поговорим	позже.

Кроме	того,	инструкцию	AND	можно	использовать	для	округления	чисел	с не
достатком	(в меньшую	сторону),	а особенно	полезна	она	при	округлении	адре
сов	с выравниванием	по	16байтовой	границе.	В дальнейшем	мы	будем	при
менять	инструкцию	AND	для	выравнивания	стеков.

Число	16	и числа,	кратные	16,	в шестнадцатеричном	формате	всегда	оканчи
ваются	0	или	0000	в двоичном	представлении.

address =    0x42444213 = 01000010010001000100001000010011
mask =       0xfffffff0 = 11111111111111111111111111110000
rounded =    0x42444210 = 01000010010001000100001000010000

Здесь	выполнено	округление	в меньшую	сторону	к самому	младшему	бай
ту	адреса.	Если	адрес	уже	заканчивается	нулевым	байтом,	то	инструкция	AND 
не изменяет	ничего.	Проверьте,	действительно	ли	округленный	адрес	делится	
нацело	на	16.	Для	этого	воспользуйтесь	онлайновой	утилитой	преобразования	
форматов	чисел	(например,	www.binaryconvert.com/convert_unsigned_int.html).

резюме
В	этой	главе	вы	узнали:

	� о	логических	операторах;
	� как	использовать	логические	операторы	в качестве	инструкций	ассемб
лера.

Глава 6
Отладчик  

Data Display Debugger

Отладчик	Data	Display	Debugger	 (DDD) –	 это	инструментальное	 средство	от
ладки	 с  графическим	пользовательским	интерфейсом	для	 систем	Linux.	Не
обходимо	установить	его	(командой	sudo apt install ddd),	так	как	отладчик	DDD	
будет	использоваться	на	протяжении	всей	этой	главы.	Программа,	написанная	
в этой	главе,	ничего	не выводит,	и мы	будем	наблюдать	за	выполнением	кода	
и за	содержимым	регистров	с помощью	DDD.

рабОта С ОтладчикОм DDD
В	листинге	6.1	показан	пример	исходного	кода.

Листинг 6.1.	Программа move.asm

; move.asm
section .data
      bNum  db    123
      wNum  dw    12345
      dNum  dd    1234567890
      qNum1 dq    1234567890123456789
      qNum2 dq    123456
      qNum3 dq    3.14
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      mov rax, -1           ; Заполнение регистра rax единицами.
      mov al, byte [bNum]   ; Верхние (старшие) биты регистра rax НЕ очищать.
      xor rax,rax           ; Очистка регистра rax.
      mov al, byte [bNum]   ; Теперь rax содержит корректное значение.
      mov rax, -1           ; Заполнение регистра rax единицами.
      mov ax, word [wNum]   ; Верхние (старшие) биты регистра rax НЕ очищать.
      xor rax,rax           ; Очистка регистра rax.
      mov ax, word [wNum]   ; Теперь rax содержит корректное значение.
      mov rax, -1           ; Заполнение регистра rax единицами.



      mov eax, dword [dNum] ; Очистить верхние (старшие) биты регистра rax.
      mov rax, -1           ; Заполнение регистра rax единицами.
      mov rax, qword [qNum1] ; Очистить верхние (старшие) биты регистра rax.
      mov qword [qNum2], rax ; Один операнд всегда должен быть регистром.
      mov rax, 123456       ; Операнд-источник – непосредственное значение.
      movq xmm0, [qNum3]    ; Инструкция для числа с плавающей точкой.
mov rsp,rbp
pop rbp
ret

Сохраните	 этот	 исходный	 код	 в  файле	move.asm,	 выполните	 сборку	 и  за
пустите	полученный	выполняемый	файл,	 чтобы	 убедиться,	 что	 он	 работает.	
При	запуске	эта	программа	не должна	ничего	выводить	на	экран.	В командной	
строке	выполните	команду

ddd move

Вы	увидите	окно	графического	пользовательского	интерфейса	со	специали
зированной	разметкой	(см.	рис. 6.1).	DDD	представляет	собой	старое	испытан
ное	инструментальное	средство	с открытым	исходным	кодом,	и вряд	ли	ктото	
пожелает	адаптировать	его	к современным	стандартам	GUI.

Рис. 6.1.	Окно отладчика DDD

На	экране отображается	окно	с исходным	кодом	и панелью,	в которой	можно	
вводить	команды	отладчика	GDB.	Также	имеется	перемещаемая	(плавающая)	
панель,	в которой	можно	щелкать	по	кнопкам	Run	(Запуск),	Step	(Шаг),	Stepi 
(i шагов)	и т.	д.	Щелкните	по	пункту	меню	Source	 (Исходный	код)	и выбери
те	отображение	номеров	строк.	В том	же	меню	можно	выбрать	создание	окна	
с ассемблированным	кодом.

Поместите	курсор	перед	меткой	main:,	затем	щелкните	правой	кнопкой	мыши	
и выберите	пункт	Break	(Точка	останова)	или	выберите	значок	Stop	(Останов)	
в меню	в верхней	части	окна.	Щелкните	по	кнопке	Run	 (Запуск)	на	переме
щаемой	панели,	и отладка	начнется.	Щелкните	по	пункту	Status	(Состояние)	
в строке	меню	в верхней	части	окна	и выберите	пункт	Registers	 (Регистры).	
В перемещаемой	панели	щелкните	по	кнопке	Step	(Шаг),	чтобы	выполнить	те
кущую	инструкцию.	Теперь	можно	наблюдать	за	изменением	содержимого	ре
гистров	при	пошаговом	проходе	по	программе.	Если	необходимо	обследовать	
адреса	памяти,	такие	как	qNum1	 или	bNum,	то	можно	воспользоваться	пунктом	
меню	Data	(Данные).	Сначала	перейдите	к пункту	View	(Вид),	чтобы	сделать	
видимым	окно	данных.	Затем	в спускающемся	меню	Data	щелкните	по	пунк
ту	Memory	(Память).	На рис. 6.2	показан	пример	наблюдения	за	содержимым	
ячеек	памяти.	Поскольку	интерфейс	DDD	не  всегда	понятен,	 использование	
окна	ввода	отладчика	GDB	иногда	позволяет	намного	быстрее	выполнить	тре
буемую	операцию,	чем	поиск	нужной	команды	в меню.

Рис. 6.2.	Наблюдение за содержимым памяти в отладчике DDD

Отладчик	DDD	создан	на	основе	GDB,	поэтому	обязательным	требованием	
является	использование	пролога	и эпилога	функции,	для	того	чтобы	избежать	
проблем.	 Обратите	 внимание:	 при	 пошаговом	 проходе	 по	 программе	 DDD	
просто	игнорирует	пролог.

Главная	цель	этого	кода –	показать,	что	происходит	с содержимым	регистров	
при	использовании	команды	mov.	Откройте	окно	регистров	в DDD	(в меню	вы

берите	Status → Registers).	Обратите	внимание:	изначально	регистр	rax	содер
жит	-1,	это	означает,	что	все	биты	в регистре	rax	равны	1.	Если	вы	не понимаете,	
почему,	то	вернитесь	к главе	2	и еще	раз	прочтите	раздел	о представлении	дво
ичных	чисел	в компьютерах.	Вы увидите,	что	если	число	помещается	в регистр	
al	или	ax,	то	верхние	биты	регистра	rax	не сбрасываются	в 0	(не очищаются),	
и в	результате	регистр	rax	содержит	значение,	не равное	значению	al	или	ax.	
В рассматриваемом	здесь	примере	rax	содержит	0xffffffffffffff7b,	т. е.	большое	
отрицательное	число.	Но в регистре	al	 содержится	 значение	0x7b –	десятич
ное	число	123,	как	ожидалось.	Это	может	совпадать	или	не совпадать	с вашими	
намерениями.	Если	при	вычислениях	вы	по	ошибке	использовали	регистр	rax 
вместо	al,	то	результат	окажется	абсолютно	неверным.	Но поскольку	вы	про
должаете	пошаговый	проход	по	коду,	то	увидите,	что	при	передаче	32битово
го	значения	в 64битовый	регистр	верхние	биты	64битового	регистра	будут	
очищены	(сброшены	в 0).	При	записи	значения	в регистр	eax	верхние	биты	ре
гистра	rax	очищаются.	Об этом	важно	помнить	всегда.

В	 конце	 рассматриваемого	 примера	 значение	 из	 регистра	 перемещается	
в qNum2.	Обратите	внимание	на	квадратные	скобки,	сообщающие	ассемблеру,	
что	qNum2 –	это	адрес	памяти.	И самое	последнее	действие –	«непосредственное	
значение»	помещается	в регистр.

резюме
В	этой	главе	вы	узнали:

	� об	отладчике	DDD,	хотя	и немного	устаревшем,	но	все	еще	используемом	
в качестве	отладчика	на	основе	GDB;

	� о	том,	что	копирование	значения	в 8битовый	и 16битовый	регистры	
не очищает	(не сбрасывает	в 0)	верхнюю	часть	64битового	регистра,	...

	� ...но	копирование	значения	в 32битовый	регистр	очищает	(сбрасывает	
в 0)	верхнюю	часть	64битового	регистра.

Глава 7
Переходы и циклы

Вероятнее	всего,	вы	согласитесь,	что	такой	отладчик,	как	DDD,	полезен,	осо
бенно	для	обследования	больших	программ.	В этой	главе	будет	представлено	
программное	средство	SASM	(Simple	ASM).	Это	кроссплатформенная	интегри
рованная	 среда	 разработки	 (integrated	 development	 environment –	 IDE)	 с  от
крытым	исходным	кодом.	SASM	поддерживает	подсветку	синтаксиса	и отлад
ку	в режиме	графического	пользовательского	интерфейса.	Это	великолепное	
инструментальное	средство	для	программиста	на	ассемблере.

уСтанОвка SImpleaSm
Необходимо	 перейти	 на	 вебстраницу	 https://dman95.github.io/SASM/English.
html,	выбрать	версию	для	ОС,	которую	вы	используете,	и начать	установку.	Для	
Ubuntu	18.04	нужно	перейти	в каталог	xUbuntu_18.04/amd64/,	затем	загрузить	
и установить	пакет	sasm_3.10.1_amd64.deb	следующей	командой:

sudo dpkg -i sasm_3.10.1_amd64.deb

Если	выводится	сообщение	об	ошибке	изза	проблем	с зависимостями,	то	
требуется	установить	отсутствующие	пакеты	и повторить	процедуру	установ
ки	SASM.	Можно	также	попробовать	выполнить	следующую	команду:

sudo apt --fix-broken install

Обычно	 это	позволяет	 установить	 все	требуемые,	 но	пока	 отсутствующие	
пакеты.

иСпОльзОвание SaSm
Запустите	SASM,	выполнив	команду	sasm	в командной	строке,	и выберите	язык	
интерфейса.	SASM	начнет	загружаться,	но	если	в командной	строке	появит
ся	сообщение	об	ошибке	«Failed to load module "canberra-gtk-module"»	(Невозможно 
загрузить модуль "canberra-gtk-module"),	то	установите	следующие	пакеты:

sudo apt install libcanberra-gtk*



После	 установки	 заданной	 группы	 файлов	 сообщение	 об	 ошибке	 больше	
не должно	появляться.

В	окне SASM	перейдите	в диалоговую	панель	Settings	(Параметры	настрой
ки),	как	показано	на	рис. 7.1.	На вкладке	Common	(Общие)	выберите	вариант	
Yes	(Да)	для	пункта	Show	all	registers	in	debug:	(Показывать	все	регистры	при	
отладке:).

Рис. 7.1.	SASM, диалоговое окно Settings (Параметры настройки), вкладка Common (Общие)

На	вкладке	Build	(Сборка)	измените	параметры	настройки,	как	показано	на	
рис. 7.2.

Рис. 7.2.	SASM, диалоговое окно Settings (Параметры настройки), вкладка Build (Сборка)

Будьте	 весьма	 внимательны:	 настройки	должны	 быть	 в точности	такими,	
как	показано	на	рис. 7.2, –	один	лишний	пробел,	даже	«спрятавшийся»	в конце	
строки,	и SASM	не будет	выполнять	то,	что	вам	нужно.	После	правильного	вво
да	всех	параметров	щелкните	по	кнопке	OK	и перезапустите	SASM.

При	начале	работы	 с новым	проектом	в  SASM	вы	обнаружите	некоторый	
исходный	код,	который	уже	находится	в окне редактора	по	умолчанию.	Этот	
код	не будет	использоваться,	поэтому	его	можно	удалить.	В командной	строке	
введите

sasm jump.asm

Если	файл	jump.asm	не существует,	то	SASM	запускается	с открытием	нового	
окна	редактора,	в котором	нужно	просто	удалить	исходный	код	по	умолчанию.	
Если	файл	существует,	то	его	содержимое	будет	загружено	в окно	редактора.

В	листинге	7.1	показан	исходный	код,	содержащийся	в файле	jump.asm.

Листинг 7.1.	Файл jump.asm

; jump.asm
extern printf
section .data
      number1    dq    42
      number2    dq    41
      fmt1  db   "NUMBER1 > = NUMBER2",10,0
      fmt2  db   "NUMBER1 < NUMBER2",10,0
section .bss
section .text
      global     main
main:
      push  rbp
      mov   rbp,rsp
      mov   rax, [number1]   ; Передача чисел в регистры.
      mov   rbx, [number2]
      cmp   rax,rbx    ; Сравнение регистров rax и rbx.
      jge   greater    ; Если rax больше или равен, то перейти к метке greater:.
mov   rdi,fmt2         ; Если rax меньше, продолжить здесь.
mov   rax,0            ; Регистр xmm не используется.
      call  printf     ; Вывод строки fmt2.
      jmp   exit       ; Переход к метке exit:.
greater:
      mov   rdi,fmt1   ; Регистр rax больше.
      mov   rax,0      ; Регистр xmm не используется.
      call  printf     ; Вывод строки fmt1.
exit:
      mov   rsp,rbp
      pop   rbp
      ret

Скопируйте	этот	исходный	код	в окно	редактора	SASM.	По умолчанию	SASM	
использует	подсветку	синтаксиса.	После	завершения	ввода	щелкните	по	кноп
ке	с зеленым	треугольником	в панели	инструментов	в верхней	части	окна.	Это	
кнопка	«Пуск».	Если	все	было	сделано	правильно,	то	в панели	Output	(Вывод)	
вы	увидите	результат	работы	программы,	как	показано	на	рис. 7.3.

Рис. 7.3.	Вывод программы в панели Output окна SASM

При	 сохранении	файла	 в  SASM	 будет	 сохранен	 введенный	 исходный	 код.	
Если	вы	хотите	сохранить	выполняемый	файл,	то	необходимо	выбрать	пункт	
Save.exe	в меню	File	(Файл).

Для	начала	отладки	щелкните	по	столбцу	нумерации	строк	слева	от	метки	
main:.	После	этого	между	меткой	main:	и соответствующим	номером	строки	по
явится	красный	кружок.	Это	символ	точки	останова.	Затем	в верхней	панели	
инструментов	щелкните	 по	 зеленому	 треугольнику	 с  изображением	 на	 нем	
жука.	 В  главном	меню	 выберите	 пункт	Debug	 (Отладка)	 и  отметьте	 пункты	
Show	Registers	(Показывать	регистры)	и Show	Memory	(Показывать	память).	
На  экране  появится	 несколько	 дополнительных	 окон:	Registers	 (Регистры),	
Memory	(Память),	а также	виджет	командной	строки	отладчика	GDB.

Теперь	с помощью	значка	Step	 (Шаг)	можно	начать	пошаговый	проход	по	
коду	и наблюдать,	как	изменяются	значения	в регистрах.	Для	того	чтобы	уви
деть	изменения	 значения	переменной,	щелкните	правой	кнопкой	мыши	по	
объявлению	переменной	в разделе	section .data	и в	контекстном	меню	выбе
рите	пункт	Watch	(Наблюдать).	Указанная	переменная	будет	добавлена	в окно	
Memory,	при	этом	SASM	попытается	автоматически	определить	ее	тип.	Если	
значение,	 выведенное	 SASM,	 не  соответствует	 ожидаемому,	 то	 необходимо	
вручную	изменить	тип	на	правильный.	При	отладке	с использованием	SASM	
для	корректной	отладки	добавляется	следующая	строка	исходного	кода:

mov rbp, rsp     ; Для корректной отладки.

Эта	строка	может	запутать	другие	отладчики,	такие	как	GDB,	поэтому	не за
бывайте	удалять	ее	перед	запуском	GDB	отдельно	из	командной	строки.

Следует	 убедиться	 в  том,	 что	 в  главном	 меню	 Settings	 (Параметры	 на
стройки)	 → Common	 (Общие)	 выбран	 вариант	 Yes	 (Да)	 для	 пункта	 Show	
all	 registers	 in	debug	 (Показывать	все	регистры	при	отладке).	При	отладке	
в SASM	выполните	прокрутку	вниз	в окне регистров.	В нижней	части	окна	
вы	увидите	16	ymmрегистров,	в каждом	из	которых	указаны	два	значения	
в  круглых	 скобках.	 Первое	 значение –	 это	 соответствующий	 xmmрегистр.	
Более	подробно	эти	регистры	будут	описаны	при	рассмотрении	SIMD.

На	рис. 7.4	показан	вывод	на	экран	после	сборки	и запуска	программы	jump 
тем	же	способом,	который	был	описан	ранее.

Рис. 7.4.	Вывод программы jump.asm

В	этой	программе	использовалась	инструкция	сравнения	cmp	и две	инструк
ции	перехода	jge	и jmp.	Инструкцию	cmp	называют	инструкцией	проверки	усло
вия	или	просто	условной	инструкцией.	Здесь	cmp	сравнивает	два	операнда,	в дан
ном	случае	два	регистра.	Один	из	операндов	также	может	быть	адресом	памяти,	
а второй	операнд	может	быть	непосредственным	значением.	В любом	случае	
размер	 обоих	 операндов	 обязательно	 должен	 быть	 одинаковым	 (байт,	 слово	
и т. п.).	Инструкция	cmp	устанавливает	или	очищает	биты	в регистре	флагов.

Флаги –	 это	 биты,	 расположенные	 в  регистре	 rflags,	 которые	 могут	 быть	
установлены	в 1	или	очищены	(сброшены)	в 0,	в зависимости	от	нескольких	
условий.	В рассматриваемом	здесь	примере	важными	являются	флаг	нуля	(ZF –	
zero	flag),	флаг	переполнения	 (OF –	overflow	flag)	и флаг	знака	 (SF –	sign	flag).	
Вы можете	воспользоваться	любым	отладчиком	для	обследования	этих	и дру
гих	флагов.	При	работе	с SASM	можно	с легкостью	увидеть,	что	происходит	во	
всех	регистрах,	в том	числе	и в	регистре	флагов,	который	в SASM	называется	
eflags.	Различные	значения	операндов	инструкции	cmp	приводят	к установке	
или	 очистке	 разных	флагов.	 Немного	 поэкспериментируйте	 со	 значениями,	
чтобы	наблюдать,	что	происходит	с флагами.

Если	необходимо	использовать	флаги,	то	вы	должны	проверять	их	сразу	же	
после	выполнения	инструкции	cmp.	Если	перед	проверкой	регистра	rflags	вы
полняются	другие	инструкции,	то	значения	флагов	могут	измениться.	В рас
сматриваемом	 здесь	 примере	 регистр	 флагов	 проверяется	 инструкцией	 jge,	
означающей	«переход,	если	больше	или	равно»	(jump	if	greater	than	or	equal).	
Если	условие	выполнено,	то	происходит	переход	(jump)	на	метку,	указанную	
после	инструкции	jge.	Если	условие	не соблюдено,	то	выполнение	продолжа
ется	с инструкции,	следующей	непосредственно	за	инструкцией	jge.	В табл. 7.1	
приведены	некоторые	наиболее	часто	используемые	условные	инструкции,	но	
вы	можете	найти	более	подробное	описание	всех	условных	инструкций	в ру
ководствах	Intel.

Таблица 7.1.	Инструкции перехода и значения флагов

Ин-
струк-

ция

Флаги Описание Использование

je ZF=1 Переход, если равно Знаковые, беззна-
ковые

jne ZF=0 Переход, если не равно Знаковые, беззна-
ковые

jg ((SF XOR OF) OR ZF) = 0 Переход, если больше Знаковые

jge (SF XOR OF) = 0 Переход, если больше или равно Знаковые

jl (SF XOR OF) = 1 Переход, если меньше Знаковые

jle (SF XOR OF) OR ZF) = 1 Переход, если меньше или равно Знаковые

ja (CF OR ZF) = 0 Переход по условию «больше» Беззнаковые

jae CF=0 Переход по условию «больше или 
равно»

Беззнаковые

jb CF=1 Переход по условию «меньше» Беззнаковые

jbe (CF OR ZF) = 1 Переход по условию «меньше или 
равно»

Беззнаковые

В	 рассматриваемой	 здесь	 программе	 также	 использовалась	 инструкция	
безу	словного	перехода	jmp.	Если	при	выполнении	программы	встречается	эта	
инструкция,	то	происходит	переход	к метке,	указанной	после	инструкции	jmp,	
независимо	от	значений	флагов	или	какихлибо	других	условий.

Более	сложной	формой	перехода	является	зацикливание	(looping),	означа
ющее	многократное	повторение	группы	инструкций,	пока	выполняется	(или	
не выполняется)	некоторое	заданное	условие.	В листинге	7.2	показан	пример 
зацикливания.

Листинг 7.2.	Программа jumploop.asm

; jumploop.asm
extern printf
section .data
      number      dq    5
      fmt         db    "The sum from 0 to %ld is %ld",10,0
section .bss
section .text
      global main
main:
      push  rbp
      mov   rbp, rsp
      mov   rbx,0        ; Счетчик.
      mov   rax,0        ; Сумма будет сохраняться в регистре rax.
jloop:
      add   rax, rbx
      inc   rbx
      cmp   rbx,[number] ; Конечное число итераций цикла достигнуто?
      jle   jloop        ; Конечное число итераций пока не достигнуто, продолжение цикла.
                         ; Конечное число итераций достигнуто, продолжить здесь.

      mov   rdi,fmt      ; Подготовка вывода результата.
      mov   rsi, [number]
      mov   rdx,rax
      mov   rax,0
      call  printf
      mov   rsp,rbp
      pop   rbp
      ret

Эта	программа	выполняет	сложение	чисел	от	0	до	значения,	содержащегося	
в переменной	number.	Регистр	rbx	используется	как	счетчик	(цикла),	а в	регист
ре	rax	 сохраняется	накапливаемая	сумма.	Здесь	создан	цикл	(loop) –	это	код	
между	меткой	jloop:	и инструкцией	jle jloop.	В этом	цикле	значение	в регистре	
rbx	прибавляется	к значению	регистра	rax,	после	чего	значение	регистра	rbx 
увеличивается	на	1,	затем	выполняется	сравнение,	чтобы	узнать,	не было	ли	
достигнуто	конечное	значение	(number).	Если	в регистре	rbx	содержится	значе
ние,	меньшее	или	равное	значению	number,	то	выполнение	цикла	продолжает
ся,	иначе	выполнение	продолжается	с инструкции,	следующей	непосредствен
но	после	цикла,	и начинается	подготовка	к выводу	результата.	Для увеличения	
значения	регистра	rbx	использована	арифметическая	инструкция	inc.	Арифме
тические	инструкции	будут	рассматриваться	в следующих	главах.

В	листинге	7.3	показан	другой	способ	создания	цикла.

Листинг 7.3.	Программа betterloop.asm

; betterloop
extern printf
section .data
      number      dq    5
      fmt         db    "The sum from 0 to %ld is %ld",10,0
section .bss
section .text
      global main
main:
      push  rbp
      mov   rbp,rsp
      mov   rcx,[number]    ; Инициализация регистра rcx значением number.
      mov   rax, 0
bloop:
      add   rax,rcx         ; Прибавление rcx для получения суммы.
      loop  bloop           ; Цикл, пока значение rcx уменьшается на 1.
                            ; До тех пор, когда выполнится условие rcx = 0.
      mov   rdi,fmt         ; rcx = 0, продолжить здесь.
      mov   rsi, [number]   ; Выводимая сумма.
      mov   rdx, rax
      mov   rax,0           ; Без использования чисел с плавающей точкой.
      call  printf          ; Вывод результата.
      mov   rsp,rbp
      pop   rbp
      ret

Здесь	можно	видеть,	что	существует	специализированная	инструкция	loop,	
которая	использует	регистр	rcx	как	уменьшающийся	счетчик	цикла.	При	каж
дом	проходе	по	циклу	значение	rcx	автоматически	уменьшается	на	1,	и пока	

значение	rcx	не равно	0,	цикл	выполняется	снова	и снова.	В этом	варианте	объ
ем	вводимого	исходного	кода	меньше.

Можно	провести	любопытный	 эксперимент:	 поместить	 1000000000	 (едини
цу	с девятью	нулями)	в переменную	number,	затем	пересобрать	и запустить	обе	
предыдущие	 программы.	 Скорость	 выполнения	 программ	 можно	 измерить	
с помощью	команды	Linux	time,	как	показано	ниже:

time ./jumploop
time ./betterloop

Обратите	 внимание:	 betterloop	 работает	 медленнее,	 чем	 jumploop 
(см. рис. 7.5).	Использование	инструкции	loop	более	удобно,	но	за	это	прихо
дится	расплачиваться	снижением	производительности.	Для	измерения	про
изводительности	 применялась	 команда	 Linux	 time,	 но	 в  дальнейшем	 будут	
продемонстрированы	 более	 эффективные	 способы	 обследования	 и  тонкой	
настройки	программного	кода.

Рис. 7.5.	Сравнение использования инструкции цикла и инструкции условного перехода

Возможно,	 вы	 удивлены,	 почему	 мы	 все	 еще	 используем	 отладчик	 DDD,	
если	существует	такое	удобное	инструментальное	средство,	как	SASM.	Немно
го	позже	вы	увидите,	что	в SASM	невозможно	наблюдать	за	состоянием	стека,	
но	это	можно	сделать	с помощью	отладчика	DDD.	Поэтому	в дальнейшем	мы	
вернемся	к работе	с DDD.

резюме
В	этой	главе	вы	узнали,	как:

	� использовать	интегрированную	среду	разработки	SASM;
	� применять	инструкции	перехода;
	� использовать	инструкцию	cmp;
	� применять	инструкцию	loop;
	� проверять	значения	флагов.

Глава 8
Память

Память	используется	процессором	как	место	для	хранения	данных	и инструк
ций.	Мы уже	рассмотрели	регистры –	средства	хранения	данных	с весьма	вы
сокой	скоростью	доступа.	Доступ	к памяти	осуществляется	значительно	мед
леннее,	чем	доступ	к регистрам.	Но количество	регистров	ограничено.	Размер	
памяти	теоретически	ограничен	264	адресами,	т. е.	18 446 744 073 709 551	616,	
или	16	Эб (эксабайт).	Но такой	огромный	объем	памяти	невозможно	использо
вать	изза	проблем	ее	практической	реализации.	Как	бы	то	ни	было,	наступило	
время	для	более	подробного	обследования	памяти.

ОбСледОвание памяти
В	листинге	8.1	показан	пример,	который	будет	использоваться	постоянно	при	
изучении	процесса	обследования	памяти.

Листинг 8.1.	Программа memory.asm

; memory.asm
section .data
      bNum        db    123
      wNum        dw    12345
      warray      times        5 dw 0      ; Массив из 5 слов, ...
                                           ; ...содержащих 0.
      dNum        dd    12345
      qNum1       dq    12345
      text1       db    "abc",0
      qNum2       dq    3.141592654
      text2       db    "cde",0
section .bss
      bvar  resb  1
      dvar  resd  1
      wvar  resw  10
      qvar  resq  3
section .text
      global main
main:
      push   rbp
      mov   rbp, rsp
      lea   rax, [bNum]     ;Загрузка адреса bNum в регистр rax.
      mov   rax, bNum       ;Загрузка адреса bNum в регистр rax



      mov   rax, [bNum]     ;Загрузка значения bNum в регистр rax
      mov   [bvar], rax     ;Загрузка из регистра rax в адрес памяти bvar.
      lea   rax, [bvar]     ;Загрузка адреса bvar в регистр rax.
      lea   rax, [wNum]     ;Загрузка адреса wNum в регистр rax.
      mov   rax, [wNum]     ;Загрузка содержимого wNum в регистр rax.
      lea   rax, [text1]    ;Загрузка адреса text1 в регистр rax.
      mov   rax, text1      ;Загрузка адреса text1 в регистр rax.
      mov   rax, text1+1    ;Загрузка второго символа в регистр rax.
      lea   rax, [text1+1]  ;Загрузка второго символа в регистр rax.
      mov   rax, [text1]    ;Загрузка, начиная с адреса text1, в регистр rax.
      mov   rax, [text1+1]  ;Загрузка, начиная с адреса text1+1, в регистр rax.
      mov   rsp,rbp
      pop   rbp
      ret

Введите	и сохраните	в файле	этот	исходный	код,	затем	выполните	сборку	
программы.	Программа	ничего	не выводит,	поэтому	необходимо	воспользо
ваться	отладчиком	для	пошагового	прохода	по	каждой	ее	инструкции.	Здесь	
будет	полезна	интегрированная	среда	разработки	SASM.

В	рассматриваемом	примере	определены	некоторые	переменные	с различ
ными	размерами,	в том	числе	массив	из	пяти	двойных	слов,	заполненных	ну
лями.	Кроме	того,	некоторые	элементы	определены	в разделе	section .bss.	В ис
пользуемом	отладчике	посмотрите	на	 содержимое	 регистра	 указателя	 стека	
rsp –	он	содержит	достаточно	большое	значение.	Указатель	стека	(stack	pointer)	
ссылается	на	адрес	в верхней	части	памяти.	Стек	(stack) –	это	область	памяти,	
используемая	для	 кратковременного	 хранения	данных.	Стек	растет	по	мере	
добавления	в него	данных	в направлении	«вниз»,	т. е.	от	более	старших	адре
сов	к младшим.	Указатель	стека	rsp	уменьшается	при	каждой	операции	записи	
данных	в стек.	Работа	стека	будет	рассматриваться	в отдельной	главе,	но	уже	
сейчас	необходимо	запомнить,	что	стек –	это	место	в области	верхней	памяти	
(в области	старших	адресов	памяти).	См.	рис. 8.1.

Рис. 8.1.	Регистр rsp содержит адрес в области верхней памяти (старших адресов)

В	рассматриваемом	здесь	примере	использовалась	инструкция	lea,	означа
ющая	«загрузить	эффективный	(действительный)	адрес»	(load	effective	address)	
и предназначенная	для	загрузки	адреса	памяти	переменной	bNum	в регистр	rax.	
Тот	же	результат	можно	получить	с помощью	инструкции	mov	без	квадратных	
скобок,	окружающих	имя	bNum.	Если	используются	квадратные	скобки	[bNum],	
то	инструкция	mov	загружает	значение,	а не адрес	переменной	bNum,	в регистр	
rax.	Но в этом	коде	в регистр	rax	загружается	не только	bNum.	Поскольку	rax	яв
ляется	64битовым	 (или	8байтовым)	регистром,	 в него	 загружается	больше	
байтов.	Значение	bNum	является	крайним	правым	байтом	в rax	(порядок	байтов	
от	младшего	к старшему),	поэтому	нас	интересует	только	содержимое	регистра	
al.	Если	требуется,	чтобы	регистр	rax	содержал	лишь	значение	123,	то	сначала	
необходимо	очистить	rax,	как	показано	ниже:

xor rax, rax

Затем	вместо	инструкции

mov rax, [bNum]

нужно	использовать	инструкцию

mov al, [bNum]

Внимательно	следите	за	размером	данных,	которые	перемещаются	в память	
и из	памяти.	Например,	рассмотрим	следующую	инструкцию:

mov [bvar], rax

В	 соответствии	 с  этой	инструкцией	 8	 байт	из	 регистра	 rax	 перемещаются	
в локацию	с адресом	bvar.	Если	вы	предполагали	запись	только	значения	123 
в блок	памяти	bvar,	то	можете	проверить	с помощью	отладчика	и убедиться	
в том,	что	вы	перезаписали	еще	7	байт	памяти	(в окне памяти	SASM	выберите	
тип	d для	переменной	bvar).	Это	может	привести	к появлению	неожиданных	
ошибок	в программе.	Чтобы	избежать	этого,	необходимо	заменить	некоррект
ную	инструкцию	на	следующую:

mov [bvar], al

При	загрузке	содержимого	из	блока	памяти	с адресом	text1	в регистр	rax	сле
дует	отметить,	что	значение	в rax	сохраняется	с порядком	байтов	от	младше
го	к старшему	(little	endian).	Выполните	пошаговый	проход	по	программе	для	
просмотра	результатов	выполнения	различных	инструкций	и изменяйте	раз
меры	и значения	данных,	чтобы	наблюдать	за	тем,	что	происходит	при	этом.

Существуют	два	способа	загрузки	адреса	памяти:	инструкции	mov	и lea.	При
менение	инструкции	lea	может	сделать	код	более	удобным	для	чтения,	так	как	
сразу	можно	видеть,	что	здесь	происходит	работа	с адресами	памяти.	Кроме	
того,	инструкцию	lea	можно	использовать	для	ускорения	вычислений,	но	здесь	
мы	не будем	применять	lea	для	этой	цели.

Выполните	команду	gdb memory,	затем	команду	disass main	и обратите	внима
ние	на	левый	столбец	с адресами	памяти	(рис. 8.2).	Не забудьте	сначала	удалить	
специальную	строку,	добавленную	SASM	для	корректной	отладки,	о которой	
упоминалось	в предыдущей	главе.	В рассматриваемом	здесь	примере	первая	
инструкция	размещается	по	адресу	0x4004a0.

Рис. 8.2.	Код функции main, дизассемблированный отладчиком GDB

Теперь	 в  командной	 строке	 воспользуемся	 утилитой	 readelf.	Напомню,	что	
NASM	был	проинформирован	о необходимости	ассемблирования	исходного	кода	
в выполняемый	формат	ELF	(см.	соответствующий	makefile).	readelf –	это	утили
та	командной	строки,	используемая	для	получения	более	подробной	информа
ции	о выполняемом	файле.	Если	у вас	возникло	непреодолимое	желание	узнать	
больше	о программах	редактирования	связей	(линкерах),	то	можно	порекомен
довать	качественный	источник	информации:	Linkers	and	Loaders,	John	R. Levine,	
1999,	The	Morgan	Kaufmann,	Series	in	Software	Engineering	and	Programming.

Более	краткое	описание	формата	ELF	можно	найти	здесь:
https://linux-audit.com/elf-binaries-on-linux-understanding-and-analysis/

или	

https://www.cirosantilli.com/elf-hello-world/.
Вероятно,	вы	уже	догадались,	что	в командной	строке	можно	ввести	следу

ющую	команду:

man elf

Для	рассматриваемого	 здесь	примера	необходимо	выполнить	 следующую	
команду:

readelf --file-header ./memory

Выводится	некоторая	общая	информация	о памяти,	используемой	выпол
няемым	 файлом	 memory.	 Обратите	 внимание	 на	 строку	 Entry point address: 
0x4003b0.	 Это	 локация	 в  памяти,	 в  которой	 расположено	 начало	 программы.	
Очевидно,	что	между	точкой	входа	(entry	point)	в программу	и началом	кода,	
которое	показано	в отладчике	GDB	(0x4004a0),	существует	некоторый	добавоч
ный	заголовок.	Этот	заголовок	предоставляет	нам	дополнительную	информа
цию	об	операционной	системе	и выполняемом	коде.	См.	рис. 8.3.

Рис. 8.3.	Вывод заголовка ELF с помощью утилиты readelf

Утилита	readelf	удобна	для	обследования	бинарных	выполняемых	файлов.	
На рис. 8.4	показано	несколько	дополнительных	примеров.

Рис. 8.4.	Вывод символов программы с помощью утилиты readelf

Используя	 утилиту	 grep,	 можно	 выполнить	 поиск	 всех	 строк,	 содержащих	
слово	main.	На рис. 8.4	можно	видеть,	что	основная	функция	main	начинается	
с адреса	0x4004a0,	как	показано	в отладчике	GDB.	В следующем	примере	выпол
няется	поиск	в таблице	символов	каждого	вхождения	метки	start.	На рис. 8.5	
можно	видеть	начальные	адреса	разделов	section .data,	section .bss,	а также	на
чальный	адрес	самой	программы.

Рис. 8.5.	Вывод символов, содержащих слово start, с помощью утилит readelf и grep

Рассмотрим	подробнее,	что	находится	в памяти,	с помощью	команды,	по
казанной	ниже:

readelf --symbols ./memory |tail +10|sort -k 2 -r

Утилита	 tail	 позволяет	исключить	 строки,	 которые	 в данный	момент	нас	
не интересуют.	Выполняется	сортировка	по	второму	столбцу	(по	адресам	па
мяти)	в обратном	порядке.	Вы сами	видите,	как	полезны	знания	о командах	
системы	Linux.

Началу	программы	соответствует	некоторый	младший	адрес,	а начало	функ
ции	main	находится	по	адресу	0x004004a0.	Найден	адрес	начала	раздела	section 
.data	(0x00601018)	с адресами	всех	его	переменных	и адрес	начала	раздела	sec-
tion .bss	 (0x00601051)	с адресами,	зарезервированными	для	переменных	этого	
раздела.

Подведем	итог	всех	поисков:	в начале	главы	мы	выяснили,	что	стек	разме
щается	в верхних	(старших)	адресах	памяти	(см.	регистр	rsp).	С помощью	ути
литы	readelf	было	обнаружено,	что	выполняемый	код	размещается	в нижней	
части	памяти	 (в области	младших	адресов).	Выше	выполняемого	кода	нахо
дится	раздел	section .data,	а над	ним	раздел	section .bss.	Стек	в верхней	памяти	
может	расти	вниз,	т. е.	в направлении	раздела	section .bss.	Доступная	свобод
ная	память	между	стеком	и другими	разделами	программы	называется	кучей	
(heap;	более	формально –	динамически	распределяемой	памятью).

Память	 в  разделе	 section .bss	 распределяется	 во	 время	 выполнения	 про
граммы –	это	легко	проверить.	Обратите	внимание	на	размер	выполняемого	
кода,	а затем	замените,	например,	строку

qvar       resq      3

на	следующую:

qvar      resq      30000

После	этого	пересоберите	программу	и снова	посмотрите	на	размер	выпол
няемого	кода.	Размер	останется	таким	же,	потому	что	дополнительная	память	
не резервируется	во	время	ассемблирования	и связывания.	См.	рис. 8.6.

Рис. 8.6.	Вывод команды readelf --symbols ./memory |tail +10|sort -k 2 -r

На	рис. 8.7	показано,	как	в итоге	выглядит	память	после	загрузки	выполня
емого	кода.

Младший адрес

Аргументы командной строки

.data

.text

Рис. 8.7.	Схема памяти, выделенной программе

Почему	так	важно	знать	структуру	памяти?	Весьма	важно	понимать,	что	стек	
растет	в направлении	нижней	памяти	(младших	адресов).	Это	понимание	по
требуется	 в  следующих	 главах	 книги,	 когда	мы	 будем	детально	 обследовать	
стек.	 Кроме	 того,	 если	 вы	 работаете	 в  сфере	 судебных	 расследований,	 свя
занных	с вредоносным	программным	обеспечением,	то	практический	навык	
анализа	памяти	чрезвычайно	важен.	Здесь	рассматривались	лишь	некоторые	
основные	положения,	но	если	вы	хотите	знать	больше,	то	обратитесь	к упомя
нутым	выше	источникам.

резюме
В	этой	главе	вы	узнали:

	� о	структуре	памяти	(выполняемого)	процесса;
	� как	избежать	непреднамеренной	перезаписи	памяти;
	� как	использовать	утилиту	readelf	для	анализа	двоичного	кода.

Глава 9
Целочисленная арифметика

В	этой	главе	рассматривается	группа	арифметических	инструкций	для	целых	
чисел.	Арифметика	чисел	с плавающей	точкой	описана	в главе	11.	Рекоменду
ется	бегло	просмотреть	главу	2,	чтобы	освежить	знания	о двоичных	(бинар
ных)	числах.

ОСнОвы иСпОльзОвания целОчиСленнОй арифметики
В	листинге	9.1	показан	пример	кода,	который	мы	будем	анализировать.

Листинг 9.1.	Программа icalc.asm

; icalc.asm
extern printf
section .data
      number1    dq    128   ; Числа, используемые для
      number2    dq    19    ; демонстрации арифметических вычислений
      neg_num    dq    -12   ; и распространения знакового разряда.
      fmt        db    "The numbers are %ld and %ld",10,0
      fmtint     db    "%s %ld",10,0
      sumi  db   "The sum is",0
      difi  db   "The difference is",0
      inci  db   "Number 1 Incremented:",0
      deci  db   "Number 1 Decremented:",0
      sali  db   "Number 1 Shift left 2 (x4):",0
      sari  db   "Number 1 Shift right 2 (/4):",0
      sariex db  "Number 1 Shift right 2 (/4) with "
            db   "sign extension:",0
      multi db   "The product is",0
      divi  db   "The integer quotient is",0
      remi  db   "The modulo is",0
section .bss
        resulti  resq  1
        modulo   resq  1
section .text
      global main
main:
      push  rbp
      mov   rbp,rsp



; Вывод чисел.
            mov   rdi, fmt
            mov   rsi, [number1]
            mov   rdx, [number2]
            mov   rax, 0
            call printf
; Сложение------------------------------------------------------------
      mov   rax, [number1]
      add   rax, [number2]            ; Сложение number2 с rax.
      mov   [resulti], rax            ; Перемещение суммы в переменную result.
      ; displaying the result
            mov   rdi, fmtint
            mov   rsi, sumi
            mov   rdx, [resulti]
            mov   rax, 0
            call printf
; Вычитание-----------------------------------------------------------
      mov   rax, [number1]
      sub   rax, [number2]            ; Вычитание number2 из rax.
      mov   [resulti], rax
      ; displaying the result
            mov   rdi, fmtint
            mov   rsi, difi
            mov   rdx, [resulti]
            mov   rax, 0
            call printf
; Инкрементирование----------------------------------------------------
      mov   rax, [number1]
      inc   rax               ; Инкрементирование rax на 1.
      mov   [resulti], rax
      ; displaying the result
            mov   rdi, fmtint
            mov   rsi, inci
            mov   rdx, [resulti]
            mov   rax, 0
            call printf
; Декрементирование-----------------------------------------------------
      mov   rax, [number1]
      dec   rax               ; Декрементирование rax на 1.
        mov [resulti], rax
      ; displaying the result
            mov   rdi, fmtint
            mov   rsi, deci
            mov   rdx, [resulti]
            mov   rax, 0
            call printf
; Арифметический сдвиг влево--------------------------------------------
      mov   rax, [number1]
      sal   rax, 2                  ; Умножение rax на 4.
      mov   [resulti], rax
      ; displaying the result
            mov   rdi, fmtint
            mov   rsi, sali
            mov   rdx, [resulti]
            mov   rax, 0
            call printf

; Арифметический сдвиг вправо-------------------------------------------
      mov   rax, [number1]
      sar   rax, 2                  ; Деление rax на 4.
      mov   [resulti], rax
      ; вывод результата
            mov   rdi, fmtint
            mov   rsi, sari
            mov   rdx, [resulti]
            mov   rax, 0
            call  printf
; Арифметический сдвиг вправо с распространением знакового разряда -----
      mov   rax, [neg_num]
      sar   rax, 2                  ; Деление rax на 4.
      mov   [resulti], rax
      ; вывод результата
            mov   rdi, fmtint
            mov   rsi, sariex
            mov   rdx, [resulti]
            mov   rax, 0
            call  printf
; Умножение----------------------------------------------------------
      mov         rax, [number1]
      imul  qword [number2]         ; Умножение rax на number2.
      mov         [resulti], rax
      ; вывод результата
            mov   rdi, fmtint
            mov   rsi, multi
            mov   rdx, [resulti]
            mov   rax, 0
            call  printf
; Деление------------------------------------------------------------
            mov   rax, [number1]
            mov   rdx, 0           ; В rdx должен быть 0 перед idiv.
            idiv  qword [number2]  ; Деление rax на number2, остаток в rdx.
            mov   [resulti], rax
            mov   [modulo], rdx    ; Запись содержимого rdx в modulo.
; Вывод результата.
            mov rdi, fmtint
            mov   rsi, divi
            mov   rdx, [resulti]
            mov   rax, 0
            call  printf
            mov   rdi, fmtint
            mov   rsi, remi
            mov   rdx, [modulo]
            mov   rax, 0
            call  printf
mov rsp,rbp
pop rbp
ret

На	рис. 9.1	показан	вывод	этой	программы.

Рис. 9.1.	Целочисленная арифметика

изучение арифметичеСких инСтрукций
В	 ассемблере	 доступно	 много	 арифметических	 инструкций,	 но	 здесь	 будут	
рассматриваться	лишь	некоторые	из	них,	а остальные	инструкции	похожи	на	
изучаемые	 в  этой	 главе.	 Перед	 детальным	 изучением	 арифметических	 ин
струкций	необходимо	отметить,	что	в рассматриваемой	здесь	программе	ис
пользуется	 функция	 printf,	 принимающая	 более	 двух	 аргументов,	 поэтому	
требуется	дополнительный	регистр:	первый	аргумент	записывается	в регистр	
rdi,	второй –	в rsi,	третий –	в rdx.	Именно	такой	порядок	ожидает	printf	при	
передаче	аргументов	в среде	Linux.	Более	подробно	об	этом	вы	узнаете	немно
го	позже,	при	обсуждении	соглашений	при	вызовах	функций.

Ниже	приведено	описание	некоторых	арифметических	инструкций:
	� первая	инструкция	add –	ее	можно	использовать	для	сложения	целых	чи
сел	со	знаком	или	без	знака.	Второй	операнд	(источник)	прибавляется	
к первому	операнду	(цели),	а результат	помещается	в первый	операнд	
(цель).	Целевым	операндом	может	быть	регистр	или	блок	памяти.	Ис
точником	может	быть	непосредственное	значение,	регистр	или	блок	па
мяти.	В одной	инструкции	сложения	нельзя	одновременно	использовать	
блоки	памяти	как	источник	и цель.	Если	полученная	в результате	сумма	
слишком	велика	и не умещается	в цели,	то	для	целых	чисел	со	знаком	
устанавливается	(в 1)	флаг	CF,	а для	чисел	без	знака	в той	же	ситуации	
устанавливается	флаг	OF.	Если	результат	равен	0,	то	флаг	ZF	устанавли
вается	в 1,	а если	результат	отрицательный,	то	устанавливается	флаг	SF;

	� действие	инструкции	вычитания	sub	аналогично	действию	инструкции	add;
	� для	инкрементирования	регистра	или	значения	в памяти	с увеличением	
на	1	используется	инструкция	inc.	Аналогичная,	но	противоположная	по	
действию	инструкция	dec	используется	для	декрементирования	регист
ра	или	значения	в памяти	с уменьшением	на	1;

	� инструкции	арифметического	сдвига –	это	особый	тип	арифметических	
инструкций.	Инструкция	сдвига	влево	sal –	это,	по	существу,	умножение:	
если	выполняется	 сдвиг	 влево	на	одну	позицию,	то	число	 умножается	
на 2.	Каждый	бит	сдвигается	влево	на	одну	позицию,	а справа	добавля

ется	0.	Рассмотрим	двоичное	число	1.	При	сдвиге	влево	на	одну	позицию	
получается	двоичное	число	10	или	2	в десятичном	представлении.	Сле
дующий	сдвиг	влево	на	одну	позицию	дает	результат	100	 (двоичный),	
т. е.	4	в десятичном	представлении.	Если	выполнить	сдвиг	влево	сразу	
на	две	позиции,	то	это	соответствует	умножению	на	4.	А что,	если	нужно	
умно	жить	на	6?	Выполняется	сдвиг	влево	два	раза,	затем	два	раза	сложе
ние	с первоначальным	источником,	именно	в таком	порядке;

	� сдвиг	вправо	sar	похож	на	сдвиг	влево,	но	означает	деление	на	2.	Каж
дый	бит	сдвигается	на	одну	позицию	вправо,	а слева	добавляется	допол
нительный	бит.	Но здесь	возникает	сложность:	если	исходное	значение	
было	отрицательным,	то	добавляемый	крайний	левый	бит	должен	быть	
равен	 1,	 а  если	 инструкция	 сдвига	 добавила	 бит	 0	 слева,	 то	 значение	
должно	стать	положительным,	т. е.	результат	становится	неверным.	По
этому	 при	 работе	 с  отрицательными	 числами	 sar	 добавляет	 слева	 бит,	
равный	 1,	 а  в	 случае	 с  положительным	 значением	 слева	 добавляются	
биты,	 равные	 0.	 Это	 называется	 распространением	 (или	 расширени
ем)	 знакового	 разряда	 (sign	 extension).	 Кстати,	 самый	 быстрый	 способ	 
узнать,	является	ли	шестнадцатеричное	число	отрицательным, –	провер
ка	байта	номер	7	(крайний	левый	байт	при	отсчете	с 0	от	крайнего	право
го	байта).	Число	отрицательное,	если	байт	номер	7	начинается	с 8,	9,	A,	B,	
C,	D,	E или	F.	Но при	этом	необходимо	принять	во	внимание	содержимое	
всех	8	байтов.	Например,	число	0xd12	 остается	положительным,	так	как	
крайний	левый	байт,	который	не показан	в этом	представлении,	равен	0;

	� существуют	также	инструкции	неарифметического	сдвига,	которые	бу
дут	рассматриваться	в главе	16;

	� далее	выполняется	умножение	целых	чисел.	Для	умножения	целых	чисел	
без	знака	можно	воспользоваться	инструкцией	mul –	беззнаковое	умно
жение,	 а инструкция	 imul	 применяется	для	 умножения	 с  учетом	 знака.	
Мы будем	использовать	imul –	инструкцию	знакового	умножения,	кото
рая	 предоставляет	 бóльшую	 гибкость:	 она	может	принимать	 один,	два	
или	три	операнда.	В рассматриваемом	здесь	примере	используется	один	
операнд,	 следующий	непосредственно	за	инструкцией	imul, –	он	умно
жается	на	значение	в регистре	rax.	Возможно,	вы	предположили,	что	ре
зультат	умножения	сохраняется	в регистре	rax,	но	это	совершенно	невер
но.	Рассмотрим	выполнение	этой	операции	на	примере:	вы	сами	можете	
проверить,	что	при	умножении,	скажем,	двузначного	числа	на	трехзнач
ное	число	произведение	будет	четырехзначным	или	пятизначным.	При	
умножении	48битового	числа	на	30битовое	число	вы	получите	77би
товое	или	78битовое	число,	а такое	значение	не уместится	в 64битовом	
регистре.	Чтобы	справиться	с этой	проблемой,	инструкция	imul	сохраня
ет	младшие	64	бит	полученного	произведения	в регистре	rax,	а старшие	
64 бит	в регистре	rdx.	Такой	подход	может	ввести	в заблуждение.
Проведем	небольшой	эксперимент:	вернемся	к исходному	коду	в SASM.	
Изменим	 значение	 number1	 так,	 чтобы	 эта	 переменная	 содержала	 число	
12345678901234567,	а переменной	number2	теперь	будет	присвоено	значение	
100.	Произведение	этих	чисел	будет	в точности	соответствовать	размеру	ре
гистра	rax –	это	можно	проверить	в режиме	отладки	SASM.	Назначим	точку	
останова	 перед	 инструкцией	 imul.	 Перезапустите	 режим	 отладки	 и  нач

ните	пошаговый	проход	по	программе.	Результат	умножения	будет	равен	
1234567890123456700,	как	можно	видеть	в регистре	rax	после	выполнения	ин
струкции	imul.	Теперь	замените	значение	number2	на	10000.	Перезапустите	
режим	отладки.	После	выполнения	инструкции	imul	проверьте	регистр	rax.	
Вы увидите,	что	произведением	теперь	является	большое	отрицательное	
число.	Причина	в том,	что	самым	значимым	(старшим)	битом	в rax	явля
ется	1,	и SASM	приходит	к выводу,	что	такое	значение	непременно	должно	
быть	отрицательным	числом.	Функция	printf	также	считает,	что	rax	содер
жит	отрицательное	число,	потому	что	в крайней	левой	позиции	rax	нахо
дится	1,	следовательно,	число	отрицательное.	Поэтому	применение	printf 
требует	особого	внимания	и осторожности.
Проведем	 более	 глубокое	 исследование:	 сразу	 после	 выполнения	 ин
струкции	imul	регистр	rax	содержит	число	0xb14e9f812f364970.	В двоичном	
представлении	оно	выглядит	как	1011000101001110100111111000000100101111
00110110010010010 –	с 1	в позиции	самого	значимого	(старшего)	бита,	сле
довательно,	является	отрицательным	числом.
Но	регистр	rdx	содержит	значение	0x6,	которому	соответствует	двоичное	
представление	 000000000000000000000000000000000000000000000000000000110 – 
с 0	в позиции	самого	значимого	(старшего)	бита,	следовательно,	это	по
ложительное	число.
В	действительности	произведением	является	число	 0x6b14e9f812f364970,	
которое	можно	найти,	объединив	регистры	rdx	и rax	в правильном	по
рядке:	rdx:rax.	Если	преобразовать	это	шестнадцатеричное	число	в деся
тичное,	то	можно	убедиться	в том,	что	полученное	произведение	соот
ветствует	ожидаемому	значению:	123456789012345670000.	См.	рис. 9.2.
В	интернете	можно	найти	приложения,	позволяющие	выполнять	опера
ции	преобразования	шестнадцатеричных	чисел	в десятичные,	например	
https://www.rapidtables.com/convert/number/hex-to-decimal.html;

Рис. 9.2.	Содержимое регистров rax и rdx

	� следующая	операция –	целочисленное	деление	idiv.	В действительности	
это	операция,	обратная	умножению	(а чего	вы	ожидали?).	Делимое	нахо
дится	в регистрах	rdx:rax,	делитель –	в операндеисточнике,	а результат	
(частное)	целочисленного	деления	 (целое	число –	 integer)	 сохраняется	
в регистре	rax.	Остаток	(modulo)	сохранен	в регистре	rdx.	Легко	забыть	
о весьма	важном	факте:	необходимо	убедиться	в том,	что	в регистре	rdx 
содержится	ноль,	перед	каждым	выполнением	инструкции	idiv,	иначе	
полученное	частное	может	оказаться	неверным.

Операциям	умножения	и деления	64битовых	целых	чисел	присущи	некото
рые	тонкости,	более	подробное	описание	которых	можно	найти	в руководствах	
Intel.	В этой	главе	приведен	лишь	краткий	обзор,	служащий	общим	введением	
в целочисленную	арифметику.	В руководствах	Intel	представлены	подробней
шие	описания	не только	перечисленных	выше	инструкций,	но	и многих	других	
арифметических	инструкций,	которые	можно	применять	в особых	ситуациях.

резюме
В	этой	главе	вы	узнали:

	� как	использовать	целочисленную	арифметику;
	� как	выполнять	арифметический	сдвиг	влево	и вправо;
	� о	том,	что	операция	умножения	использует	регистры	rax	и rdx	для	сохра
нения	произведения;

	� о	том,	что	операция	деления	использует	регистры	rax	и rdx	для	хранения	
делимого;

	� о	необходимости	внимательного	и осторожного	использования	функции	
printf	при	выводе	значений.

Глава 10
Стек

В	предыдущих	главах	мы	рассмотрели	использование	регистров,	особого	типа	
временного	 быстрого	 хранилища	 данных,	 которые	 могут	 применяться	 для	
хранения	значений	или	адресов	во	время	выполнения	инструкций.	Существу
ет	также	более	медленный	тип	хранилища –	оперативная	память,	куда	про
цессор	может	записывать	значения	для	более	длительного	хранения.	Наконец,	
существует	стек	(stack) –	непрерывный	массив	блоков	памяти.

изучение рабОты Стека
Как	было	отмечено	в главе	8,	сегмент	стека	начинается	в старших	адресах	па
мяти	 (в  верхней	памяти)	и  растет	 вниз,	 в направлении	к младшим	адресам	
памяти,	подобно	сосульке.	Элементы	данных	помещаются	в стек	инструкцией	
push,	 а  удаляются	 (извлекаются)	из	 стека	инструкцией	 pop.	При	каждом	про
талкивании	(push)	данных	стек	растет,	при	каждом	выталкивании	(pop)	стек	
уменьшается.	Это	поведение	стека	можно	проверить,	наблюдая	за	регистром	
указателя	стека	rsp,	который	указывает	на	вершину	стека	(в действительности	
на	«дно»,	т. е.	нижнюю	границу,	так	как	стек	растет	вниз).

Стек	можно	использовать	как	временное	хранилище	для	записи	значений	
из	регистров	и последующего	возврата	в регистры	или,	что	более	важно,	для	
передачи	значений	в функции.	Более	подробно	функции	или	процедуры	будут	
рассматриваться	несколько	позже.

В	примере	исходного	кода	в листинге	10.1	стек	используется	для	реверсиро
вания	строки,	т. е.	для	изменения	порядка	букв	в строке	на	обратный.

Листинг 10.1.	Программа stack.asm

; stack.asm
extern printf
section .data
      strng       db    "ABCDE",0
      strngLen    equ   $ - strng-1 ; stringlength without 0
      fmt1        db    "The original string: %s",10,0
      fmt2        db    "The reversed string: %s",10,0
section .bss
section .text
      global main
main:



push  rbp
mov   rbp,rsp
; Вывод исходной строки.
    mov    rdi, fmt1
    mov    rsi, strng
    mov    rax, 0
    call   printf
; Проталкивание (push) строки символ за символом в стек.
     xor   rax, rax
     mov   rbx, strng      ; Адрес строки strng в регистре rbx.
     mov   rcx, strngLen   ; Длина строки в регистре rcx – счетчик.
     mov   r12, 0          ; Использовать регистр r12 как указатель.
     pushLoop:
          mov   al, byte [rbx+r12] ; Запись символа в регистр rax.
          push  rax        ; Проталкивание (push) значения rax в стек.
          inc   r12        ; Увеличение указателя на символ на 1
          loop  pushLoop   ; Продолжение цикла.
; Выталкивание (pop) строки символ за символом из стека.
; Это позволяет реверсировать исходную строку.
     mov   rbx, strng      ; Адрес строки strng в регистре rbx.
     mov   rcx, strngLen   ; Длина строки в регистре rcx – счетчик.
     mov   r12, 0          ; Использовать регистр r12 как указатель.
     popLoop:
          pop   rax        ; Выталкивание (pop) символа из стека.
          mov   byte [rbx+r12], al    ; Запись символа в строку strng.
          inc   r12                   ; Увеличение указателя на символ на 1.
          loop  popLoop               ; Продолжение цикла.
          mov   byte [rbx+r12],0   ; Завершение строки значением 0.
; Вывод реверсированной строки.
    mov    rdi, fmt2
    mov    rsi, strng
    mov    rax, 0
    call   printf
mov   rsp,rbp
pop   rbp
ret

На	рис. 10.1	показан	вывод	этой	программы.

Рис. 10.1.	Реверсирование строки

Прежде	всего	следует	отметить,	что	для	вычисления	длины	строки	ее	размер	
был	уменьшен	на	1,	чтобы	не учитывать	завершающий	0.	Если	этого	не сде
лать,	то	реверсированная	строка	будет	начинаться	с 0.	Затем	исходная	стро
ка	выводится	со	следующим	за	ней	символом	перехода	на	новую	строку.	Для	
запи	си	символов	в стек	будет	использоваться	регистр	rax,	поэтому	необходи
мо	предварительно	инициализировать	rax	нулями	с помощью	инструкции	xor.	

Адрес	строки	передается	в регистр	rbx,	далее	будет	применяться	инструкция	
loop,	поэтому	в регистре	rcx	устанавливается	значение	длины	строки.	Затем	на
чинается	цикл	с поочередной	записью	символов	(по	одному	символу	на	каж
дой	итерации	цикла)	в стек,	начиная	с первого	символа.	Символ	(байт)	запи
сывается	в регистр	al.	После	этого	содержимое	rax	проталкивается	в стек.	При	
каждом	использовании	инструкции	push	в стек	перемещается	8	байт.	Если	бы	
регистр	rax	не был	предварительно	инициализирован	нулями,	то,	возможно,	
в нем	содержались	бы	некоторые	значения	в старших	байтах,	и запись	этих	
байтов	в стек	привела	бы	к непредсказуемому	результату.	После	этой	операции	
стек	содержит	переданный	символ	и дополнительные	нулевые	биты	во	всех	
разрядах,	которые	старше	al.

После	завершения	цикла	loop	последний	символ	строки	находится	на	«вер
шине»	стека,	которая	в действительности	является	самым	младшим	адресом	
этой	«сосульки»,	потому	что	стек	растет	вниз.	Начинается	второй	цикл,	в кото
ром	символы	поочередно	извлекаются	(выталкиваются)	из	стека,	затем	сохра
няются	в блоке	памяти	исходной	строки	один	за	другим.	Обратите	внимание:	
необходим	только	один	байт,	поэтому	символы	извлекаются	в регистр	rax,	но	
при	записи	в память	используется	только	регистр	al.

На	рис. 10.2	показана	общая	схема	выполняемого	процесса:	исходная	строка	
изображена	справа,	ее	символы	по	одному	записываются	в стек	и добавляются	
к предыдущему	его	содержимому.	Потом	символы	извлекаются	из	стека	и за
писываются	обратно	по	адресу	памяти	исходной	строки,	но	поскольку	стек	ра
ботает	по	схеме	«последним	пришел,	первым	вышел»	(last	in	first	out –	LIFO),	
символы	располагаются	в строке	в обратном	порядке –	строка	реверсирована.

Стек

ПОСЛЕ

Строка

ДО

Строка

Рис. 10.2.	Схема реверсирования строки

Необходимо	какимлибо	способом	отслеживать,	что	записывается	в стек	и в	
каком	порядке.	Например,	при	использовании	стека	для	временного	хранения	
содержимого	регистров	требуется	обеспечить	извлечение	значений	регистров	
в правильном	обратном	порядке,	иначе	программа	будет	работать	некоррект
но	или	в худшем	случае	может	 завершиться	аварийно.	Таким	образом,	 если	
запись	содержимого	регистров	в стек	выполняется	в последовательности,	по
казанной	ниже:

push rax
push rbx
push rcx

то	извлекать	значения	в регистры	необходимо	в соответствии	с правилом	«по
следним	пришел,	первым	вышел»	(LIFO),	как	показано	ниже:

pop rcx
pop rbx
pop rax

Кроме	регистров,	в стек	можно	записывать	содержимое	блоков	памяти	и не
посредственные	 значения.	Извлекать	данные	из	 стека	можно	 в  регистр	или	
в блок	памяти,	но	не в непосредственное	значение,	что	вполне очевидно.

Об	этом	полезно	знать,	однако	здесь	мы	не пользуемся	этими	возможно
стями.	Если	необходимо	записывать	в стек	и извлекать	из	стека	содержимое	
регистра	флагов,	то	можно	воспользоваться	инструкциями	pushf	и popf.

наблюдение за СтекОм
Как	мы	выяснили	выше,	наблюдение	за	стеком	важно,	и наш	добрый	друг	DDD 
предлагает	 удобные	 функциональные	 возможности	 для	 этого.	 Сначала	 от
кройте	текстовый	редактор	с исходным	кодом	программы	и удалите	строку,	
которую	добавила	среда	разработки	SASM,	затем	сохраните	файл	и закройте	
редактор.	В командной	строке	выполните	сборку	программы,	потом	введите	
следующую	команду:

ddd stack

В	 меню	 выберите	 пункт	Data	 (Данные)	 → Status	 Displays	 (Показать	 со
стояние)	и выполняйте	прокрутку	в окне,	пока	не найдете	строку	параметра	
Backtrace	of	the	stack	(Обратная	трассировка	стека) –	ее	необходимо	активи
зировать.	Установите	точку	останова,	например,	на	метке	main:,	 затем	щелк
ните	по	кнопке	Run	(Пуск)	в перемещаемой	панели.	После	этого	начните	про
цесс	 отладки	 и  пошагово	 перемещайтесь	 по	 программе	 с  помощью	 кнопки	
Next	(Следующий	шаг)	(нет	необходимости	построчно	проходить	по	функции	
printf).	Наблюдайте	 в  верхнем	окне,	 как	 выводится	и  обновляется	 содержи
мое	стека.	Не обращайте	внимания	на	вывод	начального	содержимого	стека.	
При	переходе	к инструкции,	следующей	за	инструкцией	push,	вы	увидите,	что	
символы	помещены	в стек	и отображены	в десятичном	представлении	ASCII	

(41, 42	и т.	д.).	Наблюдайте	за	уменьшением	стека	во	время	выполнения	вто
рого	цикла.	Это	простой	способ	наблюдения	за	тем,	что	находится	в стеке	и в	
каком	порядке.

На	рис. 10.3	показано,	как	это	выглядит	в окне отладчика	DDD.

Рис. 10.3.	Стек в окне отладчика DDD

Как	уже	было	отмечено	ранее,	DDD –	это	довольнотаки	старая	программа	
с открытым	исходным	кодом.	Нет	никакой	уверенности	в том,	что	в будущем	
она	будет	продолжать	работать	корректно,	но	сейчас	DDD	делает	то,	что	нам	
нужно,	хотя	и выглядит	не слишком	элегантно.

По	справедливости	говоря,	SASM	тоже	можно	заставить	показывать	содер
жимое	стека,	но	для	этого	потребуется	дополнительная	ручная	работа.	Вот	как	
это	делается:	сначала	напомню,	что	во	время	отладки	в SASM	можно	выводить	
значения	переменных	в памяти,	а стек –	это	просто	список	блоков	памяти	с ре
гистром	rsp,	указывающим	на	самый	нижний	блок.	Таким	образом,	необходи
мо	убедить	SASM	показать	то,	что	находится	по	адресу,	хранящемуся	в rsp,	и в	
блоках	памяти	с более	старшими	адресами.	На рис. 10.4	показан	пример	со
держимого	окна	Memory	(Память)	в SASM,	где	отображено	содержимое	стека.

Рис. 10.4.	Содержимое стека в окне SASM

Ссылка	на	регистр	rsp	выглядит	как	$rsp.	Адрес	стека	каждый	раз	увеличи
вается	на	8	($rsp + 8),	потому	что	при	выполнении	каждой	инструкции	записи	
push	в стек	передается	8	байт.	Для	столбца	Type	 (Тип)	задано	значение	Char	
(Символ),	b (байты)	и размер	8,	а также	активизирован	атрибут	Address	(Адрес).	
Тип	символов	(Char)	выбран	потому,	что	в стек	записывается	строка,	и в	таком	
представлении	ее	проще	читать,	а подтип	«байты»	выбран	потому,	что	нас	ин
тересуют	значения	байтов	(при	каждой	операции	передачи	в стек	регистр	al 
содержит	1	байт),	а на	каждой	итерации	цикла	в стек	записывается	8	байт.	Ре
гистр	rsp	содержит	адрес	(активизирован	атрибут	Address).	Выполняйте	поша
говый	проход	по	программе	и наблюдайте,	как	изменяется	содержимое	стека.

Этот	 прием	 работает,	 но	для	 наблюдения	 все	 характеристики	 и  атрибуты	
каждого	блока	памяти	стека	необходимо	устанавливать	вручную –	это	может	
оказаться	 слишком	 утомительным,	 если	 используется	 большой	 стек	 и/или	
большое	количество	переменных	в памяти,	требуемых	для	наблюдения	за	со
держимым	стека.

резюме
В	этой	главе	вы	узнали:

	� что	стек	начинается	со	старших	адресов	памяти	и растет	в направлении	
уменьшения	адресов;

	� что	инструкция	записи	в стек	push	уменьшает	указатель	стека	rsp;
	� что	инструкция	извлечения	из	стека	pop	увеличивает	указатель	стека	rsp;
	� что	инструкции	push	и pop	работают	во	взаимно	обратном	порядке;
	� как	использовать	отладчик	DDD	для	наблюдения	за	стеком;
	� как	использовать	интегрированную	среду	разработки	SASM	для	наблю
дения	за	стеком.

Глава 11
Арифметика с плавающей 

точкой

Вы	уже	знаете	о целочисленной	арифметике,	а сейчас	мы	рассмотрим	неко
торые	основы	вычислений	 с плавающей	точкой.	В  этом	нет	ничего	 сложно
го,	число	с плавающей	точкой	содержит	десятичную	точку	и ноль	или	более	
десятичных	знаков	после	точки.	Здесь	будут	рассматриваться	два	типа	чисел	
с плавающей	точкой:	с обычной	 (одиночной)	точностью	и с	двойной	точно
стью.	Числа	с двойной	точностью	более	верны,	потому	что	позволяют	работать	
с бóльшим	количеством	значимых	разрядов.	После	получения	этой	информа
ции	вы	обладаете	знаниями,	достаточными	для	выполнения	и анализа	при
меров	программ	в текущей	главе.

Сравнение чиСел С ОбычнОй и двОйнОй тОчнОСтью
Немного	истории	для	тех,	кто	ею	интересуется.

Число	с обычной	(одиночной)	точностью	(single	precision)	хранится	в 32	битах:	
1	знаковый	бит,	8	бит	показателя	степени	и 23	бита	дробной	части.

S      EEEEEEEE      FFFFFFFFFFFFFFFFFFFFFFF
0      1      8      9                     31

Число	с двойной	точностью	(double	precision)	хранится	в 64	битах:	1	знако
вый	бит,	11	бит	показателя	степени	и 52	бита	дробной	части.

S      EEEEEEEEEEE      FFFFFFFFFFFFF......FFFFFFFFFF
0      1         11     12                          63

Знаковый	бит	объясняется	просто:	если	число	положительное,	то	этот	бит	
равен	0,	если	число	отрицательное,	то	этот	бит	равен	1.

Смысл	битов	показателя	степени	объяснить	немного	сложнее.	Рассмотрим	
пример	с десятичными	числами.

200	=	2.0	×	102

5000.30	=	5.0003	×	103



Теперь	пример	с двоичными	числами:

1101010.01011	=	1.0101001011	×	26	(дробная	точка	перемещена	на	6	позиций	
влево)

Но	 показатель	 степени	 может	 быть	 положительным,	 отрицательным	 или	
равняться	нулю.	Чтобы	сделать	эти	различия	ясными,	для	чисел	с обычной	точ
ностью	к положительному	показателю	степени	прибавляется	число	127	перед	
сохранением.	Это	 означает,	 что	нулевой	показатель	 степени	будет	 сохранен	
как	127.	Это	число	127	называется	смещением	(bias).	Для	чисел	с двойной	точ
ностью	смещение	равно	1023.

В	 приведенном	 выше	примере	 число	 1.0101001011	 называется	мантиссой 
(significand;	mantissa).	Предполагается,	 что	 первый	 бит	мантиссы	 всегда	 ра
вен 1	(т. е.	мантисса	«нормализована»),	поэтому	он	не сохраняется.

Ниже	 приводится	 простой	 пример,	 показывающий,	 как	 это	 работает.	
Для  проверки	 результата	 проводимого	 здесь	 эксперимента	 воспользуйтесь,	
например,	этим	интернетресурсом:	https://babbage. cs.qc.cuny.edu/IEEE-754/.

Пример	с десятичным	числом	с обычной	точностью	10:
	� десятичное	число	10	имеет	вид	1010	как	двоичное	целое	число;
	� знаковый	бит	равен	0,	потому	что	число	положительное;
	� получено	число	в формате	b.bbbb.	Здесь	1.010 –	мантисса	с первой	1	в со
ответствии	с требованиями.	Самая	первая	1	не будет	сохранена;

	� следовательно,	показатель	 степени	равен	3,	потому	что	дробная	точка	
была	перемещена	на	три	позиции	влево.	Показатель	 степени	положи
тельный,	к нему	прибавляется	127,	в результате	показатель	степени	ра
вен	130,	или	в двоичном	формате	10000010;

	� таким	образом,	десятичное	число	с обычной	точностью	10	будет	сохра
нено	в следующем	виде:

0 10000010       01000000000000000000000
S EEEEEEEE       FFFFFFFFFFFFFFFFFFFFFFF

или	41200000	в шестнадцатеричном	формате.
Следует	отметить,	что	шестнадцатеричное	представление	значения	обыч

ной	точности	отличается	от	того	же	значения,	представленного	с двойной	точ
ностью.	Почему	бы	не использовать	всегда	числа	с двойной	точностью	и по
лучать	 соответствующие	 преимущества?	 Вычисления	 с  двойной	 точностью	
медленнее	вычислений	с обычной	точностью,	 а операнды	с двойной	точно
стью	требуют	больше	памяти.

Если	все	вышеизложенное	кажется	вам	сложным,	то	вы	правы.	Попробуйте	
найти	в интернете	подходящее	инструментальное	средство	для	выполнения	
или,	по	крайней	мере,	для	проверки	выполненных	преобразований.

В	более	старых	программах	вам	могут	встретиться	80битовые	числа	с пла
вающей	точкой,	но	для	таких	чисел	существуют	собственные	специализирован
ные	инструкции,	обозначаемые	как	FPUинструкции.	Эта	функциональность	
представляет	собой	наследие	прошлого	и не рекомендуется	к использованию	

в новых	разработках.	Тем	не менее	в статьях	в интернете	время	от	времени	вы	
будете	обнаруживать	FPUинструкции.

Теперь	приступим	к выполнению	более	интересных	действий.

кОдирОвание С применением чиСел С плавающей тОчкОй
В	листинге	11.1	показан	пример	программы,	использующей	числа 
с плавающей	точкой.

Листинг 11.1.	Программа fcalc.asm

; fcalc.asm
extern printf
section .data
      number1      dq    9.0
      number2      dq    73.0
      fmt          db    "The numbers are %f and %f",10,0
      fmtfloat     db    "%s %f",10,0
      f_sum        db    "The float sum of %f and %f is %f",10,0
      f_dif        db    "The float difference of %f and %f is %f",10,0
      f_mul        db    "The float product of %f and %f is %f",10,0
      f_div        db    "The float division of %f by %f is %f",10,0
      f_sqrt       db    "The float squareroot of %f is %f",10,0
section .bss
section .text
      global main
main:
    push    rbp
    mov     rbp,rsp
; Вывод чисел.
      movsd       xmm0, [number1]
      movsd       xmm1, [number2]
      mov   rdi,fmt
      mov   rax,2          ; Два числа с плавающей точкой.
      call  printf
; Вычисление суммы.
    movsd   xmm2, [number1]   ; Число с плавающей точкой двойной точности в xmm.
    addsd   xmm2, [number2]   ; Сложение с другим числом двойной точности, результат в xmm.
            ; Вывод результата.
            movsd xmm0, [number1]
            movsd xmm1, [number2]
            mov   rdi,f_sum
            mov   rax,3     ; Три числа с плавающей точкой.
            call  printf
; Вычисление разности.
    movsd   xmm2, [number1]      ; Число с плавающей точкой двойной точности в xmm.
    subsd   xmm2, [number2]      ; Вычитание другого числа двойной точности из xmm.
            ; Вывод результата.
            movsd xmm0, [number1]
            movsd xmm1, [number2]
            mov   rdi,f_dif
            mov   rax,3     ; Три числа с плавающей точкой.
            call  printf

; Умножение.
    movsd   xmm2, [number1]      ; Число с плавающей точкой двойной точности в xmm.
    mulsd   xmm2, [number2]      ; Умножение заданного числа на xmm.
            ; Вывод результата.
            mov   rdi,f_mul
            movsd xmm0, [number1]
            movsd xmm1, [number2]
            mov   rax,3     ; Три числа с плавающей точкой.
            call  printf
; Деление.
    movsd   xmm2, [number1]      ; Число с плавающей точкой двойной точности в xmm.
    divsd   xmm2, [number2]      ; Деление xmm0.
            ; Вывод результата.
            mov   rdi,f_div
            movsd xmm0, [number1]
            movsd xmm1, [number2]
            mov   rax,1     ; Одно число с плавающей точкой.
            call  printf
; Вычисление квадратного корня.
    sqrtsd  xmm1, [number1]      ; Квадратный корень двойной точности в xmm.
            ; Вывод результата.
            mov   rdi,f_sqrt
            movsd xmm0, [number1]
            mov   rax,2     ; Два числа с плавающей точкой.
            call  printf
; Выход из программы.
            mov   rsp, rbp
            pop   rbp            ; Операция, обратная операции push в начале программы.
            ret

Это	простая	программа –	в действительности	организация	вывода	требует	
больше	усилий,	чем	вычисления	с плавающей	точкой.	

Используйте	отладчик	для	пошагового	прохода	по	программе	и наблюдайте	
за	содержимым	регистров	и блоков	памяти.	Например,	обратите	внимание	на	
то,	как	числа	9.0	и 73.0	сохраняются	в блоках	памяти	с адресами	number1	и num-
ber2 –	это	значения	с плавающей	точкой	двойной	точности.

Следует	 напомнить,	 что	 при	 отладке	 в  SASM	 регистры	 xmm	 расположены	
в нижней	части	окна	регистров,	в крайней	левой	области	регистров	ymm.

Инструкция	 movsd	 означает	 «move	 a  double	 precision	 floating	 point	 value»	
(переместить	 значение	 с  плавающей	точкой	двойной	точности).	 Существует	
также	аналогичная	инструкция	movss	для	чисел	с обычной	точностью.	Арифме
тические	инструкции	для	чисел	с обычной	точностью:	addss,	subss,	mulss,	divss 
и sqrtss.

Все	прочее	должно	быть	абсолютно	понятно	на	текущий	момент.	На рис. 11.1	
показан	вывод	этой	программы.

Рис. 11.1.	Вывод программы fcalc.asm

Теперь,	когда	вы	знаете,	как	используется	стек,	попробуйте	сделать	следу
ющее:	закомментируйте	инструкцию	push rbp	в начале	программы	и инструк
цию	pop rbp	 в конце.	Выполните	сборку,	запустите	программу	и наблюдайте,	
что	происходит:	программа	«падает»	(завершается	аварийно).	Причина	этого	
аварийного	завершения	станет	понятной	немного	позже,	но	следует	отметить,	
что	она	связана	с выравниванием	в стеке.

резюме
В	этой	главе	вы	узнали:

	� об	основах	использования	регистров	xmm	для	вычислений	с плавающей	
точкой;

	� о	различиях	между	числами	с обычной	(одиночной)	точностью	и числа
ми	с двойной	точностью;

	� об	инструкциях	movsd,	addsd,	subsd,	mulsd,	divsd	и sqrtsd.

Глава 12
Функции

Ассемблер	 не  является	 «структурным	 языком».	 Взгляните	 на	 многочислен
ные	инструкции	jmp	и метки,	которые	позволяют	при	выполнении	программы	
кудато	переходить	и возвращаться	обратно.	В современных	языках	програм
мирования	высокого	уровня	существуют	такие	программные	конструкции,	как	
do…while,	while…do,	case	и т.	п.	Подобных	конструкций	нет	в языке	ассемблера.
Но,	как	и в	современных	языках	программирования,	в языке	ассемблера	име
ются	функции	и процедуры,	которые	помогают	сделать	исходный	код	более	
структурированным.	 Небольшое	 уточнение:	 функция	 выполняет	 инструк
ции	и возвращает	некоторое	значение.	Процедура	выполняет	инструкции,	но	
не возвращает	значение.

В	этой	книге	мы	уже	использовали	функции,	точнее	применяли	внешнюю	
функцию	с именем	printf	из	стандартной	библиотеки	языка	C.	В данной	главе	
представлены	простые	функции,	в следующих	главах	будут	рассматриваться	
важные	подробности	использования	функций,	такие	как	выравнивание	стека,	
внешние	функции	и соглашения	о вызове	функций.

СОздание прОСтОй функции
В	листинге	12.1	показан	пример	программы	на	ассемблере	с простой	функцией,	
вычисляющей	площадь	круга.

Листинг 12.1.	Программа function.asm

; function.asm
extern printf
section .data
      radius      dq    10.0
      pi          dq    3.14
      fmt         db    "The area of the circle is %.2f",10,0
section .bss
section .text
      global main
;----------------------------------------------
main:
push  rbp
mov   rbp, rsp
      call  area              ; Вызов функции.
      mov   rdi,fmt           ; Формат вывода.



      movsd xmm1, [radius]    ; Запись числа с плав.точкой в регистр xmm1.
      mov   rax,1             ; Значение площади в регистре xmm0.
      call  printf
leave
ret
;----------------------------------------------
area:
push  rbp
mov   rbp, rsp
      movsd xmm0, [radius]    ; Запись числа с плав.точкой в регистр xmm0.
      mulsd xmm0, [radius]    ; Умножение xmm0 на число с плав.точкой.
      mulsd xmm0, [pi]        ; Умножение xmm0 на (другое) число с плав.точкой.
leave
ret

На	рис. 12.1	показан	вывод	этой	программы.

Рис. 12.1.	Вывод программы function.asm

В	этой	программе	имеется	основная	часть,	определяемая,	как	и ранее,	мет
кой	main,	а за	ней	следует	функция,	определяемая	меткой	area.	В основной	части	
main	вызывается	функция	area,	вычисляющая	площадь	круга	с использованием	
radius	и pi –	переменных,	значения	которых	хранятся	в блоках	памяти.	Здесь	
можно	видеть,	что	функции	обязательно	должны	иметь	пролог	и эпилог,	так	
же,	как	и основная	часть	main.

Вычисленная	площадь	круга	сохраняется	в регистре	xmm0.	После	возврата	из	
функции	в основную	часть	main	вызывается	функция	printf	с регистром	rax,	со
держащим	1, –	это	означает,	что	один	xmmрегистр	должен	быть	выведен.	Здесь	
вводится	новая	инструкция	leave,	которая	выполняет	те	же	действия,	что	и mov 
rsp, rbp	и pop rbp	(эпилог).

Если	 из	функции	 возвращается	 значение,	то	для	 чисел	 с  плавающей	точ
кой	используется	регистр	xmm0,	а для	прочих	значений,	таких	как	целые	числа	
и адреса,	используется	регистр	rax.	Аргументы	функции	pi	и radius	размещены	
в памяти.	В рассматриваемом	здесь	примере	это	приемлемый	вариант,	но	все	
же	лучше	применять	регистры	и стек	для	хранения	аргументов	функции.	Ис
пользование	переменных	памяти	для	передачи	 значений	в функцию	может	
привести	к конфликту	имен	для	значений,	используемых	в основной	части	main 
и в	функциях,	а кроме	того,	может	сделать	код	менее	«переносимым».

еще О функциях
Рассмотрим	некоторые	характеристики	функций	на	другом	примере,	показан
ном	в листинге	12.2.

Листинг 12.2.	Программа function2.asm

; function2.asm
extern printf
section .data
      radius      dq    10.0
section .bss
section .text
;----------------------------------------------
area:
      section .data
            .pi  dq    3.141592654      ; Локальная переменная в функции area.
      section .text
push  rbp
mov   rbp, rsp
      movsd      xmm0, [radius]
      mulsd      xmm0, [radius]
      mulsd      xmm0, [.pi]
leave
ret
;----------------------------------------------
circum:
section .data
      .pi  dq     3.14        ; Локальная переменная в функции circum.
section .text
push  rbp
mov   rbp, rsp
      movsd       xmm0, [radius]
      addsd       xmm0, [radius]
      mulsd       xmm0, [.pi]
leave
ret
;----------------------------------------------
circle:
section .data
      .fmt_area  db    "The area is %f",10,0
      .fmt_circum      db    "The circumference is %f",10,0
section .text
push  rbp
mov   rbp, rsp
      call  area
      mov   rdi,.fmt_area
      mov   rax,1            ; Значение площади в регистре xmm0.
      call  printf
      call  circum
      mov   rdi,.fmt_circum
      mov   rax,1            ; Значение длины окружности в регистре xmm0.
      call  printf
leave
ret
;----------------------------------------------
      global main
main:
push  rbp
mov   rbp, rsp
      call  circle
leave
ret

В	 этой	 программе	 основная	 часть	 main	 вызывает	функцию	 circle,	 которая	
в свою	очередь	вызывает	функции	area	и circum.	Таким	образом,	функции	могут	
вызывать	другие	функции.	В действительности	основная	часть	main –	это	тоже	
функция,	вызывающая	другие	функции.	Но помните	о том,	что	функции	не мо
гут	быть	вложенными,	т. е.	функции	не должны	содержать	код	других	функций.

Функции	 также	 могут	 иметь	 собственные	 разделы,	 такие	 как	 .data,	 .bss 
и .text.	А что	означает	точка	перед	pi	и перед	переменными	fmt?	Точка	обозна
чает	локальную	переменную,	т. е.	переменную,	известную	только	внутри	функ
ции,	в которой	эта	переменная	объявлена.	В функции	area	используется	зна
чение	pi,	отличающееся	от	значения	pi	в функции	circum.	Переменная	radius,	
объявленная	в разделе	section .data	основной	функции	main,	известна	в каждой	
функции	в этом	листинге	исходного	кода,	включая	саму	функцию	main.	Весьма	
разумно	использовать	локальные	переменные	во	всех	случаях,	когда	это	воз
можно, –	это	снижает	риск	возникновения	конфликта	имен	переменных.

На	рис. 12.2	показан	вывод	этой	программы.

Рис. 12.2.	Вывод программы function2.asm

резюме
В	этой	главе	вы	узнали:

	� как	использовать	функции;
	� что	функции	могут	 иметь	 собственные	 разделы	 section .data	 и  section 
.bss;

	� что	функции	не могут	быть	вложенными;
	� что	функции	могут	вызывать	другие	функции;
	� что	main –	это	всего	лишь	одна	из	функций;
	� как	использовать	локальные	переменные.

Глава 13
Выравнивание стека  

и фрейм стека

Когда	основная	программа	вызывает	функцию,	8байтовый	адрес	возврата	за
писывается	в стек.	Этот	8байтовый	адрес	является	адресом	инструкции,	которая	
должна	быть	выполнена	после	выхода	из	функции.	Поэтому	после	завершения	
выполнения	функции	программа	находит	адрес	возврата	в стеке	и продолжа
ет	работу	после	вызова	функции.	Внутри	функции	также	может	использоваться	
стек	для	различных	целей.	Каждый	раз,	когда	в стек	записывается	какоелибо	
значение,	указатель	стека	уменьшается	на	8	байт,	а при	каждом	извлечении	зна
чения	из	стека	указатель	стека	увеличивается	на	8	байт.	Поэтому	необходимо	
обеспечить	«восстановление»	в стеке	корректного	значения	перед	выходом	из	
функции.	Если	этого	не сделать,	то	вызывающая	программа	получит	неправиль
ный	адрес	инструкции,	которая	должна	выполняться	после	вызова	функции.

выравнивание Стека
В	руководствах	Intel	вы	обнаружите	упоминание	о следующем	обязательном	
требовании:	стек	непременно	должен	выравниваться	по	16байтовой	грани
це	при	каждом	вызове	функции.	Это	требование	может	показаться	немного	
странным,	 ведь	 стек	 создается	 в  памяти	 с  8байтовыми	 (64битовыми)	 гра
ницами.	Причина	данного	требования	 состоит	в том,	что	 существуют	SIMD
инструкции,	которые	выполняют	параллельные	операции	с более	крупными	
блоками	данных,	 и  для	 этих	 SIMDинструкций	может	 потребоваться	 разме
щение	данных	по	адресам,	кратным	16	байтам.	В предыдущих	примерах	при	
использовании	функции	printf	с xmmрегистрами	выполнялось	выравнивание	
стека	по	16байтовой	границе,	но	это	действие	не объяснялось.	Вернемся	в гла
ву	11,	где	рассматривалась	арифметика	с плавающей	точкой.	В этой	главе	был	
приведен	пример	с аварийным	завершением	программы	после	комментиро
вания	инструкций	push rbp	и pop rbp.	Эта	программа	завершалась	аварийно,	по
тому	что	удаление	вышеупомянутых	инструкций	приводило	к тому,	что	стек	
не был	выровнен	по	16байтовой	границе.	Если	вы	используете	функцию	printf 
без	xmmрегистров,	то	можете	не обращать	внимания	на	требование	выравни
вания	стека,	но	однажды	вдруг	обнаружите,	что	в программе	возникают	непо
нятные	ошибки.



SIMDинструкции	и операции	выравнивания	стека	будут	рассматривать
ся	 в  следующих	 главах,	поэтому	не беспокойтесь,	 если	приведенное	выше	
описание	оказалось	не совсем	понятным.	Сейчас	просто	следует	запомнить,	
что	при	вызове	функции	необходимо	выровнять	стек	по	адресам,	кратным	
16	байтам.

С	точки	зрения	процессора	main –	одна	из	обычных	функций.	Прежде	чем	
программа	начнет	выполняться,	происходит	выравнивание	стека.	Непосред
ственно	 перед	 началом	 выполнения	 main	 8байтовый	 адрес	 возврата	 запи
сывается	в стек,	т. е.	стек	не выровнен	перед	началом	выполнения	main.	Если	
не происходит	обращений	к стеку	в интервале	между	началом	main	и вызовом	
какойлибо	функции,	то	указатель	стека	rsp	не выровнен	по	16байтовой	гра
нице.	 Это	можно	 проверить,	 посмотрев	 содержимое	 регистра	 rsp:	 если	 зна
чение	в rsp	заканчивается	нулем,	то	стек	выровнен	по	16байтовой	границе.	
Чтобы	обеспечить	это	нулевое	завершение,	нужно	записать	чтонибудь	в стек,	
чтобы	выровнять	его	по	16байтовой	границе.	Разумеется,	не следует	забывать	
в дальнейшем	выполнить	соответствующую	инструкцию	pop.

Требование	выравнивания	стека	является	одной	из	причин	использования	
пролога	и эпилога	функции.	Первая	инструкция	в main	и в	любой	другой	функ
ции	должна	записывать	чтолибо	в стек	для	его	выравнивания.	Это	обоснова
ние	применения	в прологе	инструкции	push rbp.	Регистр	rbp	также	называется	
указателем	базы	(base	pointer).

Почему	используется	именно	rbp?	В прологе,	когда	используются	фреймы	
стека	(их	описание	будет	приведено	немного	позже),	регистр	rbp	изменяется,	
поэтому	перед	использованием	rbp	в фрейме	стека	его	значение	записывается	
в стек	для	сохранения	и восстановления	при	возврате.	Даже	если	фрейм	стека	
не создается,	регистр	rbp	является	идеальным	кандидатом	для	выравнивания	
стека,	поскольку	не используется	для	передачи	аргументов	в функцию.	Пере
дача	аргументов	будет	рассматриваться	немного	позже	в этой	главе.	В прологе	
также	используется	инструкция	mov rbp, rsp.	Эта	инструкция	сохраняет	регистр	
rsp –	указатель	стека,	содержащий	необходимый	адрес	возврата.	Инструкции	
пролога	выполняются	с обратным	смыслом	и в	обратном	порядке	в эпилоге,	
разумеется,	в этом	случае	лучше	всего	не трогать	rbp.	В следующих	главах	мы	
рассмотрим	несколько	других	методов	выравнивания	стека.

В	листинге	13.1	показан	исходный	код	для	некоторых	экспериментов	со	сте
ком.	Наблюдайте	за	регистром	rsp	при	отладке	и выполните	пошаговый	проход	
по	программе	с помощью	SASM.	Уберите	символы	комментариев	из	инструк
ций	push rbp	и pop rbp	и посмотрите,	что	происходит.	Если	программа	подходит	
к  вызову	функции	 printf	 без	 выравнивания	 стека,	то	происходит	 аварийное	
завершение	(программа	«падает»).	Причина	в том,	что	функция	printf	непре
менно	требует	выравнивания	стека.

В	этой	программе	не используется	полноценный	пролог	и эпилог,	т. е.	не соз
даются	фреймы	стека.	Применяются	только	инструкции	push	и pop	для	демон
страции	выравнивания	стека.

Листинг 13.1.	Программа aligned.asm

; aligned.asm
extern printf
section .data
      fmt  db     "2 times pi equals %.14f",10,0
      pi   dq     3.14159265358979
section .bss
section .text
;-----------------------------------------------
func3:
      push  rbp
            movsd       xmm0, [pi]
            addsd       xmm0, [pi]
            mov   rdi,fmt
            mov   rax,1
            call  printf     ; Вывод числа с плав.точкой.
      pop   rbp
      ret
;-----------------------------------------------
func2:
      push  rbp
            call  func3 ; Вызов третьей функции.
      pop   rbp
      ret
;-----------------------------------------------
func1:
      push  rbp
            call  func2 ; Вызов второй функции.
      pop   rbp
      ret
;-----------------------------------------------
      global main
main:
      push  rbp
            call  func1 ; Вызов первой функции.
      pop   rbp
ret

Обратите	внимание:	если	выполняется	определенное	количество	вызовов	
(четное	или	нечетное	в зависимости	от	того,	как	вы	начинаете	программу),	то	
стек	будет	выровнен	по	16байтовой	границе,	даже	без	применения	инструк
ций	push	и pop,	и в	программе	не возникает	критический	сбой.

бОлее пОдрОбнО О фреймах Стека
Можно	различать	два	типа	функций:	функцииветви	и функциилистья.	Функ
цииветви	 содержат	 вызовы	других	функций,	 а  функциилистья	 выполняют	
некоторые	команды,	затем	возвращаются	в родительскую	функцию	без	вызова	
какойлибо	другой	функции.

Теоретически	 при	 каждом	 вызове	 функции	 необходимо	 создавать	 фрейм 
стека	(stack	frame).	Это	делается	следующим	образом:	в вызываемой	функции	
в первую	очередь	выравнивается	стек	по	16байтовой	границе,	т. е.	выполня

ется	инструкция	push rbp.	Затем	содержимое	указателя	стека	rsp	 сохраняется	
в регистре	rbp.	При	выходе	из	функции	восстанавливается	содержимое	указа
теля	стека	rsp	(из	регистра	rbp)	и выполняется	инструкция	pop rbp	для	восста
новления	rbp.	В этом	и заключается	роль	пролога	и эпилога	функции.	Теперь	
внутри	функции	регистр	rbp	 служит	точкой	привязки	для	исходной	локации	
стека.	Каждый	раз,	когда	функция	вызывает	другую	функцию,	новая	(вызыва
емая)	функция	должна	создавать	собственный	фрейм	стека.

В	функцияхлистьях,	вообще	говоря,	можно	не уделять	внимания	выравни
ванию	стека	и созданию	фрейма	стека –	в этом	нет	необходимости,	если	вы	
не работаете	со	стеком.	Следует	отметить,	что	при	вызове,	например,	printf 
вызывающая	функция	не является	функциейлистом.	Кроме	того,	если	функ
ция	не использует	SIMDинструкции,	то	нет	необходимости	заботиться	о вы
равнивании	стека.

Компиляторы	поддерживают	функции	оптимизации,	и иногда	при	просмот
ре	сгенерированного	ими	кода	можно	заметить,	что	фрейм	стека	не использу
ется.	Это	происходит,	когда	компилятор	в процессе	оптимизации	обнаружива
ет,	что	фрейм	стека	не нужен.

В	любом	случае	лучше	выработать	правильную	привычку	всегда	включать	
в функцию	создание	фрейма	стека	и выполнять	выравнивание	стека –	это	по
может	избежать	множества	проблем	в дальнейшем.	Весомой	причиной	вклю
чения	фрейма	стека	является	тот	факт,	что	GDB	и основанные	на	GDB	отладчи
ки	(такие	как	DDD	и SASM)	предполагают	наличие	фрейма	стека.	Если	в коде	
не выполнено	создание	фрейма	стека,	то	отладчик	поведет	себя	непредсказу
емо,	например	будет	игнорировать	точки	останова	или	пропускать	инструк
ции.	В одном	из	примеров	исходного	кода	из	предыдущей	главы	(например,	
alife.asm)	 закомментируйте	 пролог	 и  эпилог	 функции,	 затем	 запустите	 GDB	
и понаблю	дайте	за	происходящим.

В	качестве	дополнительного	упражнения	возьмите	исходный	код	из	преды
дущей	главы	(например,	function2.asm)	и в	отладчике	SASM	или	GDB	наблюдай
те	за	тем,	как	стек	остается	выровненным	во	время	выполнения.

Дополнительный	 упрощенный	 прием:	 пролог	 функции	 можно	 заменить	
на	инструкцию	enter 0, 0,	а эпилог	функции	на	инструкцию	leave.	Но для	en-
ter	характерна	низкая	производительность,	поэтому	лучше	продолжать	при
менять	инструкции	push rbp	и mov rbp, rsp,	если	производительность	важна.	Для	
инструкции	leave	проблемы	производительности	не существует.

резюме
В	этой	главе	вы	узнали:

	� что	такое	выравнивание	стека;
	� как	использовать	фреймы	стека;
	� как	использовать	SASM	для	проверки	указателя	стека;
	� об	инструкциях	входа	и выхода	из	функций.

Глава 14
Внешние функции

Нам	 уже	известно,	 как	 создавать	 и  использовать	функции	 в  исходном	 коде.	
Но функции	не обязательно	должны	размещаться	в том	же	файле,	в котором	
находится	 основная	 программа.	 Можно	 написать	 и  ассемблировать	 функ
ции	в отдельном	файле	и связывать	их	во	время	сборки	программы.	Функция	
printf,	которую	мы	уже	использовали	несколько	раз,	представляет	собой	при
мер	внешней	функции.	В файле	исходного	кода,	где	планируется	применение	
внешней	функции,	необходимо	объявить	ее	с помощью	ключевого	слова	extern,	
чтобы	сообщить	ассемблеру	о том,	что	исходный	код	этой	функции	не следует	
искать	в текущем	файле.	Ассемблер	предполагает,	что	эта	функция	уже	ассемб
лирована	в некотором	объектном	файле.	Внешняя	функция	будет	 вставлена	
линкером,	если	он	сможет	найти	ее	в заданном	объектном	файле.

Можно	использовать	внешние	функции	из	языка	C,	но	вы	также	можете	соз
дать	собственный	набор	внешних	функций	и при	необходимости	связывать	их	
с основной	программой.

СОздание и Связывание функций
В	листинге	14.1	показан	пример	программы,	состоящей	из	трех	файлов	исход
ного	кода	с именами:	function4.asm,	circle.asm	и rect.asm.	Для	этой	программы	
написан	новый	makefile.	Внимательно	изучите	содержимое	этих	файлов.

Листинг 14.1.	Программа function4.asm

; function4.asm
extern printf
extern c_area
extern c_circum
extern r_area
extern r_circum
global pi
section .data
      pi         dq    3.141592654
      radius     dq    10.0
      side1      dq    4
      side2      dq    5
      fmtf  db   "%s %f",10,0
      fmti  db   "%s %d",10,0
      ca    db   "The circle area is ",0



      cc    db   "The circle circumference is ",0
      ra    db   "The rectangle area is ",0
      rc    db   "The rectangle circumference is ",0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Площадь круга.
      movsd  xmm0, qword [radius]    ; Аргумент radius в регистре xmm0.
      call   c_area                  ; Площадь возвращается в регистре xmm0.
      ; Вывод площади круга.
            mov   rdi, fmtf
            mov   rsi, ca
            mov   rax, 1
            call  printf
; Длина окружности.
      movsd      xmm0, qword [radius] ; Аргумент radius в регистре xmm0.
      call  c_circum                  ; Длина окружности в регистре xmm0.
      ; Вывод длины окружности.
            mov   rdi, fmtf
            mov   rsi, cc
            mov   rax, 1
            call  printf
; Площадь прямоугольника.
      mov   rdi, [side1]
      mov   rsi, [side2]
      call  r_area               ; Площадь возвращается в регистре rax.
      ; Вывод площади прямоугольника.
            mov   rdi, fmti
            mov   rsi, ra
            mov   rdx, rax
            mov   rax, 0
            call  printf
; Периметр прямоугольника.
      mov   rdi,  [side1]
      mov   rsi,  [side2]
      call  r_circum             ; Периметр в регистре rax.
      ; Вывод периметра прямоугольника.
            mov   rdi, fmti
            mov   rsi, rc
            mov   rdx, rax
            mov   rax, 0
            call  printf
mov rsp,rbp
pop rbp
ret

В	исходном	коде	в листинге	14.1	несколько	функций	объявлено	как	extern 
(внешние),	как	это	ранее	делалось	неоднократно	при	использовании	функции	
printf.	В этом	нет	ничего	нового.	Но, кроме	того,	переменная	pi	объявлена	как	
global.	Таким	образом,	эта	переменная	будет	доступной	во	внешних	функциях.

В	 листингах	 14.2	 и  14.3	 показаны	 отдельные	 файлы,	 содержащие	 только	
внешние	функции.

Листинг 14.2.	Файл circle.asm

; circle.asm
extern pi
section .data
section .bss
section .text
;------------------------------------------------
global c_area
c_area:
      section .text
      push rbp
      mov rbp,rsp
            movsd xmm1, qword [pi]
            mulsd xmm0, xmm0       ; Радиус в регистре xmm0.
            mulsd xmm0, xmm1
      mov rsp,rbp
      pop rbp
      ret
;------------------------------------------------
global c_circum
c_circum:
      section .text
      push rbp
      mov rbp,rsp
            movsd xmm1, qword [pi]
            addsd xmm0, xmm0        ; Радиус в регистре xmm0.
            mulsd xmm0, xmm1
      mov rsp,rbp
      pop rbp
      ret

Листинг 14.3.	Файл rect.asm

; rect.asm
section .data
section .bss
section .text
;------------------------------------------------
global r_area
r_area:
      section .text
      push rbp
      mov rbp,rsp
            mov   rax, rsi
            imul  rax, rdi
            mov   rsp,rbp
      pop rbp
      ret
;------------------------------------------------
global r_circum
r_circum:
      section .text
      push rbp
      mov rbp,rsp
            mov   rax, rsi
            add   rax, rdi
            add   rax, rax

      mov rsp,rbp
      pop rbp
      ret

В	файле	circle.asm необходимо	использовать	переменную	pi,	 объявленную	
в  основном	файле	 исходного	 кода	 как	 global,	 что	 необходимо	 отметить	 как	
не самое	удачное	решение,	но	здесь	оно	применено	в демонстрационных	це
лях.	Глобальные	переменные,	подобные	pi,	трудно	отслеживать,	и они	могут	
даже	 приводить	 к  конфликту	 переменных	 с  одинаковыми	 именами.	 Самое	
лучшее	практическое	решение –	использование	регистров	для	передачи	зна
чений	в функцию.	Здесь	же	приходится	определять	pi	как	внешнюю	перемен
ную.	Каждый	файл	circle.asm	и rect.asm	содержит	две	функции:	для	вычисления	
периметра	и площади.	Необходимо	объявить,	что	эти	функции	являются	гло
бальными	 (global),	 как	и основная	функция	 main.	При	ассемблировании	этих	
функций	добавляются	необходимые	«служебные	данные»,	позволяющие	лин
керу	включать	эти	функции	в другой	объектный	код.

раСширенная верСия makefIle
Для	выполнения	всей	работы	необходима	расширенная	версия	makefile,	пока
занная	в листинге	14.4.

Листинг 14.4.	makefile

# makefile для function4, circle и rect.
function4: function4.o circle.o rect.o
      gcc -g -o function4 function4.o circle.o rect.o -no-pie
function4.o: function4.asm
      nasm -f elf64 -g -F dwarf function4.asm -l function4.lst
circle.o: circle.asm
      nasm -f elf64 -g -F dwarf circle.asm -l circle.lst
rect.o: rect.asm
      nasm -f elf64 -g -F dwarf rect.asm -l rect.lst

Читать	 makefile	 нужно	 снизу	 вверх:	 сначала	 различные	 файлы	 исходного	
кода	на	ассемблере	ассемблируются	в объектные	файлы,	затем	эти	объектные	
файлы	связываются	и объединяются	в выполняемый	файл	function4.	Здесь	про
является	истинная	мощь	утилиты	make.	При	любом	изменении	одного	из	фай
лов	исходного	кода	make	благодаря	древовидной	структуре	точно	знает,	какие	
файлы	нужно	повторно	ассемблировать	и связать.	Разумеется,	если	функции	
стабильны	и в	дальнейшем	не будут	изменяться,	то	нет	необходимости	их	по
вторного	ассемблирования	при	каждом	сеансе	обработки	makefile.	Просто	со
храните	объектные	файлы	в специально	созданном	каталоге	и в	дальнейшем	
обращайтесь	к ним	по	полному	путевому	имени	в строке	gcc	в makefile.	Объ
ектный	 файл  –	 это	 результат	 ассемблирования	 или	 компиляции	 исходного	
кода.	Объектный	файл	содержит	машинный	код	и информацию	для	линкера	
о глобальных	переменных	и внешних	функциях,	необходимую	для	создания	
корректного	выполняемого	файла.	В рассматриваемом	здесь	примере	все	объ
ектные	файлы	размещены	в том	же	каталоге,	что	и основной	файл	исходного	
кода,	поэтому	полные	путевые	имена	не используются.

Что	можно	сказать	о функции	printf?	Почему	в makefile	нет	ссылки	на	printf?	
Потому	что	компилятор	gcc	обладает	достаточным	интеллектом,	для	того	что
бы	 поискать	 еще	 и  в	 библиотеках	 языка	 C  функции,	 обращения	 к  которым	
встречаются	в исходном	коде.	Это	означает,	что	вы	не должны	использовать	
имена	функций	языка	C для	именования	своих	функций.	Это	может	запутать	
кого	угодно,	не говоря	уже	о линкере.

В	приведенном	выше	коде	регистры	использовались	для	передачи	значений	
из	основной	программы	в функции	и в	обратном	направлении –	это	самый	
правильный	практический	подход.	Например,	перед	вызовом	r_area	значение	
переменной	side1	записывается	в регистр	rdi,	а значение	side2	в регистр	rsi.	
Затем	вычисленная	площадь	прямоугольника	возвращается	в регистре	rax.	Для	
возврата	результата	можно	было	бы	воспользоваться	глобальной	переменной,	
подобной	 pi,	 объявленной	 в  разделе	 section .data	 функции	 main.	 Но,	 как	 уже	
было	отмечено	выше,	этого	следует	избегать.	Более	подробно	этот	вопрос	бу
дет	обсуждаться	в следующей	главе	о соглашениях	при	вызовах	функций.

На	рис. 14.1	показан	вывод	этой	программы.

Рис. 14.1.	Вывод программы function4

При	использовании	этого	примера	в среде	SASM	необходимо	в первую	оче
редь	ассемблировать	внешние	функции	для	получения	объектных	файлов.	За
тем	в диалоговом	окне SASM	Settings	(Параметры	настройки)	на	вкладке	Build 
(Сборка)	нужно	добавить	место	расположения	этих	объектных	файлов	в строке	
Linking	Options	 (Параметры	связывания).	Для	рассматриваемого	здесь	при
мера	строка	Linking	Options	должна	выглядеть	следующим	образом	(при	вводе	
будьте	внимательны –	не вставляйте	лишние	пробелы	в этой	строке):

$PROGRAM.OBJ$ -g -o $PROGRAM$ circle.o rect.o -no-pie

резюме
В	этой	главе	вы	узнали,	как:

	� используются	внешние	функции;
	� применяются	глобальные	переменные;
	� используется	makefile	для	внешних	функций;
	� передаются	значения	в функции	и возвращаются	из	функций.

Глава 15
Соглашения о вызовах 

функций

Соглашения	 о  вызовах	 функций	 описывают,	 как	 переменные	 передаются	
в функции	и возвращаются	из	функций.	Если	вы	будете	использовать	только	
функции,	которые	написали	сами,	то	нет	особого	смысла	беспокоиться	о со
глашениях	о вызовах.	Но если	вы	пользуетесь	функциями	из	библиотеки	язы
ка	C,	то	необходимо	знать,	в какие	регистры	требуется	записывать	значения,	
с  которыми	должны	работать	такие	 внешние	функции.	Кроме	того,	 если	 вы	
пишете	 ассемблерные	функции	для	 создания	 библиотеки,	 предназначенной	
для	использования	другими	разработчиками,	то	настоятельно	рекомендуется	
соблюдать	некоторые	соглашения	о регистрах,	в которых	передаются	конкрет
ные	аргументы	функций.	Без	 соблюдения	этих	 соглашений	будут	постоянно	
возникать	проблемы	и конфликты	с передачей	аргументов.

Вам	уже	известно,	что	при	вызове	функции	printf	первый	аргумент	записыва
ется	в регистр	rdi,	второй	аргумент	в регистр	rsi,	а следующий	аргумент	передает
ся	в регистре	xmm0.	Вызывая	функцию	printf,	мы	соблюдали	соглашения	о вызове.

Чтобы	избежать	конфликтов	и их	последствий	в виде	критических	сбоев	про
граммы,	опытные	программисты	разработали	соглашения	о вызовах	функций	
(calling	conventions),	стандартизированный	способ	вызова	функций.	Это	превос
ходный	замысел,	но,	как	можно	было	ожидать,	не все	согласны	со	сторонниками	
данного	подхода,	поэтому	существует	несколько	различных	соглашений	о вызове	
функций.	До настоящего	момента	в этой	книге	использовались	соглашения	о вы
зове	функций	System	V AMD64	ABI,	являющиеся	стандартом	для	платформ	Linux.	
Но существуют	также	и другие	соглашения	о вызове	функций,	заслуживающие	
внимания:	Microsoft	x64,	используемые	для	программирования	в ОС	Windows.

Эти	соглашения	о вызовах	функций	позволяют	использовать	внешние	функ
ции,	написанные	на	ассемблере,	а также	функции,	скомпилированные	из	ис
ходного	кода	на	других	языках,	таких	как	C,	без	необходимости	доступа	к их	
исходному	коду.	Нужно	просто	записать	правильные	аргументы	в регистры,	
определенные	в соглашении	о вызове	функций.

Более	подробную	информацию	о соглашениях	о вызовах	функций	System V	
AMD64	 ABI	 можно	 найти	 здесь:	 https://software.intel.com/sites/default/files/
article/402129/mpx-linux64-abi.pdf.	 Этот	 документ	 компании	 Intel	 содержит	
огромный	 объем	 подробнейшей	 информации	 о  прикладном	 двоичном	 (би



нарном)	интерфейсе	 (ABI)	System	V.	В данной	главе	содержится	все,	что	вам	
нужно	знать	для	того,	чтобы	на	уровне начинающего	вызывать	функции	стан
дартным	способом.

аргументы функций
Вернемся	немного	назад –	к предыдущим	файлам	исходного	кода:	при	вычис
лениях	характеристик	круга	использовался	регистр	xmm0	для	передачи	значений	
с плавающей	точкой	из	программы	main	в функцию	circle,	и тот	же	регистр	xmm0 
использовался	для	возвращения	результата	с плавающей	точкой,	полученного	
в этой	функции,	в основную	программу.	При	вычислениях	характеристик	пря
моугольника	для	передачи	целочисленных	значений	в функцию	использовались	
регистры	rdi	и rsi,	а целочисленный	результат	возвращался	в регистре	rax.	Этот	
способ	передачи	аргументов	в функцию	и возвращения	результата	из	функции	
определяется	конкретным	соглашением	о вызовах	функций.

Аргументы,	не являющиеся	значениями	с плавающей	точкой,	такие	как	це
лые	числа	и адреса,	передаются	в следующем	порядке:

	� 1й	аргумент	передается	в регистре	rdi;
	� 2й	аргумент	передается	в регистре	rsi;
	� 3й	аргумент	передается	в регистре	rdx;
	� 4й	аргумент	передается	в регистре	rcx;
	� 5й	аргумент	передается	в регистре	r8;
	� 6й	аргумент	передается	в регистре	r9.

Дополнительные	аргументы	передаются	через	стек –	обязательно	в обрат
ном	порядке,	чтобы	можно	было	извлечь	их	в правильном	порядке.	Например,	
при	передаче	10	аргументов	порядок	следующий:

	� первым	в стек	записывается	10й	аргумент;
	� затем	в стек	записывается	9й	аргумент;
	� потом	в стек	записывается	8й	аргумент;
	� последним	в стек	записывается	7й	аргумент.

После	перехода	в функцию	необходимо	обеспечить	извлечение	значений	из	
регистров.	При	извлечении	значений	из	стека	будьте	чрезвычайно	вниматель
ны:	помните	о том,	что	адрес	возврата	также	записывается	в стек	сразу	после	
аргументов.

	� При	записи	в стек	10го	аргумента	указатель	стека	rsp	уменьшается	на	
8 байт.

	� При	записи	в стек	9го	аргумента	rsp	уменьшается	на	8	байт.
	� При	записи	в стек	8го	аргумента	rsp	уменьшается	на	8	байт.
	� При	записи	в стек	7го	аргумента	rsp	уменьшается	на	8	байт.
	� Затем	 при	 вызове	 функции	 содержимое	 регистра	 rip	 записывается	
в стек,	и указатель	стека	rsp	уменьшается	на	8	байт.

	� Далее	в начале	функции	регистр	rbp	записывается	в стек –	это	часть	про
лога,	а указатель	стека	rsp	уменьшается	на	8	байт.

	� После	этого	стек	выравнивается	по	16байтовой	границе,	поэтому,	воз
можно,	потребуется	еще	одна	операция	записи	в стек	push,	чтобы	умень
шить	rsp	на	8	байт.

Таким	образом,	после	записи	в стек	аргументов	функции,	как	минимум,	еще	
два	дополнительных	регистра	 записываются	в  стек,	т.  е.	 16	дополнительных	
байт.	Поэтому	после	входа	в функцию	для	доступа	к аргументам	необходимо	
пропустить	первые	16	байт	в стеке	или,	возможно,	больше,	если	выполнялась	
операция	выравнивания	стека.

Аргументы	с плавающей	точкой	передаются	в xmmрегистрах	в следующем	
порядке:

	� 1й	аргумент	записывается	в регистр	xmm0;
	� 2й	аргумент	записывается	в регистр	xmm1;
	� 3й	аргумент	записывается	в регистр	xmm2;
	� 4й	аргумент	записывается	в регистр	xmm3;
	� 5й	аргумент	записывается	в регистр	xmm4;
	� 6й	аргумент	записывается	в регистр	xmm5;
	� 7й	аргумент	записывается	в регистр	xmm6;
	� 8й	аргумент	записывается	в регистр	xmm7.

Дополнительные	 аргументы	 передаются	 через	 стек,	 но	 запись	 в  стек	 вы
полняется	 не  с  помощью	 инструкции	 push,	 как	 можно	 было	 предположить.	
Эта операция	будет	показана	позже,	в главах	о SIMD.

Функция	возвращает	результат	с плавающей	точкой	в регистре	xmm0,	а целое	
число	или	адрес	возвращается	в регистре	rax.

Выглядит	слишком	сложно?	В листинге	15.1	показан	пример,	который	выво
дит	несколько	аргументов	с помощью	функции	printf.

Листинг 15.1.	Программа function5.asm

; function5.asm
extern printf
section .data
      first      db    "A",0
      second     db    "B",0
      third      db    "C",0
      fourth     db    "D",0
      fifth      db    "E",0
      sixth      db    "F",0
      seventh    db    "G",0
      eighth     db    "H",0
      ninth      db    "I",0
      tenth      db    "J",0
      fmt1  db   "The string is: %s%s%s%s%s%s%s%s%s%s",10,0
      fmt2       db    "PI = %f",10,0
      pi         dq    3.14
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      mov   rdi,fmt1   ; Сначала используются регистры.
      mov   rsi, first
      mov   rdx, second
      mov   rcx, third
      mov   r8, fourth

      mov   r9, fifth
      push  tenth      ; Теперь начинается запись в стек
      push  ninth      ; в обратном порядке.
      push  eighth
      push  seventh
      push  sixth
      mov   rax, 0
      call  printf
      and   rsp, 0xfffffffffffffff0 ; Выравнивание стека по 16-байтовой границе.
      movsd       xmm0,[pi]  ; Вывод числа с плавающей точкой.
      mov   rax, 1           ; Выводится 1 число с плав.точкой.
      mov   rdi, fmt2
      call  printf
leave
ret

В	этом	примере	все	аргументы	передаются	в правильном	порядке	в функ
цию	printf.	Обратите	внимание:	в стек	аргументы	записываются	в обратном	
порядке.

Используйте	отладчик	для	проверки	содержимого	регистра	rsp	перед	вызо
вом	функции	printf.	Стек	не выровнен	по	16байтовой	границе.	В программе	
не возникает	критический	сбой,	потому	что	от	printf	не требуется	вывод	числа	
с плавающей	точкой.	Но при	следующем	вызове	printf	выполняется	именно	
такой	 вывод.	 Поэтому	 перед	 использованием	 printf	 необходимо	 выровнять	
стек,	следовательно,	используется	инструкция

and      rsp, 0xfffffffffffffff0

Эта	инструкция	сохраняет	все	байты	в регистре	rsp	без	изменений,	за	исклю
чением	самого	последнего	байта:	последние	(младшие)	4	бита	в rsp	изменяют
ся	на	0,	т. е.	числовое	значение	в rsp	уменьшается,	и указатель	стека	rsp	вырав
нивается	по	16байтовой	границе.	Если	стек	уже	был	выровнен	изначально,	то	
приведенная	выше	инструкция	and	ничего	не делает.	Тем	не менее	будьте	вни
мательны	и осторожны.	Если	необходимо	извлекать	значения	из	стека	после	
выполнения	этой	инструкции	and,	то	возникает	проблема:	необходимо	узнать,	
изменила	ли	инструкция	and	значение	rsp,	и в	итоге	выровнять	указатель	стека	
снова	до	этого	значения	перед	выполнением	инструкции	and.

На	рис. 15.1	показан	вывод	данной	функции.

Рис. 15.1.	Вывод функции function5

Схема Стека
Рассмотрим	пример,	в котором	можно	наблюдать,	что	происходит	в стеке	при	
записи	в него	аргументов	функции.	В листинге	15.2	показана	программа,	ис
пользующая	 функцию	 для	 формирования	 строки,	 а  после	 возврата	 из	 этой	
функции	созданная	строка	выводится.

Листинг 15.2.	Программа function6.asm

; function6.asm
extern printf
section .data
      first      db    "A"
      second     db    "B"
      third      db    "C"
      fourth     db    "D"
      fifth      db    "E"
      sixth      db    "F"
      seventh    db    "G"
      eighth     db    "H"
      ninth      db    "I"
      tenth      db    "J"
      fmt        db    "The string is: %s",10,0
section .bss
      flist      resb 11    ; Длина строки + завершающий 0.
section .text
      global main
main:
push  rbp
mov   rbp, rsp
      mov   rdi, flist      ; Длина.
      mov   rsi, first      ; Заполнение регистров.
      mov   rdx, second
      mov   rcx, third
      mov   r8, fourth
      mov   r9, fifth
      push  tenth      ; Теперь начинается запись в стек
      push  ninth      ; в обратном порядке.
      push  eighth
      push  seventh
      push  sixth
      call  lfunc           ; Вызов функции.
      ; Вывод результата.
           mov   rdi, fmt
           mov   rsi, flist
           mov   rax, 0
           call  printf
leave
ret
;-----------------------------------------------
lfunc:
push  rbp
mov   rbp,rsp
      xor   rax,rax    ; Очистка rax (особенно старшие биты).
      mov   al,byte[rsi]     ; Запись значения 1-го аргумента в al.
      mov   [rdi], al        ; Сохранение al в памяти.

      mov   al, byte[rdx]    ; Запись значения 2-го аргумента в al.
      mov   [rdi+1], al      ; Сохранение al в памяти.
      mov   al, byte[rcx]    ; И т. д. для всех прочих аргументов.
      mov   [rdi+2], al
      mov   al, byte[r8]
      mov   [rdi+3], al
      mov   al, byte[r9]
      mov   [rdi+4], al
; Теперь извлечение аргументов из стека.
      push  rbx              ; Сохраняемый вызываемой функцией.
      xor   rbx,rbx
      mov   rax, qword [rbp+16]   ; Первое значение: начальный указатель стека
                                  ; + rip + rbp
      mov   bl, byte[rax]         ; Извлечение символа.
      mov   [rdi+5], bl     ; Сохранение символа в памяти.
      mov   rax, qword [rbp+24]   ; Продолжение обработки следующего значения.
      mov   bl, byte[rax]
      mov   [rdi+6], bl
      mov   rax, qword [rbp+32]
      mov   bl, byte[rax]
      mov   [rdi+7], bl
      mov   rax, qword [rbp+40]
      mov   bl, byte[rax]
      mov   [rdi+8], bl
      mov   rax, qword [rbp+48]
      mov   bl, byte[rax]
      mov   [rdi+9], bl
      mov   bl,0
      mov   [rdi+10], bl
pop   rbx                   ; Сохраняемый вызываемой функцией.
mov   rsp,rbp
pop   rbp
ret

В	 рассматриваемом	 здесь	 примере	 вместо	 вывода	 с  помощью	 функции	
printf	сразу	после	передачи	всех	аргументов,	как	это	сделано	в предыдущем	
разделе,	 вызывается	функция	 lfunc.	 Эта	 функция	 принимает	 все	 аргументы	
и формирует	строку	в памяти	(flist).	Эта	строка	будет	выведена	после	возвра
щения	в основную	функцию	main.

Рассмотрим	 подробнее	 функцию	 lfunc.	 Она	 принимает	 только	 младший	
байт	регистров	аргументов,	в которых	записаны	символы,	используя	инструк
цию,	показанную	ниже:

mov    al, byte[rsi]

Эти	символы	поочередно	сохраняются	в памяти,	начиная	с адреса,	храняще
гося	в регистре	rdi,	т. е.	с адреса	flist,	с помощью	инструкции	mov [rdi], al.	Ис
пользование	ключевого	слова	byte	не обязательно,	но	оно	повышает	удобство	
чтения	исходного	кода.

Для	нас	наиболее	интересно	начало	извлечения	значений	из	стека.	В нача
ле	функции	 lfunc	 значение	 rsp,	 т.  е.	 адрес	 стека,	 сохраняется	 в  регистре	 rbp.	
Но между	этой	инструкцией	и конечной	инструкцией	записи	значений	в стек	
в main	 содержимое	регистра	rsp	 было	изменено	дважды.	Вопервых,	при	вы

зове	lfunc	адрес	возврата	был	помещен	в стек.	Затем	в стеке	был	сохранен	ре
гистр	rbp	как	часть	пролога.	В сумме	указатель	стека	rsp	уменьшился	на	16	байт.	
Для доступа	к значениям	аргументов	в стеке	необходимо	скорректировать	их	
адреса	на	16	байт.	Именно	поэтому	для	доступа,	например,	к переменной	sixth 
используется	инструкция

mov rax, qword [rbp+16]

Каждая	следующая	переменная	располагается	на	8	байт	выше	предыдущей.	
Регистр	rbx	используется	как	временное	хранилище	при	формировании	стро
ки	в памяти	flist.	Перед	применением	rbx	его	содержимое	сохраняется	в сте
ке.	Никогда	не известно,	применяется	ли	регистр	rbx	 в основной	программе	
main для	других	целей,	поэтому	его	содержимое	сохраняется	и восстанавлива
ется	перед	выходом	из	функции.

На	рис. 15.2	показан	вывод	этой	программы.

Рис. 15.2.	Вывод программы function6

СОхранение региСтрОв
Теперь	необходимо	объяснить	смысл	инструкций

push    rbx          ; Сохраняемый вызываемой функцией.

и

pop     rbx          ; Сохраняемый вызываемой функцией.

Должна	быть	очевидной	необходимость	наблюдения	за	тем,	что	происходит	
с регистрами	во	время	вызова	функции.	Некоторые	регистры	будут	изменяться	
во	время	выполнения	функции,	другие	останутся	неизменными.	Необходимо	
предпринять	меры	предосторожности,	чтобы	избежать	непредсказуемых	ре
зультатов	изза	изменений	регистров	функциями,	используемыми	в основной	
(вызывающей)	программе	main.

В	табл. 15.1	дан	краткий	обзор	определений	назначения	регистров	из	согла
шений	о вызовах	функций.

Таблица 15.1.	Соглашения о вызовах функций

Регистр Использование Кто сохраняет

rax Возвращаемое значение Вызывающая функция

rbx Сохраняемый вызываемой функцией Вызываемая функция

rcx 4-й аргумент Вызывающая функция

rdx 3-й аргумент Вызывающая функция

rsi 2-й аргумент Вызывающая функция

rdi 1-й аргумент Вызывающая функция

rbp Сохраняемый вызываемой функцией Вызываемая функция

rsp Указатель стека Вызываемая функция

r8 5-й аргумент Вызывающая функция

r9 6-й аргумент Вызывающая функция

r10 Временный Вызывающая функция

r11 Временный Вызывающая функция

r12 Сохраняемый вызываемой функцией Вызываемая функция

r13 Сохраняемый вызываемой функцией Вызываемая функция

r14 Сохраняемый вызываемой функцией Вызываемая функция

r15 Сохраняемый вызываемой функцией Вызываемая функция

xmm0 Первый аргумент и возвращаемое значение Вызывающая функция

xmm1 Второй аргумент и возвращаемое значение Вызывающая функция

xmm2-7 Аргументы Вызывающая функция

xmm8-15 Временные Вызывающая функция

Вызываемая	функция	в документации	обозначена	как	callee.	Если	функция	
использует	регистр,	сохраняемый	вызываемой	функцией	(calee-saved register),	
то	она	обязательно	должна	поместить	содержимое	этого	регистра	в стек	перед	
его	использованием	и перед	завершением	работы	восстановить	содержимое	
в правильном	порядке.	Вызывающая	функция	полагает,	что	регистр,	сохраняе
мый	вызываемой	функцией,	остается	неизменным	после	вызова	функции.	Ре
гистры	аргументов	могут	быть	изменены	во	время	выполнения	функции,	по
этому	ответственность	за	их	сохранение	и восстановление	из	стека	остается	на	
вызывающей	функции,	если	в этом	есть	необходимость.	Временные	регистры	
также	могут	изменяться	в вызываемой	функции,	поэтому	при	необходимости	
вызывающая	функция	должна	их	 сохранять	и  восстанавливать.	 Вполне  оче
видно,	 что	 регистр	 rax,	 содержащий	 возвращаемое	 значение,	должен	 сохра
няться	и восстанавливаться	вызывающей	функцией.

Проблемы	 могут	 возникать	 при	 изменении	 существующей	 функции,	
когда	 в  ней	 начинается	 использование	 регистра,	 сохраняемого	 вызываю
щей	функцией.	Если	не добавить	инструкции	сохранения	и восстановления	
этого	регистра	в вызывающей	функции,	то	вы	получите	непредсказуемые	 
результаты.

Регистры,	сохраняемые	вызываемой	функцией,	также	называются	неразру
шаемыми,	или	долговременными	(nonvolatile).	Регистры,	которые	должна	со
хранять	вызывающая	функция,	еще	называются	разрушаемыми,	или	кратко
временными	(volatile).

Все	xmmрегистры	могут	быть	изменены	вызываемой	функцией,	поэтому	при	
необходимости	вызывающая	функция	отвечает	за	их	сохранение	и восстанов
ление.

Разумеется,	если	вы	абсолютно	уверены	в том,	что	не намерены	использо
вать	изменяемые	регистры,	то	можете	пропустить	инструкции	их	сохранения.	
Но если	в будущем	код	изменится,	то,	возможно,	возникнут	проблемы,	если	вы	
начнете	использовать	эти	регистры	без	сохранения	их	содержимого.	Можете	
верить	или	не  верить,	но	через	пару	недель	или	месяцев	 ассемблерный	код	
очень	трудно	читать,	даже	если	весь	исходный	код	был	написан	лично	вами.

И	 последнее	 замечание:	 syscall  –	 это	 тоже	 функция,	 которая	 изменяет	 
регистры,	так	что	внимательно	следите	за	тем,	что	делает	syscall.

резюме
В	этой	главе	вы	узнали,	что	такое:

	� соглашения	о вызовах	функций;
	� выравнивание	стека;
	� регистры,	сохраняемые	вызываемой	и вызывающей	функциями.

Глава 16
Операции с битами

Мы	уже	выполняли	некоторые	операции	с битами	в главе	9	при	использовании	
целочисленной	арифметики:	инструкции	арифметического	сдвига	sar	и sal –	
это	битовые	операции,	сдвигающие	биты	вправо	и влево.	Кроме	того,	инструк
ция	and,	применяемая	для	выравнивания	стека,	описанного	в предыдущей	гла
ве, –	это	также	битовая	операция.

ОСнОвные пОлОжения
В	приведенном	ниже	примере	программы	(листинг	16.1)	создается	специали
зированная	(пользовательская)	функция	языка	C с	именем	printb	для	вывода	
строки	битов.	Для	удобства	она	разделяет	строку	из	64	бит	на	8	байт,	по	8	бит	
в каждом	байте.	В качестве	дополнительного	упражнения:	после	завершения	
чтения	этой	главы	рассмотрите	код	на	языке	C –	вам	вполне по	силам	написать	
программу	на	ассемблере	для	формирования	строки	из	битов.

В	листингах	16.1,	16.2	и 16.3	показан	пример	кода	с использованием	битовых	
операций	на	ассемблере,	программа	на	языке	C printb	и makefile	соответственно.

Листинг 16.1.	Программа bits1.asm

; bits1.asm
extern printb
extern printf
section .data
      msgn1 db    "Number 1",10,0
      msgn2 db    "Number 2",10,0
      msg1  db    "XOR",10,0
      msg2  db    "OR",10,0
      msg3  db    "AND",10,0
      msg4  db    "NOT number 1",10,0
      msg5  db    "SHL 2 lower byte of number 1",10,0
      msg6  db    "SHR 2 lower byte of number 1",10,0
      msg7  db    "SAL 2 lower byte of number 1",10,0
      msg8  db    "SAR 2 lower byte of number 1",10,0
      msg9  db    "ROL 2 lower byte of number 1",10,0
      msg10 db    "ROL 2 lower byte of number 2",10,0
      msg11 db    "ROR 2 lower byte of number 1",10,0
      msg12 db    "ROR 2 lower byte of number 2",10,0
      number1     dq    -72
      number2     dq    1064



section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Вывод number1.
      mov   rsi, msgn1
      call  printmsg
      mov   rdi, [number1]
      call  printb
; Вывод number2.
      mov   rsi, msgn2
      call  printmsg
      mov   rdi, [number2]
      call  printb
; Вывод XOR (исключающего OR)---------------------
      mov   rsi, msg1
      call  printmsg
; xor и вывод.
      mov   rax,[number1]
      xor   rax,[number2]
      mov   rdi, rax
      call  printb
; Вывод OR ---------------------------------------
      mov   rsi, msg2
      call  printmsg
; or и вывод.
      mov   rax,[number1]
      or    rax,[number2]
      mov   rdi, rax
      call  printb
; Вывод AND  ---------------------------------------
      mov   rsi, msg3
      call  printmsg
; and и вывод.
      mov   rax,[number1]
      and   rax,[number2]
      mov   rdi, rax
      call  printb
; Вывод NOT  ---------------------------------------
      mov   rsi, msg4
      call  printmsg
; not и вывод.
      mov   rax,[number1]
      not   rax
      mov   rdi, rax
      call  printb
; Вывод SHL  (сдвиг влево) -------------------------
      mov   rsi, msg5
      call  printmsg
; shl и вывод.
      mov   rax,[number1]
      shl   al,2
      mov   rdi, rax
      call  printb

; Вывод SHR  (сдвиг вправо)-------------------------
      mov   rsi, msg6
      call  printmsg
; shr и вывод.
      mov   rax,[number1]
      shr   al,2
      mov   rdi, rax
      call  printb
; Вывод SAL  (арифметический сдвиг влево)-----------
      mov   rsi, msg7
      call  printmsg
; sal и вывод.
      mov   rax,[number1]
      sal   al,2
      mov   rdi, rax
      call  printb
; Вывод SAR  (арифметический сдвиг вправо)----------
      mov   rsi, msg8
      call  printmsg
; sar и вывод.
      mov   rax,[number1]
      sar   al,2
      mov   rdi, rax
      call  printb
; Вывод ROL  (вращение влево)-----------------------
      mov   rsi, msg9
      call  printmsg
; rol и вывод.
      mov   rax,[number1]
      rol   al,2
      mov   rdi, rax
      call  printb
      mov   rsi, msg10
      call  printmsg
      mov   rax,[number2]
      rol   al,2
      mov   rdi, rax
      call  printb
; Вывод ROR  (вращение вправо)----------------------
      mov   rsi, msg11
      call  printmsg
; ror и вывод.
      mov   rax,[number1]
      ror   al,2
      mov   rdi, rax
      call  printb
      mov   rsi, msg12
      call  printmsg
      mov   rax,[number2]
      ror   al,2
      mov   rdi, rax
      call  printb
leave
ret
;-----------------------------------------------------
printmsg:    ; Вывод заголовка для каждой битовой операции.

section .data
      .fmtstr            db      "%s",0
section .text
      mov   rdi,.fmtstr
      mov   rax,0
      call  printf
      ret

Листинг 16.2.	Функция printb.c

// printb.c

#include <stdio.h>

void printb(long long n){
      long long s,c;
      for (c = 63; c >= 0; c--)
      {
      s = n >> c;
            // Пробел после каждого 8-го бита.
            if ((c+1) % 8 == 0) printf(" ");
            if (s & 1)
                  printf("1");
            else
                  printf("0");
      }
      printf("\n");
}

Листинг 16.3.	makefile для программы bits1 и функции printb

# makefile для bits1 и printb
bits1: bits1.o printb.o
      gcc -g -o bits1 bits1.o printb.o -no-pie
bits1.o: bits1.asm
      nasm -f elf64 -g -F dwarf bits1.asm -l bits1.lst
printb: printb.c
      gcc -c printb.c

После	 сборки	 запустите	 эту	 программу	 и  внимательно	 изучите	 ее	 вывод.	
Если	 вы	 используете	 SASM,	 то	 не  забудьте	 сначала	 скомпилировать	 файл	
printb.c	и добавить	полученный	объектный	файл	в строку	параметров	Linking	
Options	(Параметры	связывания)	в соответствии	с указаниями	при	обсужде
нии	внешних	функций	в главе	14.

Это	достаточно	длинная	программа.	К счастью,	ее	исходный	код	не очень	
сложен.	Программа	демонстрирует	работу	различных	инструкций,	выполняю
щих	операции	с битами.	Используйте	вывод,	показанный	на	рис. 16.1,	для	того	
чтобы	лучше	ориентироваться	в исходном	коде.

Рис. 16.1.	Вывод программы bits1.asm

Первое	примечание	относится	к двоичному	представлению	значения	num-
ber1	(-72) –	здесь	1	в самом	старшем	(крайнем	справа)	бите	обозначает	отрица
тельное	число.

Инструкции	xor,	or,	and	и not	весьма	просты –	они	выполняются	так,	как	опи
сано	в главе	5.	Поэкспериментируйте	с различными	значениями,	чтобы	лучше	
понять,	как	они	работают.

Для	инструкций	shl,	shr,	sal	и sar используется	самый	младший	байт	регист
ра	rax,	чтобы	наглядно	показать,	что	происходит.	При	использовании	shl	биты	
сдвигаются	влево,	 а нули	добавляются	 справа	в регистре	al,	т.  е.	 биты	пере
мещаются	влево,	а те	биты,	которые	оказываются	слева	от	8го	бита,	просто	
отбрасываются.	При	использовании	shr	биты	сдвигаются	вправо,	а нули	добав
ляются	слева	в регистре	al.	Все	биты	перемещаются	вправо,	а те	биты,	которые	
оказываются	справа	от	самого	младшего	(нулевого)	бита,	отбрасываются.	При	
пошаговом	проходе	по	программе	внимательно	следите	за	регистром	флагов,	
особенно	за	флагами	знака	и переполнения.

Инструкция	арифметического	сдвига	влево	sal	делает	в точности	то	же,	что	
инструкция	 shl, –	 умножает	 значение.	 Инструкция	 арифметического	 сдвига	
вправо	sar	отличается	от	инструкции	shr.	Здесь	происходит	то,	что	называют	
распространением	(расширением)	знака	(sign extension).	Если	самый	левый	бит	

в регистре	al	равен	1,	то	регистр	al	содержит	отрицательное	значение.	Чтобы	
арифметические	операции	выполнялись	корректно	при	сдвиге	вправо,	слева	
вместо	нулей	добавляются	единицы	в случае	сдвига	отрицательного	значения.	
Это	называется	распространением	(расширением)	знака.

Инструкция	 вращения	 влево	 rol	 удаляет	 левый	 крайний	 бит,	 выполняет	
сдвиг	влево,	затем	добавляет	удаленный	бит	(биты)	справа.	Инструкция	вра
щения	вправо	ror	делает	то	же	самое,	но	в обратном	направлении.

арифметика
Попробуем	более	подробно	разобраться	в сдвиговой	арифметике.	Почему	су
ществуют	два	 типа	 инструкций	 сдвига	 влево	 и  два	 типа	 инструкций	 сдвига	
вправо?	При	выполнении	арифметических	операций	с отрицательными	зна
чениями	инструкции	сдвига	могут	давать	неправильные	результаты,	потому	
что	необходимо	учитывать	распространение	(расширение)	знака.	Именно	по
этому	существуют	арифметические	инструкции	сдвига	и логические	инструк
ции	сдвига.

Рассмотрим	пример	в листинге	16.4.

Листинг 16.4.	Программа bits2.asm

; bits2.asm
extern printf
section .data
      msgn1 db    "Number 1 is = %d",0
      msgn2 db    "Number 2 is = %d",0
      msg1  db    "SHL 2 = OK multiply by 4",0
      msg2  db    "SHR 2 = WRONG divide by 4",0
      msg3  db    "SAL 2 = correctly multiply by 4",0
      msg4  db    "SAR 2 = correctly divide by 4",0
      msg5  db    "SHR 2 = OK divide by 4",0
      number1     dq    8
      number2     dq    -8
      result      dq    0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
;SHL-------------------------------------------------
; Положительное число.
      mov   rsi, msg1
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number1]
      call  printnbr         ; Вывод числа number1.
      mov   rax,[number1]
      shl   rax,2      ; Умножение на 4 (логический сдвиг).
      mov   rsi, rax
      call  printres
; Отрицательное число.
      mov   rsi, msg1
      call  printmsg         ; Вывод заголовка.

      mov   rsi, [number2]
      call  printnbr         ; Вывод числа number2.
      mov   rax,[number2]
      shl   rax,2      ; Умножение на 4 (логический сдвиг).
      mov   rsi, rax
      call  printres
;SAL-------------------------------------------------
; Положительное число.
      mov   rsi, msg3
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number1]
      call  printnbr         ; Вывод числа number1.
      mov   rax,[number1]
      sal   rax,2      ; Умножение на 4 (арифметический сдвиг)
      mov   rsi, rax
      call  printres
; Отрицательное число.
      mov   rsi, msg3
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number2]
      call  printnbr         ; Вывод числа number2.
      mov   rax,[number2]
      sal   rax,2      ; Умножение на 4 (арифметический сдвиг)
      mov   rsi, rax
      call  printres
;SHR-------------------------------------------------
; Положительное число.
      mov   rsi, msg5
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number1]
      call  printnbr         ; Вывод числа number1.
      mov   rax,[number1]
      shr   rax,2      ; Деление на 4 (логический сдвиг).
      mov   rsi, rax
      call  printres
; Отрицательное число.
      mov   rsi, msg2
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number2]
      call  printnbr         ; Вывод числа number2.
      mov   rax,[number2]
      shr   rax,2      ; Деление на 4 (логический сдвиг).
      mov   [result], rax
      mov   rsi, rax
      call  printres
;SAR-------------------------------------------------
; Положительное число.
      mov   rsi, msg4
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number1]
      call  printnbr         ; Вывод числа number1.
      mov   rax,[number1]
      sar   rax,2      ; Деление на 4 (арифметический сдвиг).
      mov   rsi, rax
      call  printres
; Отрицательное число.

      mov   rsi, msg4
      call  printmsg         ; Вывод заголовка.
      mov   rsi, [number2]
      call  printnbr         ; Вывод числа number2.
      mov   rax,[number2]
      sar   rax,2      ; Деление на 4 (арифметический сдвиг).
      mov   rsi, rax
      call  printres
leave
ret
;-----------------------------------
printmsg:              ; Вывод общего заголовка.
      section .data
            .fmtstr db 10,"%s",10,0 ; Формат для вывода строки.
      section .text
            mov   rdi,.fmtstr
            mov   rax,0
            call  printf
      ret
;-----------------------------------
printnbr:              ; Вывод исходного числа.
      section .data
            .fmtstr db "The original number is %lld",10,0
      section .text
            mov   rdi,.fmtstr
            mov   rax,0
            call  printf
      ret
;-----------------------------------
printres:              ; Вывод результата.
      section .data
            .fmtstr db "The resulting number is %lld",10,0
      section .text
            mov   rdi,.fmtstr
            mov   rax,0
            call  printf
      ret

Используйте	вывод	этой	программы,	показанный	на	рис. 16.2,	для	анализа	
исходного	кода.

Обратите	внимание:	инструкции	shl	и sal	дают	одинаковые	результаты	даже	
с  отрицательными	числами.	Но будьте	 внимательны	и осторожны:	 если	ин
струкция	shl	 поместит	1	 вместо	0	 в левый	крайний	бит,	то	результат	 станет	
отрицательным,	т. е.	неверным.

Инструкции	shr	и sar	дают	одинаковые	результаты,	только	если	числа	по
ложительные.	Арифметический	результат	при	использовании	shr	для	отрица
тельных	чисел	просто	некорректен,	потому	что	инструкция	shr	не выполняет	
распространение	знака.

Вывод:	при	выполнении	арифметических	операций	необходимо	использо
вать	инструкции	sal	и sar.

Зачем	 нужны	 инструкции	 сдвига,	 если	 существуют	 арифметические	 ин
струкции,	явно	предназначенные	для	выполнения	умножения	и деления?	Ока
зывается,	 сдвиг	 выполняется	 намного	 быстрее,	 чем	 инструкции	 умножения	
и деления.	 В  общем	 случае	 битовые	инструкции	 выполняются	 чрезвычайно	
быстро,	например	xor rax, rax	намного	быстрее,	чем	mov rax, 0.

Рис. 16.2.	Вывод программы bits2.asm

резюме
В	этой	главе	вы	узнали:

	� об	использовании	ассемблерных	инструкций	для	операций	с битами;
	� о	различии	между	инструкциями	логического	и арифметического	сдвига.

Глава 17
Работа с битами

Вам	 уже	 известно	 о  возможности	 устанавливать	 или	 очищать	 (сбрасывать)	
биты	с помощью	битовых	операций,	таких	как	and,	xor,	or	и not.	Но существу
ют	 и  другие	 способы	 изменения	 отдельных	 битов:	 bts	 для	 установки	 битов	
в 1, btr для	сброса	битов	в 0	и bt	для	проверки,	установлен	ли	бит	в 1.

другие СпОСОбы изменения битОв
В	листинге	17.1	показан	пример	исходного	кода.

Листинг 17.1.	Программа bits3.asm

; bits3.asm
extern printb
extern printf
section .data
      msg1  db   "No bits are set:",10,0
      msg2  db   10,"Set bit #4, that is the 5th bit:",10,0
      msg3  db   10,"Set bit #7, that is the 8th bit:",10,0
      msg4  db   10,"Set bit #8, that is the 9th bit:",10,0
      msg5  db   10,"Set bit #61, that is the 62nd bit:",10,0
      msg6  db   10,"Clear bit #8, that is the 9th bit:",10,0
      msg7  db   10,"Test bit #61, and display rdi",10,0
      bitflags   dq      0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      ; Вывод заголовка.
      mov   rdi, msg1
      xor   rax,rax
      call  printf
      ; Вывод переменной bitflags.
      mov   rdi, [bitflags]
      call  printb
; Установить в 1 бит 4 (=5-й бит).
      ; Вывод заголовка.
      mov   rdi, msg2
      xor   rax,rax
      call  printf



      bts   qword [bitflags],4    ; Установить бит 4.
      ; Вывод переменной bitflags.
      mov   rdi, [bitflags]
      call  printb
; Установить в 1 бит 7 (=8-й бит).
      ; Вывод заголовка.
      mov   rdi, msg3
      xor   rax,rax
      call  printf
      bts   qword [bitflags],7    ; Установить бит 7.
      ; Вывод переменной bitflags.
      mov   rdi, [bitflags]
      call  printb
; Установить в 1 бит 8 (=9-й бит).
      ; Вывод заголовка.
      mov   rdi, msg4
      xor   rax,rax
      call  printf
      bts   qword [bitflags],8    ; Установить бит 8.
      ; Вывод переменной bitflags.
      mov   rdi, [bitflags]
      call  printb
; Установить в 1 бит 61 (=62-й бит).
      ; Вывод заголовка.
      mov   rdi, msg5
      xor   rax,rax
      call  printf
      bts   qword [bitflags],61   ; Установить бит 61.
      ; Вывод переменной bitflags.
      mov   rdi, [bitflags]
      call  printb
; Очистить бит 8 (=9-й бит).
      ; Вывод заголовка.
      mov   rdi, msg6
      xor   rax, rax
      call  printf
      btr   qword [bitflags],8    ; Сброс бита 8.
      ; Вывод переменной bitflags.
      mov   rdi, [bitflags]
      call  printb
; Проверка бита 61 (будет установлен в 1 флаг переноса CF, если этот бит равен 1).
      ; Вывод заголовка.
      mov   rdi, msg7
      xor   rax, rax
      call  printf
      xor   rdi,rdi
      mov   rax,61           ; Нужно проверить бит 61.
      xor   rdi,rdi          ; Чтобы все биты были равны 0.
      bt    [bitflags],rax   ; Проверка бита.
      setc  dil              ; Установить dil (=младший rdi) в 1, если CF установлен.
      call  printb           ; Вывод регистра rdi.
leave
ret

Здесь	снова	используется	программа	printb.c,	поэтому	необходимо	изменить	
соответствующим	образом	makefile	или	параметры	настройки	сборки	в SASM.

В	рассматриваемом	примере	переменная	bitflags	является	объектом	иссле
дования,	в ней	выполняются	различные	операции	с отдельными	битами.

переменная bitflags
Напомню,	что	отсчет	битов	(индексирование)	начинается	с 0.	Это	значит,	что	
в байте,	состоящем	из	8	бит,	первый	бит	находится	в позиции	0,	а последний	
бит –	в позиции	7.	Установка	битов	в 1	с помощью	инструкции	bts	и сброс	би
тов	в 0	с помощью	инструкции	btr	выполняются	просто:	нужно	только	указать	
индекс	изменяемого	бита	во	втором	операнде.

Проверка	бита	выполняется	немного	сложнее.	Необходимо	поместить	индекс	
проверяемого	бита	в регистр	rax	и использовать	инструкцию	bt.	Если	бит	равен	
1,	то	флаг	переноса	CF	будет	установлен	в 1,	иначе	CF	будет	равен	0.	На основе	зна
чения	флага	CF	можно	перенаправить	программу	для	выполнения	определен
ных	инструкций	или	продолжить	выполнение	без	изменения	порядка.	В этом	
случае	используется	специальная	инструкция	setc	для	установки	значения	по	
условию.	В рассматриваемом	здесь	примере	инструкция	setc	устанавливает	dil 
в 1,	если	флаг	переноса	равен	1.	dil –	это	младшая	(нижняя)	часть	регистра	rdi.	
Необходимо	обязательно	сбросить	регистр	rdi в	0	перед	использованием	ин
струкции	setc	для	условной	установки	dil.	Вполне может	случиться,	что	старшие	
биты	регистра	rdi	установлены	во	время	выполнения	предыдущей	инструкции.

Инструкция	setc	представляет	собой	пример	из	группы	инструкций	setCC.	Ин
струкции	setCC	устанавливают	в 1	байт,	указанный	в операнде,	если	выполнено	
условие	CC,	где	CC –	это	флаг,	например	CF	(инструкция	setc),	ZF	(инструкция	setz),	
SF	(инструкция	sets)	и т.	д.	Более	подробно	об	этом	см.	в руководствах	Intel.

На	рис. 17.1	показан	вывод	этой	программы.

Рис. 17.1.	Вывод программы bits3.asm

резюме
В	этой	главе	вы	узнали:

	� об	установке,	 сбросе	и проверке	битов	с помощью	инструкций	bts,	btr 
и bt;

	� о	группе	инструкций	setCC.

Глава 18
Макрокоманды

При	многократном	использовании	в программе	одного	и того	же	набора	ин
струкций	можно	создать	функцию	и вызывать	ее	каждый	раз,	когда	необхо
димо	выполнить	эти	инструкции.	Но для	функций	характерно	снижение	про
изводительности:	при	каждом	вызове	функции	поток	выполнения	переходит	
к вызываемой	функции	в некоторой	области	памяти,	а после	ее	завершения	
возвращается	обратно	в вызывающую	программу.	Вызов	функции	и возврат	
из	нее	требуют	определенного	времени.

Чтобы	 избежать	 подобных	 проблем	 с  производительностью,	 можно	 вос
пользоваться	макрокомандами	(или	просто	макросами –	macros).	Как	и функ
ция,	 макрос	 представляет	 собой	 последовательность	 инструкций.	 Макросу	
присваивается	имя,	и когда	нужно	выполнить	этот	макрос	в коде,	вы	просто	
указываете	его	имя,	возможно,	сопровождаемое	аргументами.

Важное	отличие	от	функции:	во	время	ассемблирования	везде,	где	в исход
ном	 коде	 встречается	 «вызов»	 макроса,	 NASM	 заменяет	 имя	 этого	 макроса	
на	набор	инструкций,	записанных	в его	определении.	Во время	выполнения	
не происходит	никаких	переходов	и возвратов,	потому	что	NASM	уже	вставил	
машинный	код	во	всех	необходимых	местах.

Макросы	не являются	функциональным	свойством	языка	ассемблера	Intel,	
но	 эта	 функциональность	 предоставляется	 NASM	 (или	 другой	 версией	 про
граммы	ассемблирования).	Макросы	создаются	с помощью	директив	препро
цессора,	 и  NASM	 использует	макропроцессор	 для	 преобразования	макросов	
в машинный	язык	и вставляет	инструкции	машинного	языка	в соответствую
щие	места	в коде	программы.

Макросы	позволят	улучшить	скорость	выполнения	кода,	но	они	увеличива
ют	объем	кода,	потому	что	во	время	ассемблирования	инструкции	из	опреде
ления	макроса	вставляются	в каждую	локацию,	где	используется	макрос.

Более	подробную	информацию	о макросах	NASM	можно	получить	из	руко
водства	NASM,	глава	4	«The	NASM	Preprocessor»	(для	версии	NASM	2.14.02).

СОздание макрОСа
В	листинге	18.1	показано	несколько	примеров	определения	и использования	
макросов.



Листинг 18.1.	Программа macro.asm

; macro.asm
extern printf

%define     double_it(r)    sal r, 1    ; Однострочный макрос.

%macro      prntf 2    ; Макрос из нескольких строк с 2 аргументами.
      section .data
            %%arg1    db    %1,0            ; Первый аргумент.
            %%fmtint  db    "%s %ld",10,0   ; Строка формата.
      section .text                 ; Аргументы для внешней функции printf.
            mov   rdi,%%fmtint
            mov   rsi,%%arg1
            mov   rdx,[%2]     ; Второй аргумент.
            mov   rax,0        ; Числа с плав. точкой не используются.
            call  printf
%endmacro

section .data
      number      dq    15
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      prntf      "The number is", number
      mov        rax, [number]
      double_it(rax)
      mov        [number],rax
      prntf      "The number times 2 is", number
leave
ret

Существует	два	типа	макросов:	 однострочные	макросы	и макросы	из	 не
скольких	 строк	 (многострочные).	 	 Однострочный	макрос	 начинается	 с  клю
чевого	слова	%define.	Многострочный	макрос	определяется	между	ключевыми	
словами	%macro	и %endmacro.	Ключевые	слова	%define,	%macro	и %endmacro	называ
ются	директивами	препроцессора	ассемблера	(assembler preprocessor directives).

Однострочный	макрос	чрезвычайно	прост:	во	время	ассемблирования	ин
струкция	double_it(rax)	 заменяется	на	машинный	код,	 соответствующий	ин
струкции	sal r, 1,	где	r –	значение	в регистре	rax.

Многострочный	макрос	немного	более	сложен:	здесь	макрос	prntf	вызывает
ся	с двумя	аргументами.	В определении	макроса	можно	видеть,	что	за	именем	
prntf	следует	число	2,	определяющее	количество	аргументов.	Чтобы	аргументы	
можно	было	использовать	 внутри	макроса,	 они	обозначаются	 символами	%1 
для	первого	аргумента,	%2	для	второго	и т.	д.	Обратите	внимание:	аргумент	%1 
позволяет	использовать	строку,	но	аргумент	[%2]	 (в квадратных	скобках)	по
зволяет	обратиться	к числовому	значению	по	аналогии	с требованиями	к коду	
без	использования	макросов.

Внутри	макроса	можно	использовать	переменные,	при	этом	рекомендуется	

давать	им	имена	с префиксом	%%,	например	%%arg1	и %%fmtint.	Если	пропустить	
символы	%%,	то	NASM	успешно	создает	переменные	макро	при	первом	вызове	
prntf,	но	сгенерирует	ошибку	ассемблирования	при	втором	вызове	prntf,	 со
общая,	что	вы	пытаетесь	повторно	определить	переменные	arg1	и fmtint.	Сим
волы	%%	информируют	NASM	о необходимости	создания	новых	экземпляров	
переменных	при	каждом	вызове	макроса.	(Дополнительное	упражнение:	уда
лите	символы	%%	и попробуйте	ассемблировать	программу.)

Для	ассемблерных	макросов	существует	одна	крупная	проблема:	их	сложно	
отлаживать.	Попробуйте	отладить	программу	macro.asm	в GDB	или	в отладчике	
на	основе	GDB,	например	в SASM,	и понаблюдайте	за	ее	поведением.

На	рис. 18.1	показан	вывод	программы	macro.asm.

Рис. 18.1.	Вывод программы macro.asm

иСпОльзОвание objdump
Проверим	тот	факт,	что	ассемблированный	код	макроса	действительно	встав
лен	в соответствующих	местах	в исполняемом	файле,	т. е.	там,	где	использо
вался	макрос.	Для	этого	воспользуемся	утилитой	командной	строки	с именем	
objdump.	Если	вы	установили	инструментальные	средства	разработки	в соответ
ствии	с рекомендациями	в начале	этой	книги,	то	утилита	objdump	уже	установ
лена	и готова	к работе.	В командной	строке	введите	следующую	строку:

objdump -M intel -d macro

Флаг	-M intel	позволит	получить	код	с синтаксисом	ассемблера	Intel,	а флаг	
-d macro	 обеспечивает	дизассемблирование	выполняемого	кода	макроса.	Вы
полните	прокрутку	по	полученному	коду	до	раздела	<main>.

На	рис. 18.2	можно	видеть,	что	код	макроса	prntf	вставлен	в код	main,	начиная	
с адреса	памяти	4004f4	до	адреса	400515,	а также	с адреса	40052d	до	адреса	40054e.	
Код	макроса	double_it	размещен	по	адресу	400522.	Программа	ассемблирования	
принимает	на	себя	ответственность	при	замене инструкции	sal	на	shl	по	со
ображениям	улучшения	производительности.	Как	вы	уже	знаете	из	главы 16	
об	инструкциях	сдвига,	в большинстве	случаев	такая	замена	не приводит	к ка
кимлибо	проблемам.	Для	эксперимента	замените	инструкцию	sal	на	инструк
цию	sar.	Вы увидите,	что	программа	ассемблирования	не заменяет	sar	на	shr,	
чтобы	избежать	проблем.

Утилита	командной	строки	objdump	полезна	при	исследовании	кода,	даже	если	
код	написан	не вами.	Вы можете	получить	огромный	объем	информации	о вы
полняемом	файле,	используя	objdump,	но	мы	не будем	углубляться	в подробности	

в этой	книге.	Если	вы	хотите	узнать	больше,	то	введите	команду	man objdump	в ко
мандной	строке	или	найдите	требуемую	информацию	в интернете.

Рис. 18.2.	Результат выполнения команды objdump -M intel -d macro

резюме
В	этой	главе	вы	узнали:

	� когда	следует	использовать	макросы,	а когда –	функции;
	� об	однострочных	макросах;
	� о	многострочных	макросах;
	� о	передаче	аргументов	в многострочные	макросы;
	� о	проблемах	отладки	в GDB	при	использовании	ассемблерных	макросов;
	� об	использовании	утилиты	objdump.

Глава 19
Ввод и вывод в консоли

Вам	уже	известно,	как	выполняется	вывод	в консоли	с использованием	систем
ных	вызовов	или	внешней	функции	printf.	В  этой	 главе	снова	будут	приме
няться	системные	вызовы,	но	не только	для	вывода	на	экран,	а еще	и для	при
ема	ввода	с клавиатуры.

иСпОльзОвание СредСтв ввОда/вывОда
Можно	было	бы	просто	позаимствовать	необходимые	функции	из	стандартной	
библиотеки	языка	C,	но	это	испортило	бы	удовольствие	от	работы	с ассембле
ром.	В листинге	19.1	показан	пример	исходного	кода.

Листинг 19.1.	Программа console1.asm

; console1.asm
section .data
      msg1       db    "Hello, World!",10,0
      msg1len    equ   $-msg1
      msg2       db    "Your turn: ",0
      msg2len    equ   $-msg2
      msg3       db    "You answered: ",0
      msg3len    equ   $-msg3
      inputlen equ     10   ; Длина буфера ввода.
section .bss
      input resb inputlen+1 ; Обеспечение места для завершающего 0.
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      mov   rsi, msg1       ; Вывод первой строки.
      mov   rdx, msg1len
      call  prints
      mov   rsi, msg2       ; Вывод второй строки, без NL.
      mov   rdx, msg2len
      call  prints
      mov   rsi, input      ; Адрес буфера ввода inputbuffer.
      mov   rdx, inputlen   ; Длина буфера ввода inputbuffer.
      call  reads           ; Ожидание ввода.
      mov   rsi, msg3       ; Вывод третьей строки.
      mov   rdx, msg3len



      call  prints
      mov   rsi, input      ; Вывод содержимого буфера ввода inputbuffer.
      mov   rdx, inputlen   ; Длина буфера ввода inputbuffer.
      call  prints
leave
ret
;----------------------------------------------------
prints:
push  rbp
mov   rbp, rsp
; rsi содержит адрес строки.
; rdx содержит длину строки.
      mov   rax, 1          ; 1 = запись
      mov   rdi, 1          ; 1 = stdout – стандартный вывод.
      syscall
leave
ret
;----------------------------------------------------
reads:
push  rbp
mov   rbp, rsp
; rsi содержит адрес буфера ввода inputbuffer.
; rdi содержит длину буфера ввода inputbuffer.
      mov   rax, 0         ; 0 = чтение
      mov   rdi, 1         ; 1 = stdin – стандартный ввод.
      syscall
leave
ret

Это	не очень	сложная	программа,	в ней	создается	буфер	ввода	с именем	in-
put	для	хранения	вводимых	символов.	Кроме	того,	определяется	длина	этого	
буфера	в переменной	inputlen.	После	вывода	нескольких	приветственных	со
общений	вызывается	функция	reads,	которая	принимает	символы,	вводимые	
с клавиатуры,	и возвращает	их	в вызывающую	программу	после	нажатия	кла
виши	Enter.	Затем	вызывающая	программа	использует	функцию	prints	для	вы
вода	ранее	введенных	символов.	На рис. 19.1	показан	вывод	этой	программы.

Рис. 19.1.	Вывод программы console1.asm

Но	 и  здесь	 возникают	 некоторые	 проблемы.	 Для	 буфера	 ввода	 зарезер
вировано	 10	 байт.	 Что	 произойдет,	 если	 будет	 введено	 более	 10	 символов?	
На рис. 19.2	показан	результат	в этом	случае.

Рис. 19.2.	Вывод программы console1.asm при вводе слишком большого количества символов

Эта	программа	принимает	только	10	символов	и не знает,	что	делать	с осталь
ными	«лишними»	символами,	поэтому	возвращает	их	в операционную	систе
му.	Операционная	система	пытается	определить	их	смысл	и интерпретирует	
эти	символы	как	команды,	но	не находит	соответствующие	файлы	или	имена	
встроенных	команд,	поэтому	выводит	сообщения	об	ошибках.

Результат	неутешителен,	но	он	даже	хуже,	чем	кажется	на	первый	взгляд.	
Такой	 способ	 обработки	 вывода	 может	 стать	 причиной	 серьезной	 уязвимо
сти	(бреши)	в системе	обеспечения	безопасности,	через	которую	хакер	может	
взломать	программу	и получить	доступ	к операционной	системе.

ОбрабОтка перепОлнений
В	листинге	19.2	показана	другая	версия	программы,	в которой	ведется	подсчет	
вводимых	символов,	а лишние	символы	просто	отбрасываются.	Дополнитель
ное	условие:	разрешен	ввод	только	алфавитных	символов	в нижнем	регистре,	
от	a до	z.

Листинг 19.2.	Программа console2.asm

; console2.asm
section .data
      msg1  db    "Hello, World!",10,0
      msg2  db    "Your turn (only a-z): ",0
      msg3  db    "You answered: ",0
      inputlen    equ   10   ; Длина буфера ввода inputbuffer.
      NL    db    0xa
section .bss
      input resb inputlen+1     ; Обеспечение места для завершающего 0.
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      mov   rdi, msg1   ; Вывод первой строки.

      call  prints
      mov   rdi, msg2   ; Вывод второй строки, без NL.
      call  prints
      mov   rdi, input       ; Адрес буфера ввода inputbuffer.
      mov   rsi, inputlen    ; Длина буфера ввода inputbuffer.
      call  reads            ; Ожидание ввода.
      mov   rdi, msg3        ; Вывод третьей строки и добавление введенной строки.
      call  prints
      mov   rdi, input       ; Вывод содержимого буфера ввода inputbuffer.
      call  prints
      mov   rdi,NL           ; Вывод символа перехода на новую строку NL.
      call  prints
leave
ret
;----------------------------------------------------------
prints:
push  rbp
mov   rbp, rsp
push  r12      ; Регистр, сохраняемый вызываемой функцией.
; Подсчет символов.
      xor   rdx, rdx    ; Длина в rdx.
      mov   r12, rdi
.lengthloop:
      cmp   byte [r12], 0
      je    .lengthfound
      inc   rdx
      inc   r12
      jmp   .lengthloop
.lengthfound:           ; Вывод строки, ее длина в rdx.
      cmp   rdx, 0      ; Строка отсутствует (длина 0).
      je    .done
      mov   rsi,rdi     ; rdi содержит адрес строки.
      mov   rax, 1      ; 1 = запись.
      mov   rdi, 1      ; 1 = stdout – стандартный вывод.
      syscall
.done:
pop r12
leave
ret
;----------------------------------------------------------
reads:
section .data
section .bss
      .inputchar       resb       1
section .text
push  rbp
mov   rbp, rsp
      push  r12              ; Регистр, сохраняемый вызываемой функцией.
      push  r13              ; Регистр, сохраняемый вызываемой функцией.
      push  r14              ; Регистр, сохраняемый вызываемой функцией.
      mov   r12, rdi    ; Адрес буфера ввода inputbuffer.
      mov   r13, rsi    ; Максимальная длина в r13.
      xor   r14, r14    ; Счетчик символов.
.readc:
      mov   rax, 0           ; Чтение.
      mov   rdi, 1           ; stdin – стандартный ввод.

      lea   rsi, [.inputchar]     ; Адрес источника ввода.
      mov   rdx, 1           ; Число считываемых символов.
      syscall
      mov   al, [.inputchar]      ; Введен символ NL?
      cmp   al, byte[NL]
      je    .done            ; NL – конец ввода.
      cmp   al, 97           ; Код символа меньше a?
      jl    .readc           ; Отбросить символ.
      cmp   al, 122          ; Код символа больше z?
      jg    .readc           ; Отбросить символ.
      inc   r14                    ; Увеличить счетчик символов на 1.
      cmp   r14, r13
      ja    .readc           ; Максимальное заполнение буфера, отбросить лишнее.
      mov   byte [r12], al   ; Сохранить символ в буфере.
      inc   r12              ; Переместить указатель на следующий символ в буфере.
      jmp   .readc
.done:
      inc   r12
      mov   byte [r12],0     ; Добавить завершающий 0 в буфер ввода inputbuffer.
      pop   r14              ; Регистр, сохраняемый вызываемой функцией.
      pop   r13              ; Регистр, сохраняемый вызываемой функцией.
      pop   r12              ; Регистр, сохраняемый вызываемой функцией.
leave
ret

Функция	 prints	 изменена:	 сначала	 она	подсчитывает	 количество	 выводи
мых	символов,	т. е.	счет	ведется	до	тех	пор,	пока	не встретится	байт	0.	После	
определения	длины	prints	выводит	строку	с помощью	syscall.

Функция	reads	ожидает	ввода	одного	символа	и проверяет,	не является	ли	
он	символом	перехода	на	новую	строку	(NL).	Если	получен	символ	перехода	на	
новую	строку,	то	чтение	символов	с клавиатуры	прекращается.	Регистр	r14	со
держит	счетчик	вводимых	символов.	Функция	следит	за	тем,	чтобы	количество	
введенных	символов	не превышало	длину	буфера	inputlen.	Если	это	количест
во	меньше	длины	 буфера,	 то	 символ	добавляется	 в  буфер	 input.	 Если	длина	
inputlen	 превышена,	то	 символ	отбрасывается,	но	чтение	 с  клавиатуры	про
должается.	 Введено	 дополнительное	 требование:	ASCIIкод	 символа	 должен	
быть	не меньше	97	(a)	и не больше	122	(z).	Это	требование	обеспечивает	прием	
только	алфавитных	символов	в нижнем	регистре.	Обратите	внимание	на	со
хранение	 и  восстановление	 регистров,	 сохраняемых	 вызываемой	 функцией	
(callee saved), –	регистр	r12	используется	в обеих	функциях	prints	и reads.	В рас
сматриваемом	здесь	примере	отсутствие	инструкций	сохранения	таких	(callee 
saved)	регистров	не должно	приводить	к проблемам,	но	легко	представить	себе	
ситуацию,	 когда	одна	из	функций	вызывает	другую,	 а  вызываемая	функция	
обращается	к следующей, –	тогда	могут	возникать	проблемы.

На	рис. 19.3	показан	вывод	этой	программы.

Рис. 19.3.	Вывод программы console2.asm

Отладка	 консольного	 ввода	 с  использованием	 SASM	 затруднена,	 потому	
что	ввод	обеспечивается	через	системный	вызов	syscall.	SASM	предоставляет	
собственные	функциональные	возможности	для	ввода/вывода,	но	мы	не поль
зовались	 ими,	 так	 как	 необходимо	 было	 продемонстрировать,	 как	 работает	
ассемблер	и машинный	язык,	не скрывая	подробностей.	Если	возникли	серьез
ные	затруднения	при	отладке	в SASM,	то	рекомендуется	вернуться	к старому	
доброму	отладчику	GDB.

резюме
В	этой	главе	вы	узнали:

	� о	вводе	с клавиатуры	с использованием	системного	вызова	syscall;
	� о	проверке	корректности	ввода	с клавиатуры;
	� об	отладке	при	вводе	с клавиатуры,	которая	может	оказаться	затруднен
ной.

Глава 20
Файловый ввод/вывод

При	разработке	программного	обеспечения	операции	с файлами	могут	быть	
сложными.	Различные	операционные	системы	предоставляют	разнообразные	
методы	 управления	 файлами,	 и  для	 каждого	 метода	 существует	 обширный	
список	параметров.	В этой	главе	рассматривается	файловый	ввод/вывод	для	
систем	Linux.	В главе	43	вы	увидите,	что	файловый	ввод/вывод	в ОС	Windows	
выполняется	совершенно	другими	методами.

В	 Linux	методы	 управления	файлами	 сложны	и  включают	 создание	 и  от
крытие	файла	только	для	чтения	или	для	чтения/записи,	для	записи	в новый	
файл	или	для	добавления	в существующий	файл,	а также	удаление	файлов...	
не  говоря	 уже	о параметрах	 защиты	файлов	 с  разделением	на	 «пользовате
лявладельца»	(user),	«группу»	(group)	и «прочих»	(other).	При	необходимости	
придется	освежить	в памяти	свои	навыки	администратора	файловой	системы	
Linux	и обратиться	к руководству	системного	администратора	Linux.	В исход
ном	коде	примера	этой	главы	определяются	только	флаги	для	текущего	поль
зователя	 (user),	но	вы	можете	самостоятельно	добавить	определения	флагов	
для	группы	(group)	и прочих	(other).	Если	вы	не совсем	понимаете,	о чем	идет	
речь,	то	прямо	сейчас	необходимо	узнать	немного	больше	об	основах	управле
ния	файлами	в Linux.

иСпОльзОвание СиСтемных вызОвОв
Файлы	создаются,	открываются,	закрываются	и т.	д.	через	системные	вызовы.	
В этой	главе	используются	многие	системные	вызовы,	поэтому	сейчас	требу
ются	некоторые	объяснения,	чтобы	упростить	понимание.	В начальной	части	
исходного	кода	будут	определяться	константы,	для	того	чтобы	в дальнейшем	
проще	было	обращаться	к номерам	системных	вызовов.	В приведенном	ниже	
примере	исходного	кода	константы	номеров	системных	вызовов	легко	узнать	
по	префиксу	NR_.	Применение	таких	констант	NR_syscall	делает	исходный	код	
более	удобным	для	чтения.	Полный	список	символьных	имен	системных	вы
зовов	для	вашей	системы	можно	найти	в файле

/usr/include/asm/unistd_64.h



Эти	имена	будут	использоваться	в приведенной	ниже	программе.	Следует	
отметить,	что	здесь	также	используется	заголовочный	файл	unistd_32.h	для	
обеспечения	обратной	совместимости	с 32битовыми	системными	компо
нентами.

Кроме	того,	созданы	константы	для	флагов	создания,	состояния	и режимов	
доступа	 к  файлам.	 Эти	флаги	 определяют,	 должен	 ли	 файл	 создаваться	 или	
использоваться	существующий	файл	для	добавления	в него	данных,	предна
значен	ли	файл	только	для	чтения,	только	для	записи	и т.	п.	Список	и краткие	
описания	этих	флагов	можно	найти	в файле

/usr/include/asm-generic/fcntl.h

Значения	флагов	в этом	файле	приведены	в восьмеричном	формате	(напри
мер,	O_CREAT = 00000100).	Значение,	начинающееся	с 0x,	является	шестнадцате
ричным,	 а  значение,	начинающееся	 с 0	 без	x, –	 восьмеричное.	Для	удобства	
чтения	к восьмеричному	числу	будет	добавляться	символ	q.

При	 создании	 файла	 необходимо	 обязательно	 определить	 права	 доступа	
к нему.	Напомню,	что	 в  Linux	определяются	права	на	чтение,	 запись	и  вы
полнение	файла	для	пользователя	(владельца),	группы	и прочих.	Общий	об
зор	прав	доступа	и огромное	количество	подробностей	можно	получить	с по
мощью	команды

man 2 open

Права	доступа	к файлу	также	определяются	в восьмеричном	формате	и хо
рошо	 знакомы	 любому	 системному	 администратору	 Linux.	 Для	 сохранения	
логической	согласованности	мы	позаимствуем	символические	имена,	исполь
зуемые	в описанных	выше	файлах.

Пример	 исходного	 кода	 программы	 достаточно	 длинный,	 но	 мы	 будем	
анализировать	программу	постепенно,	шаг	 за	шагом –	такой	подход	можно	
применить,	 используя	 условное	 ассемблирование	 (conditional assembly).	 Это	
дает	возможность	постепенного	анализа	программы	по	ее	последовательным	
фрагментам.

ОбрабОтка файла
В	рассматриваемом	здесь	примере	программы	выполняются	следующие	опе
рации:	

1)	создание	файла,	затем	запись	данных	в этот	файл;
2)	перезапись	(замещение)	части	содержимого	того	же	файла;
3)	добавление	данных	в файл;
4)	запись	данных	в определенную	позицию	файла;
5)	чтение	данных	из	файла;
6)	чтение	данных	из	определенной	позиции	файла;
7)	удаление	файла.
В	листинге	20.1	показан	исходный	код	программы.

Листинг 20.1.	Программа file.asm

; file.asm
section .data
; Выражения, используемые для условного ассемблирования.
      CREATE     equ   1
      OVERWRITE  equ   1
      APPEND     equ   1
      O_WRITE    equ   1
      READ       equ   1
      O_READ     equ   1
      DELETE     equ   1

; Символические имена системных вызовов.
      NR_read    equ   0
      NR_write   equ   1
      NR_open    equ   2
      NR_close   equ   3
      NR_lseek   equ   8
      NR_create  equ   85
      NR_unlink  equ   87

; Флаги создания и состояния файлов.
      O_CREAT    equ   00000100q
      O_APPEND   equ   00002000q

; Режимы доступа.
      O_RDONLY   equ   000000q
      O_WRONLY   equ   000001q
      O_RDWR     equ   000002q

; Режимы при создании файла (права доступа).
      S_IRUSR    equ   00400q      ; Право на чтение для владельца.
      S_IWUSR    equ   00200q      ; Право на запись для владельца.
      NL         equ   0xa
      bufferlen        equ   64
      fileName   db    "testfile.txt",0
      FD         dq    0    ; Дескриптор файла.

      text1 db   "1. Hello...to everyone!",NL,0
      len1  dq   $-text1-1       ; Удаление 0.
      text2 db   "2. Here I am!",NL,0
      len2  dq   $-text2-1       ; Удаление 0.
      text3 db   "3. Alife and kicking!",NL,0
      len3  dq   $-text3-1       ; Удаление 0.
      text4 db   "Adios !!!",NL,0
    len4    dq   $-text4-1

    error_Create db "error creating file",NL,0
    error_Close  db "error closing file",NL,0
    error_Write  db "error writing to file",NL,0
    error_Open   db "error opening file",NL,0
    error_Append db "error appending to file",NL,0
    error_Delete db "error deleting file",NL,0
    error_Read   db "error reading file",NL,0
    error_Print  db "error printing string",NL,0
    error_Position db "error positioning in file",NL,0

    success_Create    db "File created and opened",NL,0
    success_Close     db "File closed",NL,NL,0
    success_Write     db "Written to file",NL,0
    success_Open      db "File opened for R/W",NL,0
    success_Append    db "File opened for appending",NL,0
    success_Delete    db "File deleted",NL,0
    success_Read      db "Reading file",NL,0
    success_Position  db "Positioned in file",NL,0

section .bss
      buffer resb bufferlen
section .text
      global main
main:
      push rbp
      mov  rbp,rsp
%IF CREATE
; СОЗДАНИЕ И ОТКРЫТИЕ ФАЙЛА, ЗАТЕМ ЗАКРЫТИЕ. --------------------
; Создание и открытие файла.
      mov   rdi, fileName
      call  createFile
      mov   qword [FD], rax ; Сохранение дескриптора.
; Запись в файл #1.
      mov   rdi, qword [FD]
      mov   rsi, text1
      mov   rdx, qword [len1]
      call  writeFile
; Закрытие файла.
      mov   rdi, qword [FD]
      call  closeFile
%ENDIF
%IF OVERWRITE
; ОТКРЫТИЕ И ПЕРЕЗАПИСЬ ФАЙЛА, ЗАТЕМ ЗАКРЫТИЕ. -----------------
; Открытие файла.
      mov   rdi, fileName
      call  openFile
      mov   qword [FD], rax     ; Сохранение дескриптора файла.
; Запись в файл #2 – ПЕРЕЗАПИСЬ (ЗАМЕЩЕНИЕ)!
      mov   rdi, qword [FD]
      mov   rsi, text2
      mov   rdx, qword [len2]
      call  writeFile
; Закрытие файла.
      mov   rdi, qword [FD]
      call  closeFile
%ENDIF
%IF APPEND
; ОТКРЫТИЕ И ДОБАВЛЕНИЕ В ФАЙЛ, ЗАТЕМ ЗАКРЫТИЕ. ----------------
; Открытие файла для добавления в него данных.
      mov   rdi, fileName
      call  appendFile
      mov   qword [FD], rax    ; Сохранение дескриптора файла.
; Запись в файл #3 – ДОБАВЛЕНИЕ!
      mov   rdi, qword [FD]
      mov   rsi, text3
      mov   rdx, qword [len3]

      call  writeFile
; Закрытие файла.
      mov   rdi, qword [FD]
      call  closeFile
%ENDIF
%IF O_WRITE
; ОТКРЫТИЕ И ПЕРЕЗАПИСЬ ПО ПОЗИЦИИ СМЕЩЕНИЯ В ФАЙЛЕ, ЗАТЕМ ЗАКРЫТИЕ. ----
; Открытие файла для записи.
      mov   rdi, fileName
      call  openFile
      mov   qword [FD], rax    ; Сохранение дескриптора файла.
; Позиция в файле, определяемая по смещению.
      mov   rdi, qword[FD]
      mov   rsi, qword[len2]  ; Смещение в заданную позицию.
      mov   rdx, 0
      call  positionFile
; Запись в файл в позицию с заданным смещением.
      mov   rdi, qword[FD]
      mov   rsi, text4
      mov   rdx, qword [len4]
      call  writeFile
; Закрытие файла.
      mov   rdi, qword [FD]
      call  closeFile
%ENDIF
%IF READ
; ОТКРЫТИЕ И ЧТЕНИЕ ИЗ ФАЙЛА, ЗАТЕМ ЗАКРЫТИЕ. ----------------
; Открытие файла для чтения.
      mov   rdi, fileName
      call  openFile
      mov  qword [FD], rax    ; Сохранение дескриптора файла.
; Чтение из файла.
      mov   rdi, qword [FD]
      mov   rsi, buffer
      mov   rdx, bufferlen
      call  readFile
      mov   rdi,rax
      call  printString
; Закрытие файла.
      mov  rdi, qword [FD]
      call  closeFile
%ENDIF
%IF O_READ
; ОТКРЫТИЕ И ЧТЕНИЕ В ПОЗИЦИИ, ЗАДАННОЙ СМЕЩЕНИЕМ В ФАЙЛЕ, ЗАТЕМ ЗАКРЫТИЕ. -----
; Открытие файла для чтения.
      mov   rdi, fileName
      call  openFile
      mov   qword [FD], rax     ; Сохранение дескриптора файла.
; Позиция в файле, определяемая по смещению.
      mov   rdi, qword[FD]
      mov   rsi, qword[len2]      ;skip the first line
      mov   rdx, 0
      call  positionFile
; Чтение из файла (в заданной позиции).
      mov   rdi, qword [FD]
      mov   rsi, buffer

      mov   rdx, 10        ; Количество считываемых символов.
      call  readFile
      mov   rdi,rax
      call  printString
; Закрытие файла.
      mov   rdi, qword [FD]
      call  closeFile
%ENDIF
%IF DELETE
; УДАЛЕНИЕ ФАЙЛА. ----------------------------------
; Удаление файла.   ДЛЯ ИСПОЛЬЗОВАНИЯ УБРАТЬ СИМВОЛЫ КОММЕНТАРИЕВ В СЛЕДУЮЩИХ СТРОКАХ.
      mov   rdi, fileName
      call  deleteFile
%ENDIF
leave
ret
; ФУНКЦИИ ОБРАБОТКИ ФАЙЛА. --------------------
;------------------------------------------------
global readFile
readFile:
      mov   rax, NR_read
      syscall          ; rax содержит количество считываемых символов.
      cmp   rax, 0
      jl    readerror
      mov   byte [rsi+rax],0 ; Добавление завершающего нуля.
      mov   rax, rsi
      mov   rdi, success_Read
      push  rax        ; Регистр, сохраняемый вызывающей функцией.
      call  printString
      pop   rax        ; Регистр, сохраняемый вызывающей функцией.
      ret
readerror:
      mov   rdi, error_Read
      call  printString
      ret
;--------------------------------------------------
global deleteFile
deleteFile:
      mov   rax, NR_unlink
      syscall
      cmp   rax, 0
      jl    deleteerror
      mov   rdi, success_Delete
      call  printString
      ret
deleteerror:
      mov   rdi, error_Delete
      call  printString
      ret
;-------------------------------------------------
global appendFile
appendFile:
      mov   rax, NR_open
      mov   rsi,  O_RDWR|O_APPEND
      syscall
      cmp   rax, 0

      jl    appenderror
      mov   rdi, success_Append
      push  rax        ; Регистр, сохраняемый вызывающей функцией.
      call  printString
      pop   rax        ; Регистр, сохраняемый вызывающей функцией.
      ret
appenderror:
      mov   rdi, error_Append
      call  printString
      ret
;------------------------------------------------
global openFile
openFile:
      mov   rax, NR_open
      mov   rsi, O_RDWR
      syscall
      cmp   rax, 0
      jl    openerror
      mov   rdi, success_Open
      push  rax        ; Регистр, сохраняемый вызывающей функцией.
      call  printString
      pop   rax        ; Регистр, сохраняемый вызывающей функцией.
      ret
openerror:
      mov   rdi, error_Open
      call  printString
      ret
;----------------------------------------------
global writeFile
writeFile:
      mov   rax, NR_write
      syscall
      cmp   rax, 0
      jl    writeerror
      mov   rdi, success_Write
      call  printString
      ret
writeerror:
      mov   rdi, error_Write
      call  printString
      ret
;---------------------------------------------
global positionFile
positionFile:
      mov   rax, NR_lseek
      syscall
      cmp   rax, 0
      jl    positionerror
      mov   rdi, success_Position
      call  printString
      ret
positionerror:
      mov   rdi, error_Position
      call  printString
      ret
;---------------------------------------------

global closeFile
closeFile:
      mov   rax, NR_close
      syscall
      cmp   rax, 0
      jl    closeerror
      mov   rdi, success_Close
      call  printString
      ret
closeerror:
      mov   rdi, error_Close
      call  printString
      ret
;------------------------------------------------
global createFile
createFile:
      mov   rax, NR_create
      mov   rsi, S_IRUSR |S_IWUSR
      syscall
      cmp   rax, 0             ; Дескриптор файла в регистре rax.
      jl    createerror
      mov   rdi, success_Create
      push  rax               ; Регистр, сохраняемый вызывающей функцией.
      call  printString
      pop   rax               ; Регистр, сохраняемый вызывающей функцией.
      ret
createerror:
      mov   rdi, error_Create
      call  printString
      ret
; ВЫВОД ИНФОРМАЦИИ О ФАЙЛОВЫХ ОПЕРАЦИЯХ (ОБРАТНОЙ СВЯЗИ).
;--------------------------------------------------------
global printString
printString:
; Счетчик символов.
      mov   r12, rdi
      mov   rdx, 0
strLoop:
      cmp   byte [r12], 0
      je    strDone
      inc   rdx             ; Длина строки в регистре rdx.
      inc   r12
      jmp   strLoop
strDone:
      cmp   rdx, 0          ; Строка отсутствует (длина 0).
      je    prtDone
      mov   rsi,rdi
      mov   rax, 1
      mov   rdi, 1
      syscall
prtDone:
      ret

Обработка файла  151



уСлОвнОе аССемблирОвание
Поскольку	программа	весьма	велика,	для	упрощения	ее	анализа	используется	
условное	ассемблирование	(conditional assembly).	С этой	целью	созданы	различ
ные	переменные,	такие	как	CREATE,	WRITE,	APPEND	и т.	п.	Если	для	такой	перемен
ной	установлено	значение	1,	то	конкретный	фрагмент	исходного	кода,	распо
ложенный	между	строками	%IF 'ИМЯ_ПЕРЕМЕННОЙ'	и %ENDIF,	будет	ассемблирован.	
Если	для	переменной	установлено	значение	0,	то	программа	ассемблирования	
пропустит	этот	фрагмент	кода.	Строки	%IF	и %ENDIF	называются	директивами	
препроцессора	ассемблера	(assembler preprocessor directives).	Начнем	с перемен
ной	CREATE equ 1	и установим	для	всех	других	переменных	значение	0,	затем	ас
семблируем,	запустим	и проанализируем	программу.	Далее	двигаемся	посте
пенно	сверху	вниз.	Продолжаем	с установкой	значений	CREATE equ 1	и OVERWRITE 
equ 1	(все	прочие	переменные	равны	0)	для	второй	сборки	программы	и т.	д.

NASM	 предоставляет	 большой	 набор	 директив	 препроцессора,	 здесь	 мы	
используем	директивы	условного	ассемблирования.	Для	определения	макро
сов,	как	описано	в главе	18,	также	применялись	директивы	препроцессора.	
В главе	4	руководства	по	NASM	вы	найдете	полное	описание	директив	пре
процессора.

инСтрукции для ОбрабОтки файлОв
Начнем	с создания	файла.	Необходимо	поместить	имя	файла	в регистр	rdi	и вы
звать	функцию	createFile.	В функции	createFile	символьная	переменная	NR_cre-
ate	записывается	в регистр	rax,	а в	регистре	rsi	определяются	флаги	для	создания	
файла.	В рассматриваемом	здесь	примере	пользователю	(владельцу)	предостав
ляются	права	на	чтение	и запись,	затем	выполняется	системный	вызов.

Если	по	какимлибо	причинам	файл	не может	быть	создан,	то	функция	cre-
ateFile	 возвращает	 отрицательное	 значение	 в  регистре	 rax,	 и  в	 этом	 случае	
необходимо	вывести	сообщение	об	ошибке.	Если	требуются	подробности,	то	
отрицательное	значение	в rax	определяет	тип	возникшей	ошибки.	Если	файл	
создан,	то	функция	возвращает	дескриптор	файла	в rax.	В вызывающей	про
грамме	этот	дескриптор	файла	сохраняется	в переменной	FD	для	дальнейшей	
работы	с файлом.	Очевидно,	что	необходима	аккуратность	при	сохранении	со
держимого	регистра	rax	перед	вызовом	функции	printString.	Вызов	printString 
уничтожает	 содержимое	 rax,	 поэтому	необходимо	 записать	 его	 в  стек	перед	
вызовом	функции.	В соответствии	с соглашениями	о вызовах	функций	rax	яв
ляется	регистром,	сохраняемым	вызывающей	функцией	(caller	saved).

Далее	в рассматриваемом	коде	в файл	записывается	некоторый	текст,	затем	
файл	закрывается.	Обратите	внимание:	операция	создания	файла	создает	но
вый	файл,	если	файл	с таким	именем	уже	существует,	то	он	будет	удален.

Выполните	 сборку	 и  запустите	 программу	 с  условием	 CREATE equ 1,	 дру
гие	 переменные	 условного	 ассемблирования	 должны	 быть	 равны	 0.	 Затем	 
перейдите	 в  командную	 строку	и проверьте,	 создан	ли	файл	 testfile.txt	 и  со
держится	ли	в нем	сообщение.	Если	требуется	просмотреть	содержимое	файла	
в шестнадцатеричном	формате,	что	иногда	полезно,	то	воспользуйтесь	коман
дой	xxd testfile.txt	в командной	строке.

Продолжайте	постепенно	присваивать	переменным	условного	ассемблиро
вания	значение	1	и проверяйте	по	содержимому	testfile.txt,	что	происходит.

Следует	отметить,	что	в рассматриваемом	здесь	примере	создаются	и ис
пользуются	функции	 без	 пролога	 и  эпилога.	 На  рис.  20.1	 показан	 итоговый	
вывод	программы	при	всех	переменных	условного	ассемблирования,	установ
ленных	в 1.

Рис. 20.1.	Итоговый вывод программы file.asm

резюме
В	этой	главе	вы	узнали:

	� об	операциях	создания,	открытия,	закрытия	и удаления	файла;
	� об	операциях	записи	и добавления	данных	в файл,	а также	об	операции	
записи	в файл	в заданной	позиции;

	� об	операции	чтения	из	файла;
	� о	различных	параметрах	для	работы	с файлом.

Глава 21
Командная строка

Иногда	 необходимо	 при	 запуске	 программы	 из	 командной	 строки	 указать	
аргументы,	которые	будет	использовать	эта	программа.	Это	может	оказаться	
полезным	при	разработке	 собственных	инструментальных	 средств	команд
ной	строки.	Системные	администраторы	постоянно	применяют	инструмен
ты	командной	строки,	потому	что,	как	правило,	такие	инструменты	работают	
быст	рее	в руках	опытного	пользователя.

дОСтуп к аргументам кОманднОй СтрОки
В	примере	программы	в листинге	21.1	показано,	как	можно	получить	доступ	
к аргументам	командной	строки	в ассемблерной	программе.	Здесь	рассмат
риваются	самые	простые	действия:	поиск	аргументов	и их	вывод.

Листинг 21.1.	Программа cmdline.asm

;cmdline.asm
extern printf
section .data
     msg   db      "The command and arguments: ",10,0
     fmt   db      "%s",10,0
section .bss
section .text
     global main
main:
push rbp
mov  rbp,rsp
     mov   r12, rdi     ; Количество аргументов.
     mov   r13, rsi     ; Адрес массива аргументов.
; Вывод заголовка.
     mov   rdi, msg
     call  printf
     mov   r14, 0
; Вывод имени команды и аргументов.
.ploop:               ; Цикл прохода по массиву аргументов и их вывода.
     mov   rdi, fmt
     mov   rsi, qword [r13+r14*8]
     call  printf
     inc   r14
     cmp   r14, r12   ; Достигнуто максимальное количество аргументов?
     jl    .ploop
leave
ret



При	выполнении	этой	программы	количество	аргументов,	включая	имя	са
мой	программы,	сохраняется	в регистре	rdi.	Регистр	rsi	содержит	адрес	памя
ти	массива,	содержащего	адреса	аргументов	командной	строки,	при	этом	пер
вым	 аргументом	 является	 имя	 самой	программы.	Использование	 регистров	
rdi	 и  rsi	 соответствует	 соглашениям	 о  вызовах	функций.	 Напомню,	 что	мы	
работаем	в системе	Linux	и используем	соглашения	о вызовах	функций	System	
V AMD64	ABI.	На других	платформах,	например	в ОС	Windows,	применяются	
иные	соглашения	о вызовах	функций.	Информация	из	этих	регистров	копиру
ется,	потому	что	rdi	и rsi	позже	будут	использоваться	для	функции	printf.

В	 этом	 коде	 существует	 цикл	 для	 прохода	 по	 массиву	 аргументов	 до	 тех	
пор,	пока	не будет	достигнуто	максимальное	число	аргументов.	В цикле	.plo-
op	регистр	r13	указывает	на	массив	аргументов.	Регистр	r14	используется	как	
счетчик	 аргументов.	На  каждой	итерации	цикла	 вычисляется	 адрес	 следую
щего	аргумента	и  записывается	в rsi.	Число	8	 в  выражении	qword [r13+r14*8] 
соответствует	длине указываемых	адресов:	8	байт	×	8	бит	=	64битовый	адрес.	
Регистр r14	на	каждой	итерации	цикла	сравнивается	с регистром	r12,	содержа
щим	общее	количество	аргументов.

На	рис. 21.1	показан	вывод	этой	программы,	вызванной	с некоторыми	слу
чайно	заданными	аргументами.

Рис. 21.1.	Вывод программы cmdline.asm

Отладка прОграммы С аргументами кОманднОй СтрОки
В	 настоящее	 время	 невозможно	 использовать	 SASM	 для	 отладки	 программ	
с аргументами	командной	строки,	поэтому	придется	воспользоваться	отлад
чиком	GDB.	Ниже	показан	один	из	способов	сделать	это:

gdb --args ./cmdline arg1 arg2 abc 5
break main
run
info registers rdi rsi rsp

С	помощью	приведенных	выше	инструкций	можно	проверить,	что	регистр	
rdi	 содержит	 общее	 количество	 аргументов	 (включая	 имя	 самой	 выполняе
мой	команды),	а регистр	rsi	указывает	на	адрес	в старшей	(верхней)	памяти,	
даже	более	старшей,	чем	указатель	на	стек,	как	уже	было	отмечено	в главе 8	
(см. рис. 8.7).	На рис. 21.2	показан	вывод	отладчика	GDB.

Рис. 21.2.	Вывод отладчика GDB с программой cmdline

На	рис. 21.2	видно,	что	массив	с адресами	аргументов	начинается	с адреса	
0x7fffffffde58.	Продолжим	исследования,	чтобы	добраться	до	действительных	
значений	аргументов.	Адрес	первого	аргумента	можно	найти	с помощью	сле
дующей	команды:

x/1xg 0x7fffffffde58

Здесь	выполняется	запрос	к одному	огромному	слову	(8	байт)	в шестнадца
теричном	формате	по	адресу	0x7fffffffde58.	Ответ	показан	на	рис. 21.3.

Рис. 21.3.	GDB возвращает адрес первого аргумента

Теперь	посмотрим,	что	находится	по	этому	адресу	(см.	рис. 21.4).

x/s 0x7fffffffe204

Рис. 21.4.	GDB выводит значение первого аргумента

Разумеется,	 это	первый	аргумент,	т.  е.	имя	 самой	выполняемой	команды.	
Чтобы	найти	второй	аргумент,	нужно	увеличить	адрес	0x7fffffffde58	на	8	байт,	
чтобы	 получить	 0x7fffffffde60,	 затем	 определить	 адрес	 второго	 аргумента	
и т. д.	На рис. 21.5	показан	результат.

Рис. 21.5.	GDB выводит значение второго аргумента

Таким	способом	можно	отлаживать	и проверять	программы	с аргументами	
командной	строки.	

резюме
В	этой	главе	вы	узнали,	как:

	� получить	доступ	к аргументам	командной	строки;
	� использовать	регистры	для	аргументов	командной	строки;
	� отлаживать	программы	с аргументами	командной	строки.

Глава 22
Использование ассемблера 

в коде C

В	предыдущих	главах	иногда	для	удобства	использовались	функции	языка	C,	
например	стандартная	функция	вывода	printf	или	разработанная	нами	версия	
printb.	В этой	главе	будет	показано,	как	использовать	ассемблерные	функции	
в коде	на	языке	C.	При	этом	сразу	же	становится	очевидным	важное	значение	
соглашений	о вызовах	функций.	В этой	главе	используются	соглашения	о вы
зове	функций	System	V AMD64	ABI,	потому	что	мы	работаем	в системе	Linux.	
Для	ОС	Windows	определены	другие	соглашения	о вызовах	функций.	Если	вы	
внимательно	изучили	предыдущие	главы	и примеры	исходного	кода	в них,	то	
эта	глава	не вызовет	у вас	затруднений.

СОздание файла иСхОднОгО кОда на языке C
Почти	весь	ассемблерный	код	должен	быть	хорошо	знаком	вам	по	предыдущим	
главам.	Новым	является	только	код	программы	на	языке	C.	Здесь	вычисляются	
площадь	и периметр	прямоугольника	и круга.	Затем	принимается	некоторая	
строка	и реверсируется	(порядок	букв	в ней	меняется	на	обратный),	наконец,	
вычисляется	сумма	элементов	массива,	элементы	массива	удваиваются,	и вы
числяется	сумма	удвоенных	элементов	массива.	Рассмотрим	различные	фай
лы	исходного	кода.

Начнем	с файла	исходного	кода	на	языке	C,	показанного	в листинге	22.1.

Листинг 22.1.	Программа fromc.c

// fromc.c

#include <stdio.h>
#include <string.h>

extern int rarea(int, int);
extern int rcircum(int, int);
extern double carea(double);
extern double ccircum(double);
extern void sreverse(char *, int );
extern void adouble(double [], int );
extern double asum(double [], int );



int main()
{
      char rstring[64];
      int side1, side2, r_area, r_circum;
      double radius, c_area, c_circum;
      double darray[] = {70.0, 83.2, 91.5, 72.1, 55.5};
      long int len;
      double sum;

// Вызов ассемблерной функции с аргументами типа int.
      printf("Compute area and circumference of a rectangle\n");
      printf("Enter the length of one side : \n");
      scanf("%d", &side1 );
      printf("Enter the length of the other side : \n");
      scanf("%d", &side2 );

      r_area = rarea(side1, side2);
      r_circum = rcircum(side1, side2);

      printf("The area of the rectangle = %d\n", r_area);
      printf("The circumference of the rectangle = %d\n\n", r_circum);

// Вызов ассемблерной функции с аргументом типа double (float).
      printf("Compute area and circumference of a circle\n");
      printf("Enter the radius : \n");
      scanf("%lf", &radius);

      c_area = carea(radius);
      c_circum = ccircum(radius);

      printf("The area of the circle = %lf\n", c_area);
      printf("The circumference of the circle = %lf\n\n", c_circum);

// Вызов ассемблерной функции со строковым аргументом.
      printf("Reverse a string\n");
      printf("Enter the string : \n");
      scanf("%s", rstring);
      printf("The string is = %s\n", rstring);
      sreverse(rstring,strlen(rstring));
      printf("The reversed string is = %s\n\n", rstring);

// Вызов ассемблерной функции с аргументом-массивом.
      printf("Some array manipulations\n");
      len = sizeof (darray) / sizeof (double);

      printf("The array has %lu elements\n",len);
      printf("The elements of the array are: \n");
      for (int i=0;i<len;i++){
            printf("Element %d = %lf\n",i, darray[i]);
      }

      sum = asum(darray,len);
      printf("The sum of the elements of this array = %lf\n", sum);

      adouble(darray,len);
      printf("The elements of the doubled array are: \n");
      for (int i=0;i<len;i++){
            printf("Element %d = %lf\n",i, darray[i]);
      }

      sum = asum(darray,len);
      printf("The sum of the elements of this doubled array = %lf\n", sum);
      return 0;
}

СОздание файла иСхОднОгО кОда на аССемблере
Начнем	 с  объявлений	 ассемблерных	функций.	Это	 внешние	функции,	 и для	
них	объявляются	типы	данных	возвращаемых	значений	и аргументов.

Программа	будет	предлагать	пользователю	ввести	большинство	использу
емых	данных,	за	исключением	массива,	значения	в котором	для	удобства	мы	
определим	предварительно.

В	листингах	с 22.2	по	22.7	показаны	соответствующие	ассемблерные	функции.

Листинг 22.2.	Функция rect.asm

;rect.asm
section .data
section .bss
section .text
global rarea
rarea:
      section .text
            push  rbp
            mov   rbp, rsp
            mov   rax, rdi
            imul  rsi
            leave
            ret
global rcircum
rcircum:
      section .text
            push  rbp
            mov   rbp, rsp
            mov   rax, rdi
            add   rax, rsi
            imul  rax, 2
            leave
            ret

Листинг 22.3.	Функция circle.asm

;circle.asm
section .data
      pi   dq    3.141592654
section .bss
section .text
global carea
carea:
      section .text
            push  rbp
            mov   rbp, rsp
            movsd xmm1, qword [pi]
            mulsd xmm0,xmm0       ; Радиус в регистре xmm0.
            mulsd xmm0, xmm1
            leave
            ret
global ccircum
ccircum:
      section .text
            push  rbp

            mov   rbp, rsp
            movsd xmm1, qword [pi]
            addsd xmm0,xmm0       ; Радиус в регистре xmm0.
            mulsd xmm0, xmm1
            leave
            ret

Листинг 22.4.	Функция sreverse.asm

;sreverse.asm
section .data
section .bss
section .text
global sreverse
sreverse:
push  rbp
mov   rbp, rsp
pushing:
      mov rcx, rsi
      mov rbx, rdi
      mov r12, 0
      pushLoop:
            mov rax, qword [rbx+r12]
            push rax
            inc r12
            loop pushLoop
popping:
      mov rcx, rsi
      mov rbx, rdi
      mov r12, 0
      popLoop:
            pop rax
            mov byte [rbx+r12], al
            inc r12
            loop popLoop
mov rax, rdi
leave
ret

Листинг 22.5.	Функция asum.asm

; asum.asm
section .data
section .bss
section .text
global asum
asum:
      section .text
; Вычисление суммы.
            mov   rcx, rsi   ; Длина массива.
            mov   rbx, rdi   ; Адрес массива.
            mov   r12, 0
            movsd xmm0, qword [rbx+r12*8]
            dec rcx    ; На один проход в цикле меньше, так как
                       ; первый элемент уже в регистре xmm0.
      sloop:
            inc r12

            addsd xmm0, qword [rbx+r12*8]
            loop sloop
ret         ; Сумма возвращается в регистре xmm0.

Листинг 22.6.	Функция adouble.asm

; adouble.asm
section .data
section .bss
section .text
global adouble
adouble:
      section .text
; Удвоение элементов.
            mov   rcx, rsi     ; Длина массива.
            mov   rbx, rdi     ; Адрес массива.
            mov   r12, 0
      aloop:
            movsd xmm0, qword [rbx+r12*8]      ; Взять элемент.
            addsd xmm0,xmm0                    ; Удвоить его.
            movsd qword [rbx+r12*8], xmm0      ; Поместить результат обратно в массив.
            inc r12
            loop aloop
ret

Листинг 22.7.	makefile

fromc: fromc.c rect.o circle.o sreverse.o adouble.o asum.o
      gcc -o fromc fromc.c rect.o circle.o sreverse.o \
      adouble.o asum.o -no-pie
rect.o: rect.asm
      nasm -f elf64 -g -F dwarf rect.asm -l rect.lst
circle.o: circle.asm
      nasm -f elf64 -g -F dwarf circle.asm -l circle.lst
sreverse.o: sreverse.asm
      nasm -f elf64 -g -F dwarf sreverse.asm -l sreverse.lst
adouble.o: adouble.asm
      nasm -f elf64 -g -F dwarf adouble.asm -l adouble.lst
asum.o: asum.asm
      nasm -f elf64 -g -F dwarf asum.asm -l asum.lst

В	 ассемблерном	 коде	 нет	 ничего	 особенного –	 просто	 нужно	 вниматель
но	 следить	 за	 типами	 данных	 переменных,	 принимаемых	 из	 вызывающей	
Cпрограммы.	Ассемблерные	функции	 принимают	 аргументы	 из	 вызываю
щей	программы	и сохраняют	их	в регистрах	в соответствии	с соглашениями	
о  вызовах	 функций.	 В  вызывающую	 программу	 результаты	 возвращаются	
в регистре	rax	(целочисленное	значение)	или	в регистре	xmm0	(значение	с пла
вающей	точкой).	Теперь	вы	можете	разрабатывать	собственные	библиотеки	
функций	для	использования	в программах	на	ассемблере	или	на	C,	а благо
даря	соглашениям	о вызовах	функций	не нужно	беспокоиться	о том,	как	пере
даются	аргументы.	Просто	внимательно	следите	за	использованием	правиль
ных	типов	данных.

Обратите	внимание:	в makefile	применяется	символ	обратного	слеша	(\)	для	
разделения	длинных	строк,	а для	смещения	инструкций	вправо	используется	
символ	табуляции.

На	рис. 22.1	показан	вывод	этой	программы.

Рис. 22.1.	Вывод программы fromc.c

резюме
В	этой	главе	вы	узнали:

	� как	вызывать	ассемблерные	функции	из	исходного	кода	на	языке	более	
высокого	уровня,	в данном	случае	из	исходного	кода	на	языке	C;

	� о	важном	значении	соглашений	о вызовах	функций.

Глава 23
Встроенный ассемблер

В	этой	главе	используется	язык	программирования	C для	объяснения	смыс
ла	встроенного	 (inline)	 ассемблерного	кода.	Существует	возможность	писать	
инструкции	 на	 ассемблере	 в  исходном	 коде	 Cпрограммы.	 В  большинстве	
случаев	такой	подход	не рекомендуется,	так	как	 современные	компиляторы	
языка	 C  настолько	 хорошо	 спроектированы	 и  реализованы,	 что	 вы	должны	
быть	весьма	высокопрофессиональным	программистом	на	ассемблере,	чтобы	
действительно	улучшить	производительность	кода	на	C.	В действительности	
применение	встроенного	ассемблерного	кода	делает	более	сложной	для	ком
пиляторов	C или	C++	оптимизацию	кода	с такими	ассемблерными	вставками.	

Кроме	того,	компилятор	языка	C не найдет	какиелибо	ошибки	в таких	ас
семблерных	инструкциях –	вам	придется	самостоятельно	искать	и исправлять	
их.	 Более	того,	доступ	 к  памяти	и  регистрам,	 используемым	Cпрограммой,	
может	создавать	дополнительные	потенциальные	опасности.	Тем	не менее	во	
многих	статьях	в интернете	встроенный	ассемблерный	код	используется	для	
описания	функциональности	на	более	низком	уровне,	так	что	умение	читать	
такой	код	может	оказаться	полезным.

Существует	два	типа	встроенного	ассемблерного	кода:	простой	и расширен
ный.

прОСтОй вСтрОенный аССемблерный кОд
Начнем	 с  примера	 простого	 встроенного	 ассемблерного	 кода.	 См.	 листин
ги 23.1	и 23.2.

Листинг 23.1.	Программа inline1.c

// inline1.c
#include <stdio.h>

int x=11, y=12, sum, prod;
int subtract(void);
void multiply(void);

int main(void)
{
      printf("The numbers are %d and %d\n",x,y);
      __asm__(
            ".intel_syntax noprefix;"



            "mov rax,x;"
            "add rax,y;"
            "mov sum,rax"
            );
      printf("The sum is %d.\n",sum);
      printf("The difference is %d.\n",subtract());
      multiply();
      printf("The product is %d.\n",prod);
}

int subtract(void)
{
      __asm__(
            ".intel_syntax noprefix;"
            "mov rax,x;"
            "sub rax,y"                // Возвращаемое значение в rax.
            );
}

void multiply(void)
{
      __asm__(
            ".intel_syntax noprefix;"
            "mov rax,x;"
            "imul rax,y;"
            "mov prod,rax"   // Нет возвращаемого значения, результат в prod.
            );
}

Листинг 23.2.	makefile

# makefile inline1.c
inline1: inline1.c
      gcc -o inline1 inline1.c -masm=intel -no-pie

Обратите	внимание	на	дополнительный	параметр	компиляции	в makefile:	
-masm=intel.	Этот	параметр	обязателен	при	использовании	встроенного	ассемб
лерного	кода.

В	рассматриваемом	здесь	примере	показан	вызов	простой	(basic)	встроен
ной	 ассемблерной	 подпрограммы.	 В  основную	 программу	 main	 добавляются	
две	переменные,	затем	вызывается	функция	вычитания	этих	двух	перемен
ных.	Далее	 вызывается	другая	функция	для	 умножения	тех	же	переменных.	
Если	требуется	доступ	к таким	переменным	в простой	встроенной	ассемблер
ной	подпрограмме,	то	необходимо	объявить	их	как	глобальные,	т. е.	вынести	их	
объявление	за	пределы	любой	функции.	Если	такие	переменные	не объявлены	
глобальными,	то	компилятор	gcc	сообщит,	что	не может	найти	их.	Но глобаль
ные	 переменные	 являются	 потенциальным	 источником	 ошибок,	 таких	 как	
конфликты	имен.	Кроме	того,	при	изменении	регистров	в ассемблерном	коде,	
возможно,	потребуется	их	сохранение	перед	вызовом	встроенного	ассемблер
ного	кода	и восстановление	их	исходных	значений	при	выходе	из	встроенного	
ассемблерного	кода,	иначе	возникает	опасность	аварийного	завершения	про
граммы.	Регистры,	изменяемые	встроенным	ассемблерным	кодом,	называют
ся	затираемыми	(clobber)	регистрами.

В	ассемблерной	части,	заключенной	в блоке	__asm__(…),	первая	инструкция	
определяет,	что	здесь	используется	синтаксис	ассемблера	Intel	без	префиксов.	
(Вспомните	обсуждение	вариантов	синтаксиса	Intel	и AT&T	в главе	3.)	Затем	
используются	инструкции	ассемблера	как	обычно,	завершаемые	символом	; 
или	\n.	Последнюю	ассемблерную	инструкцию	не обязательно	завершать	сим
волом	;	или	\n.	Обратите	внимание	на	использование	глобальных	переменных.	
Нам	повезло,	потому	что	затирание	(clobbering)	регистров	не привело	к аварий
ному	завершению	программы.	Чтобы	избежать	затирания	регистров	и приме
нения	 глобальных	 переменных,	 необходимо	 воспользоваться	 расширенным	
встроенным	ассемблерным	кодом,	рассматриваемым	в следующем	разделе.

На	рис. 23.1	показан	вывод	этой	программы.

Рис. 23.1.	Вывод программы inline1.c

раСширенный вСтрОенный аССемблерный кОд
В	листингах	23.3	и 23.4	показан	пример	расширенного	встроенного	ассемблер
ного	кода.

Листинг 23.3.	Программа inline2.c

// inline2.c

#include <stdio.h>

      int a=12;    // Глобальные переменные.
      int b=13;
      int bsum;

int main(void)
{
printf("The global variables are %d and %d\n",a,b);

__asm__(
      ".intel_syntax noprefix\n"
      "mov rax,a \n"
      "add rax,b \n"
      "mov bsum,rax \n"
      :::"rax"
      );

printf("The extended inline sum of global variables is %d.\n\n", bsum);
int x=14,y=16, esum, eproduct, edif;         // Локальные переменные.
printf("The local variables are %d and %d\n",x,y);

__asm__(
      ".intel_syntax noprefix;"
      "mov rax,rdx;"
      "add rax,rcx;"
      :"=a"(esum)
      :"d"(x), "c"(y)
      );
     printf("The extended inline sum is %d.\n", esum);

__asm__(
      ".intel_syntax noprefix;"
      "mov rbx,rdx;"
      "imul rbx,rcx;"
      "mov rax,rbx;"
      :"=a"(eproduct)
      :"d"(x), "c"(y)
      :"rbx"
      );
     printf("The extended inline product is %d.\n", eproduct);

__asm__(
      ".intel_syntax noprefix;"
      "mov rax,rdx;"
      "sub rax,rcx;"
      :"=a"(edif)
      :"d"(x), "c"(y)
      );
     printf("The extended inline asm difference is %d.\n", edif);
}

Листинг 23.4.	makefile

# makefile inline2.c
inline2: inline2.c
      gcc -o inline2 inline2.c -masm=intel -no-pie

Здесь	ассемблерные	инструкции	выглядят	подругому –	для	них	используется	
шаблон,	показанный	ниже:

      asm (
           ассемблерный код
           : операнды вывода                  /* необязательные */
           : операнды ввода                   /* необязательные */
           : список затираемых регистров      /* необязательный */
           );

После	 ассемблерного	 кода	 записывается	 используемая	 дополнительная,	
но не обязательная	информация.	В качестве	примера	рассмотрим	встроенный	
код	вычисления	произведения	(повторно	воспроизведенный	ниже):

    __asm__(
           ".intel_syntax noprefix;"
           "mov rbx,rdx;"
           "imul rbx,rcx;"
           "mov rax,rbx;"

           :"=a"(eproduct)
           :"d"(x), "c"(y)
           :"rbx"
           );
          printf("The extended inline product is %d.\n", eproduct);

Каждая	дополнительная	(но	не обязательная)	строка	начинается	с символа	
двоеточия	 (:),	 при	 этом	 непременно	 должен	 соблюдаться	 порядок	 инструк
ций.	 Элементы	 a,	 d  и	 c  называются	 регистровыми	 ограничениями	 (register 
constraints) –	они	отображаются	в регистры	rax,	rdx	и rcx	соответственно.	Ниже	
показано,	как	регистровые	ограничения	отображаются	в соответствующие	ре
гистры.

a -> rax, eax, ax, al
b -> rbx, ebx, bx, bl
c -> rcx, ecx, cx, cl
d -> rdx, edx, dx, dl
S -> rsi, esi, si
D -> rdi, edi, di
r -> любой регистр

Выражение	 :"=a"(eproduct)	 в  первой	дополнительной	 строке	 означает,	 что	
вывод	должен	находиться	в регистре	rax,	а rax	ссылается	на	переменную	eprod-
uct.	Регистр	rdx	ссылается	на	x,	регистр	rcx	ссылается	на	y,	т. е.	на	переменные	
ввода.

В	 завершение	 следует	 отметить,	 что	 регистр	 rbx	 считается	 затираемым	
в приведенном	коде,	поэтому	необходимо	восстанавливать	его	исходное	зна
чение,	потому	что	он	был	объявлен	в списке	затираемых	(clobbered)	регистров.	
В рассматриваемом	здесь	примере	оставление	этого	регистра	в затертом	со
стоянии	не приводит	к аварийному	завершению	программы,	но	такой	подход	
применен	в программе	только	в демонстрационных	целях.	Намного	больше	
информации	о встроенном	ассемблерном	коде	можно	найти	в интернете,	но,	
как	уже	было	отмечено	ранее,	необходимо	использовать	встроенный	ассемб
лер	только	в особых	случаях.	Также	следует	помнить	о том,	что	применение	
встроенного	ассемблерного	кода	сделает	ваш	код	на	языке	C менее	переноси
мым.	См.	рис. 23.2.

Рис. 23.2.	Вывод программы inline2.c

В	следующих	главах	будет	рассматриваться	использование	ассемблера	в ОС	
Windows.	Полезно	знать	о том,	что	встроенный	ассемблерный	код	не поддер
живается	для	процессоров	x64	в Visual	Studio,	поддержка	обеспечивается	толь
ко	для	процессоров	x86.	Но компилятор	gcc	не имеет	такого	ограничения.

резюме
В	этой	главе	вы	узнали	об	использовании:

	� простого	встроенного	ассемблерного	кода;
	� расширенного	встроенного	ассемблерного	кода.

Глава 24
Строки

С	точки	 зрения	 человека	 строки	 обычно	 представляются	 в  виде	 последова
тельности	символов,	образующих	слова	и фразы,	которые	мы	можем	понять.	
Но в языке	ассемблера	любой	список	или	массив	непрерывных	локаций	па
мяти	считается	строкой	вне зависимости	от	того,	может	ли	понять	ее	человек.	
Ассемблер	предоставляет	несколько	мощных	инструкций	для	обработки	таких	
блоков	данных	 эффективным	 способом.	 В  приведенных	ниже	 примерах	 ис
пользуются	читаемые	символы,	но	следует	помнить,	что	в действительности	
ассемблер	не заботится	о том,	чтобы	символы	были	читаемыми.	Здесь	будет	
показано,	как	можно	перемещать	строки	в другие	локации,	как	сканировать	
и сравнивать	строки.

Какими	бы	мощными	ни	были	эти	инструкции	обработки	строк,	более	эф
фективная	 функциональность	 будет	 продемонстрирована	 при	 обсуждении	
SIMDинструкций	в следующих	главах.	Но сейчас	мы	начнем	с самых	простых	
инструкций.	

ОбрабОтка СтрОк
В	листинге	24.1	показан	пример	исходного	кода.

Листинг 24.1.	Программа move_strings.asm

; move_strings.asm
%macro prnt 2
    mov     rax, 1       ; 1 = запись.
    mov     rdi, 1       ; 1 = в стандартный поток вывода stdout.
    mov     rsi, %1
    mov     rdx, %2
    syscall
      mov rax, 1
      mov rdi, 1
      mov rsi, NL
      mov rdx, 1
    syscall
%endmacro

section .data
      length       equ 95
      NL db 0xa



      string1 db "my_string of ASCII:"
      string2 db 10,"my_string of zeros:"
      string3 db 10,"my_string of ones:"
      string4 db 10,"again my_string of ASCII:"
      string5 db 10,"copy my_string to other_string:"
      string6 db 10,"reverse copy my_string to other_string:"
section .bss
      my_string  resb  length
      other_string resb length
section .text
      global main
main:
push rbp
mov  rbp, rsp
;--------------------------------------------------
; Заполнение строки видимыми ascii-символами.
            prnt string1,18
            mov rax,32
            mov rdi,my_string
            mov rcx, length
str_loop1: mov byte[rdi], al          ; Простой метод.
            inc rdi
            inc al
            loop str_loop1
            prnt my_string,length
;--------------------------------------------------
; Заполнение строки ascii-символами 0.
            prnt string2,20
            mov rax,48
            mov rdi,my_string
            mov rcx, length
str_loop2: stosb                  ; Здесь уже не нужна инструкция inc rdi.
            loop str_loop2
            prnt my_string,length
;--------------------------------------------------
; Заполнение строки ascii-символами 1.
            prnt string3,19
            mov rax, 49
            mov rdi,my_string
            mov rcx, length
            rep stosb  ; Здесь уже не нужна инструкция inc rdi и не нужен цикл.
            prnt my_string,length
;--------------------------------------------------
; Повторное заполнение строки видимыми ascii-символами.
            prnt string4,26
            mov rax,32
            mov rdi,my_string
            mov rcx, length
str_loop3:  mov byte[rdi], al         ; Простой метод.
            inc rdi
            inc al
            loop str_loop3
            prnt my_string,length
;--------------------------------------------------
; Копирование строки my_string в другую строку other_string.
            prnt string5,32

            mov rsi,my_string         ; Источник в rsi.
            mov rdi,other_string      ; Цель в rdi.
            mov rcx, length
            rep movsb
            prnt other_string,length
;--------------------------------------------------
; Копирование в обратном порядке строки my_string в другую строку other_string.
            prnt string6,40
            mov rax, 48               ; Очистка строки other_string.
            mov rdi,other_string
            mov rcx, length
            rep stosb
            lea rsi,[my_string+length-4]
            lea rdi,[other_string+length]
            mov rcx, 27     ; Копирование только 27-1 символов.
            std             ; std устанавливает флаг DF, cld сбрасывает флаг DF.
            rep movsb
            prnt other_string,length
leave
ret

В	 этой	 программе	 применяется	 макрос	 (более	 подробно	 о  макросах	 см.	
главу  18)	 для	 вывода	 строк,	 но	можно	 было	 бы	 применить	функцию	 языка	
C printf,	которую	мы	использовали	многократно	в предыдущих	примерах.

Мы	начинаем	с создания	строки,	содержащей	95	видимых	символов	из	таб
лицы	ASCII, –	код	первого	символа	строки	32	(пробел),	код	последнего	симво
ла 126	(тильда	~).	В этих	символах	нет	ничего	необычного.	Сначала	выводится	
заголовок,	потом	ASCIIкод	первого	символа	записывается	в регистр	rax,	а ре
гистр	rdi	указывает	на	адрес	в памяти	строки	my_string.	Затем	длина	формиру
емой	строки	передается	в регистр	rcx	для	использования	в цикле.	В этом	цикле	
один	ASCIIкод	копируется	из	регистра	al	в строку	my_string,	затем	выполня
ется	переход	к следующему	ASCIIкоду,	который	записывается	по	очередному	
адресу	памяти	в my_string	и т.	д.	После	завершения	цикла	выводится	получен
ная	строка.	И здесь	нет	ничего	нового.

В	следующей	части	программы	все	содержимое	строки	my_string	заменяет
ся	полностью	на	символы	0	(ASCIIкод	48).	Длина	строки	снова	записывается	
в регистр	rcx	для	организации	цикла.	Далее	используется	инструкция	stosb	для	
записи	символов	1	(ASCIIкод	49)	в строку	my_string.	Для	инструкции	stosb	не
обходим	только	начальный	адрес	строки	в регистре	rdi	и записываемый	сим
вол	в регистре	rax,	и stosb	автоматически	переходит	к следующему	адресу	па
мяти	на	каждом	шаге	цикла.	В этом	случае	можно	не заботиться	об	увеличении	
содержимого	регистра	rdi.

Далее	 в  программе	 принимаются	 дополнительные	 меры	 по	 исключению	
цикла	с использованием	регистра	rcx.	Применяется	инструкция	rep stosb	для	
повторения	stosb	заданное	количество	раз.	Количество	повторений	содержит
ся	в регистре	rcx.	Это	весьма	эффективный	способ	инициализации	памяти.

Продолжаем	 работать	 с  содержимым	 памяти.	 Строго	 говоря,	 здесь	 будет	
выполняться	копирование	блоков	памяти,	а не перемещение	их	содержимо
го.	 Сначала	 строка	 снова	инициализируется	ASCIIкодами	 видимых	 симво
лов.	Можно	было	бы	оптимизировать	эту	часть	кода,	применив	макрос	или	

функцию	вместо	простого	повторения	фрагмента	исходного	кода.	Затем	на
чинается	копирование	строки/блока	памяти:	из	my_string	в other_string.	Адрес	
строкиисточника	помещается	в регистр	rsi,	 адрес	целевой	строки	записы
вается	в регистр	rdi.	Это	легко	запомнить,	так	как	буква	s в	имени	регистра	
rsi	обозначает	source	(источник),	а буква	d в	имени	регистра	rdi –	destination	
(цель).	Затем	выполняется	инструкция	rep movsb,	и все	сделано.	Инструкция	rep 
продолжает	копирование	до	тех	пор,	пока	содержимое	rcx	не станет	равным	0.

В	 последней	 части	программы	выполняется	 перемещение	 (копирование)	
содержимого	памяти	в обратном	порядке.	Эта	концепция	может	показаться	
несколько	более	сложной,	поэтому	ниже	объясняются	некоторые	подробно
сти.	При	использовании	инструкции	movsb	учитывается	значение	DF	(флага	на
правления).	Если	DF=0,	то	значения	в регистрах	rsi	и rdi	увеличиваются	на	1,	
указывая	на	следующий	более	старший	адрес	памяти.	Если	DF=1,	то	значения	
rsi	и rdi	уменьшаются	на	1,	указывая	на	следующий	более	младший	адрес	па
мяти.	Это	означает,	что	в рассматриваемом	здесь	примере	при	DF=1	необходи
мо,	чтобы	регистр	rsi	указывал	на	самый	старший	адрес	памяти	в копируемом	
блоке	и уменьшался,	начиная	с этого	адреса.	При	копировании	предполагается	
«проход	в обратном	порядке»,	т. е.	уменьшение	rsi	и rdi	на	каждом	шаге	цик
ла.	Будьте	внимательны:	уменьшаются	оба	регистра	rsi	и rdi,	так	как	невоз
можно	использовать	флаг	DF	для	увеличения	одного	регистра	и уменьшения	
другого	(т. е.	для	реверсирования	строки).	В рассматриваемом	здесь	примере	
копируется	не вся	строка	полностью,	а только	алфавитные	символы	в нижнем	
регистре,	которые	размещаются	в старших	адресах	памяти	в целевой	строке.	
Инструкция	lea rsi, [my_string+length-4]	загружает	эффективный	адрес	стро
ки	my_string	в регистр	rsi	и пропускает	четыре	символа,	которые	не являются	
алфавитными.	Флаг	DF	можно	установить	в 1	инструкцией	std	и сбросить	в 0	
инструкцией	cld.	Затем	вызывается	мощная	инструкция	rep movsb,	и задача	вы
полнена.

Почему	 в  регистр	 rcx	 записывается	 значение	 27,	 хотя	 обрабатываются	
26 символов?	Оказывается,	инструкция	rep	уменьшает	rcx	на	1	перед	выполне
нием	любых	операций	внутри	цикла.	Это	можно	проверить	с помощью	отлад
чика,	например	SASM.	Перед	началом	отладки	закомментируйте	все	ссылки	на	
макрос	prnt,	чтобы	избежать	проблем.	Вы увидите,	что	SASM	позволяет	войти	
в цикл	rep	и проверить	память	и регистры.	Разумеется,	можно	также	обратить
ся	к руководствам	Intel	для	получения	более	подробной	информации	об	ин
струкции	rep –	в подразделе	Operation	вы	найдете	описание,	приблизительно	
соответствующее	приведенному	ниже:

IF AddressSize = 16
    THEN
            Use CX for CountReg;
            Implicit Source/Dest operand for memory use of SI/DI;
    ELSE IF AddressSize = 64
            THEN Use RCX for CountReg;
            Implicit Source/Dest operand for memory use of RSI/RDI;
    ELSE
            Use ECX for CountReg;
            Implicit Source/Dest operand for memory use of ESI/EDI;

FI;
WHILE CountReg =/ 0
        DO
                Service pending interrupts (if any);
                Execute associated string instruction;
                CountReg ← (CountReg – 1);
                IF CountReg = 0
                    THEN exit WHILE loop; FI;
                IF (Repeat prefix is REPZ or REPE) and (ZF = 0)
                or (Repeat prefix is REPNZ or REPNE) and (ZF = 1)
                    THEN exit WHILE loop; FI;
        OD;

Здесь	выражение	CountReg ← (CountReg – 1);	означает,	что	счетчик	будет	умень
шен	в первую	очередь.	Изучение	внутреннего	порядка	работы	инструкций	мо
жет	оказаться	полезным	для	полного	понимания	их	поведения.	Заключитель
ное	замечание:	инструкции	stosb	и movsb	работают	с байтами,	но	существуют	
также	инструкции	stosw,	movsw,	stosd	и movsd	для	работы	со	словами	и двойными	
словами	 соответственно,	 а  значения	 регистров	 rsi	 и  rdi	 увеличиваются	 или	
уменьшаются	соответствующим	образом:	на	1	для	байтов,	на	2	для	слов,	на	4	
для	двойных	слов.

На	рис. 24.1	показан	вывод	этого	примера	программы.

Рис. 24.1.	Вывод программы move_strings.asm

Сравнение и СканирОвание СтрОк
Перемещение	и копирование	строк –	это	важные	операции,	но	не менее	важна	
возможность	анализа	строк.	В примере	исходного	кода,	показанного	в листин
ге	24.2,	используется	инструкция	cmpsb	для	сравнения	двух	строк,	а также	ин
струкция	scasb	для	поиска	заданного	символа	в строке.

Листинг 24.2.	Программа strings.asm

; strings.asm
extern printf
section .data

      string1    db "This is the 1st string.",10,0
      string2    db "This is the 2nd string.",10,0
      strlen2    equ $-string2-2
      string21   db "Comparing strings: The strings do not differ.",10,0
      string22   db "Comparing strings: The strings differ, "
                 db "starting at position: %d.",10,0
      string3    db "The quick brown fox jumps over the lazy dog.",0
      strlen3    equ $-string3-2
      string33   db "Now look at this string: %s",10,0
      string4    db "z",0
      string44   db "The character '%s' was found at position: %d.",10,0
      string45   db "The character '%s' was not found.",10,0
      string46   db "Scanning for the character '%s'.",10,0
section .bss
section .text
      global main
main:
      push   rbp
      mov    rbp,rsp
; Вывод 2 строк.
      xor    rax,rax
      mov    rdi, string1
      call   printf
      mov    rdi, string2
      call   printf
; Сравнение 2 строк. -----------------------------------------------
      lea    rdi,[string1]
      lea    rsi,[string2]
      mov    rdx, strlen2
      call   compare1
      cmp    rax,0
      jnz    not_equal1
; Строки одинаковы, вывод.
      mov    rdi, string21
      call   printf
      jmp    otherversion
; Строки не одинаковы, вывод.
not_equal1:
      mov    rdi, string22
      mov    rsi, rax
      xor    rax,rax
      call   printf
; Сравнение 2 строк, другая версия. ------------------------------
otherversion:
      lea    rdi,[string1]
      lea    rsi,[string2]
      mov    rdx, strlen2
      call   compare2
      cmp    rax,0
      jnz    not_equal2
; Строки одинаковы, вывод.
      mov    rdi, string21
      call   printf
      jmp    scanning
; Строки не одинаковы, вывод.
not_equal2:

      mov    rdi, string22
      mov    rsi, rax
      xor    rax,rax
      call   printf
; Поиск символа (сканирование) в строке. -------------------------------
; Сначала вывод всей строки.
      mov    rdi,string33
      mov    rsi,string3
      xor    rax,rax
      call   printf
; Затем вывод искомого аргумента, которым может быть только 1 символ.
      mov    rdi,string46
      mov    rsi,string4
      xor    rax,rax
      call   printf
scanning:
      lea    rdi,[string3]     ; Строка.
      lea    rsi,[string4]     ; Искомый аргумент.
      mov    rdx, strlen3
      call   cscan
      cmp    rax,0
      jz     char_not_found
; Символ найден, вывод.
      mov    rdi,string44
      mov    rsi,string4
      mov    rdx,rax
      xor    rax,rax
      call   printf
      jmp    exit
; Символ не найден, вывод.
char_not_found:
      mov    rdi,string45
      mov    rsi,string4
      xor    rax,rax
      call   printf
exit:
leave
ret

; ФУНКЦИИ ===============================================================
; Функция сравнения 2 строк. ------------------------------------
compare1:    mov    rcx, rdx
             cld
cmpr:        cmpsb
             jne    notequal
             loop   cmpr
             xor    rax,rax
             ret
notequal:    mov    rax, strlen2
             dec    rcx           ; Вычисление положения в строке.
             sub    rax,rcx       ; Вычисление положения в строке.
             ret
             xor    rax,rax
             ret
;----------------------------------------------------------------
; Функция сравнения 2 строк. ------------------------------------

compare2:    mov    rcx, rdx
             cld
             repe   cmpsb
             je     equal2
             mov    rax, strlen2
             sub    rax,rcx       ; Вычисление положения в строке.
             ret
equal2:      xor    rax,rax
             ret
;----------------------------------------------------------------
; Функция поиска символа в строке (сканирования).
cscan:       mov    rcx, rdx
             lodsb
             cld
             repne  scasb
             jne    char_notfound
             mov    rax, strlen3
             sub    rax,rcx       ; Вычисление положения в строке.
             ret
char_notfound:      xor rax,rax
             ret

Здесь	рассматриваются	две	версии	функции	сравнения	строк.	Как	и в	пре
дыдущей	программе,	адрес	первой	строки	(источника)	записывается	в регистр	
rsi,	адрес	второй	строки	(целевой) –	в регистр	rdi,	а длина	строк	помещается	
в регистр	rcx.	Для	полной	уверенности	флаг	направления	DF	сбрасывается	в 0	
инструкцией	cld.	Итак,	выполняется	проход	по	строкам	в прямом	направлении.

Инструкция	cmpsb	сравнивает	два	байта	и устанавливает	флаг	состояния	ZF 
в 1,	если	сравниваемые	байты	равны,	или	сбрасывает	ZF	в 0,	если	эти	два	байта	
не равны.

Использование	флага	ZF	может	показаться	слегка	запутанным.	Если	ZF=1,	это	
значит,	что	при	выполнении	самой	последней	предыдущей	инструкции	был	
получен	результат	0	(байты	равны).	Если	ZF=0,	это	значит,	что	предыдущая	ин
струкция	выполнена	с результатом,	не равным	0	(байты	не равны).	Таким	обра
зом,	необходимо	определять,	когда	флаг	ZF	становится	равным	0.	Для	проверки	
ZF	и продолжения	выполнения	на	основе	результата	этой	проверки	требуется	
несколько	инструкций	условного	перехода,	как	показано	ниже:

	� jz –	переход,	если	результат	равен	нулю	(ZF=1);
	� равнозначная	инструкция	je –	переход,	если	равно	(ZF=1)	(байты	равны);

	� jnz –	переход,	если	результат	не равен	нулю	(ZF=0);
	� равнозначная	 инструкция	 jne  –	 переход,	 если	 равно	 (ZF=0)	 (байты	
не равны).

Регистры	rsi	и rdi	увеличиваются	инструкцией	cmpsb,	если	флаг	DF	не установ
лен	 (равен	0),	и уменьшаются,	если	флаг	DF	 установлен	 (равен	1).	Здесь	создан	
цикл,	который	выполняется	до	тех	пор,	пока	флаг	ZF	не станет	равным	0.	Когда	ZF 
становится	равным	0,	происходит	выход	из	цикла	и начинается	вычисление	по
зиции	отличающегося	символа	на	основе	значения	в регистре	rcx.	Но значение	
rcx	изменяется	только	в конце	цикла,	до	которого	дело	не доходит,	поэтому	необ
ходимо	скорректировать	содержимое	rcx	(уменьшить	его	на	1).	Полученная	в ре
зультате	позиция	символа	возвращается	в основную	программу	в регистре	rax.

Во	второй	версии	функции	сравнения	используется	инструкция	repe –	вариант	
инструкции	rep.	Аббревиатура	repe	означает	«repeat	while	equal»	(повторять,	пока	
равно).	Как	и в	первой	версии,	инструкция	cmpsb	устанавливает	флаг	ZF	по	резуль
тату	сравнения,	и ZF=1	означает,	что	байты	равны.	Как	только	cmpsb	установит	флаг	
ZF	равным	0,	цикл	repe	завершается,	и регистр	rcx	можно	использовать	для	вы
числения	позиции,	в которой	обнаружен	отличающийся	символ.	Если	строки	аб
солютно	одинаковые,	то	в регистре	rcx	находится	значение	0,	а флаг	ZF	установлен	
в 1.	После	завершения	цикла	repe	инструкция	je	проверяет	флаг	ZF	на	равенство	1.	
Если	ZF	равен	1,	то	строки	равны	(одинаковы),	если	ZF	равен	0,	то	строки	не равны	
(не одинаковы).	Регистр	rcx	используется	для	вычисления	позиции	отличающего
ся	символа,	но	здесь	нет	необходимости	в коррекции	rcx,	потому	что	инструкция	
repe	увеличивает	значение	rcx	в самом	начале	каждой	итерации	цикла.

Поиск	символа	(сканирование)	работает	аналогично,	но	вместо	инструкции	
repe	используется	инструкция	repne –	«repeat	while	not	equal»	(повторять,	пока	
не  равно).	 Также	 применяется	 инструкция	 lodsb,	 которая	 загружает	 байт	 из	
ячейки	памяти	с адресом,	хранящимся	в rsi,	в регистр	rax.	Инструкция	scasb 
сравнивает	байт	в регистре	al	(младший	байт	регистра	rax)	с байтом,	на	кото
рый	указывает	регистр	rdi,	и устанавливает	(1=равен)	или	сбрасывает	(0=не ра
вен)	флаг	ZF	соответствующим	образом.	Инструкция	repne	проверяет	состояние	
флага	и продолжает	выполнение,	если	ZF=0,	т. е.	если	два	сравниваемых	байта	
не равны.	Если	байты	равны,	то	scasb	устанавливает	флаг	ZF	в 1,	цикл	repne	пре
кращает	выполнение,	и регистр	rcx	можно	использовать	для	вычисления	по
зиции	найденного	байта	в строке.

Сканирование	строки	(поиск	символа)	работает	только	с одним	символом,	
заданным	как	искомый	аргумент.	Если	у вас	возник	вопрос:	как	использовать	
в качестве	искомого	аргумента	строку,	то	вам	придется	выполнять	поиск	(ска
нирование)	поочередно,	символ	за	символом.	Но лучше	подождать,	пока	мы	
доберемся	до	глав,	в которых	рассматриваются	SIMDинструкции.

На	рис. 24.2	показан	вывод	этой	программы.

Рис. 24.2.	Вывод программы strings.asm

резюме
В	этой	главе	вы	узнали,	как:

	� перемещать	и копировать	блоки	памяти	весьма	эффективным	способом;
	� использовать	инструкции	movsb	и rep;
	� сравнивать	и выполнять	поиск	символа	(сканирование)	в блоках	памяти;
	� использовать	инструкции	cmpsb,	scasb,	repe	и repne.

Глава 25
Предъявите ваш 
идентификатор

Иногда	необходимо	определить	функциональность,	предоставляемую	неко
торым	процессором.	Например,	в программе	можно	попытаться	определить	
наличие	 или	 отсутствие	 конкретной	 версии	 SSEинструкций.	 В  следующей	
главе	будут	использоваться	программы	с SSEинструкциями,	поэтому	необхо
димо	знать,	какая	именно	версия	SSE	поддерживается	процессором.	Для	про
верки	характеристик	ЦПУ	существует	специальная	инструкция:	cpuid.

иСпОльзОвание инСтрукции cpuid
Сначала	специальный	параметр	записывается	в регистр	eax,	затем	выполня
ется	инструкция	cpuid,	после	этого	проверяется	возвращаемое	значение	в ре
гистрах	ecx	и edx.	Вполне очевидно,	что	инструкция	cpuid	использует	32бито
вые	регистры.

Объем	информации,	который	можно	получить	с помощью	инструкции	cpuid,	
ошеломляет.	 В  руководствах	 Intel	 (https://software.intel.com/sites/default/files/
managed/39/c5/325462-sdm-vol-1-2abcd-3abcd.pdf)	 в  томе	 2A	 (Volume	 2A)	 най
дите	описание	инструкции	cpuid.	Там	вы	увидите	несколько	таблиц,	содержа
щих	описание	характеристик,	возвращаемых	в регистре	ecx	при	выполнении	
инструкции	cpuid	с определенным	значением	в регистре	eax.	Это	только	часть	
информации,	которую	можно	извлечь,	в другой	таблице	описана	информация,	
возвращаемая	в регистре	edx.	Внимательно	изучите	это	руководство	Intel,	что
бы	узнать	обо	всех	возможностях.

Рассмотрим	 пример	 определения	 функциональных	 характеристик	 SSE
инструкций,	 которые	 потребуются	 нам	 в  следующей	 главе.	 В  руководстве	
Intel	указано,	что	можно	использовать	биты	0,	19	и 20	регистра	ecx	и биты 25	
и 26	регистра	edx	для	определения	версии	SSE,	реализованной	в данном	про
цессоре.

В	листинге	25.1	показан	пример	программы.



Листинг 25.1.	Программа cpu.asm

; cpu.asm
extern printf
section .data
        fmt_no_sse db "This cpu does not support SSE",10,0
        fmt_sse42 db "This cpu supports SSE 4.2",10,0
        fmt_sse41 db "This cpu supports SSE 4.1",10,0
        fmt_ssse3 db "This cpu supports SSSE 3",10,0
        fmt_sse3 db "This cpu supports SSE 3",10,0
        fmt_sse2 db "This cpu supports SSE 2",10,0
        fmt_sse db "This cpu supports SSE",10,0
section .bss
section .text
      global main
main:
push rbp
mov  rbp,rsp
    call cpu_sse    ; Возвращает 1 в rax, если поддерживается sse, иначе 0.
leave
ret

cpu_sse:
      push rbp
      mov rbp,rsp
      xor r12,r12         ; Флаг SSE доступен.
      mov eax,1           ; Запрос флагов характеристик ЦПУ.
      cpuid
; Проверка поддержки SSE.
    test edx,2000000h           ; Проверка бита 25 (SSE).
    jz sse2                     ; SSE-инструкции доступны.
    mov r12,1
    xor rax,rax
    mov rdi,fmt_sse
    push rcx                    ; Изменяется функцией printf.
    push rdx                    ; Сохранение результата cpuid.
    call printf
    pop rdx
    pop rcx
sse2:
    test edx,4000000h           ; Проверка бита 26 (SSE 2).
    jz sse3                     ; Версия SSE 2 доступна.
    mov r12,1
    xor rax,rax
    mov rdi,fmt_sse2
    push rcx                    ; Изменяется функцией printf.
    push rdx                    ; Сохранение результата cpuid.
    call printf
    pop rdx
    pop rcx
sse3:
    test ecx,1                  ; Проверка бита 0 (SSE 3).
    jz ssse3                    ; Версия SSE 3 доступна.
    mov r12,1
    xor rax,rax
    mov rdi,fmt_sse3
    push rcx                    ; Изменяется функцией printf.

    call printf
    pop rcx
ssse3:
    test ecx,9h                 ; Проверка бита 0 (SSE 3).
    jz sse41                    ; Версия SSE 3 доступна.
    mov r12,1
    xor rax,rax
    mov rdi,fmt_ssse3
    push rcx                    ; Изменяется функцией printf.
    call printf
    pop rcx
sse41:
    test ecx,80000h             ; Проверка бита 19 (SSE 4.1).
    jz sse42                    ; Версия SSE 4.1 доступна.
    mov r12,1
    xor rax,rax
    mov rdi,fmt_sse41
    push rcx                    ; Изменяется функцией printf.
    call printf
    pop rcx
sse42:
   test ecx,100000h             ; Проверка бита 20 (SSE 4.2).
   jz wrapup                    ; Версия SSE 4.2 доступна.
   mov r12,1
   xor rax,rax
   mov rdi,fmt_sse42
   push rcx                     ; Изменяется функцией printf.
   call printf
   pop rcx
wrapup:
    cmp r12,1
    je sse_ok
    mov rdi,fmt_no_sse
    xor rax,rax
    call printf                 ; Вывод сообщения, если SSE-инструкции недоступны.
    jmp the_exit
sse_ok:
    mov rax,r12                 ; Возвращает 1, sse-инструкции поддерживаются.
the_exit:
leave
ret

Основная	программа	main	 вызывает	только	одну	функцию	cpu_sse,	и если	
возвращается	 значение	 1,	 то	 процессор	 поддерживает	 некоторую	 версию	
SSE.	Если	возвращается	значение	0,	то	можно	забыть	об	использовании	SSE
инструкций	на	этом	компьютере.	В функции	cpu_sse	определяется,	какие	вер
сии	SSE	поддерживаются.	Если	записать	1	в регистр	eax	и выполнить	инструк
цию	cpuid,	как	описано	выше,	то	результаты	будут	возвращены	в регистрах	
ecx	и edx.

иСпОльзОвание инСтрукции test
Регистры	ecx	и edx	проверяются	инструкцией	test,	которая	представляет	собой	
битовую	логическую	операцию	and,	выполняемую	с двумя	операндами.	Можно	

Использование инструкции test  181



было	бы	воспользоваться	инструкцией	cmp,	но	test	обеспечивает	более	высокую	
производительность.	Разумеется,	также	можно	было	бы	применить	инструк
цию	bt	(см.	главу	17).

Инструкция	test	устанавливает	флаги	SF,	ZF	и PF	в соответствии	с результата
ми	проверки.	В руководстве	Intel	можно	найти	описание	работы	инструкции	
test,	которое	приведено	ниже.

TEMP ← SRC1 AND SRC2;
SF ← MSB(TEMP);
IF TEMP = 0
     THEN ZF ← 1;
     ELSE ZF ← 0;
FI:
PF ← BitwiseXNOR(TEMP[0:7]);
CF ← 0;
OF ← 0;
(∗ AF is undefined *)

В	 рассматриваемом	 здесь	 примере	 самым	 важным	 флагом	 является	 ZF.	
Если	ZF=0,	то	результат	ненулевой,	т. е.	бит	SSE	равен	1,	значит,	процессор	под
держивает	эту	версию	SSE.	Инструкция	jz	проверяет	условие	ZF=1,	и если	оно	
выполнено,	то	данная	версия	SSE	не поддерживается,	и происходит	переход	
к следующей	части.	В противном	случае	программа	выводит	подтверждающее	
сообщение.

В	этом	примере	после	выполнения	инструкции	cpuid	проверяется	регистр	edx.	
В регистре	edx	32	бита,	но	нам	необходимо	узнать,	установлен	ли	бит	25,	озна
чающий,	что	процессор	поддерживает	SSE	(версию	1).	Поэтому	в инструкции	
test	требуется	второй	операнд,	в бите	25	которого	установлена	1,	а все	осталь
ные	биты	равны	0.	Напомню,	что	самый	младший	бит	имеет	индекс	0,	а индекс	
самого	старшего	бита	31.	В двоичном	представлении	это	выглядит	так:

0000 0010 0000 0000 0000 0000 0000 0000

В	шестнадцатеричном	формате	это	число	имеет	следующий	вид:

2000000

Поток	 выполнения	 «спускается	 каскадом»	по	 программе,	 и  если	не  обна
ружена	поддержка	SSE,	то	в регистре	r12	остается	значение	0.	Возвращаемое	
значение	здесь	не использовалось,	но	можно	было	бы	проверить	его	в регист
ре	rax,	чтобы	сделать	вывод	о поддержке	какойлибо	версии	SSE.	Или	можно	
изменить	программу	так,	чтобы	она	возвращала	номер	самой	последней	под
держиваемой	версии	SSE.

На	рис. 25.1	показан	вывод	этой	программы.

Рис. 25.1.	Вывод программы cpu_sse.asm

Можно	также	написать	аналогичную	функцию	для	извлечения	другой	ин
формации	о характеристиках	процессора	и в	зависимости	от	возвращаемого	
результата	выбирать	конкретные	функциональные	возможности	для	данного	
процессора	или	другие	функциональные	возможности	для	другого	процессора.

В	одной	из	следующих	глав,	где	рассматривается	AVX,	также	необходимо	бу
дет	определить,	поддерживает	ли	процессор	AVX.

резюме
В	этой	главе	вы	узнали,	как:

	� определить,	какие	функциональные	возможности	поддерживаются	кон
кретным	процессором,	с помощью	инструкции	cpuid;

	� использовать	биты	при	выполнении	инструкции	test.

Глава 26
SIMD

SIMD –	это	аббревиатура,	соответствующая	выражению	Single	Instruction	Stream,	
Multiple	 Data	 (один	 поток	 выполнения	 инструкций,	 много	 потоков	 данных).	
Термин	SIMD	предложил	Майкл	Дж.	Флинн	(Michael	J.	Flynn)	для	обозначения	
функциональности,	позволяющей	выполнять	одну	инструкцию	для	нескольких	
«потоков»	 данных.	 Теоретически	 SIMD	 может	 улучшить	 производительность	
программ.	SIMD –	это	одна	из	форм	параллельных	вычислений,	но	в некоторых	
случаях	выполнение	одной	инструкции	для	различных	потоков	данных	может	
происходить	последовательно	в зависимости	от	функциональных	возможностей	
аппаратного	оборудования	и конкретных	выполняемых	инструкций.	Более	по
дробно	о классификации	параллельных	архитектур	Флинна	можно	узнать	здесь:

https://ieeexplore.ieee.org/document/5009071/

и	

https://en.wikipedia.org/wiki/Flynn's_taxonomy.

Первой	реализацией	SIMD	был	набор	инструкций	MMX,	но	никто	не знает	точ
ного	смысла	аббревиатуры	MMX.	Это	могло	означать	Multi	Media	Extension	(муль
тимедийное	расширение),	или	Multiple	Math	Extension	(расширение	множества	
математических	операций),	или	Matrix	Math	Extension	(расширение	математи
ческих	операций	с матрицами).	Как	бы	то	ни	было,	набор	инструкций	MMX	был	
заменен	набором	инструкций	Streaming	SIMD	Extension	(SSE).	В дальнейшем	SSE	
был	расширен	до	набора	команд	Advanced	Vector	Extension	(AVX).	В этой	главе	
приводится	краткое	введение	в SSE	как	первоначальная	основа	для	дальнейшего	
изучения,	а в	одной	из	следующих	глав	будет	приведено	введение	в AVX.

Скалярные данные и упакОванные данные
Процессор,	поддерживающий	функциональность	SSE,	имеет	16	дополнитель
ных	 128битовых	 регистров	 (от	 xmm0	 до	 xmm15)	 и  управляющий	 регистр	 mxcsr.	
Ранее	мы	уже	пользовались	xmmрегистрами	для	вычислений	с плавающей	точ
кой,	но	 с помощью	этих	мощных	регистров	можно	 сделать	 гораздо	больше.	
Регистры	xmm	могут	содержать	скалярные	данные	(scalar data)	или	упакованные	
данные	(packed data).



Под	 скалярными	 данными	 подразумевается	 только	 одно	 значение.	 Если	
число	3.141592654	помещается	в регистр	xmm0,	то	xmm0	содержит	скалярное	зна
чение.	Но в xmm0	также	можно	записать	несколько	значений,	тогда	эти	значения	
определяются	как	упакованные	данные.	Ниже	перечислены	возможные	вари
анты	сохранения	значений	в любом	xmmрегистре:

	� два	64битовых	числа	с плавающей	точкой	двойной	точности;
	� четыре	 32битовых	 числа	 с  плавающей	 точкой	 обычной	 (одинарной)	
точности;

	� два	64битовых	целых	числа	(четверные	слова);
	� четыре	32битовых	целых	числа	(двойные	слова);
	� восемь	16битовых	коротких	(short)	целых	чисел	(слова);
	� шестнадцать	8битовых	байт	или	символов.

На	рис. 26.1	показаны	соответствующие	схемы	вариантов	размещения	зна
чений	в xmmрегистре.

16 упакованных 
байт

8 упакованных 
слов

4 упакованных 
двойных слова

2 упакованных 
четверных слова

4 упакованных числа с плавающей 
точкой двойной точности

8 упакованных чисел с плавающей 
точкой  обычной точности

1 скалярное число с плавающей 
точкой двойной точности

1 скалярное число с плавающей 
точкой обычной точности

Рис. 26.1.	Возможное содержимое любого xmm-регистра

Для	скалярных	значений	и для	упакованных	значений	существуют	различ
ные	ассемблерные	инструкции.	В руководствах	 Intel	можно	найти	огромное	
количество	описаний	доступных	SSEинструкций.	Здесь	и в	следующих	главах	
рассматривается	лишь	несколько	примеров	их	применения,	чтобы	предоста
вить	вам	возможность	для	дальнейшего	изучения.

В	следующих	главах	мы	будем	использовать	функциональность	AVX.	Регист
ры	AVX	имеют	удвоенный	размер	по	 сравнению	с регистрами	xmm.	Регистры	
AVX	называются	 ymmрегистрами	 и  содержат	 256	 бит.	 Существует	также	 рас
ширение	AVX512,	предоставляющее	регистры	AVX512,	содержащие	512	бит,	
которые	называются	zmmрегистрами.

Обладая	потенциальными	возможностями	параллельных	вычислений,	SIMD
инструкции	могут	использоваться	для	повышения	скорости	вычислений	в ши
роком	спектре	прикладных	областей,	таких	как	обработка	изображений,	звука,	
сигналов,	для	векторных	и матричных	вычислений	и т.	д.	В следующих	главах	мы	
рассмотрим	применение	SIMD	для	обработки	матриц,	но	не следует	волновать

ся	по	этому	поводу:	математическая	часть	будет	ограничена	самыми	простыми	
операциями	с матрицами.	Наша	цель –	изучение	SIMD,	а не линейной	алгебры.

невырОвненные и вырОвненные данные
Данные	в памяти	могут	быть	невыровненными	или	выровненными	по	опреде
ленным	адресам,	кратным	16,	32	и т.	д.	Выравнивание	данных	в памяти	может	
существенно	 улучшить	 производительность	 программы.	Причина	 в  следую
щем:	для	обработки	выровненных	упакованных	данных	SSEинструкции	тре
буют	извлечения	16байтовых	фрагментов	памяти	за	одну	операцию –	см. схе
му	 слева	 на	 рис.  26.2.	 Если	 данные	 в  памяти	 не  выровнены,	 то	 процессору	
потребуется	более	одной	операции	для	получения	необходимого	16байтового	
фрагмента	данных,	таким	образом,	выполнение	замедляется.	Существуют	два	
типа	 SSEинструкций:	 инструкции	 для	 выровненных	 упакованных	 данных	
и инструкции	для	невыровненных	упакованных	данных.	Инструкции	для	не
выровненных	упакованных	данных	могут	работать	с невыровненными	фраг
ментами	памяти,	но	в общем	случае	это	ухудшает	производительность.

Память выровнена Память не выровнена

Память ПамятьПотребность процессора Потребность 
процессора

Рис. 26.2.	Выравнивание данных

При	 использовании	 SSE	 выравнивание	 означает,	 что	 данные	 в  разделах	
section .data	и section .bss	должны	быть	выровнены	по	16байтовой	границе.	
В NASM	можно	воспользоваться	директивами	ассемблера	align 16	и alignb 16	пе
ред	данными,	которые	необходимо	выровнять.	В следующих	главах	будут	по
казаны	примеры	практического	применения	выравнивания.	Для	AVX	данные	
должны	выравниваться	по	32байтовой	границе,	а для	AVX512 –	по 64байто
вой	границе.

резюме
В	этой	главе	вы	узнали	о:

	� том,	что	SSE	предоставляет	16	дополнительных	128битовых	регистров;
	� различиях	между	скалярными	и упакованными	данными;
	� важности	выравнивания	данных.

Глава 27
Наблюдение за регистром 

MXCSR

Прежде	чем	начать	углубленное	изучение	программирования	с использовани
ем	SSE,	необходимо	понять,	как	используется	регистр	управления	и состояния	
SSE	для	операций	с плавающей	точкой,	который	называется	mxcsr.	Это	32бито
вый	регистр,	в котором	применяются	только	младшие	16	бит.	В табл. 27.1	опи
саны	используемые	биты	регистра	mxcsr.

Таблица 27.1.	Биты управления и состояния регистра mxcsr

Бит Мнемоническое 
обозначение

Описание

0 IE Ошибка: некорректная (недопустимая) операция

1 DE Ошибка денормализации

2 ZE Ошибка: деление на ноль

3 OE Ошибка переполнения

4 UE Ошибка: потеря значимости (машинный ноль)

5 PE Ошибка: потеря точности

6 DAZ Денормализованные разряды равны нулю

7 IM Маска некорректной операции

8 DM Маска операции денормализации

9 ZM Маска деления на ноль

10 OM Маска переполнения

11 UM Маска потери значимости (машинного нуля)

12 PM Маска точности

13 RC Управление округлением

14 RC Управление округлением

15 FZ Сброс в ноль



Биты	с 0	по	5	устанавливаются	в 1	при	возникновении	исключений	во	вре
мя	 выполнения	 операций	 с  плавающей	точкой,	 например	 при	 делении	 на	
ноль	или	при	потере	точности	(значимых	разрядов)	при	операции	с числом	
с плавающей	точкой.	Биты	с 7	по	12 –	это	маски,	управляющие	поведением,	
когда	операция	с плавающей	точкой	устанавливает	флаг	1	в битах	с 0	по	5.	
Например,	если	произошло	деление	на	ноль,	программа	обычно	должна	сге
нерировать	исключение	(ошибку)	и,	возможно,	завершиться	аварийно.	Если	
флаг	маски	деления	на	ноль	(ZM)	установлен	в 1,	то	программа	не завершается,	
и можно	 выполнить	некоторую	инструкцию,	 чтобы	 устранить	или	 хотя	 бы	
смягчить	критический	сбой.	Все	эти	маски	(биты	7–12)	по	умолчанию	уста
новлены	в 1,	поэтому	исключения	для	SIMDопераций	с плавающей	точкой	
генерироваться	не будут.	Два	бита	13	и 14	управляют	округлением,	как	по
казано	в табл. 27.2.

Таблица 27.2.	Биты управления округлением в регистре mxcsr

Биты Описание

00 Округление до ближайшего числа

01 Округление с недостатком (в меньшую сторону)

10 Округление с избытком (в большую сторону)

11 Усечение (отбрасывание разрядов)

Здесь	мы	не будем	рассматривать	в деталях	характеристики	битов	состоя
ния	и масок	регистра	mxcsr,	подробное	описание	см.	в руководствах	Intel.

рабОта С битами региСтра mxcsr
С	битами	регистра	mxcsr	можно	работать	с помощью	инструкций	ldmxcsr	и stmx-
csr.	По умолчанию	состояние	регистра	mxcsr	определяется	числом	00001F80	или	
0001 1111 1000 0000.	Все	биты	масок	установлены	в 1,	а округление	выполняется	
до	ближайшего	числа.

В	листингах	с 27.1	по	27.4	показаны	примеры	обработки	содержимого	регист
ра	mxcsr.

Листинг 27.1.	Программа mxcsr.asm

; mxcsr.asm
extern printf
extern print_mxcsr
extern print_hex
section .data
      eleven      dq    11.0
      two         dq    2.0
      three       dq    3.0
      ten         dq    10.0
      zero        dq    0.0
      hex         db    "0x",0
      fmt1        db    10,"Divide, default mxcsr:",10,0
      fmt2        db    10,"Divide by zero, default mxcsr:",10,0

Работа с битами регистра mxcsr  189



      fmt4        db    10,"Divide, round up:",10,0
      fmt5        db    10,"Divide, round down:",10,0
      fmt6        db    10,"Divide, truncate:",10,0
      f_div       db    "%.1f divided by %.1f is %.16f, in hex: ",0
      f_before    db    10,"mxcsr before:",9,0
      f_after     db    "mxcsr after:",9,0

; Значения регистра mxcsr.
      default_mxcsr     dd 0001111110000000b
      round_nearest     dd 0001111110000000b
      round_down        dd 0011111110000000b
      round_up          dd 0101111110000000b
      truncate          dd 0111111110000000b

section .bss
        mxcsr_before    resd  1
        mxcsr_after     resd  1
        xmm             resq  1
section .text
      global main
main:
push rbp
mov  rbp,rsp

; Деление.
; Значение mxcsr по умолчанию.
      mov   rdi,fmt1
      mov   rsi,ten
      mov   rdx,two
      mov   ecx, [default_mxcsr]
      call  apply_mxcsr
;----------------------------------------------
; Деление с ошибкой потери точности (значимых разрядов).
; Значение mxcsr по умолчанию.
      mov   rdi,fmt1
      mov   rsi,ten
      mov   rdx,three
      mov   ecx, [default_mxcsr]
      call  apply_mxcsr
; Деление на ноль.
; Значение mxcsr по умолчанию.
      mov   rdi,fmt2
      mov   rsi,ten
      mov   rdx,zero
      mov   ecx, [default_mxcsr]
      call  apply_mxcsr
; Деление с ошибкой потери точности (значимых разрядов).
; Округление в большую сторону.
      mov   rdi,fmt4
      mov   rsi,ten
      mov   rdx,three
      mov   ecx, [round_up]
      call  apply_mxcsr
; Деление с ошибкой потери точности (значимых разрядов).
; Округление в меньшую сторону.
      mov   rdi,fmt5

      mov   rsi,ten
      mov   rdx,three
      mov   ecx, [round_down]
      call  apply_mxcsr
; Деление с ошибкой потери точности (значимых разрядов).
; Усечение (отбрасывание разрядов).
      mov   rdi,fmt6
      mov   rsi,ten
      mov   rdx,three
      mov   ecx, [truncate]
      call  apply_mxcsr
;----------------------------------------------
; Деление с ошибкой потери точности (значимых разрядов).
; Значение mxcsr по умолчанию.
      mov   rdi,fmt1
      mov   rsi,eleven
      mov   rdx,three
      mov   ecx, [default_mxcsr]
      call  apply_mxcsr
; Деление с ошибкой потери точности (значимых разрядов).
; Округление в большую сторону.
      mov   rdi,fmt4
      mov   rsi,eleven
      mov   rdx,three
      mov   ecx, [round_up]
      call  apply_mxcsr
; Деление с ошибкой потери точности (значимых разрядов).
; Округление в меньшую сторону.
      mov   rdi,fmt5
      mov   rsi,eleven
      mov   rdx,three
      mov   ecx, [round_down]
      call  apply_mxcsr
; Деление с ошибкой потери точности (значимых разрядов).
; Усечение (отбрасывание разрядов).
      mov   rdi,fmt6
      mov   rsi,eleven
      mov   rdx,three
      mov   ecx, [truncate]
      call  apply_mxcsr
leave
ret

; Функция. --------------------------------------------
apply_mxcsr:
push  rbp
mov   rbp,rsp
      push rsi
      push  rdx
      push  rcx
      push  rbp       ; Еще одна операция записи для выравнивания стека.
      call  printf
      pop   rbp
      pop   rcx
      pop   rdx
      pop   rsi

      mov         [mxcsr_before],ecx
      ldmxcsr     [mxcsr_before]
      movsd       xmm2, [rsi] ; Число с плав. точкой двойной точности в регистр xmm2.
      divsd       xmm2, [rdx]     ; Деление xmm2.
      stmxcsr     [mxcsr_after]   ; Сохранение mxcsr в память.
      movsd       [xmm],xmm2      ; Для использования в функции print_xmm.
      mov         rdi,f_div
      movsd       xmm0, [rsi]
      movsd       xmm1, [rdx]
      call        printf
      call        print_xmm
; Вывод mxcsr.
      mov         rdi,f_before
      call        printf
      mov         rdi, [mxcsr_before]
      call        print_mxcsr
      mov         rdi,f_after
      call        printf
      mov         rdi, [mxcsr_after]
      call        print_mxcsr
leave
ret

; Функция. --------------------------------------------
print_xmm:
push rbp
mov  rbp,rsp
     mov   rdi, hex    ; Вывод префикса 0x.
     call  printf
     mov   rcx,8
.loop:
     xor   rdi,rdi
     mov   dil,[xmm+rcx-1]
     push  rcx
     call  print_hex
     pop   rcx
     loop  .loop
leave
ret

Листинг 27.2.	Программа print_hex.c

// print_hex.c

#include <stdio.h>

void print_hex(unsigned char n){
           if (n < 16) printf("0");
           printf("%x",n);
}

Листинг 27.3.	Программа print_mxcsr.c

// print_mxcsr.c

#include <stdio.h>

void print_mxcsr(long long n){
      long long s,c;
      for (c = 15; c >= 0; c--)
      {
           s = n >> c;
           // Вывод пробела после каждого 8-го бита.
           if ((c+1) % 4 == 0) printf(" ");
           if (s & 1)
                      printf("1");
           else
                      printf("0");
      }
      printf("\n");
}

Листинг 27.4.	makefile

mxcsr: mxcsr.o print_mxcsr.o print_hex.o
      gcc -o mxcsr mxcsr.o print_mxcsr.o print_hex.o -no-pie
mxcsr.o: mxcsr.asm
      nasm -f elf64 -g -F dwarf mxcsr.asm -l mxcsr.lst
print_mxcsr: print_mxcsr.c
      gcc -c print_mxcsr.c
print_hex: print_hex.c
      gcc -c print_hex.c

В	 этой	 программе	 продемонстрированы	 различные	 режимы	 округления	
и  обработка	 маскированного	 деления	 на	 ноль.	 По  умолчанию	 округление	
выполняется	до	ближайшего	числа.	Например,	в десятичном	формате	поло
жительное	 число,	 заканчивающееся	 на	 .5	 и  больше,	 должно	 округляться	 до	
следующего	большего	числа,	а отрицательное	число,	заканчивающееся	на	.5 
и больше,	должно	округляться	до	следующего	меньшего	(отрицательного)	чис
ла.	Но в примере	программы	округление	выполняется	в шестнадцатеричном	
формате,	а не в десятичном,	поэтому	не всегда	получается	тот	же	результат,	что	
и при	округлении	десятичных	чисел.

На	рис. 27.1	показан	вывод	данной	программы.

Рис. 27.1.	Вывод программы mxcsr.asm

анализ прОграммы
Проанализируем	программу	mxcsr.asm.	В ней	выполнено	несколько	операций	
деления	с применением	округления.	Операции	деления	производятся	в функ
ции	apply_mxcsr.	Перед	вызовом	этой	функции	адрес	строки	заголовка	записы

вается	в регистр	rdi,	делимое –	в rsi,	делитель –	в rdx.	Затем	требуемое	значение	
регистра	mxcsr	копируется	из	памяти	в регистр	ecx –	при	первом	вызове	функ
ции	 это	 значение	 mxcsr	 по	 умолчанию.	 После	 этого	 вызывается	 функция	 ap-
ply_mxcsr.	В этой	функции	выводится	строка	заголовка,	но	сначала	не забываем	
сохранить	необходимые	регистры	и выровнять	стек.	Далее	значение	регистра	
ecx	сохраняется	в переменной	mxcsr_before,	и это	сохраненное	значение	загру
жается	в регистр	mxcsr	с помощью	инструкции	ldmxcsr.	Инструкция	ldmxcsr	при
нимает	32битовое	значение	переменной	памяти	(двойное	слово)	как	операнд.	
Инструкция	divsd	принимает	xmmрегистр	как	первый	операнд	и xmmрегистр	или	
64битовую	переменную	как	второй	операнд.	После	выполнения	операции	де
ления	содержимое	регистра	mxcsr	сохраняется	в памяти	в переменной	mxcsr_af-
ter	с помощью	инструкции	stmxcsr.	Также	копируется	частное	из	регистра	xmm2 
в переменную	памяти	xmm,	чтобы	вывести	полученное	значение.

Сначала	выводится	частное	в десятичном	формате,	затем	нужно	вывести	его	
в шестнадцатеричном	формате	в той	же	строке.	При	использовании	функции	
printf	 в  программе	 на	 ассемблере	 невозможно	 вывести	шестнадцатеричное	
число	(по	крайней	мере,	в той	версии,	которая	используется	здесь),	поэтому	
необходимо	написать	функцию,	которая	выводит	числа	в шестнадцатеричном	
формате, –	это	функция	print_xmm.	Функция	принимает	значение	переменной	
памяти	xmm	и поочередно	загружает	байты	по	одному	в dil	в цикле.	В том	же	
цик	ле	специально	написанная	для	этого	примера	Cфункция	print_hex	вызы
вается	для	каждого	байта.	Используя	в адресе	уменьшающийся	счетчик	цик
ла	rcx,	мы	также	позаботились	о сохранении	правильного	порядка	байтов –	от	
младшего	к старшему:	значение	с плавающей	точкой	хранится	в памяти	имен
но	с таким	порядком	байтов.

Далее	 выводятся	 значения	 mxcsr_before	 и  mxcsr_after,	 чтобы	 можно	 было	
сравнить	их.	Функция	print_mxcsr	применяется	для	вывода	битов	регистра	mx-
csr –	она	похожа	на	функции	вывода	битов,	которые	мы	использовали	в пре
дыдущих	главах.

Для	некоторых	читателей	эта	программа	может	показаться	слишком	слож
ной.	В этом	случае	пройдите	по	программе	с помощью	отладчика	и наблюдай
те	за	памятью	и регистрами.

Проанализируем	вывод:	можно	видеть,	что	регистр	mxcsr	не изменяется	при	
делении	10	на	2.	При	делении	10	на	3	получается	3.333.	Здесь	mxcsr	сообщает	об	
ошибке	потери	точности	(значимых	разрядов)	в бите	5.	По умолчанию	округле
ние	выполняется	к ближайшему	числу,	поэтому	последняя	шестнадцатеричная	
цифра	увеличивается	с a	на	b.	В десятичном	формате	округление	должно	вы
полняться	в меньшую	сторону,	но	в шестнадцатеричном	представлении	цифра	
a больше	8,	поэтому	необходимо	округление	в большую	сторону	к b.

Далее	выполняется	деление	на	ноль:	mxcsr	сообщает	об	этом	установкой	бита	
2,	но	программа	не завершается	изза	критического	сбоя,	поскольку	установ
лена	маска	деления	на	ноль	ZM.	 Результат	определен	как	inf	 (бесконечность)	
или	0x7ff0000000000000.

При	 выполнении	 следующей	 операции	 деления	 и  округлении	 в  большую	
сторону	получается	тот	же	результат,	что	и при	округлении	к ближайшему	чис
лу.	Следующие	две	операции	деления	с округлением	в меньшую	сторону	и усе
чением	дают	в результате	число	с последней	шестнадцатеричной	цифрой	a.

Чтобы	продемонстрировать	различия	в методах	округления,	выполняются	
те	же	операции	деления	11	на	3.	В этом	варианте	деления	получается	частное	
с меньшей	конечной	шестнадцатеричной	цифрой.	Вы можете	сами	сравнить	
поведение	при	округлении.

Предлагается	дополнительное	 упражнение:	 очистить	 (сбросить	 в  0)	маску	
деления	на	ноль	ZM,	пересобрать	и запустить	программу.	Программа	завершит
ся	аварийно	изза	критического	сбоя.	Маска	деления	на	ноль	и другие	маски	
позволяют	перехватывать	ошибки	и переходить	к некоторой	процедуре	кор
рекции	ошибок.

резюме
В	этой	главе	вы	узнали	о:

	� предназначении	регистра	mxcsr	и его	отдельных	битов;
	� работе	с регистром	mxcsr;
	� подробностях	различных	методов	округления	чисел.

Глава 28
Выравнивание для SSE

Пора	начинать	настоящую	работу	с SSE.	Несмотря	на	то	что	в нескольких	главах	
мы	уже	имели	дело	с SSE,	до	сих	пор	эта	тема	обсуждалась	лишь	поверхност
но.	Существует	несколько	сотен	SIMDинструкций	(MMX,	SSE,	AVX),	и для	их	
подробного	описания	потребуется	отдельная	книга	или	даже	несколько	книг.	
В этой	главе	рассматривается	несколько	примеров,	для	того	чтобы	вы	поняли,	
с чего	следует	начать.	Цель	данных	примеров –	позволить	читателю	опреде
лить	собственную	методику	изучения	многочисленных	SIMDинструкций	по	
руководствам	 Intel.	 В  этой	 главе	 рассматривается	 методика	 выравнивания,	
которая	кратко	описывалась	в главе	26.

пример без выравнивания
В	 листинге	 28.1	 показана	 операция	 сложения	 векторов	 с  использованием	
данных,	которые	не выровнены	в памяти.

Листинг 28.1.	Программа sse_unaligned.asm

; sse_unaligned.asm
extern printf
section .data
; С обычной точностью.
      spvector1  dd   1.1
                 dd   2.2
                 dd   3.3
                 dd   4.4
      spvector2  dd   1.1
                 dd   2.2
                 dd   3.3
                 dd   4.4

; С двойной точностью.
      dpvector1  dq   1.1
                 dq   2.2
      dpvector2  dq   3.3
                 dq   4.4

      fmt1 db "Single Precision Vector 1: %f, %f, %f, %f",10,0
      fmt2 db "Single Precision Vector 2: %f, %f, %f, %f",10,0
      fmt3 db "Sum of Single Precision Vector 1 and Vector 2:"
           db " %f, %f, %f, %f",10,0



      fmt4 db "Double Precision Vector 1: %f, %f",10,0
      fmt5 db "Double Precision Vector 2: %f, %f",10,0
      fmt6 db "Sum of Double Precision Vector 1 and Vector 2:"
           db " %f, %f",10,0

section .bss
      spvector_res resd 4
      dpvector_res resq 4
section .text
      global main
main:
push  rbp
mov   rbp,rsp

; Сложение 2 векторов из чисел с плав. точкой с обычной точностью.
      mov   rsi,spvector1
      mov   rdi,fmt1
      call  printspfp
      mov   rsi,spvector2
      mov   rdi,fmt2
      call  printspfp
      movups     xmm0, [spvector1]
      movups     xmm1, [spvector2]
      addps      xmm0,xmm1
      movups     [spvector_res], xmm0
      mov        rsi,spvector_res
      mov        rdi,fmt3
      call       printspfp

; Сложение 2 векторов из чисел с плав. точкой с двойной точностью.
      mov   rsi,dpvector1
      mov   rdi,fmt4
      call  printdpfp
      mov   rsi,dpvector2
      mov   rdi,fmt5
      call  printdpfp
      movupd     xmm0, [dpvector1]
      movupd     xmm1, [dpvector2]
      addpd      xmm0,xmm1
      movupd     [dpvector_res], xmm0
      mov        rsi,dpvector_res
      mov        rdi,fmt6
      call       printdpfp
leave
ret

printspfp:
push  rbp
mov   rbp,rsp
      movss      xmm0, [rsi]
      cvtss2sd   xmm0,xmm0
      movss      xmm1, [rsi+4]
      cvtss2sd   xmm1,xmm1
      movss      xmm2, [rsi+8]
      cvtss2sd   xmm2,xmm2
      movss      xmm3, [rsi+12]

198	  Выравнивание для SSE



      cvtss2sd   xmm3,xmm3
      mov        rax,4           ; Четыре числа с плав.точкой.
      call       printf
leave
ret

printdpfp:
push  rbp
mov   rbp,rsp
      movsd      xmm0, [rsi]
      movsd      xmm1, [rsi+8]
      mov        rax,2          ; Два числа с плав.точкой.
      call       printf
leave
ret

Первая	SSEинструкция	movups	(означающая	«move	unaligned	packed	single	
precision» –	перемещение	невыровненных	упакованных	данных	с обычной	
точностью)	копирует	данные	из	памяти	в регистры	xmm0	и xmm1.	В результате	
регистр	xmm0	содержит	один	вектор	с четырьмя	значениями	с обычной	точно
стью,	а регистр	xmm1	содержит	второй	вектор	с четырьмя	значениями	с обыч
ной	 точностью.	 Потом	 применяется	 инструкция	 addps	 (означающая	 «add	
packed	 single	precision» –	 сложение	упакованных	данных	с обычной	точно
стью)	для	сложения	этих	векторов,	а полученный	в результате	вектор	записы
вается	в регистр	xmm0,	затем	передается	в память.	Далее	результат	выводится	
с помощью	функции	printspfp.	В этой	функции	копируется	каждое	значение	
из	памяти	в xmmрегистры	с помощью	инструкции	movss	(означающей	«move	
scalar	single	precision» –	перемещение	скалярного	значения	с обычной	точно
стью).	Поскольку	функция	printf	ожидает	передачу	аргументов	с плавающей	
точкой	двойной	точности,	выполняется	преобразование	чисел	с плавающей	
точкой	обычной	точности	в числа	двойной	точности	с помощью	инструкции	
cvtss2sd	(означающей	«convert	scalar	single	to	scalar	double» –	преобразование	
скалярного	значения	обычной	точности	в скалярное	значение	двойной	точ
ности).

Далее	выполняется	 сложение	двух	 значений	двойной	точности.	Эта	опе
рация	аналогична	операции	сложения	чисел	обычной	точности,	но	для	зна
чений	двойной	точности	используются	инструкции	movupd	и addpd.	Функция	
printdpfp	для	вывода	чисел	двойной	точности	немного	проще.	Мы работаем	
с  векторами,	 состоящими	только	 из	 двух	 элементов,	 поэтому	 сразу	 же	 ис
пользуются	значения	двойной	точности,	и нет	необходимости	в преобразо
вании	элементов	векторов.

На	рис. 28.1	показан	вывод	этой	программы.

Рис. 28.1.	Вывод программы sse_unaligned.asm

пример С выравниванием
В	листинге	28.2	показана	операция	сложения	векторов.

Листинг 28.2.	Программа sse_aligned.asm

; sse_aligned.asm
extern printf
section .data
      dummy   db      13
align 16
      spvector1 dd    1.1
                dd    2.2
                dd    3.3
                dd    4.4
      spvector2 dd    1.1
                dd    2.2
                dd    3.3
                dd    4.4

      dpvector1 dq    1.1
                dq    2.2
      dpvector2 dq    3.3
                dq    4.4
      fmt1 db "Single Precision Vector 1: %f, %f, %f, %f",10,0
      fmt2 db "Single Precision Vector 2: %f, %f, %f, %f",10,0
      fmt3 db "Sum of Single Precision Vector 1 and Vector 2:"
           db " %f, %f, %f, %f",10,0
      fmt4 db "Double Precision Vector 1: %f, %f",10,0
      fmt5 db "Double Precision Vector 2: %f, %f",10,0
      fmt6 db "Sum of Double Precision Vector 1 and Vector 2:"
           db " %f, %f",10,0

section .bss
alignb 16
        spvector_res resd 4
        dpvector_res resq 4
section .text
      global main
main:
push  rbp
mov   rbp,rsp

; Сложение 2 векторов из чисел с плав.точкой с обычной точностью.
      mov   rsi,spvector1
      mov   rdi,fmt1
      call  printspfp

      mov   rsi,spvector2
      mov   rdi,fmt2
      call  printspfp

      movaps     xmm0, [spvector1]
      addps      xmm0, [spvector2]

      movaps     [spvector_res], xmm0
      mov        rsi,spvector_res
      mov        rdi,fmt3
      call       printspfp

; Сложение 2 векторов из чисел с плав.точкой с двойной точностью.
      mov        rsi,dpvector1
      mov        rdi,fmt4
      call       printdpfp

      mov        rsi,dpvector2
      mov        rdi,fmt5
      call       printdpfp

      movapd     xmm0, [dpvector1]
      addpd      xmm0, [dpvector2]

      movapd     [dpvector_res], xmm0
      mov        rsi,dpvector_res
      mov        rdi,fmt6
      call       printdpfp
; Выход.
mov   rsp,rbp
pop   rbp    ; Восстановление результата команды push, выполненной в начале.
ret

printspfp:
push  rbp
mov   rbp,rsp
      movss      xmm0, [rsi]
      cvtss2sd   xmm0,xmm0       ; printf ожидает аргумент двойной точности.
      movss      xmm1, [rsi+4]
      cvtss2sd   xmm1,xmm1
      movss      xmm2, [rsi+8]
      cvtss2sd   xmm2,xmm2
      movss      xmm3, [rsi+12]
      cvtss2sd   xmm3,xmm3
      mov        rax,4           ; Четыре числа с плав.точкой.
      call printf
leave
ret

printdpfp:
push  rbp

mov   rbp,rsp
      movsd      xmm0, [rsi]
      movsd      xmm1, [rsi+8]
      mov        rax,2          ; Два числа с плав.точкой.
      call printf
leave
ret

В	начале	этой	программы	создается	переменная	dummy	для	полной	уверен
ности	в том,	что	память	не выровнена	по	16байтовой	границе.	Затем	исполь
зуется	директива	ассемблера	NASM align 16	в разделе	section .data	и директива	
alignb 16	в разделе	section .bss.	Необходимо	добавлять	эти	директивы	ассембле
ра	перед	каждым	блоком	данных,	который	требует	выравнивания.

SSEинструкции	немного	отличаются	от	инструкций	из	версии	без	выравни
вания.	Инструкция	movaps	(означающая	«move	aligned	packed	single	precision» –	
перемещение	 выровненных	 упакованных	 данных	 обычной	 точности)	 копи
рует	 данные	 из	 памяти	 в  регистр	 xmm0.	 После	 этого	 сразу	 можно	 выполнять	
сложение	упакованных	чисел	в памяти	с упакованными	значениями	в регист
ре	xmm0.	В этом	состоит	отличие	от	версии	без	выравнивания,	где	требовалась	
предварительная	 запись	двух	 значений	в xmmрегистры.	Если	добавить	пере
менную	dummy	в версию	без	выравнивания	и попытаться	использовать	инструк
цию	movaps	вместо	movups	с переменной	памяти	как	вторым	операндом,	то	воз
никает	опасность	нарушения	сегментации	памяти	(segmentation fault)	во	время	
выполнения.	Попробуйте	проверить	это	самостоятельно.

Регистр	xmm0	содержит	полученный	в результате	сложения	вектор	с четырь
мя	значениями	обычной	точности.	Затем	выводится	этот	результат	с по	мощью	
функции	 printspfp.	 В  этой	 функции	 каждое	 значение	 из	 памяти	 копирует
ся	 в  xmmрегистры.	 Поскольку	 функция	 printf	 ожидает	 передачу	 аргументов	
с  плавающей	точкой	двойной	точности,	 выполняется	 преобразование	 чисел	
с плавающей	точкой	обычной	точности	в числа	двойной	точности	с помощью	
инструкции	cvtss2sd	(означающей	«convert	scalar	single	to	scalar	double» –	пре
образование	 скалярного	 значения	 обычной	 точности	 в  скалярное	 значение	
двойной	точности).

Далее	выполняется	сложение	двух	значений	двойной	точности.	Эта	опера
ция	аналогична	операции	сложения	чисел	обычной	точности,	но	для	значений	
двойной	точности	используются	инструкции	movapd	и addpd.

На	рис. 28.2	показан	вывод	примера	программы	с выравниванием.

Рис. 28.2.	Вывод программы sse_aligned.asm

На	рис. 28.3	представлен	результат	выполнения	примера	программы	без	вы
равнивания	с переменной	dummy,	добавленной	как	второй	операнд	инструкции	
movaps.

Рис. 28.3.	Критический сбой при нарушении сегментации памяти  
в программе sse_unaligned.asm

резюме
В	этой	главе	вы	узнали	о:

	� скалярных	и упакованных	данных;
	� выровненных	и невыровненных	данных;
	� том,	как	выравнивать	данные	(в памяти);
	� перемещении	 (копировании)	 данных	 и  арифметических	 инструкциях	
для	упакованных	данных;

	� методах	преобразования	данных	обычной	точности	в данные	двойной	
точности.

Глава 29
SSE-инструкции для работы 

с упакованными  
целыми числами

В	предыдущей	главе	использовались	значения	с плавающей	точкой	и инструк
ции	для	их	обработки.	SSE	также	предоставляет	длинный	список	инструкций	
для	работы	с целыми	числами,	и точно	так	же,	как	в предыдущей	главе,	здесь	
рассматривается	применение	лишь	нескольких	инструкций,	чтобы	предоста
вить	читателям	основу	для	дальнейшего	обучения.

SSe-инСтрукции для рабОты С целыми чиСлами
В	листинге	29.1	показан	пример	программы.

Листинг 29.1.	Программа sse_integer.asm

; sse_integer.asm
extern printf
section .data
      dummy   db      13
align 16
      pdivector1 dd    1
                 dd    2
                 dd    3
                 dd    4
      pdivector2 dd    5
                 dd    6
                 dd    7
                 dd    8
      fmt1 db "Packed Integer Vector 1: %d, %d, %d, %d",10,0
      fmt2 db "Packed Integer Vector 2: %d, %d, %d, %d",10,0
      fmt3 db "Sum Vector: %d, %d, %d, %d",10,0
      fmt4 db "Reverse of Sum Vector: %d, %d, %d, %d",10,0
section .bss
alignb 16
      pdivector_res   resd 4
      pdivector_other resd 4



section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Вывод вектора 1.
      mov   rsi,pdivector1
      mov   rdi,fmt1
      call  printpdi
; Вывод вектора 2.
      mov   rsi,pdivector2
      mov   rdi,fmt2
      call  printpdi
; Сложение 2 векторов с выровненными целыми числами двойной точности.
      movdqa     xmm0, [pdivector1]
      paddd      xmm0, [pdivector2]
; Сохранение результата в памяти.
      movdqa     [pdivector_res], xmm0
; Вывод вектора, хранящегося в памяти.
      mov   rsi,pdivector_res
      mov   rdi,fmt3
      call  printpdi
; Копирование вектора из памяти в регистр xmm3.
      movdqa xmm3,[pdivector_res]
; Извлечение упакованных значений из xmm3.
      pextrd eax, xmm3, 0
      pextrd ebx, xmm3, 1
      pextrd ecx, xmm3, 2
      pextrd edx, xmm3, 3
; Вставка значений в регистр xmm0 в обратном порядке.
      pinsrd xmm0, eax, 3
      pinsrd xmm0, ebx, 2
      pinsrd xmm0, ecx, 1
      pinsrd xmm0, edx, 0
; Вывод реверсированного вектора.
      movdqa [pdivector_other], xmm0
      mov   rsi,pdivector_other
      mov   rdi,fmt4
      call  printpdi
; Выход.
mov   rsp,rbp
pop   rbp
ret

; Функция вывода. -----------------------------------------
printpdi:
push  rbp
mov   rbp,rsp
      movdqa xmm0, [rsi]
      ; Извлечение упакованных значений из регистра xmm0.
            pextrd esi, xmm0,0
            pextrd edx, xmm0,1
            pextrd ecx, xmm0,2
            pextrd r8d, xmm0,3
      mov   rax,0               ; Числа с плав. точкой не используются.
      call  printf
leave
ret

анализ иСхОднОгО кОда
В	рассматриваемом	здесь	примере	снова	обрабатываются	два	вектора,	на	сей	
раз	содержащие	целочисленные	значения.	Для	копирования	значений	в xmm
регистр	применяется	инструкция	 movdqa.	 Эта	инструкция	предназначена	для	
работы	 с  выровненными	 данными.	 Затем	 инструкция	 paddd	 суммирует	 зна
чения	в регистрах	и помещает	результат	в xmm0.	Для	использования	функции	
printf	 необходимо	 извлечь	 целочисленные	 значения	 из	 xmmрегистров	 и  за
писать	их	в «обычные»	регистры.	Следует	помнить	о соглашениях	о вызовах	
функций,	в соответствии	с которыми	функция	printf	считает,	что	xmmрегистр	
предназначен	для	значений	с плавающей	точкой.	Если	предварительно	не из
влечь	из	такого	регистра	целочисленные	значения,	то	printf	будет	рассматри
вать	содержимое	xmmрегистра	как	значения	с плавающей	точкой	и выводить	
неправильные	результаты.	Для	вставки	и	извлечения	упакованных	целых	чи
сел	используются	инструкции	pinsrd	и pextrd	соответственно.	Полученный	в ре
зультате	вектор	еще	и реверсируется,	чтобы	продемонстрировать,	как	вставля
ются	значения	в упакованный	вектор	в xmmрегистре.

Существуют	 также	 версии	 инструкций	 movd,	 padd,	 pinsr	 и  pextr	 для	 байтов,	
слов,	двойных	и четверных	слов.

На	рис. 29.1	показан	вывод	этой	программы.

Рис. 29.1.	Вывод программы sse_integer.asm

резюме
В	этой	главе	вы	узнали	об:

	� упакованных	целочисленных	данных;
	� инструкциях	 для	 вставки	 и  извлечения	 упакованных	 целочисленных	
данных;

	� инструкциях	для	копирования	и сложения	упакованных	целочисленных	
данных.

Глава 30
Обработка строк  
средствами SSE

В	SSE	версии	4.2	были	введены	четыре	инструкции	сравнения	строк:	две	ин
струкции	для	строк	с длиной,	не определенной	явно,	и две	инструкции	для	строк	
с явно	заданной	длиной.	Две	из	этих	четырех	инструкций	используют	маски.

Строки	с длиной,	не определенной	явно, –	это	строки	с завершающим	0.	Для	
строк	с явно	заданной	длиной	размер	должен	определяться	некоторыми	дру
гими	средствами.

В	этой	главе	немного	внимания	уделено	строкам	SSE,	потому	что	инструк
ции	 сравнения	для	них	несколько	более	 сложны	и необычны,	 особенно	при	
использовании	масок.	В табл. 30.1	перечислены	эти	инструкции.

Таблица 30.1.	Инструкции сравнения строк SSE

Строка Инструкция Аргумент1
(arg1)

Аргумент2
(arg2)

Аргумент3
(arg3)

Вывод

Неявная 
длина

pcmpistri xmm xmm/m128 imm8 Индекс в ecx

Неявная 
длина

pcmpistrm xmm xmm/m128 imm8 Маска в xmm0

Явная длина pcmpestri xmm xmm/m128 imm8 Индекс в ecx

Явная длина pcmpestrm xmm xmm/m128 imm8 Маска в xmm0

Смысловое	значение	этих	инструкций:

pcmpistri:	packed	compare	implicit	length	strings,	return	index –	сравнение	
упакованных	строк	с неявно	заданной	длиной,	возвращает	индекс;

pcmpistrm:	packed	compare	implicit	length	strings,	return	mask –	сравнение	
упакованных	строк	с неявно	заданной	длиной,	возвращает	маску;

pcmpestri:	packed	compare	explicit	length	strings,	return	index –	сравнение	
упакованных	строк	с явно	заданной	длиной,	возвращает	индекс;

pcmpestrm:	packed	compare	explicit	length	strings,	return	mask –	сравнение	
упакованных	строк	с явно	заданной	длиной,	возвращает	маску.



Эти	инструкции	сравнения	принимают	три	аргумента.	Аргумент	1	всегда	на
ходится	в xmmрегистре,	аргумент	2	может	располагаться	в xmmрегистре	или	в не
которой	локации	памяти,	аргумент	3	является	«непосредственным»	(immediate)	
и  представлен	 управляющим	 байтом	 (см.	 imm8	 в  руководствах	 Intel),	 который	
определяет,	 как	 выполняется	 инструкция.	 Управляющий	 байт	 играет	 важную	
роль,	поэтому	в следующем	разделе	он	будет	описан	более	подробно.

управляющий байт imm8
В	табл. 30.2	приведено	описание	отдельных	битов	управляющего	байта.

Таблица 30.2.	Управляющий байт imm8

Парамет-
ры

Но-
мер 
бита

Зна-
чение 
бита

Операция Описание

7 0 Зарезервирована Зарезервирован

Формат 
вывода

6 0 Битовая маска xmm0 содержит IntRes2 как 
битовую маску

1 Байтовая маска xmm0 содержит IntRes2 как 
байтовую маску

0 Самый младший индекс Самый младший индекс, най-
денный в ecx

1 Самый старший индекс Самый старший индекс, най-
денный в ecx

Знаковые 
разряды 
(ориента-
ция)

5, 4 00 + IntRes2 = IntRes1

01 − IntRes2 = ~IntRes1

10 Маскированный + IntRes2 = IntRes1

11 Маскированный − IntRes2 = ~IntRes1

Агрегация 
и сравне-
ние

3, 2 00 Равенство любому симво-
лу из набора

Совпадение символов

01 Равенство символу из диа-
пазона

Соответствие символам  
из заданного диапазона

10 Равенство в любой по-
зиции

Сравнение строк

11 Равенство в определен-
ном порядке

Поиск подстроки

Формат 
данных

1, 0 00 Упакованные беззнаковые 
байты

01 Упакованные беззнаковые 
слова

10 Упакованные знаковые 
байты

11 Упакованные знаковые 
слова

208	  Обработка строк средствами SSE 



Описанные	 выше	 инструкции	 сравнения	 принимают	 входные	 данные	 (их	
формат	определяется	в битах	1	и 0),	выполняют	операции	агрегации	и сравне
ния	(биты	3	и 2),	которые	дают	промежуточный	результат	(совпадение	между	
аргументами	 arg1	 и  arg2).	 В  руководствах	 Intel	 этот	 результат	 обозначен	 как	
IntRes1.	 Знаковые	разряды	 (биты	5	и  4)	 применяются	 к  промежуточному	ре
зультату	 IntRes1	для	получения	 IntRes2.	 Затем	 IntRes2	 используется	для	 выво
да	итогового	результата	в требуемом	формате.	Отрицание	(~IntRes1)	означает	
вычисление	дополнительного	кода	значения	IntRes1	и запись	полученного	ре
зультата	в IntRes2.	То есть	каждый	бит,	равный	1,	преобразуется	в 0,	а каждый	
бит,	равный	0,	преобразуется	в 1.	Другими	словами,	это	операция	логическо
го	отрицания	(logical	NOT).	Результат	в IntRes2	может	быть	сохранен	как	маска	
в регистре	xmm0	для	инструкций	с использованием	маски	pcmpistrm	и pcmpestrm 
или	как	индекс	в регистре	ecx	для	инструкций	pcmpistri	и pcmpestri.	Для	лучшего	
понимания	будет	полезным	рассмотрение	нескольких	примеров.

Ниже	приведены	примеры	содержимого	управляющего	байта:

00001000 or 0x08:
     00 – упакованные беззнаковые байты,
     10 – равенство в любой позиции,
     00 – положительная ориентация,
     00 – самый младший индекс в регистре ecx

01000100 or 0x44:
     00 – упакованные беззнаковые байты,
     01 – равенство в диапазоне,
     00 – положительная ориентация,
     01 – регистр xmm0 содержит байтовую маску

иСпОльзОвание управляющегО байта imm8
В	этом	разделе	показано,	как	можно	устанавливать	биты	в управляющем	байте	
imm8	для	управления	поведением	инструкциями	обработки	упакованных	строк.	
Кроме	того,	приведены	примеры,	демонстрирующие	воздействие	различных	
вариантов	установки	управляющих	битов.

Биты 0 и 1
Биты	0	и 1	определяют	формат	исходных	данных:	источником	данных	может	
быть	упакованный	байт	или	упакованное	слово	со	знаком	или	без	знака.

Биты 2 и 3
Биты	2	и 3	определяют	тип	применяемой	агрегации.	Результат	обозначается	как	
IntRes1	 (Intermediate	Result	1 –	промежуточный	результат	1).	Из второго	опе
ранда	берется	блок	из	16	байт	и сравнивается	с содержимым	первого	операнда.

Агрегация	может	выполняться	одним	из	следующих	методов:
	� равенство	любому	символу	из	набора	(00)	или	поиск	символа	из	задан
ного	набора:	это	означает	поиск	в операнде	1	любых	символов,	заданных	
в операнде	2.	При	обнаружении	совпадения	соответствующий	бит	в In-
tRes1	устанавливается	в 1.	Пример:

operand 1: "this is a joke!!"
operand 2: "i!"
IntRes1:    0010010000000011

	� равенство	символу	из	диапазона	(01)	или	поиск	символов	из	заданного	
диапазона:	это	означает	поиск	в операнде	1	любых	символов	из	диапа
зона,	определяемого	операндом	2.	При	обнаружении	совпадения	соот
ветствующий	бит	в IntRes1	устанавливается	в 1.	Пример:

operand 1: "this is a joke!!"
operand 2: "aj"
IntRes1:    0010010010100100

	� равенство	 в  любой	 позиции	 (10)	 или	 сравнение	 строк:	 это	 означает	
сравнение	каждого	символа	в операнде	1	с соответствующим	символом	
в операнде	2.	При	обнаружении	совпадения	соответствующий	бит	в In-
tRes1	устанавливается	в 1.	Пример:

operand 1: "this is a joke!!"
operand 2: "this is no joke!"
IntRes1:    1111111100000000

	� равенство	в определенном	порядке	(11)	или	поиск	подстроки:	это	озна
чает	поиск	в операнде	1	строки,	определенной	в операнде	2.	При	обна
ружении	совпадения	соответствующий	бит	в IntRes1	устанавливается	в 1.	
Пример:

operand 1: "this is a joke!!"
operand 2: "is"
IntRes1:    0010010000000000

Биты 4 и 5
Биты	4	и 5	применяют	знак	ориентации	(polarity)	и сохраняют	результат	в In-
tRes2	по	следующим	правилам:

	� положительная	ориентация	(00)	и (10):	IntRes2	равнозначен	IntRes1.	При
мер:

IntRes1: 0010010000000011
IntRes2: 0010010000000011

	� отрицательная	ориентация	(01)	и (11):	IntRes2	является	дополнительным	
кодом	или	логическим	отрицанием	IntRes1.	Пример:

IntRes1: 0010010000000011
IntRes2: 1101101111111100

Бит 6
Бит	6	определяет	формат	вывода	в двух	возможных	вариантах:

	� без	использования	маски:
	� 0:	индекс,	возвращаемый	в регистре	ecx,	является	самым	младшим	би
том,	установленным	в 1	в IntRes2.	Пример:

IntRes2: 0010010011000000
ecx = 6
В IntRes2 самый первый бит, установленный в 1, найден по индексу 6 (отсчет  
начинается с 0 с правой стороны).

	� 1:	индекс,	возвращаемый	в регистре	ecx,	является	самым	старшим	би
том,	установленным	в 1	в IntRes2.	Пример:

IntRes2: 0010010010100100
ecx = 13
В IntRes2 самый последний бит, установленный в 1, найден по индексу 13 (отсчет  
начинается с 0 с правой стороны).

	� с	использованием	маски:
	� 0:	 IntRes2	 возвращается	 как	 маска	 в  младших	 битах	 регистра	 xmm0 
(с распространением	нуля	до	128	бит).	Пример:

Поиск всех символов 'a' и 'e' в строке
string = 'qdacdekkfijlmdoz'
затем
xmm0: 024h
или в двоичном формате 0000000000100100

Обратите	внимание:	в регистре	xmm0	маска	реверсирована.
	� 1:	IntRes2	развертывается	в байтовую	маску	/	маску	слова	в регистре	
xmm0.	Пример:

Поиск всех символов 'a' и 'e' в строке
string = 'qdacdekkfijlmdoz'
затем
xmm0:  00000000000000000000ff0000ff0000

Обратите	внимание:	в регистре	xmm0	маска	реверсирована.

Зарезервированный бит 7
Бит	7	зарезервирован.

флаги
Для	инструкций	обработки	строк	с неявно	заданной	длиной	флаги	использу
ются	способом,	отличающимся	от	того,	что	вы	видели	в предыдущих	главах	
(см.	руководства	Intel).

Таблица 30.3.	Использование флагов для инструкций обработки строк с неявно заданной 
длиной

Флаг Описание

CF Сбрасывается в 0, если IntRes2 равен нулю, иначе устанавливается в 1

ZF Устанавливается в 1, если любой байт/слово в xmm2/mem128 является нулевым, 
иначе сбрасывается в 0

SF Устанавливается в 1, если любой байт/слово в xmm1 является нулевым, иначе 
сбрасывается в 0

OF IntRes2[0]

AF Сбрасывается в 0

PF Сбрасывается в 0

Для	инструкций	обработки	строк	с явно	заданной	длиной	флаги	также	ис
пользуются	другими	способами,	как	показано	в табл. 30.4.

Таблица 30.4.	Использование флагов для инструкций обработки строк с явно заданной длиной

Флаг Описание

CF Сбрасывается в 0, если IntRes2 равен нулю, иначе устанавливается в 1

ZF Устанавливается в 1, если абсолютное значение регистра EDX < 16 (8), иначе 
сбрасывается в 0

SF Устанавливается в 1, если абсолютное значение регистра EDX < 16 (8), иначе 
сбрасывается в 0

OF IntRes2[0]

AF Сбрасывается в 0

PF Сбрасывается в 0

В	примерах	следующей	главы	будет	использоваться	флаг	CF,	для	того	чтобы	
определить,	был	ли	получен	какойлибо	результат,	и флаг	ZF	для	определения	
конца	строки.

Эта	теоретическая	часть	может	показаться	немного	сложной,	поэтому	пора	
перейти	к практике.

резюме
В	этой	главе	вы	узнали	о:

	� SSEинструкциях	обработки	строк;
	� схеме	расположения	битов	в управляющем	регистре	imm8	и их	использо
вании.

Глава 31
Поиск символа в строке

В	этой	главе	мы	начнем	использовать	управляющий	байт	для	того,	чтобы	вы
полнить	поиск	заданного	символа	в строке.

Определение длины СтрОки
В	первом	примере	определяется	длина	строки	методом	поиска	завершающего	0.

В	листинге	31.1	показан	исходный	код	примера.

Листинг 31.1.	Программа sse_string_length.asm

; sse_string_length.asm
extern printf
section .data
; Шаблон             0123456789abcdef0123456789abcdef0123456789abcd  e
; Шаблон             1234567890123456789012345678901234567890123456  7
      string1 db    "The quick brown fox jumps over the lazy river.",0
      fmt1 db       "This is our string: %s ",10,0
      fmt2 db       "Our string is %d characters long.",10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      mov   rdi, fmt1
      mov   rsi, string1
      xor   rax,rax
      call  printf
      mov   rdi, string1
      call  pstrlen
      mov   rdi, fmt2
      mov   rsi, rax
      xor   rax,rax
      call  printf
leave
ret
; Функция вычисления длины строки. -------------------------
pstrlen:
push  rbp



mov   rbp,rsp
      mov   rax,  -16         ; Исключение изменений в дальнейшем.
      pxor  xmm0, xmm0        ; 0 (конец строки).
.not_found:
      add        rax, 16      ; Исключение изменений ZF в дальнейшем –
                              ; после вызова pcmpistri.
      pcmpistri  xmm0, [rdi + rax], 00001000b   ; Равенство заданному символу.
      jnz      .not_found  ; 0 найден?
      add      rax, rcx    ; rcx содержит индекс завершающего 0.
      inc      rax         ; Корректировка индекса 0 относительно начала строки.
leave
ret

В	начале	этой	программы	добавлены	два	шаблона	в комментариях,	для	того	
чтобы	упростить	подсчет	символов.	В одном	шаблоне используется	десятичная	
нумерация,	начинающаяся	с 1,	в другом	шаблоне нумерация	шестнадцатерич
ная,	начинающаяся	с индекса	0.

; Шаблон      1234567890123456789012345678901234567890123456  7
; Шаблон      0123456789abcdef0123456789abcdef0123456789abcd  e
string1 db   "The quick brown fox jumps over the lazy river.",0

Сначала,	как	обычно,	выводятся	строки.	Затем	вызывается	специально	на
писанная	для	этого	примера	функция	поиска	pstrlen.	Эта	функция	выполняет	
поиск	в строке	(сканирование)	первого	вхождения	нулевого	байта.	Инструкция	
pcmpistri	поочередно	анализирует	блоки	из	16	байт,	регистр	rax	используется	
как	счетчик	блоков.	Если	pcmpistri	находит	нулевой	байт	в текущем	блоке,	то	
устанавливается	флаг	ZF,	который	используется	для	принятия	решения	о пере
ходе.	Необходимо	исключить	 ситуацию,	 в  которой	 увеличение	 значения	 rax 
воздействует	на	флаг	ZF	 непосредственно	перед	выполнением	перехода,	по
этому	 требуется	 инкрементирование	 флага	 ZF	 перед	 инструкцией	 pcmpistri.	
Именно	 поэтому	мы	начинаем	 с  записи	 значения	 −16	 в  регистр	 rax,	 теперь	
можно	увеличивать	rax	перед	использованием	pcmpistri.	Обратите	внимание	
на	инструкцию	pxor:	это	логическая	инструкция	or	для	xmmрегистров.	Для	SIMD	
существуют	отдельные	логические	инструкции.

Управляющий	байт	imm8	содержит	значение	00001000,	которое	имеет	следую
щий	смысл:

	� 00 –	упакованные	беззнаковые	байты;
	� 10 –	равенство	любому	символу	из	заданного	набора	(в данном	случае	
равенство	одному	символу);

	� 00 –	положительная	ориентация;
	� 0 –	самый	младший	индекс;
	� 0 –	зарезервирован.

Можно	 было	 бы	 предположить	 использование	 метода	 поиска	 «равенство	
в любой	позиции»	(equal	any)	для	поиска	любого	0.	Но вместо	этого	применяется	
метод	«равенство	любому	символу	из	заданного	набора»	(equal	each).	Почему?

Необходимо	всегда	помнить	о том,	что	инструкция	pcmpistri	инициализи
рует	регистр	rcx	значением	16,	т. е.	количеством	байтов	в блоке.	Если	найден	

совпадающий	 байт,	 то	 pcmpistri	 копирует	 индекс	 этого	 совпадающего	 байта	
в регистр	rcx.	Если	совпадение	не обнаружено,	то	в rcx	содержится	значение	16.

Заглянем	в руководства	Intel,	точнее	в Volume	2B,	Section	4.1.6	«Valid/Invalid	
Override	of	Comparisons».	Этот	раздел	объясняет,	что	происходит,	когда	блок	со
держит	«некорректные»	байты,	т. е.	байты,	расположенные	после	конца	строки.

Для	описания	этой	ситуации	можно	воспользоваться	табл. 31.1.

Таблица 31.1.	Обработка «некорректных» байтов в блоке

xmm0 Память Равенство в любой 
позиции (equal any)

Равенство любому символу из 
заданного набора (equal each)

Некорректное 
значение

Некоррект-
ное значение

Принудительно вычислен-
ное значение false (ложь)

Принудительно вычисленное 
значение true (истина)

Некорректное 
значение

Корректное 
значение

Принудительно вычислен-
ное значение false (ложь)

Принудительно вычисленное 
значение false (ложь)

В	 регистре	 xmm0	 некорректное	 значение,	 потому	 что	 при	 инициализации	
в него	было	записано	значение	0	байт.	При	получении	16байтового	блока,	со
держащего	нулевой	байт,	в случае	«равенство	в любой	позиции»	(equal	any)	ин
струкция	pcmpistri	определяет,	что	один	из	16	байт	содержит	0.	В этот	момент	
значения	в xmm0	и в	памяти	некорректны.	Но инструкция	pcmpistri	спроектиро
вана	так,	чтобы	выдавать	«принудительно	вычисленное	значение	ложь	(force	
false)»	 в  случае	 «равенство	 в  любой	 позиции»	 (equal	 any).	 Поэтому	 pcmpistri 
считает,	что	совпадений	нет,	и возвращает	16	в rcx,	следовательно,	вычисление	
длины	строки	дает	неправильный	результат.

Но	при	использовании	метода	«равенство	любому	символу	из	заданного	на
бора	(equal	each)»	регистр	xmm0	содержит	некорректное	значение,	как	и в	опи
санном	выше	случае,	однако	как	только	pcmpistri	считывает	завершающий	ну
левой	байт	в блоке,	выдается	«принудительно	вычисленное	значение	истина	
(force	true)»	в соответствии	с запроектированным	поведением	этой	инструк
ции.	Индекс	нулевого	байта	записывается	в регистр	rcx.	После	этого	значение	
ecx	можно	использовать	для	правильного	вычисления	конца	строки.

Предупреждение:	программа	считывает	блоки	размером	16	байт.	Это	рабо
тает	нормально,	пока	данные,	в которых	выполняется	поиск,	находятся	в про
странстве	 памяти,	 выделенном	 этой	 программе.	 Если	 программа	 пытается	
читать	данные	за	границей	разрешенной	памяти,	то	происходит	аварийное	за
вершение.	Этого	можно	избежать,	постоянно	отслеживая	текущее	положение	
на	странице	памяти	(в большинстве	случаев	страница –	это	фрагмент	памяти	
размером	4	Кб),	и если	вы	оказываетесь	слишком	близко	к границе	страницы,	
то	необходимо	перейти	на	режим	чтения	по	одному	байту	последовательно.	
Такой	 способ	 обеспечит	 защиту	 от	 случайного	 пересечения	 границы	 разре
шенной	памяти	и вторжения	на	страницу	памяти	другого	процесса.	Здесь	эта	
методика	не реализована,	чтобы	не усложнять	пример	программы	и ее	описа
ние.	Но вы	должны	знать	о возможности	возникновения	такой	ситуации.

На	рис. 31.1	показан	вывод	этой	программы.	Можно	видеть,	что	длина	стро
ки	учитывает	завершающий	ноль.

Рис. 31.1.	Вывод программы sse_string_length.asm

пОиСк в СтрОках
Теперь,	когда	известно,	как	определить	длину	строки,	можно	перейти	к выпол
нению	поиска	в строках	(см.	листинг	31.2).

Листинг 31.2.	Программа sse_string_search.asm

; sse_string_search.asm
extern printf
section .data
; Шаблон      123456789012345678901234567890123456789012345  6
; Шаблон      0123456789abcdef0123456789abcdef0123456789abc  d
string1 db   "the quick brown fox jumps over the lazy river",0
string2      db    "e",0
fmt1         db    "This is our string: %s ",10,0
fmt2         db    "The first '%s' is at position %d.",10,0
fmt3         db    "The last '%s' is at position %d.",10,0
fmt4         db    "The character '%s' didn't show up!.",10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      mov   rdi, fmt1
      mov   rsi, string1
      xor   rax,rax
      call  printf
; Поиск первого вхождения.  
      mov   rdi, string1
      mov   rsi, string2
      call  pstrscan_f
      cmp   rax,0
      je    no_show
      mov   rdi, fmt2
      mov   rsi, string2
      mov   rdx, rax
      xor   rax,rax
      call  printf
; Поиск последнего вхождения.  
      mov   rdi, string1
      mov   rsi, string2
      call  pstrscan_l
      mov   rdi, fmt3
      mov   rsi, string2
      mov   rdx, rax
      xor   rax,rax

      call  printf
      jmp   exit
no_show:
      mov   rdi, fmt4
      mov   rsi, string2
      xor   rax, rax
      call  printf
exit:
leave
ret
;------ Поиск первого вхождения. ----------------------
pstrscan_f:
push  rbp
mov   rbp,rsp
      xor    rax, rax
      pxor   xmm0, xmm0
      pinsrb xmm0, [rsi],0
.block_loop:
      pcmpistri  xmm0, [rdi + rax], 00000000b
      jc    .found
      jz    .none
      add   rax, 16
      jmp   .block_loop
.found:
      add   rax, rcx            ; rcx содержит позицию символа.
      inc   rax                 ; Начало отсчета с 1 вместо 0.
leave
ret
.none:
      xor   rax,rax             ; Ничего не найдено, возвращается 0.
leave
ret
;------ Поиск последнего вхождения. ----------------------
pstrscan_l:
push  rbp
mov   rbp,rsp
push  rbx                 ; Регистр, сохраняемый вызываемой функцией.
push  r12                 ; Регистр, сохраняемый вызываемой функцией.
      xor    rax, rax
      pxor   xmm0, xmm0
      pinsrb xmm0, [rsi],0
      xor    r12,r12
.block_loop:
      pcmpistri  xmm0, [rdi + rax], 01000000b
      setz   bl
      jc     .found
      jz     .done
      add    rax, 16
      jmp    .block_loop
.found:
      mov    r12, rax
      add    r12, rcx    ; rcx содержит позицию символа.
      inc    r12
      cmp    bl,1
      je     .done
      add    rax,16
      jmp    .block_loop

pop r12                  ; Регистр, сохраняемый вызываемой функцией.
pop rbx                  ; Регистр, сохраняемый вызываемой функцией.
leave
ret
.done:
      mov    rax,r12
pop r12                  ; Регистр, сохраняемый вызываемой функцией.
pop rbx                  ; Регистр, сохраняемый вызываемой функцией.
leave
ret

В	начале	этой	программы	добавлены	два	шаблона	в комментариях,	для	того	
чтобы	упростить	подсчет	символов	в строке.

Здесь	string1	содержит	строку,	в которой	производится	поиск,	а в	string2	со
держится	искомый	аргумент.	Выполняется	поиск	первого	и последнего	вхож
дений	 искомого	 аргумента.	 Сначала	 выводятся	 строки,	 затем	 вызываются	
специально	созданные	для	этого	примера	функции.	Для	поиска	первого	вхож
дения	 заданного	 символа	и  его	 последнего	 вхождения	написаны	отдельные	
функции.	Функция	pstrscan_f	 выполняет	поиск	первого	вхождения	искомого	
аргумента.	 Инструкция	 pcmpistri	 поочередно	 обрабатывает	 блоки	 размером	
16 байт,	в качестве	счетчика	блоков	используется	регистр	rax.	Регистр	xmm0	очи
щается	 инструкцией	 pxor.	 Инструкция	 pinsrb	 позволяет	 поместить	 искомый	
аргумент	в младший	байт	регистра	xmm0	(байт	0).	Для	поиска	вхождений	при
меняется	метод	«равенство	в любой	позиции»	(equal	any),	и как	только	найдено	
вхождение,	в rcx	содержится	индекс	совпавшего	байта	в текущем	16байтовом	
блоке.	Если	в текущем	блоке	не обнаружено	совпадение,	то	в регистр	rcx	запи
сывается	значение	16.	С помощью	инструкции	jc	проверяется	флаг	CF=1.	Если	
флаг	установлен,	то	совпадение	найдено,	выполняется	сложение	значения	rcx 
со	значением	rax,	в котором	содержится	количество	байтов,	уже	просмотрен
ных	в предыдущих	блоках,	затем	значение	rax	возвращается	и корректируется	
для	отсчета,	начиная	с 1	вместо	0.

Если	CF=0,	то	с помощью	инструкции	jz	определяется,	находимся	ли	мы	в по
следнем	блоке.	Инструкция	pcmpistri	устанавливает	ZF=1,	если	обнаружен	нуле
вой	байт,	и регистр	rax	очищается,	потому	что	совпадение	не найдено.	В этом	
случае	функция	возвращает	0.

Разумеется,	здесь	нет	никаких	проверок	на	ошибки:	если	строка	не заверша
ется	нулевым	байтом,	то	вы	можете	получить	ошибочные	результаты.	Попро
буйте	удалить	0	в конце	строки	и посмотрите,	что	получится.

Функция	 pstrscan_l	 выполняет	 поиск	 последнего	 вхождения	 искомого	 ар
гумента.	 Это	 более	 сложная	 операция,	 чем	 поиск	 первого	 вхождения	 и  вы
ход.	Требуется	прочитать	все	16байтовые	блоки	и отследить	самое	последнее	
вхождение	искомого	символа	в блоке.	Но даже	когда	вхождение	обнаружено,	
необходимо	 продолжать	 выполнение	 цикла	 до	 тех	 пор,	 пока	 не  встретится	
завершающий	нулевой	байт.	Для	постоянного	наблюдения	за	 завершающим	
нулевым	байтом	регистр	bl	устанавливается	в 1	сразу,	как	только	обнаружен	
нулевой	байт.	Регистр	r12	используется	для	записи	индекса	самого	последнего	
найденного	совпадения.	См.	рис. 31.2.

Рис. 31.2.	Вывод программы sse_string_search.asm

резюме
В	этой	главе	вы	узнали,	как:

	� использовать	инструкцию	pcmpistri	для	поиска	символов	и определения	
длины	строки;

	� интерпретировать	 результат	 выполнения	 инструкции	 pcmpistri	 с  по
мощью	различных	значений	управляющего	байта	imm8.

Глава 32
Сравнение строк

В	предыдущей	главе	мы	работали	со	строками	с неявно	заданной	длиной,	т. е.	
со	строками,	завершающимися	нулевым	байтом.	В этой	главе	рассматривается	
сравнение	строк	с неявно	заданной	длиной	и строк	с явно	заданной	длиной.

СтрОки С неявнО заданнОй длинОй
Вместо	 поиска	 совпадений	 мы	 будем	 искать	 символы,	 которые	 отличаются	
друг	от	друга.	В листинге	32.1	показан	пример	исходного	кода.

Листинг 32.1.	Программа sse_string2_imp.asm

; sse_string2_imp.asm
; Сравнение строк с неявно заданной длиной.
extern printf
section .data
       string1    db    "the quick brown fox jumps over the lazy"
                  db    " river",10,0
       string2    db    "the quick brown fox jumps over the lazy"
                  db    " river",10,0
       string3    db    "the quick brown fox jumps over the lazy
                        " dog",10,0
       fmt1   db "Strings 1 and 2 are equal.",10,0
       fmt11  db "Strings 1 and 2 differ at position %i.",10,0
       fmt2   db "Strings 2 and 3 are equal.",10,0
       fmt22  db "Strings 2 and 3 differ at position %i.",10,0
section .bss
section .text
       global main
main:
push  rbp
mov   rbp,rsp
; Сначала выводятся строки.
      mov   rdi, string1
      xor   rax,rax
      call  printf
      mov   rdi, string2
      xor   rax,rax
      call  printf
      mov   rdi, string3
      xor   rax,rax
      call  printf



; Сравнение строк 1 и 2.
      mov   rdi, string1
      mov   rsi, string2
      call  pstrcmp
      mov   rdi,fmt1
      cmp   rax,0
      je    eql1          ; Строки равны.
      mov   rdi,fmt11     ; Строки не равны.
 eql1:
      mov   rsi, rax
      xor   rax,rax
      call  printf
 ; Сравнение строк 2 и 3.
      mov   rdi, string2
      mov   rsi, string3
      call  pstrcmp
      mov   rdi,fmt2
      cmp   rax,0
      je    eql2          ; Строки равны.
      mov   rdi,fmt22     ; Строки не равны.
eql2:
      mov   rsi, rax
      xor   rax,rax
      call  printf
; Выход.
leave
ret
; Сравнение строк. ----------------------------------------------
pstrcmp:
push  rbp
mov   rbp,rsp
      xor    rax, rax            ;
      xor    rbx, rbx            ;
.loop: movdqu    xmm1, [rdi + rbx]
      pcmpistri  xmm1, [rsi + rbx], 0x18 ; Метод equal each | отрицательная ориентация.
      jc         .differ
      jz         .equal
      add        rbx, 16
      jmp        .loop
.differ:
      mov rax,rbx
      add rax,rcx    ; Позиция отличающегося символа.
      inc rax        ; Потому что индекс начинается с 0.
.equal:
leave
ret

Как	 обычно,	 сначала	 выводятся	 сами	 строки,	 затем	 вызывается	 функция	
pstrcmp	для	сравнения	строк.	Самая	важная	информация	содержится	в функции	
pstrcmp.	Управляющий	байт	содержит	значение	0x18	или	00011000,	т. е.	справа	на
лево:	упакованные	целочисленные	байты,	метод	«равенство	любому	символу	
из	заданного	набора»	(equal	each),	отрицательная	ориентация.	Регистр	ecx	со
держит	индекс	первого	вхождения.	Инструкция	pcmpistri позволяет	восполь
зоваться	флагами –	соответствующее	описание	можно	найти	в руководствах	
Intel:

	� CFlag –	сбрасывается	в 0,	если	IntRes2	равен	0,	иначе	устанавливается	в 1;
	� ZFlag –	 устанавливается	в 1,	 если	любой	байт/слово	в xmm2/mem128	 равен	
нулю,	иначе	сбрасывается	в 0;

	� SFlag –	 устанавливается	 в  1,	 если	любой	байт/слово	 в  xmm1	 равен	нулю,	
иначе	сбрасывается	в 0;

	� OFlag –	IntRes2[0];
	� AFlag –	сбрасывается	в 0;
	� PFlag –	сбрасывается	в 0.

В	рассматриваемом	здесь	примере	инструкция	pcmpistri	 записывает	1	для	
каждого	 обнаруженного	 совпадения	 в  соответствующую	позицию	промежу
точного	 результата	 IntRes1.	 Если	 найден	 отличающийся	 байт,	 то	 в  соответ
ствующую	позицию	IntRes1	записывается	ноль.	Затем	формируется	промежу
точный	результат	IntRes2	и применяется	отрицательная	ориентация	к IntRes1.	
IntRes2	будет	содержать	1	по	индексу	отличающегося	байта	(результат	отрица
тельной	ориентации),	поэтому	IntRes2	не равен	нулю,	и флаг	CF	устанавливает
ся	в 1.	После	этого	цикл	будет	прерван,	и функция	pstrcmp	возвращает	в регист
ре  rax	 позицию	отличающегося	 символа.	 Если	флаг	 CF	 не  установлен	в  1,	 но	
инструкция	pcmpistri	обнаружила	завершающий	ноль,	то	функция	возвращает	
в регистре	rax	значение	0.

На	рис. 32.1	показан	вывод	этой	программы.

Рис. 32.1.	Вывод программы sse_string2_imp.asm

СтрОки С явнО заданнОй длинОй
В	основном	мы	используем	строки	с неявно	заданной	длиной,	но	в листин
ге 32.2	показан	пример	обработки	строк	с явно	заданной	длиной.

Листинг 32.2.	Программа sse_string3_exp.asm

; sse_string3_exp.asm
; Сравнение строк с явно заданной длиной.
extern printf
section .data
      string1      db      "the quick brown fox jumps over the "
                   db      "lazy river"
      string1Len equ $ - string1
      string2      db      "the quick brown fox jumps over the "
                   db      "lazy river"
      string2Len equ $ - string2
      dummy  db "confuse the world"
      string3      db      "the quick brown fox jumps over the "
                   db      "lazy dog"
      string3Len equ $ - string3

      fmt1  db "Strings 1 and 2 are equal.",10,0
      fmt11 db "Strings 1 and 2 differ at position %i.",10,0
      fmt2  db "Strings 2 and 3 are equal.",10,0
      fmt22 db "Strings 2 and 3 differ at position %i.",10,0

section .bss
        buffer resb 64
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Сравнение строк 1 и 2.
    mov      rdi, string1
    mov      rsi, string2
    mov      rdx, string1Len
    mov      rcx, string2Len
    call     pstrcmp
    push     rax    ; Запись результата в стек для дальнейшего использования.
; Вывод строк 1 и 2 и результата.
;-------------------------------------------------------------
; Сначала создается строка с символом перехода на новую строку
; и завершающим 0.
; string1
    mov      rsi,string1
    mov      rdi,buffer
    mov      rcx,string1Len
    rep      movsb
    mov      byte[rdi],10    ; Добавление NL в буфер.
    inc      rdi             ; Добавление завершающего 0 в буфер.
    mov      byte[rdi],0
; Вывод.
    mov      rdi, buffer
    xor      rax,rax
    call     printf
; string2
    mov      rsi,string2
    mov      rdi,buffer
    mov      rcx,string2Len
    rep      movsb
    mov      byte[rdi],10    ; Добавление NL в буфер.
    inc      rdi             ; Добавление завершающего 0 в буфер.
    mov      byte[rdi],0
; Вывод.
    mov      rdi, buffer
    xor      rax,rax
    call     printf
;-------------------------------------------------------------
; Теперь вывод результата сравнения.
    pop      rax         ; Восстановление возвращаемого значения.
    mov      rdi,fmt1
    cmp      rax,0
    je       eql1
    mov      rdi,fmt11
 eql1:
    mov      rsi, rax

Строки с явно заданной длиной  223



    xor      rax,rax
    call     printf
;-------------------------------------------------------------
;-------------------------------------------------------------
; Сравнение строк 2 и 3.
    mov      rdi, string2
    mov      rsi, string3
    mov      rdx, string2Len
    mov      rcx, string3Len
    call     pstrcmp
    push     rax
; Вывод строки string3 и результата.
;-------------------------------------------------------------
; Сначала создается строка с символом перехода на новую строку
; и завершающим 0.
; string3
    mov      rsi,string3
    mov      rdi,buffer
    mov      rcx,string3Len
    rep      movsb
    mov      byte[rdi],10    ; Добавление NL в буфер.
    inc      rdi             ; Добавление завершающего 0 в буфер.
    mov      byte[rdi],0
; Вывод.
    mov      rdi, buffer
    xor      rax,rax
    call     printf
;-------------------------------------------------------------
; Теперь вывод результата сравнения.
    pop      rax                  ; Восстановление возвращаемого значения.
    mov      rdi,fmt2
    cmp      rax,0
    je       eql2
    mov      rdi,fmt22
eql2:
    mov      rsi, rax
    xor      rax,rax
    call     printf
; Выход.
leave
ret
;-------------------------------------------------------------
pstrcmp:
push   rbp
mov    rbp,rsp
       xor     rbx, rbx
       mov     rax,rdx         ;rax содержит длину 1-й строки.
       mov     rdx,rcx         ;rdx содержит длину 2-й строки.
       xor     rcx,rcx         ;rcx как индекс.
.loop:
       movdqu      xmm1, [rdi + rbx]
       pcmpestri xmm1, [rsi + rbx], 0x18   ; Метод equal each | отр. ориентация.
       jc     .differ
       jz     .equal
       add    rbx, 16
       sub    rax,16

       sub    rdx,16
       jmp    .loop
.differ:
       mov    rax,rbx
       add    rax,rcx     ; rcx содержит позицию отличающегося символа.
       inc    rax         ; Потому что отсчет начинается с 0.
       jmp    exit
.equal:
       xor    rax,rax
exit:
leave
ret

Здесь	можно	видеть,	что	обработка	строк	с явно	заданной	длиной	иногда	
может	 становиться	 достаточно	 сложной.	 Так	 зачем	 же	 использовать	 такие	
строки?	Строки	с явно	заданной	длиной	используют	многие	протоколы	обмена	
информацией.	Или,	возможно,	в какомлибо	приложении	потребуется	исполь
зование	 данных	 с  нулевыми	 значениями.	 Так	 или	 иначе,	 необходимо	 обес
печить	представление	информации	о длине строк.	В рассматриваемом	здесь	
примере	длина	строк	вычислялась	по	локациям	памяти	в разделе	section .data.	
Но функция	printf	ожидает	передачи	строк,	завершающихся	нулевым	байтом.	
Поэтому	после	демонстрации	сравнения	строк	с явно	заданной	длиной	строки	
переформатируются	в буфере:	добавляется	символ	перехода	на	новую	строку	
и завершающий	ноль	непосредственно	в буфер,	после	чего	отредактирован
ный	буфер	передается	в функцию	printf.

Теперь	рассмотрим	функцию	сравнения	pstrcmp.	Длина	первой	строки	запи
сывается	в регистр	rax,	длина	второй	строки –	в rdx.	Затем	начинается	цикл:	
адрес	 16байтового	 блока	 памяти	 загружается	 в  регистр	 xmm1,	 и  выполняет
ся	 инструкция	 pcmpestri	 с  предварительно	 установленным	 значением	 0x18 
в управляющем	байте.	Далее	проверяем	флаги:	их	описание	можно	найти	в ру
ководствах	Intel:

	� CFlag –	сбрасывается	в 0,	если	IntRes2	равен	нулю,	иначе	устанавливается	
в 1;

	� ZFlag –	устанавливается	в 1,	если	абсолютное	значение	EDX	меньше	16	(8),	
иначе	сбрасывается	в 0;

	� SFlag –	устанавливается	в 1,	если	абсолютное	значение	EAX	меньше	16	(8),	
иначе	сбрасывается	в 0;

	� OFlag –	IntRes2[0];
	� AFlag –	сбрасывается	в 0;
	� PFlag –	сбрасывается	в 0.

Обратите	внимание:	инструкции	pcmpestri	и pcmpistri	используют	флаги	ZF 
и CF	поразному.	Вместо	флага	ZF,	сигнализирующего	о появлении	завершаю
щего	нуля,	на	каждой	итерации	цикла	уменьшаются	значения	регистров	rax 
и rdx,	и когда	значение	в одном	из	этих	регистров	становится	меньше	16,	цикл	
завершается.

На	рис. 32.2	показан	вывод	этой	программы.

Рис. 32.2.	Вывод программы sse_string3_exp.asm

резюме
В	этой	главе	вы	узнали:

	� о	строках	с неявно	и явно	заданной	длиной;
	� о	смысле	отрицательной	ориентации;
	� об	использовании	флагов.

Глава 33
Перемешиваем данные

Применяя	 инструкции	 обработки	 строк	 без	 использования	масок,	 мы	 полу
чаем	 в  свое	 распоряжение	 несколько	 возможных	 вариантов.	 Можно	 найти	
первое	или	последнее	вхождение	символа,	но	поиск	всех	вхождений	является	
более	трудной	задачей.	Можно	сравнивать	строки	и находить	различие,	но	по
иск	всех	различий	не такто	прост.	К счастью,	существуют	еще	и инструкции	
обработки	строк,	которые	используют	маски,	что	придает	им	гораздо	большую	
мощь.	Но,	прежде	чем	начать	углубленное	изучение	инструкций	с применени
ем	масок,	необходимо	рассмотреть	операции	перемешивания.

ОСнОвные принципы Операций перемешивания
Перемешивание	(shuffling;	или	тасование,	перетасовка)	означает	перестанов
ку	в произвольном	порядке	упакованных	значений.	Перестановка	может	вы
полняться	в одном	и том	же	xmmрегистре,	или	из	одного	xmmрегистра	в другой	
xmmрегистр,	или	из	128битового	блока	памяти	в xmmрегистр.

В	листинге	33.1	показан	пример	исходного	кода.

Листинг 33.1.	Программа shuffle.asm

; shuffle.asm
extern printf
section .data
      fmt0  db "These are the numbers in memory: ",10,0
      fmt00 db "This is xmm0: ",10,0
      fmt1  db "%d ",0
      fmt2  db "Shuffle-broadcast double word %i:",10,0
      fmt3  db "%d %d %d %d",10,0
      fmt4  db "Shuffle-reverse double words:",10,0
      fmt5  db "Shuffle-reverse packed bytes in xmm0:",10,0
      fmt6  db "Shuffle-rotate left:",10,0
      fmt7  db "Shuffle-rotate right:",10,0
      fmt8  db "%c%c%c%c%c%c%c%c%c%c%c%c%c%c%c%c",10,0
      fmt9  db "Packed bytes in xmm0:",10,0
      NL    db 10,0

      number1     dd 1
      number2     dd 2
      number3     dd 3
      number4     dd 4



      char  db "abcdefghijklmnop"
      bytereverse db 15,14,13,12,11,10,9,8,7,6,5,4,3,2,1,0

section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      sub  rsp,32   ; Пространство в стеке для исходного xmm0
                    ; и для измененного xmm0.
; ПЕРЕМЕШИВАНИЕ ДВОЙНЫХ СЛОВ.
; Сначала вывод чисел в обратном порядке.
      mov   rdi, fmt0
      call  printf
      mov   rdi, fmt1
      mov   rsi, [number4]
      xor   rax,rax
      call  printf
      mov   rdi, fmt1
      mov   rsi, [number3]
      xor   rax,rax
      call  printf
      mov   rdi, fmt1
      mov   rsi, [number2]
      xor   rax,rax
      call  printf
      mov   rdi, fmt1
      mov   rsi, [number1]
      xor   rax,rax
      call  printf
      mov   rdi, NL
      call  printf
; Наполнение регистра xmm0 числами.
      pxor      xmm0,xmm0
      pinsrd    xmm0, dword[number1],0
      pinsrd    xmm0, dword[number2],1
      pinsrd    xmm0, dword[number3],2
      pinsrd    xmm0, dword[number4],3
      movdqu    [rbp-16],xmm0   ; Сохранение xmm0 для дальнейшего использования.
      mov       rdi, fmt00
      call      printf          ; Вывод заголовка.
      movdqu    xmm0,[rbp-16]   ; Восстановление xmm0 после printf.
      call      print_xmm0d     ; Вывод xmm0.
      movdqu    xmm0,[rbp-16]   ; Восстановление xmm0 после printf.

; ПЕРЕМЕШИВАНИЕ В ПРОИЗВОЛЬНОМ ПОРЯДКЕ.
; Перемешивание: случайная перестановка младшего двойного слова (индекс 0).
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,00000000b   ; Перемешивание.
      mov       rdi,fmt2
      mov       rsi, 0                ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.

; Перемешивание: случайная перестановка двойного слова с индексом 1.
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,01010101b   ; Перемешивание.
      mov       rdi,fmt2
      mov       rsi, 1                ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.
; Перемешивание: случайная перестановка двойного слова с индексом 2.
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,10101010b   ; Перемешивание.
      mov       rdi,fmt2
      mov       rsi, 2                ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.
; Перемешивание: случайная перестановка двойного слова с индексом 3.
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,11111111b   ; Перемешивание.
      mov       rdi,fmt2
      mov       rsi, 3                ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.

; ПЕРЕМЕШИВАНИЕ: ПЕРЕСТАНОВКА В ОБРАТНОМ ПОРЯДКЕ.
; Реверсирование двойных слов.
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,00011011b   ; Перемешивание.
      mov       rdi,fmt4              ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.

; ПЕРЕМЕШИВАНИЕ: ВРАЩЕНИЕ.
; Вращение влево.
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,10010011b   ; Перемешивание.
      mov       rdi,fmt6              ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.
; Вращение вправо.
      movdqu    xmm0,[rbp-16]         ; Восстановление xmm0.
      pshufd    xmm0,xmm0,00111001b   ; Перемешивание.
      mov       rdi,fmt7              ; Вывод заголовка.
      movdqu    [rbp-32],xmm0         ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0d      ; Вывод содержимого xmm0.

; ПЕРЕМЕШИВАНИЕ БАЙТОВ.
      mov       rdi, fmt9
      call      printf              ; Вывод заголовка.
      movdqu    xmm0,[char]         ; Загрузка символа в xmm0.
      movdqu    [rbp-32],xmm0       ; printf уничтожает xmm0.
      call      print_xmm0b         ; Вывод байтов из xmm0.
      movdqu    xmm0,[rbp-32]       ; Восстановление xmm0 после printf.
      movdqu    xmm1,[bytereverse]  ; Загрузка маски.
      pshufb    xmm0,xmm1           ; Перемешивание байтов.
      mov       rdi,fmt5            ; Вывод заголовка.
      movdqu    [rbp-32],xmm0       ; printf уничтожает xmm0.
      call      printf
      movdqu    xmm0,[rbp-32]    ; Восстановление xmm0 после printf.
      call      print_xmm0b      ; Вывод содержимого xmm0.
leave
ret

; Функция для вывода двойных слов.--------------------
print_xmm0d:
push  rbp
mov   rbp,rsp
      mov       rdi, fmt3
      xor       rax,rax
      pextrd    esi, xmm0,3    ; Извлечение двойных слов в обратном порядке,
      pextrd    edx, xmm0,2    ; порядок байтов от младшего к старшему (прямой).
      pextrd    ecx, xmm0,1
      pextrd    r8d, xmm0,0
      call      printf
leave
ret

; Функция для вывода байтов.---------------------------
print_xmm0b:
push  rbp
mov   rbp,rsp
      mov       rdi, fmt8
      xor       rax,rax
      pextrb    esi, xmm0,0    ; В обратном порядке, прямой порядок байтов.
      pextrb    edx, xmm0,1    ; Сначала используются регистры,
      pextrb    ecx, xmm0,2    ; затем стек.
      pextrb    r8d, xmm0,3
      pextrb    r9d, xmm0,4
      pextrb    eax, xmm0,15
      push  rax
      pextrb    eax, xmm0,14
      push  rax
      pextrb    eax, xmm0,13
      push  rax
      pextrb    eax, xmm0,12
      push  rax
      pextrb    eax, xmm0,11
      push  rax
      pextrb    eax, xmm0,10
      push  rax
      pextrb    eax, xmm0,9
      push  rax

      pextrb    eax, xmm0,8
      push  rax
      pextrb    eax, xmm0,7
      push  rax
      pextrb    eax, xmm0,6
      push  rax
      pextrb    eax, xmm0,5
      push  rax
      xor       rax,rax
      call  printf
leave
ret

Сначала	 резервируется	 пространство	 в  стеке	 для	 переменных	 размером	
128	байт.	Это	пространство	необходимо	для	записи	xmmрегистров	в стек.	Для	
xmmрегистров	 нельзя	 использовать	 стандартные	 инструкции	 push/pop,	 по
этому	 непременно	 требуется	 применение	 адресов	 памяти	для	 копирования	
xmmрегистров	в стек	и для	извлечения	их	из	стека.	Мы используем	регистр	rbp,	
указатель	базового	адреса,	как	точку	отсчета.

Далее	выводятся	числа,	которые	будут	применяться	как	упакованные	значе
ния.	Затем	эти	числа	загружаются	в регистр	xmm0	как	двойные	слова	с помощью	
инструкции	pinsrd	(означающей	«packed	insert	double» –	вставка	упакованных	
двойных	слов).	Содержимое	регистра	xmm0	сохраняется	в стеке	как	локальная	
переменная	с помощью	инструкции	movdqu [rbp-16], xmm0.	(Для	этой	локальной	
переменной	было	 зарезервировано	пространство	 в  стеке	 в начале	програм
мы.)	При	каждом	выполнении	функции	printf	содержимое	регистра	xmm0	будет	
изменяться	 независимо	 от	 наших	 намерений.	 Поэтому	 придется	 сохранять	
и восстанавливать	исходное	значение	регистра	xmm0	при	необходимости.	Ин
струкция	movdqu	используется	для	перемещения	невыровненных	упакованных	
целочисленных	значений.	Для	более	наглядной	визуализации	результатов	пе
ремешивания	при	выводе	учитывается	прямой	порядок	байтов	(от	младшего	
к старшему).	Это	позволяет	увидеть	содержимое	регистра	xmm0	так,	как	вы	мог
ли	бы	наблюдать	его	в отладчике,	например	в SASM.

Для	операции	перемешивания	требуется	целевой	операнд,	операндисточ
ник	и маска	перемешивания.	Маска	определяется	в виде	непосредственного	
8битового	значения.	В следующих	разделах	будут	рассматриваться	некоторые	
полезные	примеры	операций	перемешивания	и соответствующие	маски:

	� перемешивание	в случайном	порядке;
	� перемешивание	в обратном	порядке;
	� перемешивание	вращением.

перемешивание в СлучайнОм пОрядке
Графическая	 схема	 поможет	 лучше	 понять	 этот	 тип	 перемешивания.	
На рис. 33.1	показаны	четыре	примера	перемешивания	в случайном	порядке.

Перемешивание в случайном порядке

Маска

Маска

Маска

Маска

Рис. 33.1.	Перемешивание в случайном порядке

На	рис. 33.1	источником	и целью	является	регистр	xmm0.	Младшее	двойное	
слово	d0	определено	в маске	как	00b.	Второе	младшее	двойное	слово	d1	опреде
лено	как	01b.	Третье	двойное	слово	d2	определяется	как	10b,	а четвертое	d3 –	как	
11b.	Двоичная	маска	10101010b	или	aah	в шестнадцатеричном	формате	работает	
следующим	образом:	d2	(10b)	помещается	в четыре	целевые	позиции	упакован
ных	двойных	слов.	Маска	11111111b	должна	помещать	d3	(11b)	в четыре	целевые	
позиции	упакованных	двойных	слов.

При	изучении	исходного	кода	вы	увидите	следующую	простую	инструкцию	
перемешивания:

pshufd xmm0,xmm0,10101010b

В	рассматриваемом	здесь	примере	выполняется	распространение	(«рассеи
вание»)	третьего	младшего	элемента	в регистре	xmm0.	Поскольку	функция	printf 
изменяет	регистр	xmm0,	необходимо	сохранять	его	содержимое	в памяти	перед	
вызовом	printf.	В действительности	требуется	выполнить	больший	объем	ра
боты	для	 защиты	 содержимого	 xmm0	 по	 сравнению	 с  самой	операцией	пере
мешивания.

Разумеется,	вы	не ограничены	четырьмя	масками,	представленными	в этом	
примере.	Можно	создать	любую	8битовую	маску	и перемешивать	данные	как	
угодно.

перемешивание в ОбратнОм пОрядке
На	рис. 33.2	показана	общая	схема	перемешивания	в обратном	порядке.

Перемешивание в обратном порядке

Маска

Рис. 33.2.	Перемешивание в обратном порядке

Здесь	определена	маска	00011011b	или	1bh –	она	интерпретируется	следую
щим	образом:

	� 11	(значение	в d3)	перемещается	в позицию	0;
	� 01	(значение	в d2)	перемещается	в позицию	1;
	� 10	(значение	в d1)	перемещается	в позицию	2;
	� 00	(значение	в d0)	перемещается	в позицию	3.

В	 исходном	 коде	 примера	 можно	 видеть,	 что	 операция	 перемешивания	
в обратном	порядке	весьма	просто	кодируется	на	языке	ассемблера,	как	по
казано	ниже:

pshufd xmm0,xmm0,1bh

перемешивание вращением
Существуют	две	версии	перемешивания	вращением:	вращение	влево	и вра
щение	вправо.	Это	всего	лишь	вопрос	определения	правильной	маски	как	по
следнего	аргумента	инструкции	перемешивания.	На рис. 33.3	показана	общая	
схема	перемешивания	вращением.

Перемешивание вращением

МаскаВлево

Вправо Маска

Рис. 33.3.	Перемешивание вращением

На	языке	ассемблера	перемешивание	вращением	выполняется	следующими	
инструкциями:

pshufd xmm0,xmm0,93h
pshufd xmm0,xmm0,39h

перемешивание байтОв
Двойные	 слова	 можно	 перемешивать	 с  помощью	 инструкции	 pshufd,	 а  оди
ночные	слова –	инструкцией	pshufw.	Также	можно	перемешивать	старшие	сло
ва	и младшие	слова	с помощью	инструкций	pshufhw	и pshuflw	соответственно.	
Подробности	можно	узнать	в руководствах	Intel.	Все	перечисленные	выше	ин
струкции	используют	операндисточник,	целевой	операнд	и маску,	определен
ную	непосредственным	значением.	Представление	маски	как	непосредствен
ного	значения	имеет	свои	ограничения:	такой	способ	не обладает	гибкостью,	
поскольку	приходится	определять	маску	во	время	ассемблирования,	 а не во	
время	выполнения.

Существует	решение	этой	проблемы:	перемешивание	байтов.
Перемешивание	 байтов	 выполняется	 с  помощью	 инструкции	 pshufb.	

Эта  инструкция	 принимает	 только	 два	 операнда:	 операнд	 целевого	 xmm
регистра	и маску,	хранящуюся	в xmmрегистре	или	в 128битовом	блоке	па
мяти.	В приведенном	выше	исходном	коде	(см.	листинг	33.1)	строка	'char' 
реверсируется	 с  помощью	 инструкции	 pshufb.	 Маска	 определена	 в  блоке	
памяти	bytereverse	в разделе	section .data.	Эта	маска	требует,	чтобы	байт	15	
переместился	в позицию	0,	байт	14 –	в позицию	1	и т.	д.	Перемешиваемая	
строка	копируется	в регистр	xmm0,	а маска	помещается	в регистр	xmm1,	так	что	
инструкция	перемешивания	выглядит	следующим	образом:

pshufb xmm0, xmm1

Потом	 происходит	магическое	 превращение.	 Следует	 помнить	 о  том,	 что	
маска	находится	во	 втором	операнде,	источник	 совпадает	 с целью,	поэтому	
результат	сохраняется	в первом	операнде.

Здесь	имеется	важное	преимущество:	нет	необходимости	определять	мас
ку	как	непосредственное	значение	во	время	ассемблирования.	Маска	может	
быть	сформирована	в регистре	xmm1	как	результат	вычислений	во	время	вы
полнения.

На	рис. 33.4	показан	итоговый	вывод	примера	программы	из	листинга	33.1.

Рис. 33.4.	Вывод программы shuffle.asm

резюме
В	этой	главе	вы	узнали:

	� об	использовании	инструкций	перемешивания;
	� о	масках	перемешивания;
	� о	масках,	определяемых	во	время	выполнения;
	� о	том,	как	использовать	стек	для	работы	с xmmрегистрами.

Глава 34
SSE-инструкции: маски строк

Теперь,	когда	мы	знаем,	как	выполняются	операции	перемешивания,	можно	
перейти	к рассмотрению	масок	строк.

Напомню,	что	SSE	предоставляет	две	инструкции	обработки	строк,	которые	
используют	маски:	pcmpistrm	и pcmpestrm.	Мы будем	использовать	инструкции	
обработки	строк	с явно	заданной	длиной.	Сначала	практическое	применение	
масок	кажется	сложным,	но	как	только	вы	поймете	сущность	этой	операции,	
сразу	же	убедитесь	в том,	насколько	мощным	инструментом	может	стать	мас
кирование.

пОиСк СимвОлОв
В	листингах	34.1,	34.2	и 34.3	показан	исходный	код	примера.

Листинг 34.1.	Программа sse_string4.asm

; sse_string4.asm
; Поиск символа.
extern print16b
extern printf
section .data
      string1      db    "qdacdekkfijlmdoza"
                   db    "becdfgdklkmdddaf"
                   db    "fffffffdedeee",10,0
      string2      db    "e",0
      string3      db    "a",0
      fmt          db    "Find all the characters '%s' "
                   db    "and '%s' in:",10,0
      fmt_oc       db    "I found %ld characters '%s'"
                   db    "and '%s'",10,0
      NL           db    10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp

; Вывод искомых символов. 
      mov    rdi, fmt
      mov    rsi, string2



      mov    rdx, string3
      xor    rax,rax
      call   printf
; Вывод целевой строки.
      mov    rdi, string1
      xor    rax,rax
      call   printf
; Поиск в строке и вывод маски.
      mov    rdi, string1
      mov    rsi, string2
      mov    rdx, string3
      call   pcharsrch
; Вывод числа вхождений искомых символов в строке string2.
      mov    rdi, fmt_oc
      mov    rsi, rax
      mov    rdx, string2
      mov    rcx, string3
      call   printf
; Выход.
leave
ret

;-------------------------------------------------------------
; Функция поиска в строке и вывода маски.
pcharsrch:           ; Поиск упакованного символа.
push  rbp
mov   rbp,rsp
      sub    rsp,16    ; Обеспечение пространства в стеке для сохранения xmm1.
      xor    r12,r12   ; Для вычисления общего числа вхождений.
      xor    rcx,rcx   ; Для оповещения о конце строки.
      xor    rbx,rbx   ; Для вычисления адреса.
      mov    rax,-16   ; Для счетчика байтов, чтобы избежать установки флага.
; Формирование значения xmm1, загрузка искомых символов.
      pxor   xmm1,xmm1                 ; Очистка регистра xmm1.
      pinsrb       xmm1,byte[rsi],0    ; Первый символ по индексу 0.
      pinsrb       xmm1,byte[rdx],1    ; Второй символ по индексу 1.
.loop:
      add       rax,16          ; Чтобы избежать установки флага ZF.
      mov       rsi,16          ; Если нет завершающего 0, то вывод 16 байт.
      movdqu    xmm2,[rdi+rbx]  ; Загрузка 16 байт строки в xmm2.
      pcmpistrm xmm1,xmm2,40h   ; Метод 'equal each' и 'байт маски в xmm0'.
      setz   cl                 ; Если найден завершающий 0.
; Если найден завершающий 0, то определить позицию.
      cmp    cl,0
      je     .gotoprint         ; Завершающий 0 не найден.
      ; Завершающий нулевой байт найден.
      ; Осталось менее 16 байт.
      ; rdi содержит адрес строки.
      ; rbx содержит число байтов в блоках, обработанных
      ; до настоящего момента.
      add    rdi,rbx     ; Адрес оставшейся части строки.
      push   rcx         ; Регистр, сохр. вызывающей функцией (cl используется).
      call   pstrlen     ; В rax возвращается длина.
      pop    rcx         ; Регистр, сохраняемый вызывающей функцией.
      dec    rax         ; Длина без завершающего 0.
      mov    rsi,rax     ; Длина оставшейся маски байтов.

238	  SSE-инструкции: маски строк



; Вывод маски.
.gotoprint:
      call print_mask
; Продолжение выполнения для вычисления общего числа совпадений.
      popcnt        r13d,r13d   ; Подсчет числа битов, равных 1.
      add           r12d,r13d   ; Сохранение числа вхождений в r12d.
      or            cl,cl       ; Завершающий 0 обнаружен?
      jnz           .exit
      add           rbx,16      ; Подготовка следующих 16 байт.
      jmp           .loop
.exit:
      mov    rdi, NL        ; Добавление символа перехода на новую строку.
      call   printf
      mov    rax,r12        ; Количество вхождений.
leave
ret

;-------------------------------------------------------------
; Функция поиска завершающего 0.
pstrlen:
push  rbp
mov   rbp,rsp
      sub           rsp,16        ; Для сохранения xmm0.
      movdqu        [rbp-16],xmm0 ; Запись в стек xmm0.
      mov           rax, -16      ; Чтобы далее не устанавливать флаг.
      pxor          xmm0, xmm0    ; Поиск 0 (конца строки).
.loop:  add  rax, 16              ; Чтобы не устанавливать флаг ZF.
      pcmpistri     xmm0, [rdi + rax], 0x08 ; Метод 'equal each'.
      jnz           .loop         ; Завершающий 0 найден?
      add           rax, rcx      ; rax = число уже обработанных байтов.
                                  ; rcx = байты, обработанные в завершающем цикле.
      movdqu        xmm0,[rbp-16] ; Восстановление xmm0 из стека.
leave
ret

;-------------------------------------------------------------
; Функция вывода маски.
; xmm0 содержит маску.
; rsi содержит число битов, которые нужно вывести (16 или меньше).
print_mask:
push  rbp
mov   rbp,rsp
      sub    rsp,16                ; Для сохранения xmm0.
      call   reverse_xmm0          ; Прямой порядок байтов (от мл. к ст.).
      pmovmskb      r13d,xmm0      ; Запись байта маски в r13d.
      movdqu        [rbp-16],xmm1  ; Сохранение в стеке xmm1 перед вызовом printf.
      push   rdi         ; rdi содержит string1.
      mov    edi,r13d    ; Содержит маску, которую нужно вывести.
      push   rdx         ; Содержит маску.
      push   rcx         ; Содержит флаг конца строки.
      call   print16b
      pop    rcx
      pop    rdx
      pop    rdi
      movdqu xmm1,[rbp-16] ; Восстановление xmm1 из стека.
leave
ret

;-------------------------------------------------------------
; Функция реверсирования, перемешивание xmm0.
reverse_xmm0:
section .data
;mask for reversing
        .bytereverse db 15,14,13,12,11,10,9,8,7,6,5,4,3,2,1,0
section .text
push  rbp
mov   rbp,rsp
      sub     rsp,16
      movdqu [rbp-16],xmm2
      movdqu xmm2,[.bytereverse]        ; Загрузка маски в xmm2.
      pshufb xmm0,xmm2                  ; Выполнение перемешивания.
      movdqu xmm2,[rbp-16]              ; Восстановление xmm2 из стека.
leave                             ; Возврат перемешанного значения в xmm0.
ret

Листинг 34.2.	Функция print16b.c

// print16b.c
#include <stdio.h>
#include <string.h>

void print16b(long long n, int length){
      long long s,c;
      int i=0;
      for (c = 15; c >= 16-length; c--)
      {
            s = n >> c;
            if (s & 1)
                  printf("1");
            else
                  printf("0");
      }
}

Листинг 34.3.	makefile

sse_string4: sse_string4.o print16b.o
      gcc -o sse_string4 sse_string4.o print16b.o -no-pie
sse_string4.o: sse_string4.asm
      nasm -f elf64 -g -F dwarf sse_string4.asm -l sse_string4.lst
printb: print16b.c
      gcc -c print16b.c

Основная	часть	программы	достаточно	проста,	но,	как	и в	предыдущих	при
мерах,	сложность	программы	заключается	в том,	что	необходимо	вывести	не
который	результат	на	экран.	Можно	было	бы	исключить	части,	отвечающие	за	
вывод,	и воспользоваться	отладчиком	для	наблюдения	результатов	в регист
рах	и в	памяти.	Но преодоление	трудностей,	связанных	с выводом,	само	по	себе	
занимательно,	не так	ли?

На	рис. 34.1	показан	вывод	этой	программы.

Рис. 34.1.	Вывод программы sse_string4.asm

В	 рассматриваемом	 здесь	 примере	 программы	 производится	 поиск	 двух	
символов	в строке.	Предоставлена	строка	string1,	в которой	выполняется	поиск	
символа	'e',	хранящегося	в строке	string2,	и символа	'a',	находящегося	в строке	
string3.

В	 этой	 программе	используется	 несколько	функций.	 Сначала	 рассмотрим	
функцию	reverse_xmm0.	Эта	функция	в качестве	аргумента	принимает	содержи
мое	регистра	xmm0	и изменяет	порядок	байтов	на	обратный,	используя	инструк
цию	перемешивания.	Для	этого	требуется	возможность	вывода	xmm0,	начиная	
с самых	младших	байтов,	т. е.	вывода	в формате	с прямым	порядком	байтов	(от	
младшего	к старшему).	Именно	поэтому	так	подробно	рассматривались	опера
ции	перемешивания	в предыдущей	главе.

Также	имеется	функция	вычисления	длины	строки	pstrln.	Она	необходима	
потому,	 что	 чтение	 выполняется	 16байтовыми	 блоками.	 Самый	 последний	
блок,	вероятно,	не будет	содержать	полные	16	байт,	поэтому	для	этого	послед
него	блока	нужно	определить	позицию	завершающего	нулевого	байта.	Это	по
зволит	вывести	маску,	которая	имеет	ту	же	длину,	что	и строка,	в которой	вы
полняется	поиск.

Специализированная	функция	pcharsrch	принимает	три	строки	как	аргумен
ты,	и в	ней	производится	основная	работа.	Сначала	выполняются	некоторые	
подготовительные	операции,	такие	как	инициализация	регистров.	Регистр	xmm1 
используется	как	маска –	искомые	символы	сохраняются	в xmm1	с по	мощью	ин
струкции	pinsrb	(packed	insert	bytes –	вставка	упакованных	байтов).	Затем	начи
нается	цикл	с копированием	на	каждой	итерации	16	байт	из	строки	string1	в ре
гистр	xmm2	и с	поиском	заданных	символов	или	завершающего	нулевого	байта.	
Применяется	 инструкция	 с  использованием	маски	 pcmpistrm	 (packed	 compare	
implicit	length	string	with	a mask –	сравнение	упакованной	строки	с явно	задан
ной	длиной	с использованием	маски).	Инструкция	pcmpistrm	в качест	ве	третьего	
операнда	принимает	непосредственное	значение	управляющего	байта	 (imm8),	
определяющего,	что	нужно	сделать:	в данном	случае	применяется	метод	«по
иск	любого	символа	из	заданного	набора	(equal any)»	и «байт	маски	в регистре	
xmm0».	Таким	образом,	выполняется	поиск	«любого»	символа,	который	«равен»	
заданным	искомым	строкам.	Для	каждого	совпавшего	символа	в xmm2	устанав
ливается	бит	(в	1)	в	регистре	xmm0,	который	соответствует	позиции	символа,	для	
которого	 обнаружено	 совпадение	 (в  xmm2).	 Инструкция	 pcmpistrm	 не  работает	
с регистром	xmm0	как	с операндом,	но	этот	регистр	используется	неявно.	Воз
вращаемая	маска	всегда	будет	храниться	в регистре	xmm0.

Инструкция	 pcmpistri	 отличается	 тем,	 что	 возвращает	 индекс	 единичного	
бита,	т. е.	позиции	совпадения	символов	в регистре	ecx.	А инструкция	pcmpistrm 
возвращает	все	позиции	совпадения	в регистре	xmm0	для	16байтового	блока.	

Это	позволяет	значительно	сократить	количество	шагов	для	выполнения	по
иска	всех	совпадающих	символов.

Можно	использовать	битовую	маску	или	байтовую	маску	для	xmm0	(для	этого	
необходимо	установить	или	сбросить	бит	6	в управляющем	байте).	В рассмат
риваемом	здесь	примере	использовалась	байтовая	маска,	чтобы	можно	было	
более	просто	наблюдать	за	регистром	xmm0	в отладчике –	два	значения	ff	в xmm0 
соответствуют	байту,	в котором	все	биты	установлены	в 1.

После	 обследования	 первого	 16байтового	 блока	 проверяется,	 найден	 ли	
завершающий	нулевой	блок,	и результат	этой	проверки	сохраняется	в регист
ре	cl	для	дальнейшего	использования.	Требуемый	вывод	маски,	хранящейся	
в регистре	 xmm0,	 выполняется	 с помощью	функции	 print_mask.	Обратите	 вни
мание:	в отладчике	эта	байтовая	маска	реверсирована	в регистре	xmm0,	чтобы	
соответствовать	формату	с прямым	порядком	байтов	(от	младшего	к старше
му).	Поэтому	перед	выводом	необходимо	реверсировать	маску –	это	делается	
в функции	reverse_xmm0.	Затем	вызывается	Cфункция	print16b	для	вывода	ре
версированной	маски.	Но регистр	xmm0	нельзя	передать	как	аргумент	в print16b,	
так	как	внутри	эта	функция	использует	стандартную	функцию	printf,	которая	
интерпретирует	xmm0	как	значение	с плавающей	точкой,	а не как	байтовую	мас
ку.	Поэтому	перед	вызовом	print16b	байтовая	маска	передается	из	xmm0	в r13d 
с помощью	инструкции	pmovmskb	(означающей	«move	byte	mask» –	перемещение	
байтовой	маски).	В дальнейшем	r13d	используется	как	счетчик,	а для	вывода	
его	значение	копируется	в регистр	edi.	Значение	xmm1	сохраняется	в стеке	для	
дальнейшего	использования.

Вызывается	Cфункция	print16b	для	вывода	маски.	Эта	функция	в качестве	
аргументов	 принимает	 edi	 (маска)	 и  rsi	 (длина,	 передаваемая	 вызывающей	
функцией).

Сразу	после	возвращения	в функцию	pcharsrch	подсчитывается	количество	
единиц	в r13d	с помощью	инструкции	popcnt	и обновляется	счетчик	в r12d.	Так
же	определяется,	нужно	ли	выйти	из	цикла,	потому	что	в блоке	байтов	был	
обнаружен	завершающий	нулевой	байт.

Перед	вызовом	print_mask,	если	найден	завершающий	нулевой	байт,	истин
ная	 длина	 последнего	 блока	 определяется	 с  помощью	функции	 pstrlen.	 На
чальный	адрес	этого	блока	вычисляется	сложением	значения	rbx,	содержащего	
количество	уже	проверенных	байтов	из	предыдущих	блоков,	с rdi,	где	хранит
ся	адрес	строки	string1.	Длина	строки,	возвращаемая	в rax,	используется	для	
вычисления	количества	оставшихся	байтов	маски	в xmm0,	которые	передаются	
в rsi	для	вывода.

Организация	вывода	представляет	собой	весьма	интересное	занятие,	согла
ситесь?

Не воспринимайте	часть	программы,	в которой	осуществляется	вывод,	как	
ошеломляюще	 сложную.	Сосредоточьтесь	на	том,	 как	 работают	маски, –	 это	
главная	цель	текущей	главы.

Что	можно	сделать	с маской,	возвращаемой	инструкцией	pcmpistrm?	Напри
мер,	можно	использовать	итоговую	маску	для	подсчета	всех	вхождений	иско
мого	аргумента	или	для	поиска	всех	вхождений	и замены	их	на	чтолибо	дру
гое,	создавая	собственную	версию	функции	поиска	с заменой.

Теперь	рассмотрим	другой	вариант	поиска.

пОиСк СимвОлОв из заданнОгО диапазОна
Диапазон	может	 содержать	любое	 количество	искомых	 символов,	 например	
все	символы	верхнего	регистра,	все	символы	между	a и	k,	все	символы,	пред
ставляющие	цифры,	и т.	д.

В	листинге	34.4	показан	пример	поиска	в строке	символов	верхнего	регистра.

Листинг 34.4.	Программа sse_string5.asm

; sse_string5.asm
; Поиск символов из заданного диапазона.
extern print16b
extern printf
section .data
      string1      db    "eeAecdkkFijlmeoZa"
                   db    "bcefgeKlkmeDad"
                   db    "fdsafadfaseeE",10,0
      startrange   db    "A",10,0      ; Поиск символов верхнего регистра.
      stoprange    db    "Z",10,0
      NL           db    10,0
      fmt          db    "Find the uppercase letters in:",10,0
      fmt_oc       db    "I found %ld uppercase letters",10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Сначала выводится исходная строка.
      mov    rdi, fmt        ; Заголовок.
      xor    rax,rax
      call   printf
      mov    rdi, string1    ; Строка.
      xor    rax,rax
      call   printf
; Поиск в строке.
      mov    rdi, string1
      mov    rsi, startrange
      mov    rdx, stoprange
      call   prangesrch
; Вывод количества найденных вхождений.
      mov    rdi, fmt_oc
      mov    rsi, rax
      xor    rax,   rax
      call   printf
leave
ret

;-------------------------------------------------------------
; Функция поиска и вывода маски.
prangesrch:             ; Упакованный диапазон символов для поиска.
push  rbp
mov   rbp,rsp
      sub    rsp,16     ; Пространство в стеке для сохранения xmm1.
      xor    r12,r12    ; Для числа найденных вхождений.
      xor    rcx,rcx    ; Для оповещения о конце строки.

      xor    rbx,rbx    ; Для вычисления адреса.
      mov    rax,-16    ; Чтобы избежать установки флага ZF.
; Формирование значения xmm1.
      pxor         xmm1,xmm1         ; Обеспечение очистки xmm1.
      pinsrb       xmm1,byte[rsi],0  ; startrange по индексу 0.
      pinsrb       xmm1,byte[rdx],1  ; stoprange по индексу 1.
.loop:
      add          rax,16
      mov          rsi,16            ; Если нет завершающего 0, вывод 16 байт.
      movdqu       xmm2,[rdi+rbx]
      pcmpistrm    xmm1,xmm2,01000100b   ; Метод equal each|байт.маска в xmm0.
      setz         cl                    ; Обнаружен завершающий 0.
; Если найден завершающий 0, то определить позицию.
      cmp    cl,0
      je     .gotoprint    ; Завершающий 0 не найден.
      ; Завершающий нулевой байт найден.
      ; Осталось меньше 16 байт.
      ; rdi содержит адрес строки.
      ; rbx содержит число байтов в блоках, обработанных ранее.
      add    rdi,rbx     ; Взять только хвост этой строки.
      push   rcx         ; Регистр, сохр. вызывающей функцией (используется cl).
      call   pstrlen     ; Определение позиции завершающего 0.
      pop    rcx         ; Регистр, сохраняемый вызывающей функцией.
      dec    rax         ; Длина без завершающего 0.
      mov    rsi,rax     ; Количество байтов в хвосте строки.

; Вывод маски.
.gotoprint:
      call print_mask
; Продолжать вычисление общего числа найденных совпадений.
      popcnt r13d, r13d    ; Подсчет числа битов, равных 1.
      add    r12d, r13d    ; Сохранение числа вхождений в r12.
      or     cl,cl         ; Завершающий 0 обнаружен?
      jnz    .exit
      add    rbx,16        ; Подготовка следующего блока.
      jmp   .loop
.exit:
      mov    rdi, NL
      call   printf
      mov    rax, r12      ; Возврат числа найденных вхождений.
leave
ret

;-------------------------------------------------------------
pstrlen:
push  rbp
mov   rbp,rsp
      sub       rsp,16          ; Для сохранения в стеке xmm0.
      movdqu    [rbp-16],xmm0   ; Сохранение в стеке xmm0.
      mov       rax, -16        ; Чтобы далее избежать установки флага ZF.
      pxor      xmm0, xmm0      ; Поиск 0 (конца строки).
.loop:
      add    rax, 16      ; Чтобы избежать установки ZF, если rax = 0 после pcmpistri.
      pcmpistri xmm0, [rdi + rax], 0x08   ; Метод 'equal each'.
      jnz       .loop                     ; 0 найден?
      add       rax, rcx        ; rax = число уже обработанных байтов.

                                ; rcx = число байтов, обработанных
                                ;       на последней итерации цикла.
      movdqu    xmm0,[rbp-16]   ; Восстановление из стека xmm0.
leave
ret

;-------------------------------------------------------------
; Функция вывода маски.
; xmm0 содержит маску.
; rsi содержит число выводимых битов (16 или меньше).
print_mask:
push  rbp
mov   rbp,rsp
      sub          rsp,16          ; Для сохранения в стеке xmm0.
      call         reverse_xmm0    ; Прямой порядок байтов (от мл. к ст.).
      pmovmskb     r13d,xmm0       ; Запись байтовой маски в r13d.
      movdqu       [rbp-16],xmm1   ; Сохранение в стеке xmm1 перед вызовом printf.
      push         rdi             ; rdi содержит string1.
      mov          edi, r13d       ; Содержит выводимую маску.
      push         rdx             ; Содержит маску.
      push         rcx             ; Содержит флаг конца строки.
      call         print16b
      pop          rcx
      pop          rdx
      pop          rdi
      movdqu       xmm1,[rbp-16]  ; Восстановление из стека xmm1.
leave
ret

;-------------------------------------------------------------
; Функция реверсирования, перемешивания xmm0.
reverse_xmm0:
section .data
;mask for reversing
        .bytereverse db 15,14,13,12,11,10,9,8,7,6,5,4,3,2,1,0
section .text
push  rbp
mov   rbp,rsp
      sub     rsp,16
      movdqu [rbp-16],xmm2
      movdqu xmm2,[.bytereverse]   ; Загрузка маски в xmm2.
      pshufb xmm0,xmm2             ; Выполнение перемешивания.
      movdqu xmm2,[rbp-16]         ; Восстановление из стека xmm2.
leave                              ; Возврат перемешанного значения xmm0.
ret

Эта	программа	делает	почти	то	же	самое,	что	и предыдущая	(листинг	34.1),	
просто	в данном	примере	строки	string2	и string3	получили	более	осмыслен
ные	имена.	Здесь	наиболее	важно	изменение	управляющего	байта	так,	чтобы	
в инструкцию	pcmpistrm	передавалось	значение	01000100b,	означающее	«поиск	
каждого	 символа	 из	 заданного	 диапазона»	 (equal	 range)	 и  «байтовая	 маска	
в регистре	xmm0».

Организация	вывода	точно	такая	же,	как	в предыдущем	разделе.
На	рис. 34.2	показан	вывод	этой	программы.

Поиск символов из заданного диапазона  245



Рис. 34.2.	Вывод программы sse_string5.asm

Теперь	рассмотрим	еще	один	пример.

пОиСк пОдСтрОки
В	листинге	34.5	показан	исходный	код	примера.

Листинг 34.5.	Программа sse_string6.asm

; sse_string6.asm
; Поиск подстроки.
extern print16b
extern printf
section .data
      string1      db    "a quick pink dinosour jumps over the "
                   db    "lazy river and the lazy dinosour "
                   db    "doesn't mind",10,0
      string2      db    "dinosour",0
      NL           db    10,0
      fmt          db    "Find the substring '%s' in:",10,0
      fmt_oc       db    "I found %ld %ss",10,0

section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Сначала выводятся строки.
      mov    rdi, fmt
      mov    rsi, string2
      xor    rax,rax
      call   printf
      mov    rdi, string1
      xor    rax,rax
      call   printf
; Поиск в строке.
      mov    rdi, string1
      mov    rsi, string2
      call   psubstringsrch
; Вывод числа найденных вхождений заданной подстроки.
      mov    rdi, fmt_oc
      mov    rsi, rax
      mov    rdx, string2
      call   printf
leave
ret

;-------------------------------------------------------------
; Функция поиска подстроки и вывода маски.
psubstringsrch:         ; Поиск упакованной подстроки.
push  rbp
mov   rbp,rsp
      sub    rsp,16     ; Пространство для сохранения в стеке xmm1.
      xor    r12,r12    ; Вычисление общего числа вхождений.
      xor    rcx,rcx    ; Для оповещения о конце строки.
      xor    rbx,rbx    ; Для вычисления адреса.
      mov    rax,-16    ; Чтобы избежать установки флага ZF.
; Формирование значения xmm1, загрузка подстроки.
      pxor   xmm1,xmm1
      movdqu xmm1,[rsi]
.loop:
      add    rax,16    ; Чтобы избежать установки флага ZF.
      mov    rsi,16    ; Если нет завершающего 0, вывод 16 байт.
      movdqu xmm2,[rdi+rbx]
      pcmpistrm xmm1,xmm2,01001100b ; Метод 'equal ordered'|'байтовая маска в xmm0'.
      setz   cl   ; Обнаружен завершающий 0.

; Если найден завершающий 0, то определить позицию.
      cmp    cl,0
      je     .gotoprint    ; Завершающий 0 не найден.
      ; Найден завершающий нулевой байт.
      ; Осталось меньше 16 байт.
      ; rdi содержит адрес строки.
      ; rbx содержит число байтов в блоках, обработанных ранее.
      add    rdi,rbx       ; Взять только хвост строки.
      push   rcx           ; Регистр, сохр. вызывающей функцией (используется cl).
      call   pstrlen       ; В rax возвращается позиция завершающего 0.
      push   rcx           ; Регистр, сохр. вызывающей функцией (используется cl).
      dec    rax           ; Длина без завершающего 0.
      mov    rsi,rax       ; Длина оставшихся байтов.

; Вывод маски.
.gotoprint:
      call print_mask
; Продолжение вычисления общего числа найденных совпадений.
      popcnt r13d,r13d    ; Подсчет числа битов, равных 1.
      add    r12d,r13d    ; Сохранение числа найденных вхождений в r12.
      or     cl,cl        ; Завершающий 0 обнаружен?
      jnz    .exit
      add    rbx,16       ; Подготовка следующего блока.
      jmp    .loop
.exit:
      mov    rdi, NL
      call   printf
      mov    rax, r12     ; Возврат числа найденных вхождений.
leave
ret

;-------------------------------------------------------------
pstrlen:
push  rbp
mov   rbp,rsp

      sub    rsp,16          ; Для сохранения в стеке xmm0.
      movdqu [rbp-16],xmm0   ; Сохранение в стеке xmm0.
      mov    rax, -16        ; Чтобы далее избежать установки флага ZF.
      pxor   xmm0, xmm0      ; Поиск 0 (конца строки).
.loop:
      add    rax, 16 ; Чтобы избежать установки флага ZF, если rax = 0 после pcmpistri.
      pcmpistri  xmm0, [rdi + rax], 0x08       ; Метод 'equal each'
      jnz    .loop      ; 0 найден?
      add    rax, rcx   ; rax = число уже обработанных байтов.
                        ; rcx = число байтов, обработанных на последней итерации цикла.
      movdqu xmm0,[rbp-16] ; Восстановление из стека xmm0.
leave
ret

;-------------------------------------------------------------
; Функция вывода маски.
; xmm0 содержит маску.
; rsi содержит число выводимых битов (16 или меньше).
print_mask:
push  rbp
mov   rbp,rsp
      sub          rsp,16        ; Для сохранения в стеке xmm0.
      call         reverse_xmm0  ; Прямой порядок байтов (от младшего к старшему).
      pmovmskb     r13d,xmm0     ; Запись байтовой маски в edx.
      movdqu       [rbp-16],xmm1 ; Сохранение в стеке xmm1 перед вызовом printf.
      push         rdi           ; rdi содержит string1.
      mov          edi,r13d      ; Содержит выводимую маску.
      push         rdx           ; Содержит маску.
      push         rcx           ; Содержит флаг конца строки.
      call         print16b
      pop          rcx
      pop          rdx
      pop          rdi
      movdqu       xmm1,[rbp-16] ; Восстановление из стека xmm1.
leave
ret

;-------------------------------------------------------------
; Функция реверсирования, перемешивания xmm0.
reverse_xmm0:
section .data
; Маска для реверсирования.
        .bytereverse db 15,14,13,12,11,10,9,8,7,6,5,4,3,2,1,0
section .text
push  rbp
mov   rbp,rsp
      sub     rsp,16
      movdqu [rbp-16],xmm2
      movdqu xmm2,[.bytereverse] ; Загрузка маски в xmm2.
      pshufb xmm0,xmm2           ; Выполнение перемешивания.
      movdqu xmm2,[rbp-16]       ; Восстановление из стека xmm2.
leave                            ; Возврат перемешанного значения xmm0.
ret

В	этом	примере	используется	почти	тот	же	исходный	код,	что	и в	преды
дущих	 примерах,	 изменены	 только	 строки,	 и  управляющий	 байт	 содержит	
указание:	 «поиск	 равенства	 упорядоченной	 последовательности	 символов»	
(equal	ordered)	и «байтовая	маска	в регистре	xmm0».	Достаточно	просто,	не так	
ли?

На	рис. 34.3	показан	вывод	этой	программы.

Рис. 34.3.	Вывод программы sse_string6.asm

резюме
В	этой	главе	вы	узнали:

	� об	использовании	масок	для	обработки	строк;
	� о	методах	поиска	отдельных	символов,	символов	из	заданного	диапазо
на	и подстрок;

	� о	способах	вывода	масок	из	xmmрегистров.

Глава 35
AVX

Advanced	Vector	Extensions	 (AVX) –	это	расширение	системы	команд	SSE для	
процессоров	x86.	Система	SSE	предлагает	16	xmmрегистров	размером	128	бит	
каждый,	 тогда	 как	 система	 AVX	 предоставляет	 16	 ymmрегистров	 размером	
256 бит	каждый.	Нижняя	(младшая)	половина	каждого	ymmрегистра	в действи
тельности	 представляет	 собой	 соответствующий	 xmmрегистр.	 AVX512  –	 это	
дальнейшее	расширение,	предоставляющее	32	zmmрегистра	размером	512	бит	
каждый.

В	 дополнение	 к  ymmрегистрам	 система	 AVX	 расширяет	 инструкции	 SSE	
и  предоставляет	 набор	дополнительных	новых	инструкций.	После	 изучения	
глав	этой	книги,	посвященных	SSE,	вы	не будете	испытывать	затруднения	при	
освоении	многочисленных	инструкций	SSE	и AVX.

В	этой	главе	сначала	будет	показано,	как	определить	версию	AVX,	поддержи
ваемую	конкретным	процессором,	затем	рассматривается	пример	программы	
с использованием	AVX.

прОверка пОддержки avX
В	листинге	35.1	показан	исходный	код	программы,	определяющей,	поддержи
вает	ли	процессор	расширение	AVX.

Листинг 35.1.	Программа cpu_avx.asm

; cpu_avx.asm
extern printf
section .data
      fmt_noavx    db      "This cpu does not support AVX.",10,0
      fmt_avx      db      "This cpu supports AVX.",10,0
      fmt_noavx2   db      "This cpu does not support AVX2.",10,0
      fmt_avx2     db      "This cpu supports AVX2.",10,0
      fmt_noavx512 db      "This cpu does not support AVX-512.",10,0
      fmt_avx512   db      "This cpu supports AVX-512.",10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      call   cpu_sse       ; Возврат 1 в rax, если AVX поддерживается, иначе 0.



leave
ret

cpu_sse:
push  rbp
mov   rbp,rsp
; Проверка поддержки avx.
      mov    eax,1         ; Запрос флагов характеристик ЦП.
      cpuid
      mov    eax,28        ; Проверка бита 28 в ecx.
      bt     ecx,eax
      jnc    no_avx
      xor    rax,rax
      mov    rdi,fmt_avx
      call   printf
; Проверка поддержки avx2.
      mov    eax,7         ; Запрос флагов характеристик ЦП.
      mov    ecx,0
      cpuid
      mov    eax,5         ; Проверка бита 5 в ebx
      bt     ebx,eax
      jnc    the_exit
      xor    rax,rax
      mov    rdi,fmt_avx2
      call   printf
; Проверка поддержки основных функций avx512.
      mov    eax,7         ; Запрос флагов характеристик ЦП.
      mov    ecx,0
      cpuid
      mov    eax,16        ; Проверка бита 16 в ebx
      bt     ebx,eax
      jnc    no_avx512
      xor    rax,rax
      mov    rdi,fmt_avx512
      call   printf
      jmp    the_exit
no_avx:
      mov    rdi,fmt_noavx
      xor    rax,rax
      call   printf        ; Вывод сообщения, если AVX не поддерживается.
      xor    rax,rax       ; Возврат 0, AVX не поддерживается.
      jmp    the_exit      ; Выход.
no_avx2:
      mov    rdi,fmt_noavx2
      xor    rax,rax
      call   printf        ; Вывод сообщения, если AVX не поддерживается.
      xor    rax,rax       ; Возврат 0, AVX не поддерживается.
      jmp    the_exit      ; Выход.
no_avx512:
      mov    rdi,fmt_noavx512
      xor    rax,rax
      call   printf        ; Вывод сообщения, если AVX не поддерживается.
      xor    rax,rax       ; Возврат 0, AVX не поддерживается.
      jmp    the_exit      ; Выход.
the_exit:
leave
ret

Эта	программа	похожа	на	программу,	которая	использовалась	для	проверки	
поддержки	SSE,	но	здесь	необходимо	проверять	флаги	AVX.	В этом	нет	ничего	
особенного,	а более	подробные	сведения	об	используемых	регистрах	и о	том,	
какую	информацию	можно	извлечь	из	них,	можно	найти	в руководствах	Intel,	
Volume	2,	в разделе,	описывающем	cpuid.

На	рис. 35.1	показан	вывод	этой	программы.

Рис. 35.1.	Вывод программы cpu_avx.asm

пример прОграммы С иСпОльзОванием avX
В	листинге	35.2	показан	измененный	соответствующим	образом	исходный	код	
примера	SSEпрограммы	без	выравнивания	данных	из	главы	28.

Листинг 35.2.	Программа avx_unaligned.asm

; avx_unaligned.asm
extern printf
section .data
      spvector1     dd    1.1
                    dd    2.1
                    dd    3.1
                    dd    4.1
                    dd    5.1
                    dd    6.1
                    dd    7.1
                    dd    8.1
      spvector2     dd    1.2
                    dd    1.2
                    dd    3.2
                    dd    4.2
                    dd    5.2
                    dd    6.2
                    dd    7.2
                    dd    8.2
      dpvector1     dq    1.1
                    dq    2.2
                    dq    3.3
                    dq    4.4
      dpvector2     dq    5.5
                    dq    6.6
                    dq    7.7
                    dq    8.8

      fmt1   db     "Single Precision Vector 1:",10,0
      fmt2   db     10,"Single Precision Vector 2:",10,0
      fmt3   db     10,"Sum of Single Precision Vector 1 and Vector 2:",10,0
      fmt4   db     10,"Double Precision Vector 1:",10,0

      fmt5   db     10,"Double Precision Vector 2:",10,0
      fmt6   db     10,"Sum of Double Precision Vector 1 and Vector 2:",10,0

section .bss
      spvector_res  resd   8
      dpvector_res  resq   4
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; ВЕКТОРЫ С ПЛАВАЮЩЕЙ ТОЧКОЙ ОБЫЧНОЙ ТОЧНОСТИ.
; Загрузка vector1 в регистр ymm0.
      vmovups       ymm0, [spvector1]
; Извлечение из ymm0.
      vextractf128  xmm2,ymm0,0    ; Первая часть ymm0.
      vextractf128  xmm2,ymm0,1    ; Вторая часть ymm0.
; Загрузка vector2 в регистр ymm1.
      vmovups       ymm1, [spvector2]
; Извлечение из ymm1.
      vextractf128  xmm2,ymm1,0
      vextractf128  xmm2,ymm1,1
; Сложение 2 векторов с плавающей точкой обычной точности.
      vaddps        ymm2,ymm0,ymm1
      vmovups [spvector_res],ymm2
; Вывод векторов.
      mov    rdi,fmt1
      call   printf
      mov    rsi,spvector1
      call   printspfpv
      mov    rdi,fmt2
      call   printf
      mov    rsi,spvector2
      call   printspfpv
      mov    rdi,fmt3
      call   printf
      mov    rsi,spvector_res
      call   printspfpv

; ВЕКТОРЫ С ПЛАВАЮЩЕЙ ТОЧКОЙ ДВОЙНОЙ ТОЧНОСТИ.
; Загрузка vector1 в регистр ymm0.
      vmovups       ymm0, [dpvector1]
; Извлечение из ymm0.
      vextractf128  xmm2,ymm0,0    ; Первая часть ymm0.
      vextractf128  xmm2,ymm0,1    ; Вторая часть ymm0.
; Загрузка vector2 в регистр ymm1.
      vmovups       ymm1, [dpvector2]
; Извлечение из ymm1.
      vextractf128  xmm2,ymm1,0
      vextractf128  xmm2,ymm1,1
; Сложение 2 векторов с плавающей точкой двойной точности.
      vaddpd        ymm2,ymm0,ymm1
      vmovupd       [dpvector_res],ymm2
; Вывод векторов.
      mov    rdi,fmt4
      call   printf
      mov    rsi,dpvector1

Пример программы с использованием AVX  253



      call   printdpfpv
      mov    rdi,fmt5
      call   printf
      mov    rsi,dpvector2
      call   printdpfpv
      mov    rdi,fmt6
      call   printf
      mov    rsi,dpvector_res
      call   printdpfpv
leave
ret

printspfpv:
section .data
        .NL   db    10,0
        .fmt1       db    "%.1f,  ",0
section .text
push   rbp
mov    rbp,rsp
        push       rcx
        push       rbx
        mov        rcx,8
        mov        rbx,0
        mov        rax,1
.loop:
        movss      xmm0,[rsi+rbx]
        cvtss2sd   xmm0,xmm0
        mov        rdi,.fmt1
        push       rsi
        push       rcx
        call       printf
        pop        rcx
        pop        rsi
        add        rbx,4
        loop       .loop
        xor        rax,rax
        mov        rdi,.NL
        call       printf
        pop        rbx
        pop        rcx
leave
ret

printdpfpv:
section .data
        .NL  db    10,0
        .fmt db    "%.1f,  %.1f,  %.1f,  %.1f",0
section .text
push   rbp
mov    rbp,rsp
       mov    rdi,.fmt
       mov    rax,4     ; Четыре числа с плавающей точкой.
       call   printf
       mov    rdi,.NL
       call   printf
leave
ret

В	 этой	 программе	 используются	 256битовые	 ymmрегистры	 и  несколько	
новых	 инструкций.	 Например,	 инструкция	 vmovups	 применяется	 для	 записи	
невыровненных	 данных	 в  ymmрегистр.	 Воспользуемся	 SASM	 для	 просмотра	
содержимого	 регистров.	 После	 выполнения	 инструкций	 vmovups	 регистр	 ymm0 
содержит	следующие	значения:

{0x4083333340466666400666663f8ccccd, 0x4101999a40e3333340c3333340a33333}

А	вот	как	эти	значения	выглядят	после	преобразования	в десятичный	формат:

{4.1  3.1  2.1  1.1  ,   8.1  7.1  6.1  5.1}

Посмотрите,	где	хранятся	эти	значения, –	это	может	показаться	не совсем	
понятным.

Исключительно	 в  демонстрационных	 целях	 данные	 извлекаются	 из	 ymm
регистра,	 и  используется	 инструкция	 vextractf128	 для	 записи	 упакованных	
значений	 с  плавающей	точкой	 из	 регистра	 ymm0	 в  регистр	 xmm2,	 частями	по	
128 бит.	Можно	было	бы	воспользоваться	инструкцией	extractps	для	после
дующего	извлечения	значений	с плавающей	точкой	и сохранения	их	в реги
страх	общего	назначения.

Используются	новые	инструкции	с тремя	операндами,	как	показано	ниже:

vaddps ymm2,ymm0,ymm1

Выполняется	сложение	регистров	ymm1	и ymm0,	а результат	сохраняется	в ре
гистре	ymm2.

Функции	вывода	просто	загружают	значения	из	памяти	в xmmрегистр,	вы
полняют	преобразование	чисел	с обычной	точностью	в числа	с двойной	точно
стью,	когда	это	необходимо,	затем	вызывают	функцию	printf.

На	рис. 35.2	показан	вывод	этой	программы.

Рис. 35.2.	Вывод программы avx_unaligned.asm

резюме
В	этой	главе	вы	узнали	о том:

	� как	определить,	поддерживается	ли	AVX	конкретным	процессором;
	� что	AVX	использует	16 256битовых	ymmрегистров;
	� что	 128битовые	 xmmрегистры	 соответствуют	 младшим	 частям	 ymm
регистров;

	� как	извлекать	значения	из	ymmрегистров.

Глава 36
Операции с матрицами 
с использованием AVX

Вместо	общего	обзора	многочисленных	инструкций	AVX,	возможно,	вызыва
ющих	определенный	интерес,	рассмотрим	некоторые	операции	с матрицами,	
использующие	AVX.	Это	длинная	 глава	 с несколькими	страницами	исходно
го	кода,	большая	часть	которого	уже	знакома	читателю,	но	в этом	коде	будут	
представлены	некоторые	новые	инструкции.

В	примерах	данной	главы	демонстрируется	выполнение	операций	умноже
ния	и обращения	(инверсии)	матриц.	В следующей	главе	будет	показано,	как	
транспонировать	матрицу.

пример иСхОднОгО кОда для Операций С матрицами
В	листинге	36.1	показан	исходный	код	примера.

Листинг 36.1.	Программа matrix4x4.asm

; matrix4x4.asm
extern printf

section .data
      fmt0   db    10,"4x4 DOUBLE PRECISION FLOATING POINT MATRICES",10,0
      fmt1   db    10,"This is matrixA:",10,0
      fmt2   db    10,"This is matrixB:",10,0
      fmt3   db    10,"This is matrixA x matrixB:",10,0
      fmt4   db    10,"This is matrixC:",10,0
      fmt5   db    10,"This is the inverse of matrixC:",10,0
      fmt6   db    10,"Proof: matrixC x inverse =",10,0
      fmt7   db    10,"This is matrixS:",10,0
      fmt8   db    10,"This is the inverse of matrixS:",10,0
      fmt9   db    10,"Proof: matrixS x inverse =",10,0
      fmt10  db    10,"This matrix is singular!",10,10,0
      align 32
        matrixA     dq     1.,  3.,  5.,  7.
                    dq     9., 11., 13., 15.
                    dq    17., 19., 21., 23.
                    dq    25., 27., 29., 31.
        matrixB     dq     2.,  4.,  6.,  8.



                    dq    10., 12., 14., 16.
                    dq    18., 20., 22., 24.
                    dq    26., 28., 30., 32.
        matrixC     dq     2.,          11.,        21.,      37.
                    dq     3.,          13.,        23.,      41.
                    dq     5.,          17.,        29.,      43.
                    dq     7.,          19.,        31.,      47.
        matrixS     dq      1.,      2.,      3.,         4.
                    dq      5.,      6.,      7.,         8.
                    dq      9.,     10.,     11.,        12.
                    dq     13.,     14.,     15.,        16.

section .bss
      alignb 32
      product resq 16
      inverse resq 16
section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Вывод заголовка.
      mov    rdi, fmt0
      call   printf
; Вывод матрицы matrixA.
      mov    rdi,fmt1
      call   printf
      mov    rsi,matrixA
      call   printm4x4
; Вывод матрицы matrixB.
      mov    rdi,fmt2
      call   printf
      mov    rsi,matrixB
      call   printm4x4
; Вычисление произведения матриц matrixA x matrixB.
      mov    rdi,matrixA
      mov    rsi,matrixB
      mov    rdx,product
      call   multi4x4
; Вывод произведения (результата умножения).
      mov    rdi,fmt3
      call   printf
      mov    rsi,product
      call   printm4x4
; Вывод матрицы matrixC.
      mov    rdi,fmt4
      call   printf
      mov    rsi,matrixC
      call   printm4x4
; Выполнение обращения (инверсии) матрицы matrixC.
      mov    rdi,matrixC
      mov    rsi,inverse
      call   inverse4x4
      cmp    rax,1
      je     singular
; Вывод обращенной матрицы.

      mov    rdi,fmt5
      call   printf
      mov    rsi,inverse
      call   printm4x4
; Подтверждение операции умножения matrixC и обращение.
      mov    rsi,matrixC
      mov    rdi,inverse
      mov    rdx,product
      call   multi4x4
; Вывод подтверждения.
      mov    rdi,fmt6
      call   printf
      mov    rsi,product
      call   printm4x4

; Сингулярная (вырожденная) матрица.
; Вывод matrixS.
      mov    rdi,fmt7
      call   printf
      mov    rsi,matrixS
      call   printm4x4
; Выполнение обращения (инверсии) матрицы matrixS.
      mov    rdi,matrixS
      mov    rsi,inverse
      call   inverse4x4
      cmp    rax,1
      je     singular
; Вывод обращенной матрицы.
      mov    rdi,fmt8
      call   printf
      mov    rsi,inverse
      call   printm4x4
; Подтверждение операции умножения matrixS и обращение.
      mov    rsi,matrixS
      mov    rdi,inverse
      mov    rdx,product
      call   multi4x4
; Вывод подтверждения.
      mov    rdi,fmt9
      call   printf
      mov    rsi,product
      call   printm4x4
      jmp    exit
singular:
; Вывод сообщения об ошибке.
      mov    rdi,fmt10
      call printf
exit:
leave
ret

inverse4x4:
section .data
      align 32
      .identity    dq       1., 0., 0., 0.
                   dq       0., 1., 0., 0.

                   dq       0., 0., 1., 0.
                   dq       0., 0., 0., 1.
      .minus_mask  dq      8000000000000000h
      .size        dq      4                 ; Матрицы 4×4.
      .one         dq      1.0
      .two         dq      2.0
      .three       dq      3.0
      .four        dq      4.0
section .bss
      alignb 32
      .matrix1 resq 16          ; Промежуточная матрица.
      .matrix2 resq 16          ; Промежуточная матрица.
      .matrix3 resq 16          ; Промежуточная матрица.
      .matrix4 resq 16          ; Промежуточная матрица.
      .matrixI resq 16
      .mxcsr resd 1      ; Используется для проверки деления на ноль.

section .text
push  rbp
mov   rbp,rsp
      push   rsi         ; Сохранение адреса обращенной матрицы.
      vzeroall           ; Очистка всех ymm-регистров.

; Вычисление промежуточных матриц.
; Вычисление промежуточной матрицы matrix2.
; rdi содержит адрес исходной матрицы.
      mov    rsi,rdi
      mov    rdx,.matrix2
      push   rdi
      call   multi4x4
      pop    rdi

; Вычисление промежуточной матрицы matrix3.
      mov    rsi,.matrix2
      mov    rdx,.matrix3
      push   rdi
      call   multi4x4
      pop    rdi

; Вычисление промежуточной матрицы matrix4.
      mov    rsi,.matrix3
      mov    rdx,.matrix4
      push   rdi
      call   multi4x4
      pop    rdi

; Вычисление следов матриц.
; Вычисление следа матрицы trace1.
      mov    rsi,[.size]
      call   vtrace
      movsd  xmm8,xmm0    ; След 1 в xmm8.
; Вычисление следа матрицы trace2.
      push   rdi            ; Сохранение адреса исходной матрицы.
      mov    rdi,.matrix2
      mov    rsi,[.size]
      call   vtrace
      movsd xmm9,xmm0   ; След 2 в xmm9.

; Вычисление следа матрицы trace3.
      mov    rdi,.matrix3
      mov    rsi,[.size]
      call   vtrace
      movsd  xmm10,xmm0   ; След 3 в xmm10.
; Вычисление следа матрицы trace4.
      mov    rdi,.matrix4
      mov    rsi,[.size]
      call   vtrace
      movsd  xmm11,xmm0   ;trace 4 in xmm11

; Вычисление коэффициентов.
; Вычисление коэффициента p1.
; p1 = -s1
      vxorpd        xmm12,xmm8,[.minus_mask] ; p1 в xmm12.
; Вычисление коэффициента p2.
; p2 = -1/2 * (p1 * s1 + s2)
      movsd         xmm13,xmm12     ; Копирование p1 в xmm13.
      vfmadd213sd   xmm13,xmm8,xmm9 ; xmm13=xmm13*xmm8+xmm9
      vxorpd        xmm13,xmm13,[.minus_mask]
      divsd         xmm13,[.two]    ; Деление на 2 и p2 в xmm13.
; Вычисление коэффициента p3.
; p3 = -1/3 * (p2 * s1 + p1 * s2 + s3)
      movsd         xmm14,xmm12             ; Копирование p1 в xmm14.
      vfmadd213sd   xmm14,xmm9,xmm10        ; p1*s2+s3;xmm14=xmm14*xmm9+xmm10
      vfmadd231sd   xmm14,xmm13,xmm8        ; xmm14+p2*s1;xmm14=xmm14+xmm13*xmm8
      vxorpd        xmm14,xmm14,[.minus_mask]
      divsd         xmm14,[.three]             ; p3 в xmm14.
; Вычисление коэффициента p4.
; p4 = -1/4 * (p3 * s1 + p2 * s2 + p1 * s3 + s4)
      movsd         xmm15,xmm12           ; Копирование p1 в xmm15.
      vfmadd213sd   xmm15,xmm10,xmm11     ; p1*s3+s4;xmm15=xmm15*xmm10+xmm11
      vfmadd231sd   xmm15,xmm13,xmm9      ; xmm15+p2*s2;xmm15=xmm15+xmm13*xmm9
      vfmadd231sd   xmm15,xmm14,xmm8      ; xmm15+p3*s1;xmm15=xmm15+xmm14*xmm8
      vxorpd        xmm15,xmm15,[.minus_mask]
      divsd         xmm15,[.four]         ; p4 в xmm15.
; Умножение матриц с соответствующим коэффициентом.
      mov          rcx,[.size]
      xor          rax,rax
      vbroadcastsd       ymm1,xmm12       ; p1
      vbroadcastsd       ymm2,xmm13       ; p2
      vbroadcastsd       ymm3,xmm14       ; p3
      pop rdi                      ; Восстановление адреса исходной матрицы.

.loop1:
      vmovapd            ymm0,[rdi+rax]
      vmulpd             ymm0,ymm0,ymm2
      vmovapd            [.matrix1+rax],ymm0
      vmovapd            ymm0,[.matrix2+rax]
      vmulpd             ymm0,ymm0,ymm1
      vmovapd            [.matrix2+rax],ymm0
      vmovapd            ymm0,[.identity+rax]
      vmulpd             ymm0,ymm0,ymm3
      vmovapd            [.matrixI+rax],ymm0
      add                rax,32
      loop               .loop1

Пример исходного кода для операций с матрицами  261



; Сложение четырех матриц и умножение на -1/p4.
      mov           rcx,[.size]
      xor           rax,rax
; Вычисление -1/p4.
      movsd         xmm0, [.one]
      vdivsd        xmm0,xmm0,xmm15    ; 1/p4
; Проверка деления на ноль.
      stmxcsr       [.mxcsr]
      and          dword[.mxcsr],4
      jnz           .singular

; Нет деления на ноль.
      pop          rsi         ; Восстановление адреса обращенной матрицы.
      vxorpd       xmm0,xmm0,[.minus_mask]  ; -1/p4
      vbroadcastsd ymm2,xmm0

; Цикл по строкам.
.loop2:
      ; Сложение строк.
      vmovapd      ymm0,[.matrix1+rax]
      vaddpd       ymm0, ymm0, [.matrix2+rax]
      vaddpd       ymm0, ymm0, [.matrix3+rax]
      vaddpd       ymm0, ymm0, [.matrixI+rax]
      vmulpd       ymm0,ymm0,ymm2            ; Умножение строки на -1/p4.
      vmovapd      [rsi+rax],ymm0
      add          rax,32
      loop         .loop2
      xor          rax,rax  ; Возврат 0, ошибок нет.
leave
ret

.singular:
      mov          rax,1   ; Возврат 1, вырожденная (сингулярная) матрица.
leave
ret

;------------------------------------------------------
; Вычисление следа матрицы.
vtrace:
push  rbp
mov   rbp,rsp
; Формирование матрицы в памяти.
      vmovapd      ymm0, [rdi]
      vmovapd      ymm1, [rdi+32]
      vmovapd      ymm2, [rdi+64]
      vmovapd      ymm3, [rdi+96]
      vblendpd     ymm0,ymm0,ymm1,0010b
      vblendpd     ymm0,ymm0,ymm2,0100b
      vblendpd     ymm0,ymm0,ymm3,1000b
      vhaddpd      ymm0,ymm0,ymm0
      vpermpd      ymm0,ymm0,00100111b
      haddpd       xmm0,xmm0
leave
ret

;------------------------------------------------------

printm4x4:
section .data
      .fmt db      "%f",9,"%f",9, "%f",9,"%f",10,0
section .text
push  rbp
mov   rbp,rsp
push rbx           ; Регистр, сохраняемый вызываемой функцией.
push r15           ; Регистр, сохраняемый вызываемой функцией.
      mov rdi,.fmt
      mov rcx,4
      xor rbx,rbx         ; Счетчик строк.
.loop:
      movsd  xmm0, [rsi+rbx]
      movsd  xmm1, [rsi+rbx+8]
      movsd  xmm2, [rsi+rbx+16]
      movsd  xmm3, [rsi+rbx+24]
      mov    rax,4        ; Четыре значения с плавающей точкой.
      push   rcx          ; Регистр, сохраняемый вызывающей функцией.
      push   rsi          ; Регистр, сохраняемый вызывающей функцией.
      push   rdi          ; Регистр, сохраняемый вызывающей функцией.
             ; Выравнивание стека, если необходимо.
             xor r15,r15
             test rsp,0xf     ; Последний байт равен 8 (стек не выровнен)?
             setnz r15b       ; Установить флаг, если стек не выровнен.
             shl r15,3        ; Умножение на 8.
             sub rsp,r15      ; Вычитание 0 или 8.
      call   printf
             add rsp,r15      ; Прибавление 0 или 8 для восстановления rsp.
      pop    rdi
      pop    rsi
      pop    rcx
      add    rbx,32           ; Следующая строка.
      loop   .loop
pop r15
pop rbx
leave
ret

;------------------------------------------------------
multi4x4:
push  rbp
mov   rbp,rsp
      xor rax,rax
      mov rcx,4
      vzeroall            ; Обнуление всех ymm-регистров.
.loop:
      vmovapd      ymm0, [rsi]
      vbroadcastsd ymm1,[rdi+rax]
      vfmadd231pd  ymm12,ymm1,ymm0
      vbroadcastsd ymm1,[rdi+32+rax]
      vfmadd231pd  ymm13,ymm1,ymm0
      vbroadcastsd ymm1,[rdi+64+rax]
      vfmadd231pd  ymm14,ymm1,ymm0
      vbroadcastsdymm1,[rdi+96+rax]
      vfmadd231pd  ymm15,ymm1,ymm0
      add rax,8    ; Один элемент содержит 8 байт, 64 бит.

      add rsi,32   ; Каждая строка содержит 32 байта, 256 бит.
      loop .loop
; Перемещение результата в память, строка за строкой.
      vmovapd      [rdx], ymm12
      vmovapd      [rdx+32], ymm13
      vmovapd      [rdx+64], ymm14
      vmovapd      [rdx+96], ymm15
      xor          rax,rax         ; Возврат значения.
leave
ret

Самая	интересная	часть	кода	находится	в функциях.	Основная	функция	main 
предназначена	 для	 инициализации	 программы,	 вызова	 функций	 и  вывода	
результатов.	 В  рассматриваемом	 здесь	примере	используются	матрицы	4×4,	
содержащие	значения	с плавающей	точкой	двойной	точности.	Обратите	вни
мание	на	выравнивание	матриц	по	32байтовой	границе:	применяя	систему	
команд	AVX,	мы	используем	 ymmрегистры	 размером	 32	 байта.	 В  следующих	
разделах	анализируются	отдельные	функции	этой	программы.

вывОд матрицы: printm4x4
Матрица	 считывается	 построчно,	 строки	 помещаются	 в  четыре	 xmmрегистра,	
затем	содержимое	этих	регистров	записывается	в стек.	Эти	регистры	изменяет	
функция	printf,	поэтому	необходимо	сохранить	их.	После	этого	стек	выравни
вается	по	16байтовой	границе.	При	обычной	операции	rsp	будет	выровнен	по	
8байтовой	границе.	Для	выравнивания	стека	по	16байтовой	границе	невозмож
но	использовать	прием	с применением	инструкции	and,	описанный	в главе 16.	
Причина	в том,	что	при	использовании	инструкции	and	неизвестно	заранее,	бу
дет	изменяться	rsp	или	нет.	Кроме	того,	необходимо	скорректировать	указатель	
стека,	потому	что	после	вызова	функции	printf	извлекаются	ранее	сохраненные	
в стеке	регистры.	Если	rsp	был	изменен,	то	необходимо	вернуть	его	предыдущее	
значение	перед	извлечением,	иначе	из	стека	будут	восстановлены	некорректные	
значения.	Если	rsp	не был	изменен,	то	корректировка	указателя	стека	не нужна.

Для	проверки	выравнивания	стека	используется	инструкция	test	и значение	
0xf.	Если	самая	последняя	шестнадцатеричная	цифра	в rsp	равна	0,	то	указа
тель	стека	rsp	выровнен	по	16байтовой	границе.	Если	последняя	цифра	отли
чается	от	0,	то	самый	последний	полубайт	содержит,	как	минимум,	один	бит,	
установленный	в 1.	Инструкция	test	похожа	на	инструкцию	and.	Если	самый	
последний	полубайт	rsp	содержит	один	или	несколько	битов,	установленных	
в 1,	то	результат	сравнения	будет	ненулевым,	поэтому	флаг	ZF	будет	очищен	
(сброшен	в 0).	Инструкция	setnz	(set	if	nonzero)	считывает	флаг	ZF,	и если	он	
не установлен,	то	setnz	записывает	значение	0000 0001	в r15b.	Это	означает,	что	
rsp	 не  выровнен	 по	 16байтовой	 границе,	 поэтому	 выполняется	 вычитание	
8	для	выравнивания.	Для	r15b	три	раза	выполняется	операция	 сдвига	влево,	
чтобы	получить	десятичное	значение	8	и произвести	вычитание.	После	завер
шения	работы	функции	printf	восстанавливается	корректный	адрес	стека	по
средством	прибавления	r15	к rsp,	т. е.	значения	8,	если	стек	выравнивался,	или	
значения	0,	если	стек	не выравнивался.	Таким	образом,	стек	возвращается	в то	
состояние,	в котором	он	был	перед	выравниванием,	после	чего	можно	восста
навливать	корректные	значения	регистров.

умнОжение матриц: multi4x4
В	рассматриваемом	здесь	примере	исходного	кода	и в	приведенном	ниже	опи
сании	используются	следующие	две	матрицы:

1 3 5 7 2 4 6 8
9 11 13 15 10 12 14 16

        
17 19 21 23 18 20 22 24
25 27 29 31 26 28 30 32

A B

   
   
   = =
   
   
   

;

Если	 вы	 изучали	 линейную	 алгебру,	 то,	 вероятно,	 знаете,	 что	 умножение	
мат	риц	выполняется	следующим	образом:	для	получения	элемента	c11	итого
вой	матрицы	C =	AB	требуется	вычисление	по	формуле:

a11b11	+	a12b21	+	a13b31	+	a14b41.

В	рассматриваемом	здесь	примере	вычисление	по	этой	формуле	выглядит	так:

1×2	+	3×10	+	5×18	+	7×26	=	304.

Другой	элемент	c32	должен	вычисляться,	как	показано	ниже:

a31b12	+	a32b22	+	a33b32	+	a34b42.

С	учетом	конкретных	значений	элементов	матриц	A и	B получаем:
 

17×4	+	19×12	+	21×20	+	23×28	=	1360.

Это	вполне подходит	для	вычислений	вручную,	но	мы	будем	использовать	
метод,	более	подходящий	для	компьютера.	Воспользуемся	ymmрегистрами	для	
хранения	вычисляемых	сумм	и для	обновления	сумм	в последующих	циклах.	
Здесь	мы	используем	всю	мощь	инструкций	AVX.

Сначала	производится	очистка	всех	ymmрегистров	с помощью	инструкции	
vzeroall.	Затем	выполняется	цикл	четыре	раза,	по	одной	итерации	для	каждой	
строки	матрицы	matrixB.	Строка	из	четырех	значений	двойной	точности	загру
жается	из	matrixB	в регистр	ymm0.	Далее	значение	из	последовательно	выбирае
мых	столбцов	матрицы	matrixA	распределяется	по	частям	регистра	ymm1.	Регистр	
rax	служит	счетчиком	столбцов,	при	этом	значения	в столбцах	располагаются	со	
смещением	0,	32,	64	и 96.	Распределение	(broadcast –	«рассеяние,	распыление»)	
означает,	 что	 все	 четыре	 учетверенных	 слова	 (quadwords)	 (размером	 8	 байт	
каждое)	 в регистре	будут	 содержать	 соответствующее	 значение.	После	 этого	
значения	в ymm1	умножаются	на	значения	ymm0	и суммируются	в регистре	ymm12.	
Умножение	и суммирование	выполняется	одной	инструкцией	vfmadd231pd,	ко
торая	означает	«vector	fused	multiply	add	packed	double»	(умножение	и сумми
рование	объединенного	вектора	из	упакованных	значений	двойной	точности).	
Набор	цифр	231	показывает,	как	используются	регистры.	Существует	несколько	
вариантов	инструкции	vfmadd	(132,	213,	231),	кроме	того,	имеются	варианты	для	

значений	с двойной	и обычной	(одиночной)	точностью.	Здесь	используется	ва
риант	231,	который	означает	умножение	второго	операнда	на	третий	операнд	
и сложение	с первым	операндом,	а результат	помещается	в первый	операнд.	
Эти	действия	выполняются	для	каждого	значения	столбца	матрицы	matrixA,	за
тем	итерация	продолжается;	загружается	следующая	строка	матрицы	matrixB,	
и вычисления	повторяются.

Пройдите	по	программе	по	шагам	с помощью	отладчика,	который	вы	пред
почитаете	использовать.	Наблюдайте	за	тем,	как	в регистрах	ymm12,	ymm13,	ymm14 
и ymm15	сохраняются	текущие	результаты	суммирования	и формируется	итого
вое	произведение.	Вероятнее	всего,	ваш	отладчик	будет	показывать	значения	
в ymmрегистрах	в шестнадцатеричном	формате	и с	прямым	порядком	байтов	
(от	младшего	к старшему).	Чтобы	вам	было	проще	разобраться,	в табл. 36.1–36.5	
показаны	все	подробности	того,	что	происходит	на	каждом	шаге	цикла.

Таблица 36.1.	Начальные значения элементов умножаемых матриц

rdi rsi

32 байта 32 байта

8 байт 8 байт 8 байт 8 байт 8 байт 8 байт 8 байт 8 байт 8 байт

0–31 1 3 5 7 0–31 2 4 6 8

32–63 9 11 13 15 32–63 10 12 14 16

64–95 17 19 21 23 64–95 18 20 22 24

96–127 25 27 29 31 96–127 26 28 30 32

Первая	итерация	цикла	показана	в табл. 36.2.

Таблица 36.2.	Первая итерация цикла

Инструкция Регистр Значения

vmovapd ymm0, [rsi] ymm0 2 4 6 8

vbroadcastsd ymm1, [rdi+0] ymm1 1 1 1 1

vfmadd231pd ymm12, ymm1, ymm0 ymm12 2 4 6 8

vbroadcastsd ymm1, [rdi+32+0] ymm1 9 9 9 9

vfmadd231pd ymm13, ymm1, ymm0 ymm13 18 36 54 72

vbroadcastsd ymm1, [rdi+64+0] ymm1 17 17 17 17

vfmadd231pd ymm14, ymm1, ymm0 ymm14 34 68 102 136

vbroadcastsd ymm1, [rdi+96+0] ymm1 25 25 25 25

vfmadd231pd ymm15, ymm1, ymm0 ymm15 50 100 150 200

Вторая	итерация	цикла	показана	в табл. 36.3.

Таблица 36.3.	Вторая итерация цикла

Инструкция Регистр Значения

vmovapd ymm0, [rsi+32] ymm0 10 12 14 16

vbroadcastsd ymm1, [rdi+8] ymm1 3 3 3 3

vfmadd231pd ymm12, ymm1, ymm0 ymm12 32 40 48 56

vbroadcastsd ymm1, [rdi+32+8] ymm1 11 11 11 11

vfmadd231pd ymm13, ymm1, ymm0 ymm13 128 168 208 248

vbroadcastsd ymm1, [rdi+64+8] ymm1 19 19 19 19

vfmadd231pd ymm14, ymm1, ymm0 ymm14 224 296 368 440

vbroadcastsd ymm1, [rdi+96+8] ymm1 27 27 27 27

vfmadd231pd ymm15, ymm1, ymm0 ymm15 320 424 528 632

Третья	итерация	цикла	показана	в табл. 36.4.

Таблица 36.4.	Третья итерация цикла

Инструкция Регистр Значения

vmovapd ymm0, [rsi+32+32] ymm0 18 20 22 24

vbroadcastsd ymm1, [rdi+8+8] ymm1 5 5 5 5

vfmadd231pd ymm12, ymm1, ymm0 ymm12 122 140 158 176

vbroadcastsd ymm1, [rdi+32+8+8] ymm1 13 13 13 13

vfmadd231pd ymm13, ymm1, ymm0 ymm13 362 428 494 560

vbroadcastsd ymm1, [rdi+64+8+8] ymm1 21 21 21 21

vfmadd231pd ymm14, ymm1, ymm0 ymm14 602 716 830 944

vbroadcastsd ymm1, [rdi+96+8+8] ymm1 29 29 29 29

vfmadd231pd ymm15, ymm1, ymm0 ymm15 842 1004 1166 1328

Четвертая	(и последняя)	итерация	цикла	показана	в табл. 36.5.

Таблица 36.5.	Четвертая (последняя) итерация цикла

Инструкция Регистр Значения

vmovapd ymm0, [rsi+32+32+32] ymm0 26 28 30 32

vbroadcastsd ymm1, [rdi+8+8+8] ymm1 7 7 7 7

vfmadd231pd ymm12, ymm1, ymm0 ymm12 304 336 368 400

vbroadcastsd ymm1, [rdi+32+8+8+8] ymm1 15 15 15 15

vfmadd231pd ymm13, ymm1, ymm0 ymm13 752 848 944 1040

vbroadcastsd ymm1, [rdi+64+8+8+8] ymm1 23 23 23 23

vfmadd231pd ymm14, ymm1, ymm0 ymm14 1200 1360 1520 1680

vbroadcastsd ymm1, [rdi+96+8+8+8] ymm1 31 31 31 31

vfmadd231pd ymm15, ymm1, ymm0 ymm15 1648 1872 2096 2320

Обращение матрицы: inverse4x4
Математики	разработали	 группу	 алгоритмов	для	 эффективного	 вычисления	
обращения	матрицы.	Целью	этого	раздела	является	не предоставление	полной	
программы	обращения	матрицы	со	всеми	атрибутами	и деталями,	а простая	
демонстрация	использования	системы	инструкций	AVX.

В	 рассматриваемом	 здесь	 примере	 будет	 использоваться	метод,	 основан
ный	на	теореме	Гамильтона–Кэли	 (Cayley–Hamilton theorem)	 о  характеристи
ческих	 многочленах.	 Более	 подробную	 информацию	 о  характеристических	
многочленах	 можно	 найти	 здесь:	 http://www.mcs.csueastbay.edu/~malek/Class/
Characteristic.pdf.

Теорема Гамильтона–Кэли
По	теореме	Гамильтона–Кэли	для	матрицы	A существует	следующий	характе
ристический	многочлен:

An	+	p1A
n−1	+	...	+	pn−1A	+	pnI	=	0,

где	An –	это	A в	степени	n.	Например,	A3	равно	AAA,	т. е.	матрица	A умножается	
сама	на	себя	три	раза.	Значения	p –	это	определяемые	коэффициенты,	I –	еди
ничная	матрица	(матрица	тождественности),	0 –	нулевая	матрица.

Если	приведенную	выше	формулу	умножить	на	A−1,	разделить	на	−pn	и пре
образовать	(перегруппировать	члены),	то	получим	следующую	формулу	обра
щения	матрицы:

(1/−pn)[A
n−1	+	p1A

n−2	+	...	+	pn−2A	+	pn−1I]	=	A
−1.

Таким	образом,	для	вычисления	обращения	матрицы	A необходимо	выпол
нить	несколько	операций	умножения	матриц,	а также	требуется	метод	опреде
ления	коэффициентов	p.

Для	матрицы	A размером	4×4	получаем	следующую	формулу:

(1/−p4)[A
3	+p1A

2	+	p2A	+	p3I]	=	A
−1.

Алгоритм Фаддеева–Леверье
Для	 вычисления	 коэффициентов	 p  используется	 алгоритм	 Фаддеева–Ле
верье,	 также	 описанный	 здесь:	 http://www.mcs.csueastbay.edu/~malek/Class/
Characteristic.pdf.	Сначала	определяются	следы	матриц,	т. е.	суммы	элементов	
главных	диагоналей	(от	верхнего	левого	угла	к нижнему	правому).	Пусть	sn –	
след	матрицы	An.

Для	матрицы	A размером	4×4	вычисляются	следы	степеней	матрицы	A,	как	
показано	ниже:

s1	для	A,
s2	для	AA,
s3	для	AAA,
s4	для	AAAA.

Тогда	по	алгоритму	Фаддеева–Леверье	получим	следующие	формулы:

p1	=	−s1,

p2	=	−1/2(p1s1	+	s2),

p3	=	−1/3(p2s1	+	p1s2	+	s3),

p4	=	−1/4(p3s1	+	p2s2	+	p1s3	+	s4).

Достаточно	просто,	не так	ли?	Если	не принимать	во	внимание	более	слож
ные	операции	умножения	матриц	для	вычисления	следов,	разумеется.

Исходный код
В	предлагаемой	функции	inverse4x4	существует	отдельный	раздел	section .data,	
в  котором	 определяется	 единичная	 матрица	 и  некоторые	 переменные,	 ис
пользуемые	в дальнейшем.	Сначала	вычисляются	степени	матрицы	и сохра
няются	в переменных	matrix2,	matrix3	и matrix4.	Переменная	matrix1	пока	не ис
пользуется.	Затем	вызывается	функция	vtrace	для	вычисления	следа	каждой	
(степени)	матрицы.	В функции	vtrace	сначала	формируется	требуемая	матрица	
в ymmрегистрах	(ymm0,	ymm1,	ymm2,	ymm3),	при	этом	каждый	регистр	содержит	стро
ку	матрицы.	Далее	применяется	инструкция	vblendpd	с четырьмя	операндами:	
два	операндаисточника,	один	целевой	операнд	и управляющая	маска.	Необ
ходимо	извлечь	диагональные	элементы	из	строк	2,	3,	4	и поместить	их	как	
упакованные	значения	в регистр	ymm0	по	индексам	локаций	1,	2	и 3.	В локации	
с индексом	0	сохраняется	элемент	следа	ymm0.

Маска	определяет,	какие	упакованные	значения	выбираются	из	операндов
источников.	Единица	(1)	в маске	означает,	что	в этой	локации	выбирается	зна
чение	из	второго	операндаисточника.	Ноль	(0)	в маске	означает,	что	в этой	ло
кации	выбирается	значение	из	первого	операндаисточника.	Общая	схема	этого	
процесса	показана	на	рис. 36.1,	но	следует	отметить,	что	на	этой	схеме	значения	
в регистрах	показаны	в соответствии	с номерами	битов	в маске.	В отладчике,	
возможно,	позиции	в регистре	ymm0	будут	отображаться	как	a1,	a0,	a3,	a2.

Маска смешивания (наложения)

Источник 1: ymm0

Источник 2: ymm1

Маска 0010

Цель: ymm0

Источник 2: ymm2

Маска 0100

Цель: ymm0

Источник 2: ymm3

Маска 1000

Цель: ymm0

Рис. 36.1.	Маска смешивания (наложения)

При	вычислении	первого	следа	после	операции	смешивания	(наложения)	
(blend)	регистр	ymm0	содержит	элементы	следа	2,	13,	29,	47.	Это	можно	прове
рить	с помощью	отладчика	SASM.	Пусть	вас	не смущает	порядок	представле
ния	значений	ymm0	в отладчике:	13,	2,	47,	29.	Теперь	нужно	вычислить	сумму	
этих	значений.	Проще	всего	это	можно	сделать,	извлекая	и суммируя	требу
емые	значения,	но	для	наглядной	демонстрации	здесь	будут	использоваться	
инструкции	AVX.	Применяется	инструкция	горизонтального	сложения	vhad-
dpd.	 После	 этого	 регистр	 ymm0	 содержит	 значения	 15,	 15,	 76,	 76,	 т.  е.	 сумму	
двух	младших	значений	и сумму	двух	старших	значений.	Затем	выполняется	
инструкция	перестановки	vpermpd	с маской	00100111.	Каждое	двухбитовое	зна
чение	в этой	маске	выбирает	значение	в операндеисточнике,	схема	данной	
операции	показана	на	рис. 36.2.	Теперь	младшая	часть	регистра	ymm0,	т. е.	ре
гистр	xmm0	содержит	два	значения,	которые	нужно	сложить,	чтобы	получить	
след	 матрицы.	 Выполняется	 горизонтальное	 сложение	 для	 регистра	 xmm0 

с помощью	инструкции	haddpd.	Вычисляемые	следы	сохраняются	в регистрах	
xmm8,	xmm9,	xmm10,	xmm11	для	дальнейшего	использования.

Вам	не кажется	слишком	сложным	и запутанным	такое	вычисление	следа	
матрицы?	Этот	способ	был	применен	только	для	демонстрации	использования	
нескольких	инструкций	AVX	и соответствующих	масок.

Маска перестановок

Источник 1: ymm0

Маска 00100111

Цель: ymm0

Рис. 36.2.	Маска перестановок

После	 получения	 всех	 требуемых	 следов	 матриц	 можно	 начать	 вычисле
ние	 коэффициентов	 p.	 Обратите	 внимание	 на	 изменение	 знака	 следа	 с  по
мощью	применения	отрицательной	маски	и инструкции	 vxorpd.	Инструкции	
vfmadd213sd	и vfmadd231sd	используются	для	объединения	операций	умножения	
и сложения	в одной	инструкции.	Инструкция	vfmadd213sd	означает	умножение	
первого	и второго	операндов,	прибавление	третьего	операнда	и запись	резуль
тата	в первый	операнд.	Инструкция	vfmadd231sd	означает	умножение	второго	
и  третье	го	 операндов,	 прибавление	 первого	 операнда	 и  запись	 результата	
в  первый	 операнд.	 Полный	 список	 аналогичных	 инструкций	 см.	 в  руковод
ствах	Intel.	Внимательно	изучите	их	описание.

После	вычисления	всех	требуемых	коэффициентов	выполняется	скалярное	
умножение	матриц	matrix,	matrix2,	matrix3	и matrixI	с коэффициентами	в соот
ветствии	с приведенной	выше	формулой.	Результат	умножения	с использова
нием	matrix	сохраняется	в matrix1.	Матрица	matrix4	больше	не нужна,	поэтому	из	
соображений	экономии	памяти	можно	использовать	пространство	для	inverse 
как	временную	память	вместо	matrix4.

Требуется	деление	на	коэффициент	p4,	поэтому	необходима	проверка	p4	на	
ненулевое	значение.	Можно	было	бы	выполнить	эту	простую	операцию	раньше,	
сразу	после	вычисления	p4,	но	в нашем	случае	нужно	показать,	как	используется	
регистр	mxcsr.	В регистре	mxcsr	устанавливается	в 1	бит	деления	на	ноль	и вы
полняется	операция	деления	с помощью	инструкции	vdivsd.	Если	после	деления	
третий	бит	(индекс	2)	в регистре	mxcsr	установлен	в 1,	то	произошло	деление	на	
ноль,	следовательно,	матрица	является	сингулярной	(вырожденной)	и не может	
быть	обращена.	В инструкции	and	применяется	десятичное	значение	4,	соответ
ствующее	двоичному	значению	0000 0100,	так	что	очевидно	проверяется	третий	

бит.	При	обнаружении	деления	на	ноль	происходит	перенаправление	к выходу	
с 1	в регистре	rax,	т. е.	сообщение	об	ошибке	для	вызывающей	функции.

Если	матрица	вырожденная	 (сингулярная),	то	программа	не «падает»,	по
тому	что	по	умолчанию	деление	на	ноль	маскировано	в регистре	mxcsr.	После	
завершения	подробного	анализа	этого	кода	удалите	комментарии	в той	части,	
которая	проверяет	деление	на	ноль,	и понаблюдайте,	что	происходит.

Если	коэффициент	p4	не равен	нулю,	то	выполняется	сложение	четырех	по
лученных	матриц,	потом	скалярное	умножение	результата	на	−1/p4.	Сложение	
и умножение	выполняются	в одной	итерации	цикла.	Если	все	идет,	как	предпо
лагалось,	то	мы	получаем	обращенную	матрицу,	а вызывающей	функции	воз
вращается	0	в регистре	rax.

На	рис. 36.3	показан	вывод	этой	программы.

Рис. 36.3.	Вывод программы matrix4x4.asm

резюме
В	этой	главе	вы	узнали	об:

	� операциях	с матрицами	с использованием	инструкций	AVX;
	� инструкциях	AVX	с тремя	операндами;
	� объединенных	операциях	AVX;
	� использовании	масок	для	смешивания	(наложения)	и перестановок.

Глава 37
Транспонирование матриц

Рассмотрим	 последнюю	 полезную	 операцию	 с  матрицами:	 транспонирова
ние.	В этой	главе	представлен	исходный	код	двух	версий:	в первой	использу
ются	неупакованные	данные,	во	второй	применяется	метод	перемешивания	
(shuffling).

пример иСхОднОгО кОда для транСпОнирОвания матриц
В	листинге	37.1	показан	исходный	код	программы.

Листинг 37.1.	Программа transpose4x4.asm

; transpose4x4.asm
extern printf

section .data
      fmt0   db   "4x4 DOUBLE PRECISION FLOATING POINT MATRIX TRANSPOSE",10,0
      fmt1   db   10,"This is the matrix:",10,0
      fmt2   db   10,"This is the transpose (unpack):",10,0
      fmt3   db   10,"This is the transpose (shuffle):",10,0
      align  32
      matrix dq       1.,     2.,     3.,     4.
             dq       5.,     6.,     7.,     8.
             dq       9.,    10.,    11.,    12.
             dq      13.,    14.,    15.,    16.

section .bss
      alignb       32
      transpose    resd   16

section .text
      global main
main:
push  rbp
mov   rbp,rsp

; Вывод заголовка.
      mov    rdi, fmt1
      call   printf



; Вывод матрицы.
      mov    rdi,fmt1
      call   printf
      mov    rsi,matrix
      call   printm4x4

; Вычисление транспонированной матрицы с неупакованными данными.
      mov    rdi, matrix
      mov    rsi, transpose
      call   transpose_unpack_4x4

; Вывод результата.
      mov    rdi, fmt2
      xor    rax,rax
      call   printf
      mov    rsi, transpose
      call   printm4x4

; Вычисление транспонированной матрицы с применением перемешивания.
      mov    rdi, matrix
      mov    rsi, transpose
      call   transpose_shuffle_4x4

; Вывод результата.
      mov    rdi, fmt3
      xor    rax,rax
      call   printf
      mov    rsi, transpose
      call   printm4x4
leave
ret

;--------------------------------------------------------
transpose_unpack_4x4:
push  rbp
mov   rbp,rsp
; Загрузка матрицы в регистры.
      vmovapd       ymm0,[rdi]    ;  1   2   3   4
      vmovapd       ymm1,[rdi+32] ;  5   6   7   8
      vmovapd       ymm2,[rdi+64] ;  9  10  11  12
      vmovapd       ymm3,[rdi+96] ; 13  14  15  16
; Распаковка.
      vunpcklpd ymm12,ymm0,ymm1         ;  1   5   3   7
      vunpckhpd ymm13,ymm0,ymm1         ;  2   6   4   8
      vunpcklpd ymm14,ymm2,ymm3         ;  9  13  11  15
      vunpckhpd ymm15,ymm2,ymm3         ; 10  14  12  16
; Перестановка.
      vperm2f128 ymm0,ymm12,ymm14, 00100000b    ; 1   5   9  13
      vperm2f128 ymm1,ymm13,ymm15, 00100000b    ; 2   6  10  14
      vperm2f128 ymm2,ymm12,ymm14, 00110001b    ; 3   7  11  15
      vperm2f128 ymm3,ymm13,ymm15, 00110001b    ; 4   8  12  16
; Запись в память.
      vmovapd       [rsi],   ymm0
      vmovapd       [rsi+32],ymm1
      vmovapd       [rsi+64],ymm2
      vmovapd       [rsi+96],ymm3
leave
ret

;--------------------------------------------------------
transpose_shuffle_4x4:
push  rbp
mov   rbp,rsp
; Загрузка матрицы в регистры.
      vmovapd       ymm0,[rdi]    ;  1   2   3   4
      vmovapd       ymm1,[rdi+32] ;  5   6   7   8
      vmovapd       ymm2,[rdi+64] ;  9  10  11  12
      vmovapd       ymm3,[rdi+96] ; 13  14  15  16
; Перемешивание.
      vshufpd       ymm12,ymm0,ymm1, 0000b    ;  1   5   3   7
      vshufpd       ymm13,ymm0,ymm1, 1111b    ;  2   6   4   8
      vshufpd       ymm14,ymm2,ymm3, 0000b    ;  9  13  11  15
      vshufpd       ymm15,ymm2,ymm3, 1111b    ; 10  14  12  16
; Перестановка.
      vperm2f128 ymm0,ymm12,ymm14, 00100000b    ; 1   5   9  13
      vperm2f128 ymm1,ymm13,ymm15, 00100000b    ; 2   6  10  14
      vperm2f128 ymm2,ymm12,ymm14, 00110001b    ; 3   7  11  15
      vperm2f128 ymm3,ymm13,ymm15, 00110001b    ; 4   8  12  16
; Запись в память.
      vmovapd       [rsi],   ymm0
      vmovapd       [rsi+32],ymm1
      vmovapd       [rsi+64],ymm2
      vmovapd       [rsi+96],ymm3
leave
ret

;--------------------------------------------------------
printm4x4:
section .data
      .fmt   db     "%.f",9,"%.f",9, "%.f",9,"%.f",10,0
section .text
push  rbp
mov   rbp,rsp
push  rbx                 ; Регистр, сохраняемый вызываемой функцией.
push  r15                 ; Регистр, сохраняемый вызываемой функцией.
      mov    rdi,.fmt
      mov    rcx,4
      xor    rbx,rbx      ; Счетчик строк.
.loop:
      movsd  xmm0, [rsi+rbx]
      movsd  xmm1, [rsi+rbx+8]
      movsd  xmm2, [rsi+rbx+16]
      movsd  xmm3, [rsi+rbx+24]
      mov          rax,4  ; Четыре числа с плавающей точкой.
        push rcx          ; Регистр, сохраняемый вызывающей функцией.
        push rsi          ; Регистр, сохраняемый вызывающей функцией.
        push rdi          ; Регистр, сохраняемый вызывающей функцией.
        ; Требуется выравнивание стека.
        xor   r15,r15
        test  rsp,0fh      ; Последний байт равен 8 (стек не выровнен)?
        setnz r15b         ; Установить флаг, если стек не выровнен.
        shl   r15,3        ; Умножение на 8.
        sub   rsp,r15      ; Вычитание 0 или 8.
      call   printf
        add  rsp,r15      ; Прибавление 0 или 8.

        pop  rdi
        pop  rsi
        pop  rcx
        add  rbx,32       ; Следующая строка.
        loop .loop
pop r15
pop rbx
leave
ret

На	рис. 37.1	показан	вывод	этой	программы.

Рис. 37.1.	Вывод программы transpose.asm

верСия С иСпОльзОванием неупакОванных данных
Сначала	необходимо	сделать	замечание	относительно	прямого	порядка	байтов	
и упакованных	значений	в ymmрегистрах.	Если	в примере	имеются	строки	1,	2,	
3,	4,	то	в формате	с прямым	порядком	байтов	(от	младшего	к старшему)	они	
должны	выглядеть	так:	4,	3,	2,	1.	Но поскольку	в рассматриваемом	здесь	при
мере	в ymmрегистре	 хранятся	 упакованные	 значения,	 содержимое	 этого	ymm
регистра	в SASM	будет	представлено	следующим	образом:	2,	1,	4,	3.	Это	можно	
проверить	 в  любом	 отладчике.	 Такое	 представление	 данных	 может	 сбивать	
с толку	при	отладке	программы.	В дальнейшем	всюду	будет	использоваться	
формат	с прямым	порядком	байтов	4,	3,	2,	1,	а не «формат	отладчика» 2, 1,	4,	3.

С	 учетом	 приведенного	 выше	 замечания	 после	 загрузки	 матрицы	 в  ymm
регистры	расположение	данных	в  этих	регистрах	будет	таким,	как	показано	
в табл. 37.1	(в круглых	скобках	приведены	значения	примера).

Таблица 37.1.	Содержимое ymm-регистров после загрузки матрицы

ymm0 Старшее слово 
qword2 (4)

Младшее слово 
qword2 (3)

Старшее слово 
qword1 (2)

Младшее слово 
qword1 (1)

ymm1 Старшее слово 
qword4 (8)

Младшее слово 
qword4 (7)

Старшее слово 
qword3 (6)

Младшее слово 
qword3 (5)

. . .

Инструкция	 vunpcklpd ymm12,ymm0,ymm1	 принимает	 первые	 младшие	 учетве
ренные	слова	из	операндов	2	и 3	и сохраняет	их	в операнде	1,	затем	таким	же	
способом	принимает	вторые	(самые)	младшие	учетверенные	слова	для	полу
чения	результата,	показанного	в табл. 37.2.

Таблица 37.2.	Распаковка данных из младших учетверенных слов

ymm12 Младшее слово 
qword4 (7)

Младшее слово 
qword2 (3)

Младшее слово 
qword3 (5)

Младшее слово 
qword1 (1)

Аналогичным	образом	инструкция	vunpckhpd ymm13,ymm0,ymm1	принимает	стар
шие	учетверенные	слова	из	операндов	2	и 3,	затем	сохраняет	их	в операнде	1	
таким	же	способом	(см.	табл. 37.3).

Таблица 37.3.	Распаковка данных из старших учетверенных слов

ymm13 Старшее слово 
qword4 (8)

Старшее слово 
qword2 (4)

Старшее слово 
qword3 (6)

Старшее слово 
qword1 (2)

Цель	 этого	 метода	 распаковки	 данных –	 замена	 пар	 в  столбцах	 на	 пары

в	строках.	Например,	
1
5
 
 
 

	превращается	в [1,	5].

После	распаковки	ymmрегистры	выглядят	так,	как	показано	в табл. 37.4	(фор
мат	с прямым	порядком	байтов).

Таблица 37.4.	Содержимое ymm-регистров после распаковки

ymm12 7 3 5 1

ymm13 8 4 6 2

ymm14 15 11 13 9

ymm15 16 12 14 10

Вместо	 формата	 с  прямым	 порядком	 байтов	 (от	 младшего	 к  старшему)	
в табл. 37.5	показан	удобный	для	чтения	человеком	формат.

Таблица 37.5.	Матрица в формате, удобном для чтения человеком

1 5 3 7

2 6 4 8

9 13 11 15

10 14 12 16

Теперь	необходимо	переставить	значения	в строках,	чтобы	получить	их	пра
вильный	порядок.	В формате	с прямым	порядком	байтов	требуется	получить	
результат,	показанный	в табл. 37.6.

Таблица 37.6.	Матрица после перестановки в правильном порядке

13 9 5 1

14 10 6 2

15 11 7 3

16 12 8 4

Возможно,	вы	заметили,	что	два	младших	значения	в регистрах	ymm12	и ymm13 
находятся	на	правильных	позициях.	Точно	так	же	два	старших	значения	в ре
гистрах	ymm14	и ymm15	расположены	правильно.

Необходимо	переместить	два	младших	значения	регистра	ymm14	в позиции	
старших	значений	регистра	ymm12,	а два	младших	значения	регистра	ymm15	в по
зиции	старших	значений	регистра	ymm13.

Два	старших	значения	из	ymm12	должны	переместиться	в позиции	младших	
значений	ymm14,	а два	старших	значения	из	ymm13 –	в позиции	младших	значе
ний	ymm15.

Требуется	выполнение	операции,	называемой	перестановкой	(permutation).	
С помощью	инструкции	vperm2f128	можно	выполнять	перестановку	двух	значе
ний	(128	бит).	Для	управления	перестановкой	применяется	маска:	например,	
маска	00110001	читается,	начиная	с младших	битов.	Ниже	приведено	описание	
смысла	маски	с учетом	того,	что	индексация	начинается	с 0:

	� 01 –	взять	128битовое	старшее	поле	из	источника	1	и поместить	его	в по
зицию	0	цели;

	� 00 –	эти	биты	имеют	особый	смысл;	см.	описание	ниже;
	� 11 –	взять	128битовое	старшее	поле	из	источника	2	и поместить	его	в по
зицию	128	цели;

	� 00 –	эти	биты	имеют	особый	смысл;	см.	описание	ниже.
Здесь	 также	 используется	 формат	 с  прямым	 порядком	 байтов	 (4,	 3,	 2,	 1)	

и не учитывается	порядок,	в котором	эти	значения	хранятся	в ymmрегистрах.
Таким	образом,	в действительности	два	128битовых	поля	из	двух	источни

ков	нумеруются	последовательно:

	� младшее	поле	источника	1	=	00;
	� старшее	поле	источника	1	=	01;
	� младшее	поле	источника	2	=	10;
	� старшее	поле	источника	2	=	11.

Упомянутый	выше	«особый	смысл»	заключается	в следующем:	если	устанав
ливается	в 1	третий	бит	(индекс	3)	в маске,	то	младшее	поле	цели	будет	обнуле
но,	а если	устанавливается	в 1	седьмой	бит	(индекс	7)	в маске,	то	старшее	поле	
цели	будет	обнулено.

Второй,	третий,	шестой	и седьмой	биты	здесь	не используются.	В большин
стве	случаев	можно	читать	маску	00110001	как	00110001.

Именно	это	и происходит	в программе:

vperm2f128    ymm0, ymm12, ymm14, 00100000b

Здесь	используется	маска	00100000.
	� Младшие	биты	00	означают:	взять	младшее	поле	ymm12	(5,	1)	и поместить	
его	в младшее	поле	ymm0.

	� Старшие	биты	10	означают:	взять	младшее	поле	ymm14	(13,	9)	и поместить	
его	в старшее	поле	ymm0.

Таблица 37.7.	ymm-регистры после первой перестановки

Регистр Значения

ymm12 7 3 5 1

ymm14 15 11 13 9

ymm0 13 9 5 1

Теперь	регистр	ymm0	содержит	завершенную	строку.	Далее	формируется	сле
дующая	строка.

vperm2f128    ymm1, ymm13, ymm15, 00100000b

Здесь	используется	маска	00100000.
	� Младшие	биты	00	означают:	взять	младшее	поле	ymm13	(6,	2)	и поместить	
его	в младшее	поле	ymm1.

	� Старшие	биты	10	означают:	взять	младшее	поле	ymm15	(14,	10)	и помес
тить	его	в старшее	поле	ymm1.

Таблица 37.8.	ymm-регистры после второй перестановки

Регистр Значения

ymm13 8 4 6 2

ymm15 16 12 14 10

ymm1 14 10 6 2

280	  Транспонирование матриц



Теперь	регистр	ymm1	содержит	завершенную	строку.	Формируется	следующая	
строка:

vperm2f128    ymm2, ymm12, ymm14, 00110001b

Здесь	используется	маска	00110001.
	� Младшие	биты	01	означают:	взять	старшее	поле	ymm12	(7,	3)	и поместить	
его	в младшее	поле	ymm2.

	� Старшие	биты	11	означают:	взять	старшее	поле	ymm14	(15,	11)	и поместить	
его	в старшее	поле	ymm2.

Таблица 37.9.	ymm-регистры после третьей перестановки

Регистр Значения

ymm12 7 3 5 1

ymm14 15 11 13 9

ymm2 15 11 7 3

Теперь	регистр	ymm2	содержит	завершенную	строку.	Формируется	последняя	
строка:

vperm2f128    ymm3, ymm13, ymm15, 00110001b

Здесь	используется	маска	00110001.
	� Младшие	биты	01	означают:	взять	старшее	поле	ymm13	(8,	4)	и поместить	
его	в младшее	поле	ymm3.

	� Старшие	биты	11	означают:	взять	старшее	поле	ymm15	(16,	12)	и поместить	
его	в старшее	поле	ymm3.

Таблица 37.10.	ymm-регистры после четвертой перестановки

Регистр Значения

ymm13 8 4 6 2

ymm15 16 12 14 10

ymm3 16 12 8 4

Перестановка	 выполнена.	 Осталось	 лишь	 скопировать	 строки	 из	
ymmрегистров	в память	в правильном	порядке.

верСия С применением перемешивания
Мы	уже	использовали	инструкцию	перемешивания	pshufd	в главе	33.	Здесь	бу
дет	применяться	инструкция	vshufpd,	для	которой	также	требуется	маска	для	
управления	 операцией	 перемешивания.	 Не  перепутайте  –	 для	 инструкции	
pshufd	использовалась	8битовая	маска.	В масках,	которые	применяются	здесь,	
значимыми	являются	только	4	бита.

В	этой	версии	примера	также	используется	формат	с прямым	порядком	бай
тов	(4,	3,	2,	1)	и не учитывается	порядок	хранения	упакованных	значений	в ymm
регистрах.	Это	забота	процессора.

В	табл.  37.11	 приведены	примеры,	 соответствующие	 следующему	 описа
нию.	Два	младших	бита	в управляющей	маске	определяют,	какие	упакован
ные	значения	записываются	в две	младшие	позиции	цели.	Два	старших	бита	
в управляющей	маске	определяют,	какие	упакованные	значения	записыва
ются	в две	 старшие	позиции	цели.	Биты	0	и 2	определяют,	какое	значение	
берется	из	источника	1,	биты	1	и 3	определяют,	какое	значение	берется	из	
источника	2.

Таблица 37.11.	Примеры значений битов управляющей маски

Выбор из двух 
старших значений 
в источнике 2

Выбор из двух 
старших значений 
в источнике 1

Выбор из двух 
младших значений 
в источнике 2

Выбор из двух 
младших значений 
в источнике 1

0 = младшее значе-
ние источника 2

0 = младшее значе-
ние источника 1

0 = младшее значе-
ние источника 2

0 = младшее значе-
ние источника 1

1 = старшее значе-
ние источника 2

1 = старшее значе-
ние источника 1

1 = старшее значе-
ние источника 2

1 = старшее значе-
ние источника 1

Два	 младших	 значения	 в  каждом	 источнике	 никогда	 не  могут	 оказаться	
в старших	позициях	цели,	а два	старших	значения	в каждом	источнике	никог
да	не могут	оказаться	в младших	позициях	цели.	На рис. 37.2	показаны	схемы	
для	нескольких	примеров	управляющих	масок.

Маска, управляющая перемешиванием

Источник 1: ymm0

Источник 2: ymm1

Маска 0000

Цель: ymm3

Источник 1: ymm0

Источник 2: ymm1

Маска 1111

Цель: ymm3

Источник 1: ymm0

Источник 2: ymm1

Маска 0110

Источник 1: ymm0

Источник 2: ymm1

Маска 0011

Рис. 37.2.	Примеры масок, управляющих перемешиванием

В	табл. 37.12–37.15	показано,	как	это	работает	в рассматриваемом	здесь	при
мере	программы.

vshufpd    ymm12,ymm0,ymm1, 0000b

Таблица 37.12

Регистр Значения

ymm0 4 3 2 1

ymm1 8 7 6 5

ymm12 Младший старший 
ymm1
7

Младший стар-
ший ymm0
3

Младший стар-
ший ymm1
5

Младший стар-
ший ymm0
1

vshufpd    ymm13,ymm0,ymm1, 1111b

Таблица 37.13

Регистр Значения

ymm0 4 3 2 1

ymm1 8 7 6 5

ymm13 Старший старший 
ymm1
8

Старший старший 
ymm0
4

Старший млад-
ший ymm1
6

Старший млад-
ший ymm0
2

vshufpd    ymm14,ymm2,ymm3, 0000b

Таблица 37.14

Регистр Значения

ymm2 12 11 10 9

ymm3 16 15 14 13

ymm14 Младший стар-
ший ymm3
15

Младший стар-
ший ymm2
11

Младший млад-
ший ymm3
14

Младший млад-
ший ymm2
9

Последний	пример:

vshufpd    ymm15,ymm2,ymm3, 1111b

Таблица 37.15

Регистр Значения

ymm2 12 11 10 9

ymm3 16 15 14 13

ymm15 Старший старший 
ymm3
16

Старший старший 
ymm2
12

Старший млад-
ший ymm3
14

Старший млад-
ший ymm2
10

После	 применения	маски	 управления	 смешиванием	мы	 получаем	 восемь	
пар	 значений	в  ymmрегистрах.	 Регистры	выбраны	так,	 чтобы	получать	те	же	
промежуточные	результаты,	что	и в	версии	с использованием	неупакованных	
данных.	После	этого	пары	значений	должны	быть	перемещены	в правильные	
позиции,	 чтобы	 сформировать	 транспонированную	 матрицу.	 Это	 делается	
точно	таким	же	способом,	что	и в	версии	с использованием	неупакованных	
данных:	перестановкой	полей	(блоков)	размером	128	бит	с помощью	инструк
ции	vperm2f128.

резюме
В	этой	главе	вы	узнали:

	� о	двух	способах	транспонирования	матриц;
	� об	использовании	инструкций	перемешивания,	распаковки	и переста
новки;

	� о	 существовании	 различных	 масок	 для	 перемешивания,	 распаковки	
и перестановки.

Глава 38
Оптимизация 

производительности

Вы,	 вероятно,	 согласитесь,	 что	 подавляющее	 большинство	 инструкций	 AVX	
нельзя	назвать	интуитивно	понятными,	особенно	разнообразные	схемы	ма
сок,	которые	делают	исходный	код	весьма	трудным	для	чтения	и понимания.	
Более	того,	битовые	маски	иногда	записываются	в шестнадцатеричном	фор
мате,	поэтому	приходится	сначала	преобразовывать	их	в двоичный	формат,	
чтобы	понять,	что	они	означают.

В	этой	главе	будет	продемонстрировано,	как	применение	инструкций	AVX 
может	 существенно	 улучшить	производительность,	так	 что	трудозатраты	на	
освоение	и  практическое	 использование	AVX	 окупаются	 во	многих	 случаях.	
Интересный	документ	по	тестированию	производительности	кода	можно	най
ти	 здесь:	 https://www.intel.com/content/dam/www/public/us/en/documents/white-
papers/ia-32-ia-64-benchmark-code-execution-paper.pdf.

В	примерах	этой	главы	будет	использоваться	методика	измерения	произво
дительности,	представленная	в вышеуказанном	документе.

прОизвОдительнОСть вычиСления транСпОнирОваннОй 
матрицы
В	примере	исходного	кода,	показанного	в листинге	38.1,	применяются	два	ме
тода	вычисления	транспонированной	матрицы:	один	с использованием	«клас
сических»	инструкций	ассемблера,	во	втором	применяются	инструкции	AVX.	
Также	добавлен	код	для	измерения	времени	выполнения	обоих	алгоритмов.

Листинг 38.1.	Программа transpose.asm

; transpose.asm
extern printf

section .data
      fmt0   db   "4x4 DOUBLE PRECISION FLOATING POINT MATRIX TRANSPOSE",10,0
      fmt1   db   10,"This is the matrix:",10,0
      fmt2   db   10,"This is the transpose (sequential version): ",10,0
      fmt3   db   10,"This is the transpose (AVX version): ",10,0
      fmt4   db   10,"Number of loops: %d",10,0



      fmt5   db   "Sequential version elapsed cycles: %d",10,0
      fmt6   db   "AVX Shuffle version elapsed cycles: %d",10,0
      align  32
      matrix       dq     1.,     2.,     3.,     4.
                   dq     5.,     6.,     7.,     8.
                   dq     9.,    10.,    11.,    12.
                   dq    13.,    14.,    15.,    16.
      loops  dq    10000

section .bss
      alignb       32
      transpose    resq        16
      bahi_cy      resq   1  ; Таймеры для версии avx.
      balo_cy      resq   1
      eahi_cy      resq   1
      ealo_cy      resq   1
      bshi_cy      resq   1  ; Таймеры для версии с последоват. вычислениями.
      bslo_cy      resq   1
      eshi_cy      resq   1
      eslo_cy      resq   1

section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Вывод заголовка.
      mov   rdi, fmt0
      call  printf
; Вывод матрицы.
      mov   rdi,fmt1
      call  printf
      mov   rsi,matrix
      call  printm4x4

; ВЕРСИЯ С ПОСЛЕДОВАТЕЛЬНЫМИ ВЫЧИСЛЕНИЯМИ.
; Вычисление транспонированной матрицы.
      mov   rdi, matrix
      mov   rsi, transpose
      mov   rdx, [loops]
; Начало отсчета циклов процессора.
      cpuid
      rdtsc
      mov   [bshi_cy],edx
      mov   [bslo_cy],eax
      call seq_transpose
; Окончание отсчета циклов процессора.
      rdtscp
      mov   [eshi_cy],edx
      mov   [eslo_cy],eax
      cpuid
; Вывод результата.
      mov   rdi,fmt2
      call  printf
      mov   rsi,transpose
      call  printm4x4

; ВЕРСИЯ AVX.
; Вычисление транспонированной матрицы.
      mov   rdi, matrix
      mov   rsi, transpose
      mov   rdx, [loops]
; Начало отсчета циклов процессора.
      cpuid
      rdtsc
      mov   [bahi_cy],edx
      mov   [balo_cy],eax
      call AVX_transpose
; Окончание отсчета циклов процессора.
      rdtscp
      mov   [eahi_cy],edx
      mov   [ealo_cy],eax
      cpuid

; Вывод результата.
      mov   rdi,fmt3
      call  printf
      mov   rsi,transpose
      call  printm4x4
; Вывод программных циклов.
      mov   rdi,fmt4
      mov   rsi,[loops]
      call  printf
; Вывод счетчика циклов процессора.
; Циклы версии с последовательными вычислениями.
      mov   rdx,[eslo_cy]
      mov   rsi,[eshi_cy]
      shl   rsi,32
      or    rsi,rdx         ; rsi содержит время окончания.
      mov   r8,[bslo_cy]
      mov   r9,[bshi_cy]
      shl   r9,32
      or    r9,r8           ; r9 содержит время начала.
      sub   rsi,r9          ; rsi содержит прошедшее время (интервал).
    ; Вывод результата измерения времени.
      mov   rdi,fmt5
      call  printf
; Счетчик циклов процессора для версии AVX blend.
      mov   rdx,[ealo_cy]
      mov   rsi,[eahi_cy]
      shl   rsi,32
      or    rsi,rdx          ; rsi содержит время окончания.
      mov   r8,[balo_cy]
      mov   r9,[bahi_cy]
      shl   r9,32
      or    r9,r8            ; r9 содержит время начала.
      sub   rsi,r9           ; rsi содержит прошедшее время (интервал).
    ; Вывод результата измерения времени.
      mov   rdi,fmt6
      call  printf
leave
ret

;-----------------------------------------------------
seq_transpose:
push  rbp
mov   rbp,rsp
.loopx:                     ; Количество циклов.
    pxor     xmm0,xmm0
    xor      r10,r10
    xor      rax,rax
    mov      r12,4
    .loopo:
             push  rcx
             mov   r13,4
             .loopi:
                    movsd  xmm0, [rdi+r10]
             movsd  [rsi+rax], xmm0
             add           r10,8
             add           rax,32
             dec           r13
             jnz    .loopi
             add    rax,8
             xor    rax,10000000b    ; rax - 128
             inc    rbx
             dec    r12
      jnz    .loopo
      dec rdx
jnz .loopx
leave
ret

;---------------------------------------------------------------
AVX_transpose:
push  rbp
mov   rbp,rsp
.loopx:                    ; Количество циклов.
; Загрузка матрицы в регистры.
      vmovapd       ymm0,[rdi]    ;  1   2   3   4
      vmovapd       ymm1,[rdi+32] ;  5   6   7   8
      vmovapd       ymm2,[rdi+64] ;  9  10  11  12
      vmovapd       ymm3,[rdi+96] ; 13  14  15  16
; Перемешивание.
      vshufpd       ymm12,ymm0,ymm1, 0000b     ;  1   5   3   7
      vshufpd       ymm13,ymm0,ymm1, 1111b     ;  2   6   4   8
      vshufpd       ymm14,ymm2,ymm3, 0000b     ;  9  13  11  15
      vshufpd       ymm15,ymm2,ymm3, 1111b     ; 10  14  12  16
; Перестановка.
      vperm2f128    ymm0,ymm12,ymm14,     00100000b    ; 1   5   9  13
      vperm2f128    ymm1,ymm13,ymm15,     00100000b    ; 2   6  10  14
      vperm2f128    ymm2,ymm12,ymm14,     00110001b    ; 3   7  11  15
      vperm2f128    ymm3,ymm13,ymm15,     00110001b    ; 4   8  12  16
; Запись в память.
      vmovapd       [rsi],   ymm0
      vmovapd       [rsi+32],ymm1
      vmovapd       [rsi+64],ymm2
      vmovapd       [rsi+96],ymm3
      dec rdx
      jnz .loopx

leave
ret

;---------------------------------------------------------------
printm4x4:
section .data
      .fmt   db     "%f",9,"%f",9, "%f",9,"%f",10,0
section .text
push  rbp
mov   rbp,rsp
      push   rbx               ; Регистр, сохраняемый вызываемой функцией.
      push   r15               ; Регистр, сохраняемый вызываемой функцией.
      mov           rdi,.fmt
      mov           rcx,4
      xor           rbx,rbx     ; Счетчик строк.
.loop:
      movsd  xmm0, [rsi+rbx]
      movsd  xmm1, [rsi+rbx+8]
      movsd  xmm2, [rsi+rbx+16]
      movsd  xmm3, [rsi+rbx+24]
      mov          rax,4        ; Четыре числа с плавающей точкой.
        push       rcx    ; Регистр, сохраняемый вызывающей функцией.
        push       rsi    ; Регистр, сохраняемый вызывающей функцией.
        push       rdi    ; Регистр, сохраняемый вызывающей функцией.
        ; Выравнивание стека, если необходимо.
        xor  r15,r15
        test rsp,0fh       ; Последний байт равен 8 (стек не выровнен)?
        setnz      r15b         ; Установить флаг, если стек не выровнен.
        shl  r15,3              ; Умножение на 8.
        sub  rsp,r15            ; Вычитание 0 или 8.
             call   printf
        add  rsp,r15            ; Прибавление 0 или 8.
        pop  rdi
        pop  rsi
        pop  rcx
        add  rbx,32             ; Следующая строка.
        loop .loop
pop r15
pop rbx
leave
ret

Перед	вызовом	функции	транспонирования	матрицы	начинается	процесс	
измерения	времени.	Современные	процессоры	поддерживают	внеочередное	
исполнение	 кода,	 результатом	 которого	может	 стать	 выполнение	 «рабочих»	
инструкций	в неподходящий	момент:	перед	началом	процесса	измерения	вре
мени	или	уже	после	завершения	этого	процесса.	Чтобы	избежать	этого,	необ
ходимо	использовать	«сериализацию»	инструкций,	которая	гарантирует,	что	
инструкции	контроля	времени	измеряют	тот	интервал	выполнения,	который	
нам	нужен.	Более	подробное	описание	 этой	методики	можно	найти	в доку
менте,	 ссылка	 на	 который	приведена	 выше.	Одной	из	 инструкций,	 которую	
можно	 применить	для	 сериализации,	 является	 cpuid.	 Перед	 пуском	таймера	
инструкцией	rdtsc	выполняется	инструкция	cpuid.	Инструкция	rdtsc	использу
ется	для	записи	метки	времени	как	счетчик	«младших	циклов»	в регистр	eax 

и «старших	циклов»		в регистр	edx,	потом	эти	значения	сохраняются	в памя
ти.	Инструкция	rdtsc	использует	эти	два	регистра	по	историческим	причинам:	
в 32битовых	процессорах	одного	регистра	не хватало	для	хранения	показаний	
таймера.	Один	32битовый	регистр	содержал	младшую	часть	значения	счет
чика	времени,	в другом	регистре	сохранялась	старшая	часть	этого	значения.	
После	записи	начальных	показаний	таймера	выполняется	«измеряемый»	код,	
а инструкция	rdtscp	используется	для	останова	процесса	измерений.	Итоговые	
значения	счетчиков	«старших	циклов»	и «младших	циклов»	еще	раз	сохраня
ются	в памяти,	и снова	выполняется	инструкция	cpuid	для	полной	уверенности	
в том,	что	процессор	не будет	выполнять	какиелибо	отложенные	инструкции.

В	рассматриваемом	здесь	примере	используется	рабочая	среда	64бито
вого	процессора,	поэтому	выполняется	сдвиг	влево	на	32	разряда	старших	
значений	метки	времени,	затем	выполняется	инструкция	xor	для	старшего	
и младшего	значений	метки	времени,	чтобы	получить	полные	метки	време
ни	в 64битовом	регистре.	Разность	между	начальными	и конечными	зна
чениями	счетчика	равна	количеству	использованных	циклов	процессора.

Функция	seq_transpose	использует	«классические»	инструкции,	а функция	
AVX_transpose –	это	функция	transpose_shuffle4x4	из	предыдущей	главы.	Эти	
функции	выполняются	многократно –	столько	раз,	сколько	задано	в пере
менной	loops.

На	рис. 38.1	показан	вывод	данной	программы.

Рис. 38.1.	Вывод программы transpose.asm

Здесь	можно	 видеть,	 что	 обработка	 с  использованием	 инструкций	AVX	
выполняется	заметно	быстрее.

В	 комплекте	руководств	 Intel	 есть	 отдельный	том,	посвященный	оптими
зации	 кода:	 https://software.intel.com/sites/default/files/managed/9e/bc/64-ia-32-
architectures-optimization-manual.pdf.

Этот	том	руководства	содержит	большой	объем	полезной	информации	о мето
дах	улучшения	производительности	ассемблерного	кода.	Ищите	фразу	«handling 
port 5 pressure»	(упомянутую	в главе	14).	В этом	разделе	вы	найдете	несколько	
версий	алгоритма	транспонирования	матриц	размером	8×8,	а также	оценку	воз
действия	различных	инструкций	на	производительность.	В предыдущей	главе	
демонстрировались	два	метода	транспонирования	матриц:	с использованием	
операций	распаковки	и перемешивания	данных.	В руководствах	Intel	эта	тема	
рассматривается	более	глубоко	и подробно,	и если	производительность	важна	
для	вас,	то	этот	том	станет	для	вас	«кладезем	бесценной	информации».

прОизвОдительнОСть вычиСления Следа матрицы
Пример	в листинге	38.2	показывает,	что	инструкции	AVX	не всегда	обеспечи
вают	более	высокую	скорость,	чем	«классические»	ассемблерные	инструкции.	
В этом	примере	вычисляется	след	матрицы	размером	8×8.

Листинг 38.2.	Программа trace.asm

; trace.asm
extern printf
section .data
      fmt0   db      "8x8 SINGLE PRECISION FLOATING POINT MATRIX TRACE",10,0
      fmt1   db      10,"This is the matrix:",10,0
      fmt2   db      10,"This is the trace (sequential version): %f",10,0
      fmt5   db      "This is the trace (AVX blend version): %f",10,0
      fmt6   db      10,"This is the tranpose: ",10,0
      fmt30  db      "Sequential version elapsed cycles: %u",10,0
      fmt31  db      "AVX blend  version elapsed cycles: %d",10,10,0
      fmt4   db      10,"Number of loops: %d",10,0

      align  32
      matrix dd 1.,     2.,     3.,     4.,     5.,     6.,     7.,     8.
             dd 9.,     10.,    11.,    12.,    13.,    14.,    15.,    16.
             dd 17.,    18.,    19.,    20.,    21.,    22.,    23.,    24.
             dd 25.,    26.,    27.,    28.,    29.,    30.,    31.,    32.
             dd 33.,    34.,    35.,    36.,    37.,    38.,    39.,    40.
             dd 41.,    42.,    43.,    44.,    45.,    46.,    47.,    48.
             dd 49.,    50.,    51.,    52.,    53.,    54.,    55.,    56.
             dd 57.,    58.,    59.,    60.,    61.,    62.,    63.,    64.
      loops  dq    1000
      permps dd    0,1,4,5,2,3,6,7  ; Маска для перестановки значений sp в ymm.

section .bss
      alignb       32
      transpose    resq   16
      trace        resq   1
      bbhi_cy      resq   1
      bblo_cy      resq   1

      ebhi_cy      resq   1
      eblo_cy      resq   1
      bshi_cy      resq   1
      bslo_cy      resq   1
      eshi_cy      resq   1
      eslo_cy      resq   1

section .text
      global main
main:
push  rbp
mov   rbp,rsp
; Вывод заголовка.
      mov    rdi, fmt0
      call   printf
; Вывод матрицы.
      mov    rdi,fmt1
      call   printf
      mov    rsi,matrix
      call   printm8x8

; ВЕРСИЯ С ПОСЛЕДОВАТЕЛЬНЫМИ ВЫЧИСЛЕНИЯМИ.
; Вычисление следа матрицы.
      mov    rdi, matrix
      mov    rsi, [loops]
; Начало отсчета циклов процессора.
      cpuid
      rdtsc
      mov    [bshi_cy],edx
      mov    [bslo_cy],eax
      call   seq_trace
; Останов отсчета циклов процессора.
      rdtscp
      mov    [eshi_cy],edx
      mov    [eslo_cy],eax
      cpuid
; Вывод результата.
      mov    rdi, fmt2
      mov    rax,1
      call   printf

; ВЕРСИЯ СО СМЕШИВАНИЕМ.
; Вычисление следа матрицы.
      mov    rdi, matrix
      mov    rsi, [loops]
; Начало отсчета циклов процессора.
      cpuid
      rdtsc
      mov    [bbhi_cy],edx
      mov    [bblo_cy],eax
      call   blend_trace
; Останов отсчета циклов процессора.
      rdtscp
      mov    [ebhi_cy],edx
      mov    [eblo_cy],eax
      cpuid

; Вывод результата.
      mov    rdi, fmt5
      mov    rax,1
      call   printf
; Вывод количества программных циклов.
      mov    rdi,fmt4
      mov    rsi,[loops]
      call   printf
; Вывод количества циклов процессора.
; Количество циклов в версии с последовательными вычислениями.
      mov    rdx,[eslo_cy]
      mov    rsi,[eshi_cy]
      shl    rsi,32
      or     rsi,rdx
      mov    r8,[bslo_cy]
      mov    r9,[bshi_cy]
      shl    r9,32
      or     r9,r8
      sub    rsi,r9        ; rsi содержит кол-во выполненных циклов.
    ; Вывод.
      mov    rdi,fmt30
      call printf
; Количество циклов в версии AVX со смешиванием.
      mov   rdx,[eblo_cy]
      mov   rsi,[ebhi_cy]
      shl   rsi,32
      or    rsi,rdx
      mov   r8,[bblo_cy]
      mov   r9,[bbhi_cy]
      shl   r9,32
      or    r9,r8
      sub   rsi,r9
    ; Вывод.
      mov   rdi,fmt31
      call  printf
leave
ret

;--------------------------------------------------------------
seq_trace:
push  rbp
mov   rbp,rsp
.loop0:
      pxor  xmm0,xmm0
      mov   rcx,8
      xor   rax,rax
      xor   rbx,rbx
      .loop:
      addss xmm0, [rdi+rax]
      add   rax,36   ; В каждой строке 32 байта.
      loop  .loop
      cvtss2sd     xmm0,xmm0
      dec          rsi
      jnz          .loop0
leave
ret

;--------------------------------------------------------------
blend_trace:
push   rbp
mov     rbp,rsp
.loop:
    ; Формирование матрицы в памяти.
      vmovaps       ymm0, [rdi]
      vmovaps       ymm1, [rdi+32]
      vmovaps       ymm2, [rdi+64]
      vmovaps       ymm3, [rdi+96]
      vmovaps       ymm4, [rdi+128]
      vmovaps       ymm5, [rdi+160]
      vmovaps       ymm6, [rdi+192]
      vmovaps       ymm7, [rdi+224]

      vblendps      ymm0,ymm0,ymm1,00000010b
      vblendps      ymm0,ymm0,ymm2,00000100b
      vblendps      ymm0,ymm0,ymm3,00001000b
      vblendps      ymm0,ymm0,ymm4,00010000b
      vblendps      ymm0,ymm0,ymm5,00100000b
      vblendps      ymm0,ymm0,ymm6,01000000b
      vblendps      ymm0,ymm0,ymm7,10000000b

      vhaddps       ymm0,ymm0,ymm0
      vmovdqu       ymm1,[permps]
      vpermps       ymm0,ymm1,ymm0
      haddps        xmm0,xmm0
      vextractps    r8d,xmm0,0
      vextractps    r9d,xmm0,1
      vmovd         xmm0,r8d
      vmovd         xmm1,r9d
      vaddss        xmm0,xmm0,xmm1
      dec           rsi
      jnz           .loop
cvtss2sd xmm0,xmm0
leave
ret

printm8x8:
section .data
.fmt db  "%.f,",9,"%.f,",9,"%.f,",9,"%.f,",9,"%.f,",9,"%.f,",9,"%.f,", 9,"%.f",10,0
section .text
push  rbp
mov   rbp,rsp
      push   rbx            ; Регистр, сохраняемый вызываемой функцией.
      mov    rdi,.fmt
      mov    rcx,8
      xor    rbx,rbx        ; Счетчик строк.
      vzeroall
.loop:
      movss         xmm0, dword[rsi+rbx]
        cvtss2sd    xmm0,xmm0
      movss         xmm1, [rsi+rbx+4]
        cvtss2sd    xmm1,xmm1
      movss         xmm2, [rsi+rbx+8]
        cvtss2sd    xmm2,xmm2

      movss         xmm3, [rsi+rbx+12]
        cvtss2sd    xmm3,xmm3
      movss         xmm4, [rsi+rbx+16]
        cvtss2sd    xmm4,xmm4
      movss         xmm5, [rsi+rbx+20]
        cvtss2sd    xmm5,xmm5
      movss         xmm6, [rsi+rbx+24]
        cvtss2sd    xmm6,xmm6
      movss         xmm7, [rsi+rbx+28]
        cvtss2sd    xmm7,xmm7
      mov     rax,8      ; 8 чисел с плавающей точкой.
      push    rcx        ; Регистр, сохраняемый вызывающей функцией.
      push    rsi        ; Регистр, сохраняемый вызывающей функцией.
      push    rdi        ; Регистр, сохраняемый вызывающей функцией.
        ; Выравнивание стека, если необходимо.
      xor     r15,r15
      test    rsp,0fh        ; Последний байт равен 8 (стек не выровнен)?
      setnz   r15b           ; Установка флага, если стек не выровнен.
      shl     r15,3          ; Умножение на 8.
      sub     rsp,r15        ; Вычитание 0 или 8.
      call    printf
      add     rsp,r15        ; Прибавление 0 или 8.
      pop     rdi
      pop     rsi
      pop     rcx
      add     rbx,32         ; Следующая строка.
      loop    .loop
pop rbx                      ; Регистр, сохраняемый вызываемой функцией.
leave
ret

Функция	blend_trace	представляет	собой	расширение	функции	вычисления	
следа	матрицы	с размера	4×4	до	размера	8×8.	Исходная	версия	функции	ис
пользовалась	в главе	36	в коде	обращения	матрицы	с применением	инструк
ций	AVX.	Функция	seq_trace	реализует	последовательный	проход	по	матрице,	
определяет	элементы	следа	и выполняет	их	сложение.	При	запуске	этой	про
граммы	 вы	 увидите,	 что	 функция	 seq_trace	 работает	 намного	 быстрее,	 чем	
функция	blend_trace.

На	рис. 38.2	показан	вывод	данной	программы.

Рис. 38.2.	Вывод программы trace.asm

Если	вы	хотите	более	подробно	узнать	об	оптимизации,	обратитесь	к упо
мянутому	выше	тому	руководства	Intel.	Еще	один	превосходный	источник	ин
формации	находится	здесь:	https://www.agner.org/.

резюме
В	этой	главе	вы	узнали	о:

	� методах	измерения	и вычисления	выполненных	циклов	процессора;
	� том,	что	инструкции	AVX	могут	существенно	ускорить	обработку	данных;
	� том,	что	инструкции	AVX	не всегда	эффективны	(их	эффективность	за
висит	от	конкретной	задачи).

Глава 39
Приветствуем мир Windows

В	 этой	 главе	 и  в	 нескольких	 следующих	 главах	 мы	 начинаем	 изучать	 про
граммирование	на	ассемблере	в ОС	Windows.	Если	вашей	основной	системой	
является	Linux,	то	наилучшим	решением	будет	установка	виртуальной	маши
ны	Windows.	Пробную	версию	Windows	10	с лицензией	на	90	дней	можно	за
грузить	 здесь:	 https://www.microsoft.com/en-us/evalcenter/evaluate-windows-10-
enterprise.	После	установки	пробной	версии	Windows	10	загрузите	и установите	
все	сущест	вующие	на	текущий	момент	обновления	этой	системы.

начинаем изучение
Компания	 Microsoft	 разработала	 собственный	 ассемблер	 под	 названием	
MASM,	который	включен	в программную	среду	Visual	Studio.	Возможность	ис
пользования	Visual	Studio,	несомненно,	является	преимуществом,	так	как	это	
полнофункциональное	 инструментальное	 средство	 разработки.	 Инструкции	
ассемблера,	используемые	в MASM,	те	же	самые,	что	и в	NASM,	но	директивы	
ассемблера	совершенно	другие.	Конфигурирование	и обучение	практической	
работе	с Visual	Studio	характеризуются	кривой	обучения,	зависящей	от	вашего	
опыта	как	разработчика	в среде	Windows.

Для	смягчения	воздействия	«культурного	шока»	при	встрече	с другой	сре
дой	в этой	книге	используется	версия	ассемблера	NASM	для	Windows	с ин
терфейсом	командной	строки	(CLI).	В предыдущих	главах	вы	вполне освоили	
NASM	в  среде	 Linux,	 и  это	 является	неплохой	форой	для	начала	изучения.	
Кроме	того,	переход	на	MASM	не должен	вызвать	какихлибо	серьезных	за
труднений.

Если	вы	планируете	вести	разработки	для	Windows,	то	освоение	практи
ческого	использования	Visual	Studio	стоит	затраченных	усилий.	В интернете	
можно	даже	найти	информацию	о том,	как	использовать	NASM	в Visual	Studio.

Найдите	в интернете	NASM	для	Windows	(в настоящее	время	версия	NASM	для	
Windows	находится	здесь: https://www.nasm.us/pub/nasm/releasebuilds/2.14.03rc2/
win64/)	и установите	эту	версию.	Проверьте	и убедитесь	в том,	что	переменная	
среды	PATH	содержит	запись,	указывающую	на	каталог	(папку),	в которую	уста
новлен	NASM	(см.	рис. 39.1).	Проверить	правильность	установки	NASM	можно	
в командной	строке,	выполнив	команду	nasm -v.



Рис. 39.1.	Проверка переменной среды PATH в Windows 10

Мы	также	будем	использовать	версию	MinGW	(Minimalist	GNU	for	Windows),	
представляющую	 собой	 комплект	 инструментальных	 средств	 разработки	
Linux,	 адаптированный	для	 среды	Windows.	MinGW	позволяет	 использовать	
такие	инструменты	разработки,	как	make	и gcc,	которыми	мы	часто	пользова
лись	в предыдущих	 главах.	Необходимо	установить	версию	MinGWw64.	Пе
ред	 началом	 загрузки	 и  установки,	 если	 вы	 планируете	 использовать	 SASM 
в Windows,	следует	знать,	что	SASM	устанавливает	NASM	и некоторые	инстру
менты	MinGWw64	в собственные	подкаталоги	(за	исключением	make).	Если	вы	
вручную	устанавливаете	SASM	и MinGWw64,	то	в итоге	получите	дублирован
ные	 установленные	 компоненты.	С  помощью	параметров	настройки	можно	
сконфигурировать	SASM	так,	чтобы	он	использовал	уже	установленные	вер
сии	NASM	и GCC	вместо	более	старых	версий,	включенных	в дистрибутивный	
комплект	SASM.

В	настоящее	время	загружаемые	файлы	для	MinGWw64	расположены	здесь:	
http://mingw-w64.org/doku.php/download.	Выберите	версию	MinGW-w64-builds,	 за
грузите	и установите	ее,	в окне установки	выберите	вариант	x86_64.

Далее	 необходимо	 открыть	 окно	 переменных	 среды	Windows	 и  добавить	
путь	к каталогу	bin	в иерархии	MinGWw64	в переменную	среды	PATH,	как	пока
зано	на	рис. 39.1.	Каталог	bin	содержит	компилятор	gcc.	После	обновления	пе
ременной	среды	PATH	откройте	окно	командной	строки	PowerShell	и проверьте	
правильность	установки	командой	gcc -v.

Загрузите	 версию	 SASM	 для	 win64	 (https://dman95.github.io/SASM/english.
html),	и если	необходимо,	чтобы	SASM	использовал	новые	версии	NASM	и GCC,	
то	 измените	 параметры	 сборки	 (build settings)	 так,	 чтобы	 они	 указывали	 на	
только	что	установленные	NASM	и GCC.	Не забудьте	добавить	в переменную	
среды	PATH	запись	для	SASM.

Если	вы	пока	еще	не выбрали	текстовый	редактор	для	работы	в Windows,	ре
комендуется	установить	Notepad++.	Это	простой	текстовый	редактор,	поддер
живающий	 подсветку	 синтаксиса	 многих	 языков	 программирования,	 вклю
чая	 ассемблер.	В нем	можно	без	 затруднений	 установить	любую	кодировку:	
UTF8,	UTF16	и т.	д.	Настройку	подсветки	синтаксиса	ассемблера	можно	найти	
в пунк	те	основного	меню	Language	(Язык).

Некоторое	неудобство	 состоит	в том,	 что	 в MinGWw64	нет	 команды	make,	
а существует	только	выполняемый	файл	mingw32-make.exe,	имя	которого	слиш
ком	длинное	для	ввода	в командной	строке.	Для	решения	этой	проблемы	в ре
дакторе	Notepad++	(требуется	запуск	от	имени	администратора)	создайте	файл	
make.bat,	содержащий	следующую	строку:

mingw32-make.exe

Сохраните	этот	файл	в кодировке	UTF8	в каталоге	bin	в иерархии	MinGWw64.
Ниже	приведены	некоторые	советы	и рекомендации,	полезные	при	работе	

в Windows:
	� чтобы	открыть	приложение	от	имени	администратора,	щелкните	правой	
кнопкой	по	значку	приложения	и в	контекстном	меню	выберите	пункт	
Run	as	administrator	(Запуск	от	имени	администратора);

	� всегда	полезно	иметь	быстрый	доступ	к PowerShell,	интерфейсу	команд
ной	строки	в Windows.	На панели	задач	в поле	поиска	введите	PowerShell,	
затем	щелкните	по	кнопке	Open	 (Открыть).	На панели	задач	появится	
значок	PowerShell,	щелкните	по	нему	правой	кнопкой	мыши	и выберите	
пункт	Pin	to	taskbar	(Закрепить	на	панели	задач);

	� в	окне,	где	отображаются	значки	файлов	и каталогов,	нажмите	и удержи
вайте	клавишу	Shift	и одновременно	щелкните	правой	кнопкой	мыши,	
далее	 в  появившемся	 контекстном	 меню	 можно	 выбрать	 пункт	Open	
PowerShell	window	here	(Открыть	окно	PowerShell	здесь);

	� для	отображения	скрытых	файлов	и каталогов	щелкните	по	значку	Про-
водник	(File	Explorer)	на	панели	задач.	Откройте	пункт	меню	Вид	(View)	
и поставьте	метку	(галочку)	для	параметра	Скрытые	элементы	(Hidden	
items);

	� чтобы	найти	переменные	среды,	в поле	поиска	на	панели	задач	введите	
строку	«переменные	среды»	(environment	variables).

пишем кОд в wInDowS
Теперь	вы	готовы	к началу	кодирования	на	ассемблере.	В листингах	39.1	и 39.2	
показана	первая	программа.

Листинг 39.1.	Программа hello.asm

; hello.asm
extern printf
section .data
      msg   db 'Hello, Windows World!',0
      fmt   db "Windows 10 says: %s",10,0
section .text
      global main
main:
push rbp
mov rbp,rsp
      mov   rcx, fmt
      mov   rdx, msg
      sub   rsp,32
      call  printf
      add   rsp,32
leave
ret

Листинг 39.2.	makefile

hello.exe: hello.obj
      gcc -o hello.exe hello.obj
hello.obj: hello.asm
      nasm -f win64 -g -F cv8 hello.asm -l hello.lst

В	этом	ассемблерном	коде	нет	ничего	примечательного.	Или	всетаки	есть?
Вопервых,	здесь	используется	инструкция	sub rsp,32,	которая	в Linux	приме

нялась	для	создания	стековых	переменных.	С помощью	этой	инструкции	соз
дается	скрытое	пространство	(shadow space)	в стеке	перед	вызовом	функции.	
Более	подробно	мы	обсудим	это	немного	позже.	После	выполнения	функции	
printf	 стек	 восстанавливается	 инструкцией	 add rsp,32,	 которая	 в  рассматри
ваемом	здесь	примере	не является	обязательной,	так	как	 стек	будет	восста
новлен	инструкцией	leave.	Регистры,	используемые	для	передачи	аргументов	
в функцию	printf,	 отличаются	от	регистров,	используемых	в Linux.	Причина	
в том,	что	соглашения	о вызовах	функций	для	Windows	отличаются	от	соглаше
ний	о вызовах	функций	для	Linux.	В Windows	требуется	соблюдение	соглаше
ний	Microsoft	x64,	тогда	как	Linux	требует	использования	соглашений	System	
V Application	Binary	Interface,	также	называемой	System	V ABI.

Общий	обзор	соглашений	о вызовах	функций	Microsoft	x64	можно	найти	здесь:	
https://docs.microsoft.com/en-us/cpp/build/x64-calling-convention?view=vs-2019.	
Мес	то	расположения	этой	страницы	иногда	изменяется,	так	что	если	вы	не мо
жете	ее	найти,	то	поищите	на	сайте	Microsoft	«x64	calling	convention».	Ниже	при
водится	краткое	содержание	данного	обзора:

	� целочисленные	аргументы	передаются	в регистрах	rcx,	rdx,	r8,	r9	в ука
занном	здесь	порядке;

	� если	необходимо	передать	большее	количество	аргументов,	то	они	по
мещаются	в стек;

	� аргументы	с плавающей	точкой	передаются	в регистрах	xmm0-xmm3.	Боль
шее	количество	аргументов	передается	через	стек;

	� регистры	rcx,	rdx,	r8,	r9	и дополнительно	rax,	r10,	r11,	xmm4	и xmm5	являются	
изменяемыми,	т. е.	вызывающая	сторона	должна	сохранять	их,	если	это	
необходимо.	Прочие	регистры	должны	сохраняться	вызываемой	сторо
ной	(функцией);

	� вызывающая	 сторона	 должна	 обеспечить	 32байтовое	 пространство	
в стеке	(скрытое	пространство –	shadow space)	для	четырех	аргументов	
функции,	 передаваемых	 вызываемой	 стороне,	 даже	 если	 вызываемая	
сторона	не принимает	так	много	аргументов;

	� как	и в	Linux,	стек	обязательно	должен	быть	выровнен	по	16байтовой	
границе.

На	рис. 39.2	показан	вывод	первой	программы	в Windows.

Рис. 39.2.	Вывод программы hello.asm

Отладка
После	запуска	GDB для	отладки	первой	Windowsпрограммы	вы	будете	удивле
ны.	Вы сможете	выполнить	некоторые	команды,	но	пошаговый	проход	по	коду	
не работает.	Выводится	следующее	сообщение:

Single stepping until exit from function main,
which has no line number information.
0x0000000000402a60 in printf ()

Это	означает,	что	в Windows	использование	GDB	ограничено.	Но на	помощь	
приходит	SASM.	В SASM	такой	проблемы	не существует.	Тем	не менее	в makefile 
сохранены	флаги	отладки:	возможно,	в следующей	версии	GDB	проблема	по
шагового	прохода	будет	решена.	В makefile	для	Windows	задан	флаг	cv8	(Microsoft	
CodeView	8),	определяющий	формат	отладки.	

СиСтемные вызОвы
В	 рассматриваемом	 здесь	 примере	 исходного	 кода	 использовалась	функция	
printf	вместо	инструкции	системного	вызова	syscall,	примененной	в первой	
ассемблерной	программе	в Linux.	Причина	в том,	что	в Windows	не использу
ются	системные	вызовы.	Системные	вызовы	существуют,	но	предназначены	
они	«только	для	внутреннего	пользования»	Windows.	Если	необходим	доступ	
к  системным	 ресурсам,	 то	 необходимо	 воспользоваться	 прикладным	 про
граммным	интерфейсом	Windows	API.	Разумеется,	можно	покопаться	в коде	
Windows	или	поискать	 в интернете	имена	 системных	вызовов,	но	при	 этом	

следует	помнить	о том,	что	в новых	версиях	Windows	использование	систем
ных	вызовов	может	измениться,	и если	вы	решились	применить	их,	то	ваш	код,	
возможно,	перестанет	работать.

резюме
В	этой	главе	вы	узнали:

	� как	установить	и использовать	NASM,	SASM	и комплект	инструменталь
ных	средств	разработки	Linux	в операционной	системе	Windows;

	� о	том,	что	соглашения	о вызовах	функций	Windows	отличаются	от	согла
шений	о вызовах	функций	Linux;

	� о	том,	что	в Windows	лучше	не использовать	системные	вызовы.

Глава 40
Использование Windows API

Прикладной	 программный	 интерфейс	 Windows  –	 Windows	 API  –	 это	 набор	
инструкций,	 которые	 может	 использовать	 разработчик	 для	 взаимодействия	
с операционной	системой.	Как	было	отмечено	в предыдущей	главе,	системные	
вызовы	не являются	надежным	способом	обмена	информации	с операцион
ной	 системой,	 но	 компания	Microsoft	 предоставляет	 обширный	 набор	 при
кладных	программных	интерфейсов	API	для	выполнения	практически	любых	
операций,	которые	могут	потребоваться	разработчику.	Windows	API	написан	
с учетом	свойств	языка	программирования	C,	но	если	соблюдать	соглашения	
о  вызовах	 функций,	 то	 можно	 без	 затруднений	 использовать	 Windows	 API	
в ассемблерных	программах.	В настоящее	время	(на	момент	написания	кни
ги)	описание	Windows	API	можно	найти	здесь:	https://docs.microsoft.com/en-us/
windows/win32/api/.

вывОд в кОнСОли
В	листинге	40.1	показана	версия	программы	Hello	World,	в которой	использу
ется	Windows	API	для	вывода	сообщения	на	экран.

Листинг 40.1.	Программа helloc.asm

; helloc.asm
%include "win32n.inc"
      extern WriteFile
      extern WriteConsoleA
      extern GetStdHandle

section .data
      msg            db       'Hello, World!!',10,0
      msglen  EQU    $-msg-1            ; Отбросить NULL.

section .bss
      hFile                   resq      1      ; Идентификатор файла.
      lpNumberOfBytesWritten  resq      1

section .text
      global main
main:
push  rbp
mov   rbp,rsp



; Связать идентификатор файла со стандартным выводом stdout.
; HANDLE WINAPI GetStdHandle(
;   _In_ DWORD nStdHandle
; );
      mov   rcx, STD_OUTPUT_HANDLE
      sub   rsp,32           ; Обеспечение скрытого пространства.
      call  GetStdHandle     ; Возврат INVALID_HANDLE_VALUE, если попытка неудачна.
      add   rsp,32
      mov   qword[hFile],rax ; Сохранить полученный идентификатор в памяти.

; BOOL WINAPI WriteConsole(
;   _In_                    HANDLE      hConsoleOutput,
;   _In_      const VOID    *lpBuffer,
;   _In_            DWORD   nNumberOfCharsToWrite,
;   _Out_           LPDWORD        lpNumberOfCharsWritten,
;   _Reserved_      LPVOID         lpReserved
; );
      sub   rsp, 8                ; Выравнивание стека.
      mov   rcx, qword[hFile]
      lea   rdx, [msg]            ; lpBuffer
      mov   r8, msglen            ; nNumberOfBytesToWrite
      lea   r9, [lpNumberOfBytesWritten]
      push  NULL                  ; lpReserved
      sub   rsp, 32
      call  WriteConsoleA   ; Возврат ненулевого значения, если успешное выполнение.
      add   rsp,32+8

; BOOL WriteFile(
;            HANDLE       hFile,
;            LPCVOID      lpBuffer,
;            DWORD        nNumberOfBytesToWrite,
;            LPDWORD      lpNumberOfBytesWritten,
;            LPOVERLAPPED lpOverlapped
; );
      mov   rcx, qword[hFile]    ; Идентификатор файла.
      lea   rdx, [msg]           ; lpBuffer
      mov   r8, msglen           ; nNumberOfBytesToWrite
      lea   r9, [lpNumberOfBytesWritten]
      push  NULL                 ; lpOverlapped
      sub   rsp,32
      call  WriteFile  ; Возврат ненулевого значения, если успешное выполнение.
leave
ret

В	документации	Windows	API	описано	использование	нескольких	тысяч	сим
вольных	констант.	Это	делает	исходный	код	более	удобным	для	чтения	и упро
щает	применение	Windows	API,	поэтому	в исходный	код	включен	файл	win32n.
inc	в начале	программы.	Это	список	всех	символьных	констант	и их	значений.	
Файл	win32n.inc	можно	скачать	здесь:	http://rs1.szif.hu/~tomcat/win32/3.	Но сле
дует	помнить	о том,	что	включение	данного	файла	в исходный	код	значитель
но	увеличивает	размер	выполняемого	файла.	Если	размер	важен,	то	 следует	

3	 Внимание:	при	попытке	перехода	по	этой	ссылке	выводится	сообщение,	предупреж
дающее	о том,	что	этот	сайт,	возможно,	содержит	вредоносные	программы	и атакует	
сетевые	компьютеры. –	Прим. перев.

выборочно	включить	в исходный	код	только	те	константы,	которые	действи
тельно	необходимы	в программе.	Если	используется	 SASM,	найдите	каталог	
(папку),	 в  котором	установлен	SASM,	и вручную	скопируйте	файл	win32n.inc 
в подкаталог	include	в иерархии	SASM	в вашей	системе.

В	 исходный	 код	 примера	 включены	 копии	 структур	 вызовов	 функций	
Windows	в виде	комментариев,	поэтому	можно	с легкостью	наблюдать	за	тем,	
что	происходит.	Аргументы	записываются	в регистры	в соответствии	с согла
шением	о вызовах	функций,	обеспечивается	наличие	скрытого	пространства	
(shadow space)	 в  стеке,	вызывается	функция,	 затем	восстанавливается	указа
тель	стека.

Функция	GetStdHandle	возвращает	идентификатор	(handle),	если	все	идет	пра
вильно,	иначе	возвращается	специальное	значение	INVALID_HANDLE_VALUE.	Чтобы	
не усложнять	программу,	мы	не обеспечиваем	контроль	ошибок,	но	в реаль
ных	программных	продуктах	рекомендуется	реализовать	полноценную	систе
му	контроля	ошибок.	Критический	сбой	может	не только	привести	к краху	про
граммы,	но,	что	значительно	хуже,	может	стать	причиной	нарушения	системы	
безопасности.

После	получения	идентификатора	вызывается	функция	WriteConsoleA,	в нее	
передается	идентификатор,	выводимая	строка,	длина	этой	строки,	шаблон	для	
резервирования	заданного	количества	записываемых	байтов	и NULL	как	заре
зервированный	аргумент.	Первые	четыре	аргумента	передаются	в регистрах,	
пятый	 аргумент	 помещается	 в  стек.	 Эта	 операция	 записи	 в  стек	 приводит	
к тому,	что	стек	становится	невыровненным, –	об	этом	нужно	помнить	перед	
записью	аргумента	в стек.	Если	выравнивание	выполнено	после	записи	в стек	
(после	инструкции	push),	то	вызываемая	функция	не найдет	аргумент	в стеке.	
Поэтому	непосредственно	перед	вызовом	функции	в стеке	создается	скрытое	
пространство.

Рассматриваемая	здесь	программа	использует	два	метода	записи	в консоль:	
первый	метод	применяет	функцию	WriteConsoleA,	второй –	функцию	WriteFile.	
Функция	WriteFile	использует	тот	же	идентификатор	и рассматривает	консоль	
как	обычный	файл	для	записи	в него.	После	выполнения	WriteConsoleA	стек	вос
станавливается	с учетом	предварительного	создания	скрытого	пространства	
выравнивания	 стека.	После	 выполнения	 WriteFile	 стек	не  восстанавливается	
явно,	потому	что	это	будет	сделано	инструкцией	leave.

Если	вы	не нашли	описания	функции	WriteConsoleA	в документации	Windows	
API,	 то	 попробуйте	 поискать	 WriteConsole.	 В  документации	 объясняется,	 что	
существуют	две	версии:	WriteConsoleA	для	вывода	символов	в кодировке	ANSI	
и WriteConsoleW	для	вывода	символов	в кодировке	Unicode.

Если	начать	выполнение	этого	кода	в SASM,	то	вы	увидите,	что	первый	ме
тод	с использованием	функции	WriteConsoleA	не работает.	Функция	возвращает	
0	в регистре	rax,	сообщая	о какихто	неполадках.	Причина –	в возникновении	
конфликта	между	системной	консолью	и собственной	консолью	SASM.	Метод	
с использованием	функции	WriteFile	работает	нормально.

На	рис. 40.1	показан	вывод	этой	программы.

306	  Использование Windows API



Рис. 40.1.	Вывод программы helloc.asm

СОздание ОкОн wInDowS
Теперь	 вместо	 вывода	 в  консоль	 воспользуемся	 графическим	 пользователь
ским	 интерфейсом	 (GUI)	 Windows.	 Здесь	 не  будет	 рассматриваться	 полно
функциональная	 программа	 Windows,	 необходимо	 всего	 лишь	 продемон
стрировать,	как	выводится	окно.	Если	вы	хотите	большего,	то	придется	более	
подробно	изучать	документацию	Windows	API.	Когда	вы	поймете,	как	работать	
с окном,	останется	только	найти	требуемую	функцию	в документации	Windows	
API	и передать	в нее	необходимые	аргументы	в регистрах	и в	стеке.

В	листинге	40.2	показан	пример	исходного	кода.

Листинг 40.2.	Программа hellow.asm

; hellow.asm
%include "win32n.inc"
extern ExitProcess
extern MessageBoxA

section .data
      msg    db 'Welcome to Windows World!',0
      cap    db "Windows 10 says:",0

section .text
      global main
main:
push    rbp
mov     rbp,rsp

; int MessageBoxA(
;        HWND hWnd,             Владелец окна.
;        LPCSTR lpText,         Выводимый текст.
;        LPCSTR lpCaption,      Заголовок окна.
;        UINT    uType          Поведение окна.
;       )
      mov     rcx,0              ; У окна нет владельца.
      lea     rdx,[msg]          ; lpText
      lea     r8,[cap]           ; lpCaption
      mov     r9d,MB_OK          ; Окно с кнопкой OK.
      sub     rsp,32             ; Скрытое пространство в стеке.
      call    MessageBoxA        ; Возврат IDOK=1, если выбрана кнопка OK.
      add     rsp,32
leave
ret

На	рис. 40.2	показан	вывод	этой	программы.

Рис. 40.2.	Вывод программы hellow.asm

Разумеется,	у вас	может	возникнуть	вопрос:	является	ли	ассемблер	подходя
щим	языком	программирования	для	создания	GUI	для	программы	Windows?	
Для	этой	цели	гораздо	проще	использовать	C или	C++	и вызывать	ассемблер
ные	функции	(или	применять	ассемблерные	вставки	непосредственно	в код)	
в тех	частях	программы,	в  которых	требуется	особенно	высокая	производи
тельность.

В	любом	случае	вы	можете	взять	качественную	книгу	по	программирова
нию	в Windows	на	языке	C или	C++,	где	подробно	описывается	Windows	API,	
и переписать	все	вызовы	функций	на	ассемблере	с использованием	правиль
ных	регистров,	а затем	вызывать	эти	функции,	как	показано	в данном	разделе.	
Разумеется,	в программе	с GUI	необходима	более	сложная	функциональность,	
например	 подсистема	 контроля	 ошибок,	 но	 подобные	 подсистемы	 гораздо	
проще	разрабатывать	на	языке	высокого	уровня.

резюме
В	этой	главе	вы	узнали,	как:

	� использовать	Windows	API;
	� вывести	сообщение	в окне командной	строки	(PowerShell)	Windows;
	� использовать	инструкции	GetStdHandle,	WriteConsole	и WriteFile;
	� создать	окно	с кнопкой.

Глава 41
Функции в Windows

Передача	аргументов	в функции	выполняется	просто,	если	необходимо	пере
дать	в них	четыре	или	меньше	аргументов,	не являющихся	значениями	с пла
вающей	точкой.	Для	этого	используются	регистры	rcx,	rdx,	r8,	r9	и обеспечи
вается	наличие	скрытого	пространства	(shadow space)	в стеке	перед	вызовом	
функции.	После	вызова	необходимо	скорректировать	стек	с учетом	скрытого	
пространства,	и все	будет	в порядке.	Вызов	функции	становится	более	слож
ным,	если	необходимо	передать	более	четырех	аргументов.

иСпОльзОвание бОлее четырех аргументОв функции
Сначала	рассмотрим,	почему	вызов	функции	становится	более	сложным	при	
необходимости	передать	в нее	более	четырех	аргументов,	как	показано	в лис
тинге	41.1.

Листинг 41.1.	Программа arguments1.asm

; arguments1.asm
extern printf
section .data
      first      db    "A",0
      second     db    "B",0
      third      db    "C",0
      fourth     db    "D",0
      fifth      db    "E",0
      sixth      db    "F",0
      seventh    db    "G",0
      eighth     db    "H",0
      ninth      db    "I",0
      tenth      db    "J",0
      fmt   db   "The string is: %s%s%s%s%s%s%s%s%s%s",10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      sub   rsp,8
      mov   rcx, fmt
      mov   rdx, first
      mov   r8, second



      mov   r9, third
      push  tenth         ; Здесь начинается запись аргументов в стек
      push  ninth         ; в обратном порядке.
      push  eighth
      push  seventh
      push  sixth
      push  fifth
      push  fourth
      sub   rsp,32        ; Скрытое пространство в стеке.
      call  printf
      add   rsp,32+8      ; Восстановление стека.
leave
ret

Обратите	 внимание	на	инструкцию	 sub rsp,8 –	 она	нужна	потому,	 что	 вы
зывается	функция	printf,	поэтому	стек	должен	быть	выровнен	по	16байтовой	
границе.	Почему	 бы	не  использовать	 одну	 инструкцию,	 например,	 sub rsp,40 
непосредственно	перед	вызовом	функции	printf?	В  этом	случае	 стек	был	бы	
выровнен	по	16байтовой	границе,	но	функция	printf,	вероятнее	всего,	будет	
выполнена	некорректно.	Если	уменьшить	указатель	стека	на	40	вместо	32	не
посредственно	перед	вызовом	функции,	то	аргументы	будут	находиться	в стеке	
не в тех	позициях,	где	их	предполагает	найти	функция	printf,	т. е.	прямо	над	
скрытым	пространством.	Поэтому	стек	необходимо	выровнять	перед	началом	
записи	 аргументов.	 Следует	 отметить,	 что	 аргументы	 должны	 записываться	
в стек	в обратном	порядке.	После	вызова	функции	стек	восстанавливается	с уче
том	его	предварительного	выравнивания	и создания	скрытого	пространства.

На	рис. 41.1	показан	вывод	этой	программы.

Рис. 41.1.	Вывод программы arguments1.asm

Правильно	сформировать	стек	можно	другим	способом.	В листинге 41.2	по
казано,	как	это	делается.

Листинг 41.2.	Программа arguments2.asm

;arguments2.asm
extern printf
section .data
      first         db    "A",0
      second        db    "B",0
      third         db    "C",0
      fourth        db    "D",0
      fifth         db    "E",0
      sixth         db    "F",0

      seventh       db    "G",0
      eighth        db    "H",0
      ninth         db    "I",0
      tenth         db    "J",0
      fmt     db    "The string is: %s%s%s%s%s%s%s%s%s%s",10,0
section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      sub   rsp,32+56+8 ; Скрытое пространство + 7 аргументов в стеке + выравнивание.
      mov   rcx, fmt
      mov   rdx, first
      mov   r8, second
      mov   r9, third
      mov   qword[rsp+32],fourth
      mov   qword[rsp+40],fifth
      mov   qword[rsp+48],sixth
      mov   qword[rsp+56],seventh
      mov   qword[rsp+64],eighth
      mov   qword[rsp+72],ninth
      mov   qword[rsp+80],tenth
      call  printf
      add   rsp, 32+56+8   ; Это не обязательно перед инструкцией leave.
leave
ret

Сначала	используется	инструкция	sub rsp,32+56+8	для	выравнивания	стека:
	� 32	байта	для	скрытого	пространства;
	� в	стек	записывается	7	аргументов	по	8	байт,	всего	56	байт.

Затем	начинается	формирование	стека,	и когда	вы	видите,	что	необходимо	
выровнять	стек,	еще	8	байт	требуется	вычесть	из	указателя	стека.

Теперь	в нижней	части	стека	имеется	32	байта	для	скрытого	пространства,	
а непосредственно	над	ним	расположен	четвертый	аргумент,	выше	него	на
ходится	пятый	аргумент	и т.	д.	Стек,	сформированный	здесь,	выглядит	так	же,	
как	стек	в предыдущей	программе	(листинг	41.1).	Решайте	сами,	какую	версию	
предпочесть.

На	рис. 41.2	показан	вывод	данной	программы.

Рис. 41.2.	Вывод программы arguments2.asm

Как	это	работает	в вызываемой	функции?	В листинге	41.3	показан	пример	
исходного	кода,	в котором	используется	функция	lfunc	для	создания	буфера	
строки,	выводимой	функцией	printf.

Листинг 41.3.	Программа stack.asm

; stack.asm
extern printf
section .data
      first        db    "A"
      second       db    "B"
      third        db    "C"
      fourth       db    "D"
      fifth        db    "E"
      sixth        db    "F"
      seventh      db    "G"
      eighth       db    "H"
      ninth        db    "I"
      tenth        db    "J"
       fmt         db    "The string is: %s",10,0
section .bss
      flist  resb  14       ; Длина строки плюс завершающий 0.
section .text
      global main
main:
push rbp
mov rbp,rsp
      sub rsp, 8
      mov rcx, flist
      mov rdx, first
      mov r8, second
      mov r9, third
      push tenth            ; Здесь начинается запись аргументов в стек
      push ninth            ; в обратном порядке.
      push eighth
      push seventh
      push sixth
      push fifth
      push fourth
      sub rsp,32            ; Скрытое пространство.
      call lfunc
      add rsp,32+8
; Вывод результата.
      mov rcx, fmt
      mov rdx, flist
      sub rsp,32+8
      call printf
      add rsp,32+8
leave
ret

;-------------------------------------------------------------------------
lfunc:
push  rbp
mov   rbp,rsp
      xor rax,rax             ; Очистка rax (особенно старших битов).

      ; Обработка аргументов в регистрах.
      mov al,byte[rdx]           ; Запись содержимого аргумента в al.
      mov [rcx], al              ; Запись al в память.
      mov al, byte[r8]
      mov [rcx+1], al
      mov al, byte[r9]
      mov [rcx+2], al
      ; Обработка аргументов в стеке.
      xor rbx,rbx
      mov rax, qword [rbp+8+8+32] ; rsp + rbp + адрес возврата + скрытое пространство.
      mov bl,[rax]
      mov [rcx+3], bl
      mov rax, qword [rbp+48+8]
      mov bl,[rax]
      mov [rcx+4], bl
      mov rax, qword [rbp+48+16]
      mov bl,[rax]
      mov [rcx+5], bl
      mov rax, qword [rbp+48+24]
      mov bl,[rax]
      mov [rcx+6], bl
      mov rax, qword [rbp+48+32]
      mov bl,[rax]
      mov [rcx+7], bl
      mov rax, qword [rbp+48+40]
      mov bl,[rax]
      mov [rcx+8], bl
      mov rax, qword [rbp+48+48]
      mov bl,[rax]
      mov [rcx+9], bl
      mov bl,0                ; Завершающий ноль.
      mov [rcx+10], bl
leave
ret

Основная	функция	main	та	же,	что	в программе	arguments1.asm,	но	вызыва
емой	функцией	здесь	становится	lfunc,	а не printf,	которая	вызывается	позже.

В	функции	lfunc	обратите	внимание	на	инструкцию	mov rax, qword [rbp+8+8+32],	
которая	загружает	четвертый	аргумент	из	стека	в регистр	rax.	Регистр	rbp	со
держит	копию	указателя	стека.	Первое	8байтовое	значение	в стеке –	это	содер
жимое	rbp,	сохраненное	в прологе	функции	lfunc.	Выше	находится	8байтовое	
значение	адреса	возврата	в функцию	main,	которое	автоматически	сохраняется	
в стеке	при	вызове	функции	lfunc.	Затем	следует	скрытое	пространство	раз
мером	32	байта.	Наконец,	мы	добрались	до	аргументов,	передаваемых	в сте
ке.	Таким	образом,	четвертый	и все	последующие	аргументы	можно	найти	по	
адресу	rbp+48	и выше	по	стеку.

После	возврата	в основную	функцию	main	стек	снова	выравнивается,	и вы
зывается	функция	printf.

На	рис. 41.3	показан	вывод	этой	программы,	который,	разумеется,	полно
стью	совпадает	с выводом	предыдущей	программы.

Рис. 41.3.	Вывод программы stack.asm

ОбрабОтка значений С плавающей тОчкОй
Для	работы	со	значениями	с плавающей	точкой	применяется	другая	методика.	
В листинге	41.4	показан	пример	исходного	кода.

Листинг 41.3.	Программа stack_float.asm

; stack_float.asm
extern printf
section .data
      zero   dq    0.0        ;0x0000000000000000
      one    dq    1.0        ;0x3FF0000000000000
      two    dq    2.0        ;0x4000000000000000
      three  dq    3.0        ;0x4008000000000000
      four   dq    4.0        ;0x4010000000000000
      five   dq    5.0        ;0x4014000000000000
      six    dq    6.0        ;0x4018000000000000
      seven  dq    7.0        ;0x401C000000000000
      eight  dq    8.0        ;0x4020000000000000
      nine   dq    9.0        ;0x4022000000000000
section .bss
section .text
      global main
main:
push rbp
mov rbp,rsp
      movq  xmm0, [zero]
      movq  xmm1, [one]
      movq  xmm2, [two]
      movq  xmm3, [three]

      movq  xmm4, [nine]
      sub   rsp, 8
      movq  [rsp], xmm4

      movq  xmm4, [eight]
      sub   rsp, 8
      movq  [rsp], xmm4

      movq  xmm4, [seven]
      sub   rsp, 8
      movq  [rsp], xmm4

      movq  xmm4, [six]

      sub   rsp, 8
      movq  [rsp], xmm4

      movq  xmm4, [five]
      sub   rsp, 8
      movq  [rsp], xmm4

      movq  xmm4, [four]
      sub   rsp, 8
      movq  [rsp], xmm4

      sub   rsp,32      ; Скрытое пространство.
      call  lfunc
      add   rsp,32
leave
ret

;------------------------------------------------
lfunc:
push  rbp
mov   rbp,rsp
      movsd xmm4,[rbp+8+8+32]
      movsd xmm5,[rbp+8+8+32+8]
      movsd xmm6,[rbp+8+8+32+16]
      movsd xmm7,[rbp+8+8+32+24]
      movsd xmm8,[rbp+8+8+32+32]
      movsd xmm9,[rbp+8+8+32+40]
leave
ret

В	 этой	 небольшой	 программе	 вывод	 не  выполняется	 изза	 некоторого	 
курьеза,	который	будет	описан	в следующей	главе.	Здесь	потребуется	отлад
чик	для	наблюдения	за	xmmрегистрами.	Для	удобства	значения	с плавающей	
точкой	в комментариях	представлены	в шестнадцатеричном	формате.	Первые	
четыре	значения	передаются	в функцию	в регистрах	xmm0-xmm3.	Остальные	ар
гументы	записываются	в стек.	Следует	помнить,	что	xmmрегистры	могут	содер
жать	одно	скалярное	значение	двойной	точности,	два	упакованных	значения	
двойной	точности	или	четыре	 упакованных	 значения	обычной	 (одиночной)	
точности.	 В  рассматриваемом	 здесь	 примере	 используется	 одно	 скалярное	
значение	двойной	точности,	а для	наглядности	значения	сохраняются	в стеке	
без	применения	инструкции	push.	Это	один	из	возможных	способов	сохране
ния	упакованных	значений	в стеке	с выравниванием	указателя	стека	rsp	со
ответствующим	образом	после	каждой	операции	записи.	Более	эффективным	
способом	является	 запись	 скалярного	 значения	непосредственно	из	памяти	
в стек,	как	показано	ниже:

push qword[nine]

В	 вызываемой	функции	необходимо	 копировать	 значения	из	 стека	 в  xmm
регистры,	в которых	можно	в дальнейшем	выполнять	обработку	этих	значений.

резюме
В	этой	главе	вы	узнали,	как:

	� передавать	в функции	аргументы	в регистрах	и в	стеке;
	� использовать	скрытое	пространство	в стеке;
	� получить	доступ	к аргументам,	переданным	в стеке;
	� сохранять	значения	с плавающей	точкой	в стеке.

Глава 42
Функции с переменным 

числом аргументов

Функция	с переменным	числом	аргументов	 (variadic function) –	это	функция,	
которая	может	принимать	 различное	 (меняющееся)	 количество	 аргументов.	
Хорошим	примером	такой	функции	является	printf.	Напомню,	что	в ассемб
лерных	программах	для	Linux	при	работе	функции	printf	с xmmрегистрами	тре
бовалось	соблюдение	соглашения	о том,	что	регистр	rax	содержит	количество	
xmmрегистров,	которые	должна	использовать	функция	printf.	Число	регистров	
можно	также	извлечь	из	инструкции	формата	printf,	так	что	зачастую	можно	
обойтись	без	использования	rax.	Например,	следующая	строка	формата	опре
деляет,	что	требуется	вывод	четырех	значений	с плавающей	точкой,	каждое	из	
которых	содержит	девять	десятичных	цифр:

fmt     db    "%.f",9,"%.f",9, "%.f",9,"%.f",10,0

Даже	 если	 соглашение	 об	 определении	 числа	 обрабатываемых	 значений	
с плавающей	точкой	в регистре	rax	не соблюдается,	функция	printf	все	равно	
должна	вывести	эти	четыре	значения.

функции С переменным чиСлОм аргументОв в wInDowS
В	Windows	применяется	другой	процесс	обработки	переменного	числа	аргу
ментов.	Если	в xmmрегистрах	содержатся	первые	четыре	аргумента,	то	необхо
димо	скопировать	их	в соответствующие	регистры	для	аргументов.	В листин
ге 42.1	показан	пример	исходного	кода.

Листинг 42.1.	Программа variadic1.asm

; variadic1.asm
extern printf
section .data
      one    dq    1.0
      two    dq    2.0
      three  dq    3.0

      fmt    dq    "The values are: %.1f %.1f %.1f",10,0



section .bss
section .text
      global main
main:
push  rbp
mov   rbp,rsp
      sub   rsp,32           ; Скрытое пространство в стеке.
      mov   rcx, fmt
      movq  xmm0, [one]
      movq  rdx,xmm0
      movq  xmm1, [two]
      movq  r8,xmm1
      movq  xmm2, [three]
      movq  r9,xmm2
      call  printf
      add   rsp, 32         ; Это не обязательно перед инструкцией leave.
leave
ret

При	создании	скрытого	пространства	перед	вызовом	функции	рекоменду
ется	установить	для	себя	полезное	правило:	всегда	удалять	это	скрытое	про
странство	после	завершения	выполнения	функции.	В рассматриваемом	здесь	
примере	в инструкции	add rsp,32	нет	необходимости,	потому	что	прямо	за	ней	
следует	инструкция	leave,	которая	в любом	случае	восстанавливает	указатель	
стека.	Здесь	вызывается	только	одна	функция	(printf),	но	если	в программе	вы
зывается	несколько	функций,	то	никогда	не следует	забывать	об	обязательном	
создании	 скрытого	 пространства	 в  стеке	 перед	 вызовом	 и  об	 обязательном	
удалении	скрытого	пространства	каждый	раз,	когда	продолжается	выполне
ние	после	вызова	функции.

Здесь	можно	видеть,	что	значения	с плавающей	точкой	копируются	в xmm
регистры	и в	регистры	общего	назначения,	предназначенные	для	передачи	
аргументов.	Это	требование	Windows.	Подробное	описание	данного	процес
са	выходит	за	рамки	этой	книги,	но	соблюдение	такого	требования	обяза
тельно	при	использовании	функций	языка	C без	прототипа	или	с перемен
ным	числом	аргументов.	Если	закомментировать	инструкции	копирования	
значений	в регистры	общего	назначения,	то	printf	не сможет	вывести	пра
вильные	результаты.

На	рис. 42.1	показан	вывод	этой	программы.

Рис. 42.1.	Вывод программы variadic1.asm

На	рис. 42.2	представлен	вывод	той	же	программы	без	использования	регист
ров	общего	назначения.

Рис. 42.2.	Некорректный вывод программы vriadic1.asm

ОбрабОтка Смешанных значений
В	листинге	42.2	показан	пример	обработки	значений	с плавающей	точкой,	объ
единенных	со	значениями	других	типов.

Листинг 42.2.	Программа variadic2.asm

; variadic2.asm
extern printf

section .data
      fmt    db    "%.1f %s %.1f %s %.1f %s %.1f %s %.1f %s",10,0
      one    dq    1.0
      two    dq    2.0
      three  dq    3.0
      four   dq    4.0
      five   dq    5.0
      A      db    "A",0
      B      db    "B",0
      C      db    "C",0
      D      db    "D",0
      E      db    "E",0

section .bss
section .text
    global main
main:
push rbp
mov rbp,rsp
      sub   rsp,8               ; Сначала выравнивается стек.
      mov   rcx,fmt             ; Первый аргумент.
      movq  xmm0,[one]          ; Второй аргумент.
      movq  rdx,xmm0
      mov   r8,A                ; Третий аргумент.
      movq  xmm1,[two]          ; Четвертый аргумент.
      movq  r9,xmm1
; Теперь запись аргументов в стек в обратном порядке.
      push  E                   ; 11-й аргумент.

      push  qword[five]         ; 10-й аргумент.

      push  D                   ; 9-й аргумент.

      push  qword[four]         ; 8-й аргумент.

      push  C                   ; 7-й аргумент.

      push  qword[three]        ; 6-й аргумент.

      push  B                   ; 5-й аргумент.

; Вывод результата.
      sub rsp,32
      call printf
      add rsp,32
leave
ret

Здесь	можно	видеть,	что	главное	внимание	необходимо	сосредоточить	на	
соблюдении	правильного	порядка	передачи	аргументов,	а также	на	копирова
нии	xmmрегистров	в регистры	общего	назначения	и на	записи	в стек	остальных	
аргументов	в обратном	порядке.

На	рис. 42.3	показан	вывод	этой	программы.

Рис. 42.3.	Вывод программы variadic2.asm

резюме
В	этой	главе	вы	узнали	о том,	что:

	� значения	с плавающей	точкой	в xmmрегистрах,	соответствующие	первым	
четырем	аргументам,	необходимо	обязательно	скопировать	в соответ
ствующие	регистры	общего	назначения;

	� если	требуется	передать	более	четырех	аргументов	с плавающей	точкой	
или	других	типов,	то	они	должны	быть	записаны	в стек	в обратном	по
рядке.

Глава 43
Работа с файлами в Windows

В	Linux	для	работы	с файлами	мы	использовали	системные	вызовы	(syscalls).	
В Windows	необходимо	соблюдать	другие	правила	и требования.	Как	уже	было	
отмечено	 в  нескольких	 предыдущих	 главах,	 вместо	 системных	 вызовов	 ис
пользуются	функции	Windows	API.

В	листинге	43.1	показан	пример	исходного	кода.

Листинг 43.1.	Программа files.asm

%include "win32n.inc"
extern printf
extern CreateFileA
extern WriteFile
extern SetFilePointer
extern ReadFile
extern CloseHandle

section .data
    msg      db 'Hello, Windows World!',0
    nNumberOfBytesToWrite equ $-msg
    filename db  'mytext.txt',0
    nNumberOfBytesToRead  equ 30
    fmt      db "The result of reading the file: %s",10,0

section .bss
    fHandle                resq 1
    lpNumberOfBytesWritten resq 1
    lpNumberOfBytesRead    resq 1
    readbuffer             resb 64

section .text
      global main
main:
push   rbp
mov    rbp,rsp

; HANDLE CreateFileA(
;  LPCSTR                lpFileName,
;  DWORD                 dwDesiredAccess,
;  DWORD                 dwShareMode,
;  LPSECURITY_ATTRIBUTES lpSecurityAttributes,
;  DWORD                 dwCreationDisposition,



;  DWORD                 dwFlagsAndAttributes,
;  HANDLE                hTemplateFile
; );
      sub    rsp,8
      lea    rcx,[filename]                   ; Имя файла.
      mov    rdx, GENERIC_READ|GENERIC_WRITE  ; Требуемые права доступа.
      mov    r8,0                             ; Совместное использование запрещено.
      mov    r9,0                             ; Защита по умолчанию.
; Запись в стек в обратном порядке.
      push   NULL                      ; Шаблона нет.
      push   FILE_ATTRIBUTE_NORMAL     ; Флаги и атрибуты.
      push   CREATE_ALWAYS             ; Диспозиция файла.
      sub    rsp,32                    ; Скрытое пространство в стеке.
      call   CreateFileA
      add    rsp,32+8
      mov    [fHandle],rax

; BOOL WriteFile(
;  HANDLE       hFile,
;  LPCVOID      lpBuffer,
;  DWORD        nNumberOfBytesToWrite,
;  LPDWORD      lpNumberOfBytesWritten,
;  LPOVERLAPPED lpOverlapped
; );
      mov    rcx,[fHandle]                  ; Идентификатор (дескриптор) файла.
      lea    rdx,[msg]                      ; Выводимое сообщение.
      mov    r8,nNumberOfBytesToWrite       ; Число выводимых байтов.
      mov    r9,[lpNumberOfBytesWritten]    ; Возврат числа записанных байтов.
      push   NULL
      sub    rsp,32                         ; Скрытое пространство в стеке.
      call   WriteFile
      add    rsp,32

; DWORD SetFilePointer(
;  HANDLE hFile,
;  LONG   lDistanceToMove,
;  PLONG  lpDistanceToMoveHigh,
;  DWORD  dwMoveMethod
; );
      mov    rcx,[fHandle]              ; Идентификатор (дескриптор) файла.
      mov    rdx, 7                     ; Младшие биты позиции.
      mov    r8,0                       ; В позиции не используются старшие биты.
      mov    r9,FILE_BEGIN              ; Начать запись с начала файла.
      call   SetFilePointer

; BOOL ReadFile(
;  HANDLE       hFile,
;  LPCVOID      lpBuffer,
;  DWORD        nNumberOfBytesToRead,
;  LPDWORD      lpNumberOfBytesRead,
;  LPOVERLAPPED
; );
      sub    rsp,8                      ; Выравнивание стека.
      mov    rcx,[fHandle]              ; Идентификатор (дескриптор) файла.
      lea    rdx,[readbuffer]           ; Буфер считывания.
      mov    r8,nNumberOfBytesToRead    ; Число считываемых байтов.

      mov    r9,[lpNumberOfBytesRead]   ; Число реально прочитанных байтов.
      push   NULL
      sub    rsp,32                     ; Скрытое пространство в стеке.
      call   ReadFile
      add    rsp,32+8

; Вывод результата выполнения функции ReadFile.
      mov    rcx, fmt
      mov    rdx, readbuffer
      sub    rsp,32+8
      call   printf
      add    rsp,32+8

; BOOL WINAPI CloseHandle(
;  _In_ HANDLE hObject
; );
      mov    rcx,[fHandle]
      sub    rsp,32+8
      call   CloseHandle
      add    rsp,32+8
leave
ret

Как	и ранее,	мы	просто	используем	шаблон	на	языке	C для	функции	Windows	
API,	 чтобы	 сформировать	 собственные	вызовы	на	 ассемблере.	Для	 создания	
файла	используются	простейшие	параметры	настройки	 (принятые	по	умол
чанию)	 для	 доступа	 к  файлу	 и  его	 защиты.	 При	 успешном	 создании	 файла	
функция	CreateFileA	возвращает	идентификатор	(дескриптор)	созданного	фай
ла.	Обратите	внимание	на	параметры.	Подробное	описание	всех	параметров	
можно	найти	в официальной	документации	Microsoft.	Существуют	некоторые	
возможности,	которые	могут	помочь	более	точно	управлять	свойствами	файла.

Идентификатор	 (дескриптор)	файла	используется	в функции	WriteFile	для	
записи	некоторого	текста	в файл.	В главе	40	функция	WriteFile	уже	использова
лась	для	вывода	сообщения	в консоли.

После	записи	текста	в файл	нужно	прочитать	этот	текст	обратно	в память,	
начиная	с позиции	7,	при	этом	первый	байт	имеет	индекс	0.	С помощью	функ
ции	SetFilePointer	указатель	перемещается	в позицию,	с которой	необходимо	
начать	чтение.	Если	параметр	lpDistanceToMoveHigh	равен	NULL,	то	параметр	lDis-
tanceToMove	является	32битовым	значением,	определяющим	количество	бай
тов	смещения	от	начала	файла.	Иначе	lpDistanceToMove	и lDistanceToMove	 в  со
вокупности	формируют	64битовое	значение	количества	байтов	смещения	от	
начала	файла.	В регистре	r9	определяется	позиция,	с которой	должен	начаться	
отсчет	перемещения,	возможными	вариантами	являются	FILE_BEGIN,	FILE_CUR-
RENT	и FILE_END.

После	 установки	 указателя	 в  требуемую	 позицию	 используется	 функция	
ReadFile	для	начала	чтения	с этой	позиции.	Считываемые	байты	сохраняются	
в буфере	памяти,	затем	выводятся	на	экран.	В завершающей	части	программы	
файл	закрывается.	Проверьте	свой	рабочий	каталог,	и вы	увидите	созданный	
текстовый	файл	mytext.txt.

На	рис. 43.1	показан	вывод	этой	программы.

Рис. 43.1.	Вывод программы files.asm

резюме
В	этой	главе	вы	узнали	о:

	� средствах	работы	с файлами	в Windows;
	� существовании	огромного	количества	параметров	для	точной	настрой
ки	при	обработке	файлов.

Послесловие. Что дальше?

После	изучения	этой	книги	вы	овладели	основами	программирования	на	со
временном	языке	ассемблера.	Следующий	шаг	зависит	от	того,	что	вам	нужно.	
В этом	послесловии	содержатся	некоторые	рекомендации.

Инженерыаналитики,	 работающие	 в  области	 обеспечения	 безопасности,	
могут	 использовать	 приобретенные	 знания	 для	 исследования	 вредоносного	
программного	 обеспечения,	 вирусов	 и  прочих	 средств	 взлома	 компьютеров	
и сетей.	Вредоносные	программы	в двоичном	формате	пытаются	проникнуть	
в компьютеры	и сети.	Можно	взять	их	двоичный	код,	выполнить	реверсинжи
ниринг	и попробовать	определить,	что	именно	делает	этот	код.	Разумеется,	это	
следует	делать	в изолированной	лабораторной	системе.	Изучайте	средства	ре
версинжиниринга	и приобретайте	необходимые	инструментальные	средства.	
Также	следует	отметить	необходимость	изучения	ассемблера	ARM	для	анализа	
кода	для	смартфонов.

Программист	на	одном	из	языков	высокого	уровня	может	рассмотреть	воз
можность	создания	собственной	библиотеки	высокопроизводительных	функ
ций	для	связывания	с кодом	основной	программы.	Изучайте	возможности	оп
тимизации	кода,	ведь	исходный	код	в этой	книге	написан	не для	достижения	
высокой	производительности,	 а для	демонстрации	возможностей	современ
ного	ассемблера.	В этой	книге	приведено	несколько	ссылок	на	документы,	ко
торые	помогут	вам	писать	оптимизированный	код.

Если	вам	необходимо	более	глубокое	понимание	работы	процессоров	Intel,	
то	загрузите	руководства	Intel	и внимательно	изучайте	их.	В этих	руководствах	
вы	найдете	огромный	объем	интересной	информации,	а знания	о совместной	
работе	аппаратного	и программного	обеспечения	позволят	выйти	на	перед
ний	край	разработки	системного	программного	обеспечения	или	инструмен
тальных	средств	диагностики	критических	отказов	систем.

Как	программист	на	языке	высокого	уровня	с хорошим	знанием	ассемблера	
вы	гораздо	лучше	подготовлены	к отладке	собственного	кода.	Анализируйте	
объектные	файлы	 (.obj)	 и файлы	листинга	 (.lst)	 и применяйте	реверсинжи
ниринг,	чтобы	узнать,	что	происходит	в коде	программы.	Наблюдайте,	как	ис
пользуемый	 вами	 компилятор	 преобразовывает	 исходный	 код	 в машинный	
язык.	Возможно,	использование	других	инструкций	ассемблера	окажется	бо
лее	эффективным.



Предметный указатель

.gdbinit,	файл	конфигурации	 
отладчика	GDB		39

A
Аргумент
дополнительный		113
извлечение	из	регистра		113
извлечение	из	стека		113
не	являющийся	значением	
с плавающей	точкой		113

с	плавающей	точкой		114
Арифметическая	инструкция		84
Ассемблер
встроенный	код
простой		164
расширенный		166

синтаксис	инструкций 
	встроенного	кода		166

Ассемблерная	функция		160

Б
Бит		30

В
Ввод/вывод	в	консоли
пример	использования		138

Внеочередное	исполнение	кода		290
Внешняя	функция		99,	160
Внешняя	функция	из	языка	C		107
Выравнивание	данных	в	памяти		186
Выравнивание	по	адресам	памяти		29

Г
Гамильтона–Кэли	теорема		268

Д
Данные
выровненные		186
невыровненные		186

Двоичное	число		30
Десятичное	представление	чисел		30
Директива	препроцессора	ассемблера		

135,	152
Дополнительный	двоичный	код	 

(twos	complement)		31
Доступ	к	аргументам	командной	

строки		154

И
Инструкция	перехода	и	значения	

флагов		70
Инструкция	проверки	условия		69
Интегрированная	среда	разработки		

21,	64

К
Калькулятор	с	функциями	

преобразования	форматов	чисел		31

Л
Линкер		23

М
Макрокоманда		134
Макрос		134,	172
многострочный		135
однострочный		135

Маска,	управляющая	 
перемешиванием		283

Матрица	вырожденная	 
(сингулярная)		272

Машинный	язык		28
Метка	времени		290

О
Обратный	порядок	байтов 

(от	старшего	к	младшему		big	
endian)		52

Объектный	файл		110
Операция	с	битами		124,	130
арифметическая		126

Оптимизация	кода	программы		292
Отладка	программы		36
Отладка	программы	с	аргументами	

командной	строки		155
Отладчик		36
с	графическим	пользовательским	
интерфейсом		37

с	интерфейсом	командной	строки		36

П
Память		73
структура	в	программе		79

Переменная
локальная		102



Переменная	среды	Windows		299
Перемешивание		227
байтов	в	слове		234
в	обратном	порядке		233
вращением		234
в	случайном	порядке		231
маска		231,	235

Переносимый	язык	ассемблера		19
Переполнение
обработка		140

Порядок	следования	байтов	
(endianness)		52

Пролог	функции		49,	54,	106
Простая	функция,	пример		99
Процессор		32
поддержка	AVX		250

Прямой	порядок	байтов	 
(от	младшего	к	старшему)		277

Прямой	порядок	байтов	(от	младшего	
к	старшему		little	endian)		52

Р
Распространение	знакового	разряда		85
Распространение,	 

или	расширение	знака		125
Регистр		33
затираемый	(clobbered)		165,	168
общего	назначения		33
сохраняемый	вызываемой	 
функцией		119,	120

Регистровое	ограничение		168
Редактор	связей		23

С
Сдвиг
арифметический		84

Сериализация	инструкций		290
Синтаксис	ассемблера	версии	Intel		37
Системный	вызов		144
Скалярные	данные		185
Соглашение	о	вызовах	функций		112
Microsoft	x64  301
System	V	AMD64	ABI		112,	155
выравнивание	стека	по	16байтовой	
границе		115

использование	регистров		118
регистр
неразрушаемый,	или	
долговременный	 
(nonvolatile)		120

разрушаемый,	или	
кратковременный	 
(volatile)		120

Сравнение	использования	инструкции	
цикла	и	инструкции	условного	
перехода		72

Стек		74
выравнивание	по	16байтовой	
границе		103,	115,	310

скрытое	пространство		301,	309
схема
пример	использования		116

схема	«последним	пришел,	первым	
вышел»	(last	in	first	out		LIFO)		90

фрейм		105
Страница	памяти		215
Строка
анализ		174
обработка		170
определение	длины		213
поиск		174,	216
реверсирование	(изменение	порядка	
букв	на	обратный)		88

с	длиной,	не	определенной	явно		207
с	неявно	заданной	длиной
сравнение		220

сравнение		174
с	явно	заданной	длиной		207
сравнение		222

Счетчик
младших	циклов	процессора		290
старших	циклов	процессора		291

Т
Тип	данных		25
строка		26

У
Указатель	базы	(стека)		104
Указатель	стека		74
Упакованные	данные		185
Условная	инструкция		69
Условное	ассемблирование		145,	152

Ф
Фаддеева–Леверье	алгоритм		268
Файловый	ввод/вывод		144
инструкция		152

Фрейм	стека		105
Функция
ветвь		105

Предметный	указатель  327



внешняя		99,	107
лист		105
с	переменным	числом	
аргументов	 317

Функция	языка	C		158

Ц
Центральное	процессорное 

устройство	(ЦПУ)		32
Цикл		70
счетчик		71

Ч
Число
восьмеричное		145
с	двойной	точностью		94
с	обычной	(одиночной)	 
точностью		94

с	плавающей	точкой		32
знаковый	бит		94
мантисса		95
показатель	степени		94
пример	программы		96

целое		31
беззнаковое		31
восьмеричное		32
знаковое		31
отрицательное		31
шестнадцатеричное		32

Э
Эпилог	функции		49,	54,	106

A
addpd,	инструкция		199,	202
addps,	инструкция		199
addss,	инструкция		97
add,	инструкция		84
adouble.asm,	файл	функции		162
Advanced	Vector	Extension	(AVX)		184
Advanced	Vector	Extensions	(AVX)		250
alive.asm,	программа		48
анализ		49
вывод		53

AND,	логический	оператор		58
ASCII		26,	40,	172
Assembler	preprocessor	directive		135,	152
asum.asm,	файл	функции		161
AVX		257
инструкция		292
производительность		286

маска,	управляющая	
перемешиванием		282

матрица
вывод		264
маска	перестановок		271
маска	смешивания	(наложения)		269
обращение		268
операция		257
операция,	пример	исходного	
кода	 257

оценка	производительности	
вычисления	следа		292

оценка	производительности	
транспонирования		286

транспонирование		274
транспонирование	методом	
перемешивания		282

транспонирование	с	
использованием	неупакованных	
данных		277

умножение		265
поддержка	процессором		250
пример	программы		252

B
Base	pointer		104
betterloop.asm,	программа		71
bitflags,	переменная		132
blend_trace,	функция		296
bss,	Block	Started	by	Symbol		26
btr,	инструкция		130
bts,	инструкция		130
bt,	инструкция		130

C
Caleesaved	register		119
Calling	convention		112
circle.asm,	файл,	содержащий	только	

внешние	функции		108
circle.asm,	файл	функции		160
cld,	инструкция		177
cmpsb,	инструкция		174,	177,	178
cmp,	инструкция		69,	182
Conditional	assembly		145,	152
CountReg		174
cpuid,	инструкция		179,	290
CreateFileA,	функция	Windows	API		323
createFile,	функция		152
cvtss2sd,	инструкция		199,	202
Cтек		88

328	  Предметный указатель



D
Data	Display	Debugger	(DDD),	

отладчик	 60
DDD,	отладчик		91
использование	команд	 
отладчика	GDB		62

исследование	памяти		62
DDD,	отладчик
наблюдение	за	изменением	
регистров		62

dec,	инструкция		84
DF,	флаг	направления		173,	177
divsd,	инструкция		195
divss,	инструкция		97
Double	precision		94
dwarf,	формат	отладки		23

E
eflags,	регистр	флагов	в	SASM		69
elf64,	Executable	and	Linkable	Format	 

for	64bit		23
ELF,	формат	выполняемого	файла
поддержка	в	NASM		76

extern,	ключевое	слово,	объявление	
внешней	функции		108

extractps,	инструкция		255

F
FPUинструкция		95
function2.asm,	 

программа		100
function4.asm,	программа		107
function.asm,	программа		99

G
GCC,	GNU	Compiler	Collection		21
gdb	memory,	команда		76
GDB,	отладчик		36,	155,	302
информация	о	регистрах		42
команда
tui	enable		45

прерывание	работы	программы		42
процесс	отладки	программы		37

gedit,	текстовый	редактор		20
GetStdHandle,	функция	Windows	

API	 306
global,	ключевое	слово,	 

объявление	глобальной	
переменной		108

grep,	утилита		77

H
haddpd,	инструкция		270
handling	port	5	pressure,	в	руководстве	

Intel		292
hello	world,	программа		19
улучшенная	версия		45

I
icalc.asm,	программа		81
IDE,	integrated	development	

environment		21
idiv,	инструкция		87
IEEE754,	международный	стандарт		32
imm8,	управляющий	байт	для	строк	

SSE		208
IntRes1		209
IntRes2		210
использование	битов		209

imul,	инструкция		85
inc,	инструкция		84

J
jge,	инструкция		69
jmp,	инструкция		70
jnz,	инструкция		177
jump.asm,	программа		66
jumploop.asm,	программа		70
jz,	инструкция		177,	218

L
ldmxcsr,	инструкция		189,	195
leave,	инструкция		100
lea,	инструкция		75,	173
lodsb,	инструкция		178
loop,	инструкция		71,	90

M
makefile		22
расширенная	версия		110

MASM		298
memory.asm,	программа		73
MinGW	(Minimalist	GNU	for	

Windows)	 299
MinGWw64		299
MMX		184
movapd,	инструкция		202
movaps,	инструкция		202
movdqa,	инструкция		206
movdqu,	инструкция		231
move.asm,	программа		60
movq,	инструкция		55

Предметный	указатель  329



movsb,	инструкция		173,	174
movsd,	инструкция		97,	174
movss,	инструкция		97,	199
movsw,	инструкция		174
movupd,	инструкция		199
movups,	инструкция		199
mov,	инструкция		27,	75
mulss,	инструкция		97
mul,	инструкция		85
mxcsr,	регистр		271
mxcsr,	регистр	управления	и	состояния	

SSE		188,	195
значения	битов		188

MXCSR,	регистр	флагов	для	SIMD		35

N
NASM		21,	134,	152,	202,	298
nasm	v,	команда		298
NASM,	версия	ассемблера		20
neg,	инструкция		58
nop,	инструкция		40
NOT,	логический	оператор		57

O
objdump,	утилита	командной	строки		

136
OR,	логический	оператор		57

P
paddd,	инструкция		206
PATH,	переменная	среды	Windows		298
pcharsrch,	функция		241
pcmpestri,	инструкция		207,	225
pcmpestrm,	инструкция		207,	237
pcmpistri,	инструкция		207,	214,	218,	

221,	241
pcmpistrm,	инструкция		207,	237,	241
pextrd,	инструкция		206
PIE,	position	independent	executable,	

перемещаемый	код		24
pinsrb,	инструкция		218,	241
pinsrd,	инструкция		206,	231
pmovmskb,	инструкция		242
popcnt,	инструкция		242
popf,	инструкция		91
pop,	инструкция		88
PowerShell		299
printb,	функция		121
printdpfp,	функция		199
printf,	функция		53,	86,	99,	108,	111,	

112, 115

printspfp,	функция		199,	202
printString,	функция		152
prints,	функция		142
print_xmm,	функция		195
pshufb,	инструкция		235
pshufd,	инструкция		234,	282
pshufhw,	инструкция		234
pshuflw,	инструкция		234
pshufw,	инструкция		234
pstrcmp,	функция		221,	225
pstrlen,	функция		214
pstrln,	функция		241
pstrscan_f,	функция		218
pstrscan_l,	функция		218
pushf,	инструкция		91
push,	инструкция		88
pxor,	инструкция		214,	218

R
rax,	регистр		27,	63,	75,	85,	100,	125
rbp,	регистр	указателя	базы	стека		104
rbx,	регистр		71
rdi,	регистр		28,	152
rdtscp,	инструкция		291
rdtsc,	инструкция		290
rdx,	регистр		28
readelf,	утилита		76
ReadFile,	функция	Windows	API		323
reads,	функция		142
rect.asm,	файл	функции		160
Register	constraint		168
repe,	инструкция		178
repne,	инструкция		178
reverse_xmm0,	функция		241
rflags,	регистр	флагов	процессора		34,	69
rip,	регистр		43
rip,	регистр	счетчика	команд		34
rol,	инструкция		126
ror,	инструкция		126
rsp,	регистр	указателя	стека		74,	88,	93

S
sal,	инструкция		84,	125
sar,	инструкция		85,	125
SASM		36,	64,	143,	299,	302
диалоговая	панель	Settings	
(Параметры	настройки)		65

строка	Linking	Options	(Параметры	
связывания)		111

scasb,	инструкция		174,	178
section	.bss,	раздел	программы		27

330	  Предметный указатель



section	.data,	раздел	программы		25

